Emacs Lisp によるプログラミング

初心者のための入門

Revised Third Edition
日本語版 0.97, 10 January 2000

Robert J. Chassell 著　松田茂樹 訳

これは、初心者のための 『*Emacs Lisp* によるプログラミング』 の入門です。

Edition 3.10, 28 October 2009 日本語版 0.97, 10 January 2000

Published by the:

GNU Press,
a division of the
Free Software Foundation, Inc.
51 Franklin Street, Fifth Floor
Boston, MA 02110-1301 USA

`http://www.fsf.org/licensing/gnu-press/`
email: `sales@fsf.org`
Tel: +1 (617) 542-5942
Fax: +1 (617) 542-2652

ISBN 1-882114-43-4

この Info ファイルは、Programming in Emacs Lisp (An introduction) 1.05 版の日本語訳です。翻訳元のファイルは GNU の配布の中に含まれている `emacs-lisp-intro-1.05.tar.gz` です。

Short Contents

Table of Contents

序文

GNU Emacs というエディタは大部分が Emacs Lisp と呼ばれるプログラミング言語で書かれている。このプログラミング言語の中に書かれているコードは、計算機がコマンドが与えられた時何をすべきかを教えるソフトウェア—命令の集まり—である。Emacs は、あなたが Emacs Lisp で新しいコードを書き、簡単に拡張機能としてエディタに組み込むことが出来るよう設計されている。これが Emacs が「extensible editor (拡張可能エディタ)」と呼ばれる所以である。

(実際には、Emacs は通常エディタが出来ると思われている機能をはるかに上回ることをやってのけるので、むしろ「extensible computing environment (拡張可能計算機環境)」とでもいうべきだろう。まあ、ちょっと長ったらしいフレーズになってしまうし、あなたが Emacs の中でする全てのこと—マヤ暦や月の満ち欠けを調べたり、多項式を整理したり、コードをデバッグしたり、ファイルを扱ったり、手紙を読んだり書いたりといったこと—はもっとも広い意味では編集 (edit) の範疇に属するものではあるが。)

Emacs Lisp はテキストエディタに関する言語と思われがちであるが、実際は計算機全体に関するプログラミング言語である。あなたはこの言語を他のプログラミング言語と同じ様に使うことが出来るのだ。

多分、あなたはプログラミングを理解したいと思っていることだろう。あるいは、Emacs の機能を拡張したいとか、プログラマになりたいと思っているかもしれない。この Emacs Lisp についての文書は、その出発点を示すことを意図している。つまり、あなたをプログラミングの基本的な事柄に慣れさせ、更により大切なことだが、あなたが自分自身でより先に進むにはどうしたら良いかを示すことを目的としている。

このテキストを読むにあたって

この文書の中には、Emacs の中で走らせることの出来る小さなサンプルプログラムが登場する。この文書を GNU Emacs の Info の中で読めば、そのプログラムをその場で走らせることが出来る。(これは簡単に実行出来る。その方法はそのプログラムが出てきた時に説明する。) あるいは、この文書を印刷製本し Emacs が走っている計算機のそばに座りながら読むことだって出来る。(私はこの方法が好きだ。私は印刷された本の方が好みなので。) たとえあなたの近くで Emacs が動いていなくても、この本を読むことは出来る。が、その場合はこれを小説か、まだ行ったことのない国の旅行ガイドのように扱うのが良いだろう。面白いにしても実際とは違うものだ。

この文書の多くの部分は、GNU Emacs の中で使われるコードのガイドツアーというべきことに費やされている。これらのツアーの目的は次の二つである。一つ目は、あなたを実際に役立つコード (あなたが毎日使うコード) に親しませること。二つ目は、あなたを Emacs を活用する方法に親しませることである。エディタがどのように実装されているかを見ることは面白いものである。また、私はあなたがソースコードを眺める習慣を身に付けることを望んでいる。あなたはそこからいろいろなこと学び、そしてアイディアを発掘することが出来る。GNU Emacs を手に入れるということは、宝が隠されたドラゴンの洞窟を発見したようなものなのである。

これらの例やガイドツアーを通して、エディタとしての Emacs を学んだり、プログラミング言語としての Emacs Lisp を学習するだけでなく、Emacs が Lisp プログラミング環境であることを理解するきっかけが与えられることだろう。GNU Emacs はプログラミングをサポートし、あなたが快適に使いこなしたくなるような様々な道具を与えてくれる。例えば `M-.` (`find-tag` コマンドを発生させるキー) なんかがそうだ。また、エディタ環境の一部であるバッファやその他のものを学べる。Emacs のこれらの特徴を学ぶことは、あなたが故郷に帰る新しいルートを探すことに例えられよう。

最後に、私はあなたがまだ知らないプログラミングの側面を学ぶために Emacs を利用して出来る幾つかのテクニックを伝えることが出来ればと思う。難解な事柄を理解したり、何か新しいことをしたりする際、その解決方法にしばしば Emacs そのものを利用することが出来る。こういった自己完結性があることは、単に気分が良いというだけでなく、実際に非常に便利なものである。

これは誰のために書かれたものか

この文書は、プログラマではない人々への、初歩的な入門書として書かれている。もしあなたがプログラマであるなら、あなたはこのような簡単なものでは満足出来まい。というのも、あなたは既にリファレンスマニュアルを読むのに熟達しているかもしれないし、その場合この文書のような書き方ではまどろっこしく感じるだろうからだ。

例えば、この文書を読んだある熟練したプログラマは、私に次のように言った。

> 僕は、リファレンスマニュアルから学ぶ方が好きなんだ。各々のパラグラフに「ダイブして」そして各パラグラフの間で「息継ぎをする」感じだ。
>
> 僕はあるパラグラフの最後に辿り着いたら、その主題については終了したものと見做すんだよ。つまり、*(*次のパラグラフでより詳しいことが説明されるような場合を除いて*)* 必要なことは全て分ったと考えるわけなんだ。だから無駄な沢山の繰り返しがなくって、必要な情報が載っている個所へのポインタがきちんと整備されているようなのがいいな。

この入門書はこのような人のために書かれたのではない！

まず一つ目に、私は全てのことを少なくとも３回は繰り返すようにした。まずはそれを紹介し、次にそれがどんな場合に使われるかを見て、そして、別の使われ方を見るか、それを復習するといった具合だ。

二つ目に、私はその主題についての全ての情報を一箇所にまとめるようなことは殆どしなかった。ましてや、一つのパラグラフに押し込めるようなことは避けた。これは個人的な考えだが、そういうやり方は読者に過剰な負担を強いることになる。その代わりに、私は各々の場合に必要なことだけを説明するように心がけた。(時々は、後で正式に説明する時に戸惑うことのないよう、ちょっと先走った事柄の説明も含めたりしたが。)

この文書を読む時は、あなたは全てのことを初めて学ぶものと考えられている。取り上げられた幾つかの項目については、言わば、会えば会釈する程度の浅い付き合いしかする必要がないものもある。あなたが本当に大切なことが何かに気付き、それに集中することが出来るだけの十分なヒントを提供出来るように、この文書をうまく構成出来ていればと思う。

あなたは、幾つかのパラグラフには「ダイブする」必要がある。それらには、別のもっと楽な読み方があるわけではない。しかし、私はそのようなパラグラフの数はなるべく押さえたつもりだ。この本は、そびえたつ山ではなく、ちゃんと登ることができる丘であるように書かれている。

この『Emacs Lisp によるプログラミング』—初心者のための入門— は次の本と兄弟関係にある。

『*The GNU Emacs Lisp Reference Manual*』.[1] リファレンスマニュアルには、この入門書よりも詳しいことが載っている。またリファレンスマニュアルでは、あるトピックについての全ての情報は一箇所に固まっている。もしあなたが、上に挙げたプログラマのようなタイプなら、こっちを読むべきだろう。また、勿論この入門書を読み終えた後、自分自身でプログラミングをする場合には、リファレンスマニュアルがいろいろと便利であることが納得出来るだろう。

Lisp の歴史

Lisp は最初、1950 年代の終わりに、マサチューセッツ工科大学で人工知能の研究のために開発された。Lisp 言語が持つ素晴しい能力は、エディタのコマンドを書くことは勿論のこと、他の目的についても優秀さを発揮した。

GNU Emacs Lisp は Maclisp の影響を多く受けている。Maclisp は 1960 年代に MIT で書かれた。Common Lisp の影響もいく分受けている。こちらは 1980 年代に標準となった。しかしながら、Emacs Lisp は Common Lisp と比べて非常に単純である。(標準的な Emacs の配布には、オプションとして `cl.el` が含まれている。これは Emacs Lisp に多くの Common Lisp の機能を付け加えるためのものである。

初心者の人へ

もしあなたが GNU Emacs について知らないとしても、この文書には何かしら役に立つことが書かれているだろう。しかしながら、たとえ、コンピュータのスクリーンの中を動き回るということだけでも良いから、Emacs を学ぶことを薦める。あなたはオンラインのチュートリアルを使って自分自身でその使い方を学べる。そのためには、*`C-h t`* とタイプするだけでよい。(これは `CTRL` キーと *`h`* キーを同時に押して離し、次に、*`t`* キーを押して離すことを意味する。) (訳註：Mule ならば、*`C-h T`* で日本語や韓国語、タイ語のチュートリアルが行える。)

また、私はしばしば Emacs の標準的なコマンドを、そのコマンドを引き起こすキーを書き、その後に、そのコマンドの名前を括弧でくくって書くことで示したりする。例えば *`M-C-\`* (`indent-region`) といった感じだ。(もし望むなら、そのコマンドを引き起こすためにタイプするキーを交換することも

[1] 日本語訳もある。『GNU Emacs Lisp リファレンスマニュアル』Bil Lewis, Dan LaLiberte and the GNU Manual Group 著、榎並嗣智 井田昌之監訳、発売 丸善、発行 透土社、定価 6796 円、ISBN 4-924828-39-4

出来る。これは、リバインディング (*rebinding*) と呼ばれる。Section 16.8 “キーマップ”, page 149, 参照)。この `M-C-\` という省略形は、`META` キーと `CTRL` キーと `\` キーを同時にタイプすることを示している。このような組み合わせはよくキーコードと呼ばれる。理由は、ピアノでコードを弾くのに似ているからである。もし、キーボードに `META` キーが無ければ、`ESC` キーがその代わりになる。この場合は、`M-C-\` は、まず `ESC` を押して離し、ついで、`CTRL` と `\` を同時にタイプすることを意味する。

もし、この文書を GNU Emacs の Info を使って読んでいるなら、この文書全てをただ単にスペースバー、`SPC` を押し続けることで読んでしまうことが出来る。(Info について学ぶには、`C-h i` とタイプし、ついで Info を選択すれば良い。)

言葉の使い方についての注意だが、私が単独で Lisp という単語を使う場合は、大概、Lisp の様々な方言にも通用する、一般的な事柄を述べている。しかし、Emacs Lisp と言うときは特に GNU Emacs Lisp のことを言及している。

謝辞

執筆に際し助けて頂いた全ての人々に感謝します。特に、Jim Blandy, Noah Friedman, Jim Kingdon, Roland McGrath, Frank Ritter, Randy Simith, Richard M.Stallman, そして、Melissa Weisshaus に感謝します。また、Philip Johnson と David Stampe の忍耐強い励ましにも感謝します。この文書の間違いは、全て私の責任です。

訳者まえがき

この文書は Robert J. Chassell 氏の『Programming in Emacs Lisp: A simple introduction』を訳したものです。個人用の訳なので拙い所も多く、また形式上もあまりしっかりしてはいません。が、一応実用には耐えられるのではないかと判断しています。また、訳者の実力不足のために訳に間違い等が含まれていると思われますが、お気付きの方は連絡していただけると嬉しく思います。

また、飯田義朗様、木村浩一様、幸田薫様、本田博通様、山下健司様には、以前の版での誤り、不具合等について指摘して頂きました。この場をかりて深く感謝いたします。

Robert J. Chassell

bob@gnu.org

1 List 処理

訓練を受けてない人にとっては、Lisp は奇妙なプログラミング言語である。Lisp のコードの中には到る所に括弧が見受けられる。中には、Lisp と言う名前は、'Lots of Isolated Silly Parentheses' を表わしているのだとほざく人までいる。しかし、これは根拠のない主張である。Lisp は LISt Processing を表わし、リスト (や、リストのリスト) を両側を括弧で狭んで扱うプログラミング言語である。括弧は、リストの境界を示している。時々、リストの頭にアポストロフィ、即ち引用符 '' が付いていたりすることもある。リストは Lisp の基礎である。

1.1 Lisp のリスト

Lisp の中では、リストは `'(rose violet daisy buttercup)` という格好をしている。このリストの頭には一つのアポストロフィが付いている。これは、よりあなたが親しんでいるであろう次のような形のリストに書くことも出来る。

```
'(rose
  violet
  daisy
  buttercup)
```

このリストの要素は四つの異なる花の名前であり、各々が空白で区切られ、括弧に囲まれている。これは石の壁に囲まれた広場の花を連想させる。

リストはまた、要素として数値を持つことも出来る。例えば、`(+ 2 2)` なんかがそうだ。このリストはプラスの符号 '+' とその後に続く二つの '2' を持っていて、各々は空白で区切られている。

Lisp の中では、データとプログラムの両方が同じ様に表現される。つまり、これらは共に、空白で区切られ括弧で囲まれた単語や数値や他のリストからなるリストである。(プログラムがデータと似ているために、あるプログラムは容易に他のプログラムのデータとして役に立つ; これが Lisp の極めて強力な特徴の一つである.) (ついでに言っておくと、この直前の括弧で囲まれた補足は Lisp のリストではない。というのも、これらは ';' と '.' という句読点を中に含んでいるからだ。)

(訳註：日本語としては変かもしれないが、つじつまを合わせるために ';' と '.' を使った。)

ここで別のリストの例を挙げよう。今度は中にリストを含んでいる。

```
'(this list has (a list inside of it))
```

このリストの構成要素は、'`this`', '`list`', '`has`', という単語と、'`(a list inside of it)`' というリストである。内部にある方のリストは、'`a`', '`list`', '`inside`', '`of`', '`it`' という単語からなっている。

1.1.1 Lisp のアトム

Lisp の中では、今まで単語と呼んで来たものは、*atom* (アトム) と呼ばれる。この言葉は歴史的には「これ以上分解出来ない」という意味を表わす単語アトムから来ている。Lisp について話している限りは、我々がリストの中で使ってきた単語は、それ以上小さなプログラムのパートには分解出来ない。それは数字や '+' みたいな一つの文字からなる記号でも同じである。一方リストは、アトムと違って幾つかの部分に分解することが出来る。(Chapter 7 "`car cdr` & `cons`：基本関数", page 55, 参照。)

リストの中では各々のアトムは互いに空白で区切られている。また括弧のすぐ隣りに位置することが出来る。

技術的な言い方をすると、Lisp のリストとは、空白で区切られた、いくつかのアトムないしは他のリスト、あるいはアトムと他のリストの両方を括弧でくくったものである。中には一つしかアトムが無くても良いし、全く無くても良い。中に何も入っていないリストというのは `()` であるが、これは空リスト (*null list*) と呼ばれる。空リストは同時にアトムでもリストでもある唯一のものである。

アトムやリストを表示したものは、*symbolic-expression*, もしくはもっと簡単に S 式と呼ばれる。*Expression* という単語自身は、表示された表現か、あるいは計算機の内部に保持されているアトムやリストのどちらかを表わす。*Expression* という言葉はしばしば曖昧に用いられる。(また、多くの本では形式 (*form*) という言葉が expression の同意語として使われている。)

(訳註：この訳では以下、expression を式、ないしは S 式と訳していることが多い。)

ついでだが、我々の世界を構成しているアトム (原子) がそう名付けられたのは、それらが分割不可能と考えられていた時のことである。しかしその後、物理で言うアトムは分割不可能ではないことが発見された。物質は、一つのアトムかほぼ同じサイズの二つの部分に分裂する。(訳註：意味不明。) 物理で言うアトムはまだその本当の性質が解明されぬままに時期尚早にして名付けられてしまった。Lisp

の中では、ある種のアトム、例えば配列 (array) は、複数の部分に分解出来る。しかし、そのメカニズムはリストを分解する時のメカニズムとは違う。リストの操作に関する限り、リストの中のアトムは分解不可能である。

英語におけるのと同様、Lisp のアトムを構成する文字要素は、一つの単語としての文字とは違う。例えば、南アメリカナマケモノ (South American sloth) を表わす ‘ai’ は ‘a’ と ‘i’ という二つの単語とは全く異なる。

自然の中には沢山の種類のアトムがあるが、Lisp の中には数種類のアトムしかない。例えば、37、511、1729 といった数値 (*number*)、そして ‘+’、‘foo’、や ‘forward-line’ といったシンボル (*symbol*) なんかである。今まで挙げたリストの例の中に出てきた単語は全てシンボルである。普段の Lisp の会話では、アトムという単語はあまり使われないが、それはプログラマは普通今使っているアトムがどの種類のアトムかをより限定的にみようとしているからである。Lisp のプログラミングはもっぱらリストの中のシンボル (と、ときどき数値) についてのものである。

それに加えて、二つの二重引用符で挟まれたテキストは—それが一つの文であろうと、あるいはパラグラフであろうとも—アトムである。例を挙げておこう。

```
'(this list includes "text between quotation marks.")
```

Lisp では句読点が入っていようが空白があろうが、全ての二重引用符で囲まれた文は一つのアトムである。この種のアトムは 文字列 (*string*) と呼ばれ、計算機が人間が読める形でメッセージを出す用途等に使われる。文字列は数値ともシンボルとも違う種類のアトムであり、異なる使い方をされる。

1.1.2 リストの中の空白

The amount of whitespace in a list does not matter. From the point of view of the Lisp language,

```
'(this list
  looks like this)
```

は、次のリストと全く同じである。

```
'(this list looks like this)
```

どちらの例も、Lisp にとっては ‘this’、‘list’、‘looks’、‘like’、そして ‘this’ というシンボルがこの順番で並んでいる全く同じリストである。

余分な空白と改行は、人間がよりリストを見やすいようにするために用いられている。Lisp が表現を読み取る時は、全ての余分な空白を除いてしまう (ただし、最低一個の空白は残している。そうでないとアトムとアトムの区切りが分らなくなってしまう)。

奇妙に思えるかもしれないが、今まで見てきた例で殆どの Lisp のリストがカバーされてしまう。Lisp における他の全てのリストは、多かれ少なかれ、これらの例のどれかに似ている。まあ、もっと長くなったり、複雑になったりはするが。簡単にまとめると、リストは括弧に挟まれ、文字列は二重引用符に挟まれ、シンボルとは単語のようなもので、数値とは数のようなものである。(ある場合には、角括弧や、ドット、それに他の特別な文字が使われたりもする。しかし、それらが無くても我々はかなり先まで進める。)

1.1.3 GNU Emacs によるリストのタイプの支援

もし、あなたが Lisp の S 式を GNU Emacs の Lisp Interaction mode もしくは Emacs Lisp mode の中でタイプしているなら、幾つかのコマンドを使って、Lisp の S 式を整形し読みやすいものにすることが出来る。例えば TAB キーを押すと、カーソルがある行を自動的に適当な分だけインデントしてくれる。リージョン内のコードをきちんとインデントしてくれるコマンドは、*M-C-* にバインドされている。インデントは、あなたが、どの要素がどのリストに属しているか見やすいように—サブリストの要素がそれを含んでいるようなリストの要素よりも深くインデントされるように—設計されている。

更に、あなたが括弧を閉じる時に、Emacs は一時的にその括弧にマッチする開き括弧の位置にジャンプする。そうすることでどの括弧に対応するかを確かめることが出来る。これは大変便利である。というのも、あなたが Lisp でタイプする全てのリストについて、開き括弧には必ず閉じ括弧がきちんと対応していなければならないからである。(Section “Major Modes” in *The GNU Emacs Manual*, に Emacs の mode のより詳しい説明が書かれている。)

1.2 プログラムの実行

Lisp におけるリストは、たとえどんなリストであってもそのまま走らせることが出来る。実際に走らせてみると (Lisp の専門用語では 評価 (*evaluate*) するという)、計算機は次の 3 つのうちいずれかの動作をする。まずは、そのリストをそのまま返すだけで何もしない、もしくは、あなたにエラーメッセージを返す、あるいは、リストの先頭のシンボルをコマンドとして扱い、なにがしかのことをする。(普通は勿論、最後の動作があなたの求めるものであろう。)

以前のセクションで挙げた例の中での、リストの頭に付けた単独のアポストロフィ ' は、引用符ないしは クウォート (*quote*) と呼ばれる。これがリストの頭に付いている場合は、何も付いていない場合と異なり、Lisp はこのリストについては何もしなくていいんだなと解釈する。しかし、リストの頭に引用符が付いていない場合は、リストの最初の項が特別な意味を持つ。つまり、これは計算機が従うべき命令になるのだ。(Lisp ではこれらの命令は関数 (*function*) と呼ばれる。) 以前挙げた `(+ 2 2)` というリストには頭に引用符は付いていないので、Lisp は `+` が残りのリストに関する何らかの命令であると解釈する。この場合は、残りの数を加えるというものである。

もし、この文章を GNU Emacs の Info の中で読んでいるのなら、次のようにしてこのようなリストを評価することが出来る。カーソルを次のリストの右側の括弧のすぐ後に持ってきて、*C-x C-e* とタイプするのである。

```
(+ 2 2)
```

エコー領域に `4` が見えたはずだ。(専門用語では、このことを「リストを評価する」と言う。エコー領域というのはスクリーンの最下行のことであり、テキストを表示ないしは「エコー」する。) さて、同じことを頭に引用符が付いたリストでやってみよう。カーソルを下のリストのすぐ後に移動させ、*C-x C-e* とタイプする。

```
'(this is a quoted list)
```

この場合はエコー領域に `(this is a quoted list)` という文が見えたはずだ。

いずれの場合にも、あなたはコマンドを GNU Emacs の *Lisp* インタプリタ (*Lisp interpreter*) と呼ばれるプログラムに渡している。つまり、インタプリタにその S 式を評価しろという命令を与えているのだ。Lisp インタプリタという名前は、S 式の意味を提供する—つまりそれを解釈する—という人間の仕事を表わす単語から来ている。

リストの一部ではない、単独のアトム—括弧に囲まれていないもの—を評価することも出来る。この場合も Lisp インタプリタは人間が読むことの出来る S 式から、計算機の内部の言語への翻訳をする。しかし、これについて議論する前に (Section 1.7 "変数", page 7, 参照)、まず我々がエラーを犯した場合に Lisp インタプリタが何をするかを見ていこう。

1.3 エラーメッセージの出力

まず、Lisp インタプリタがエラーメッセージを出すようなコマンドを与えてみよう。偶然このようなことをしてしまっても何も心配することはない。これは、無害な操作である。実際、我々はしばしば意図的にエラーメッセージを出させることがある。一旦専門用語を理解してしまえば、エラーメッセージから多くの情報を得ることが出来る。そういう意味ではエラーメッセージと言うよりはヘルプメッセージと言うべきだろう。言わば他の国から来た旅行者にとっての道標のようなものである。解読するのは大変だけれども、一度理解してしまえば正しい道を教えてくれるというわけだ。

The error message is generated by a built-in GNU Emacs debugger. We will 'enter the debugger'. You get out of the debugger by typing `q`.

What we will do is evaluate a list that is not quoted and does not have a meaningful command as its first element. Here is a list almost exactly the same as the one we just used, but without the single-quote in front of it. Position the cursor right after it and type *C-x C-e*:

```
(this is an unquoted list)
```

`*Backtrace*` ウィンドウが開いて、以下のように表示される:

```
---------- Buffer: *Backtrace* ----------
Debugger entered--Lisp error: (void-function this)
  (this is an unquoted list)
  eval((this is an unquoted list))
  eval-last-sexp-1(nil)
  eval-last-sexp(nil)
  call-interactively(eval-last-sexp)
---------- Buffer: *Backtrace* ----------
```

(未訳) Your cursor will be in this window (you may have to wait a few seconds before it becomes visible). To quit the debugger and make the debugger window go away, type:

```
q
```

(未訳) Please type `q` right now, so you become confident that you can get out of the debugger. Then, type `C-x C-e` again to re-enter it.

(未訳) Based on what we already know, we can almost read this error message.

(未訳) You read the `*Backtrace*` buffer from the bottom up; it tells you what Emacs did. When you typed `C-x C-e`, you made an interactive call to the command `eval-last-sexp`. `eval` is an abbreviation for 'evaluate' and `sexp` is an abbreviation for 'symbolic expression'. The command means 'evaluate last symbolic expression', which is the expression just before your cursor.

(未訳) Each line above tells you what the Lisp interpreter evaluated next. The most recent action is at the top. The buffer is called the `*Backtrace*` buffer because it enables you to track Emacs backwards.

(未訳) At the top of the `*Backtrace*` buffer, you see the line:

```
Debugger entered--Lisp error: (void-function this)
```

(未訳) The Lisp interpreter tried to evaluate the first atom of the list, the word '`this`'. It is this action that generated the error message '`void-function this`'.

(未訳) The message contains the words '`void-function`' and '`this`'.

'関数' という単語は前に一度出て来た。これは極めて重要な単語である。ここでは 関数 (*function*) とは計算機に何かやらせるための命令の集まりである、とでも定義しておけば十分だろう。

以上で、'`void-function this`' というエラーメッセージを理解することが出来るようになる。これは、シンボル (つまり、'`this`' という単語) には計算機が実行出来るような命令セットは定義されていないということである。

この、'`void-function`' というちょっと奇妙なメッセージは Emacs Lisp の実装の仕方を表わすように書かれている。つまり、そのシンボルに対して関数が定義されていない場合、命令が含まれているはずの場所が空 (void) だというわけである。

一方で、`(+ 2 2)` を評価した時にちゃんと 2 に 2 を加えることが出来るということから、シンボル `+` は計算機が従うべき命令セットを持っており、それらの命令は `+` に続く数を加えるという物であるに違いないと推測出来る。

(未訳) It is possible to prevent Emacs entering the debugger in cases like this. We do not explain how to do that here, but we will mention what the result looks like, because you may encounter a similar situation if there is a bug in some Emacs code that you are using. In such cases, you will see only one line of error message; it will appear in the echo area and look like this:

```
Symbol's function definition is void: this
```

The message goes away as soon as you type a key, even just to move the cursor.

We know the meaning of the word '`Symbol`'. It refers to the first atom of the list, the word '`this`'. The word '`function`' refers to the instructions that tell the computer what to do. (Technically, the symbol tells the computer where to find the instructions, but this is a complication we can ignore for the moment.)

The error message can be understood: '`Symbol's function definition is void: this`'. The symbol (that is, the word '`this`') lacks instructions for the computer to carry out.

1.4 シンボルの名前と関数定義

我々は今まで議論してきたことに基づいて、別の Lisp の特徴—それも極めて大切な特徴—を述べることが出来る。即ち `+` のようなシンボルはそれ自身は計算機が実行出来るような命令ではないということである。その代わりシンボルは、多分、一時的に、命令の定義ないしは命令の集まりの位置を見つける方法として使うことが出来る。我々に見えているのは、それを通して命令を見つけることの出来る名前なのである。人間の名前も同じ働きをする。私は '`Bob`' として言及される。しかしながら、私は '`B`'、'`o`'、'`b`' という文字ではなく、一貫して特定の生命体に付随しているある意識である。名前は私そのものではないが、私を示すのに使われる。

Lisp では、ある命令セットには幾つかの名前が付随している。例えば、数を加えろという計算機の命令には、+ というシンボルと同様、plus というシンボルも付随させることが出来る。(他の幾つかの Lisp の方言では実際にそうなっている。) 人間の場合でも、私は 'Bob' だけではなく 'Robert' として呼ばれることもあるし、他の呼び方をされることもある。

一方で、一つのシンボルは、一度にただ一つの関数定義しか持つことが出来ない。そうでないと、計算機がどの定義を使うべきか迷ってしまうからだ。仮にこれが人間の場合であれば、'Bob' という名前は世界で一人の人間にしかつけてはいけないということである。ただし、名前に付随する関数定義は簡単に変更することが出来る。(Section 3.2 "関数定義のインストール", page 20, 参照。)

Emacs Lisp はでかいので、関数が属する Emacs のパートが識別出来るようなシンボルの名前を付け方をするのが普通である。というわけで、Texinfo を扱うための関数の名前は全て 'texinfo-' で始まっているし、メールを読むための (reading mail) 関数の名前は 'rmail-' で始まっている。

1.5 Lisp インタプリタ

これまでに見てきたことから、リストを評価するよう命令した時に Lisp インタプリタが何をするかを理解することが出来る。まずインタプリタは、list の頭に、引用符が付いているか見る。もし付いていれば、単にそのリストを我々に渡す。一方、付いていない場合はリストの先頭の要素を見に行き、それが関数の定義を持っているかどうかを調べる。定義されている場合はその定義にある命令を実行し、そうでなければエラーメッセージを表示する。

これが Lisp の動作である。単純だ。これに加えて若干複雑なこともある。それについてはすぐ後で説明するが、基本はこれだけである。勿論、Lisp のプログラムを書く際には、関数の定義を書いたりそれに名前を付けるにはどうすればいいか、あるいはそういうことをあなた自身や計算機が混乱しないように行うにはどうすればいいか等、他にも知らなければいけないことはあるが。

さて、さっき述べた「複雑なこと」の一つ目は、Lisp インタプリタは、リストに加えて引用符が付いておらず、括弧に囲まれてもいない単独のシンボルも評価することが出来るということである。この場合、Lisp インタプリタはそのシンボルの値を変数 (*variable*) として評価しようとする。このことについては変数のセクションで説明することにする。(Section 1.7 "変数", page 7, 参照。)

複雑なことの二つ目は、幾つかの関数はその特殊性のために普通のやり方ではうまく働かないことによる。このような関数は 特殊形式 (*special form*) と呼ばれる。これらは、例えば関数を定義したりといった特別な用途を持つ。次の幾つかのセクションでもっと大切な特殊形式を紹介するつもりだ。

As well as special forms, there are also *macros*. A macro is a construct defined in Lisp, which differs from a function in that it translates a Lisp expression into another expression that is to be evaluated in place of the original expression. (See Section 8.2.2 "Lisp macro", page 66.)

For the purposes of this introduction, you do not need to worry too much about whether something is a special form, macro, or ordinary function. For example, `if` is a special form (see Section 3.7 "if", page 26), but `when` is a macro (see Section 8.2.2 "Lisp macro", page 66). In earlier versions of Emacs, `defun` was a special form, but now it is a macro (see Section 3.1 "defun", page 19). It still behaves in the same way.

三つ目の、そして最後の複雑なことはこうだ。もし現在 Lisp インタプリタが見ている関数が特殊形式ではなく、しかもそれがリストの一部であるなら、Lisp インタプリタはリストが内部にリストを含んでいないかどうかを見る。もし内部にリストを含んでいれば、それがその内部でどういうことをするかを見る。そして、その後にその外側のリストを処理する。また、もしその内部のリストが更に他のリストを中に含んでいれば、まず、それを先に処理してからということになる。即ち、Lisp インタプリタは常に一番内部のリストから処理していく。これは、内部のリストの結果をそれを包む外側の S 式で使う場合、内部のリストの結果がどうなるかを見てから外側のリストを処理しなければいけないためである。

それ以外の場合は、インタプリタは左から右へ S 式を一つずつ処理していく。

1.5.1 バイトコンパイル

解釈の仕方について、もう一つ興味深いことがある。 Lisp インタプリタは二つの種類の物を解釈出来るということである。一つは人間が読めるコードであり、我々は専らこれに焦点を当てている。もう一つは特別なプロセスコードであり、これはバイトコンパイル (*byte compile*) されたコードで、人間に読める代物ではない。バイトコンパイルされたコードは我々が読めるコードよりも速く走らせることが出来る。

人間が読めるコードをバイトコンパイルされたコードに変換することも出来る。これは `byte-compile-file` 等の、コンパイルコマンドを走らせることによって行う。バイトコンパイルされたコードは普通、`.el` ではなく、`.elc` という拡張子で終わるファイルに保存される。`emacs/lisp` などといったディレクトリィを覗けば、この両方の種類のファイルが見つかる。我々が読むのは `.el` という拡張子の付いた方である。

実際問題として、あなたが Emacs をカスタマイズしたり拡張したりする場合には大抵はバイトコンパイルする必要はない。そして私もここではこの話題については述べない。Section “Byte Compilation” in *The GNU Emacs Lisp Reference Manual*, を見ればバイトコンパイルについての全ての情報が載っている。

1.6 評価

Lisp インタプリタがある S 式を処理している時、その動作は 評価 (*evaluation*) と呼ばれる。つまり、「インタプリタは S 式を評価する」のように言うわけだ。私はこの用語を以前にも何回か用いた。この単語はこの言葉の日常会話での意味、*Webster's New Collegiate Dictionary* (及び、小学館 『プログレッシブ英和中辞典』) によれば「価値や量を確かめる、概算する、見積もる」という意味、から来ている。

Lisp インタプリタは S 式を評価した後、大抵その関数の定義の中にある計算機に対する命令を実行した時に出力する値を 返し (*return*)、そうでない場合は、おそらくその関数を処理するのを止めてエラーメッセージを出力する。(インタプリタは、急に別の関数を渡されることもあり得るし、あるいは、いわゆる「無限ループ」に入って現在の処理を何回でも繰り返そうとするかもしれない。が、これらの動作をすることはそれ程多くはないので、我々はこれらの場合は無視することにする。) つまり殆どの場合、インタプリタは値を返す。

インタプリタは値を返すと同時に、カーソルを動かしたり、ファイルをコピーしたりといった他の動作も行う。このような動作は、副作用 (*side effect*) と呼ばれる。例えば結果を表示するとか、我々人間が重要と考える動作もしばしば、Lisp インタプリタにとっては「副作用」でしかない。この専門用語には異和感があるかもしれないが、副作用自体の使い方を学ぶのはかなり簡単だということがその内分るだろう。

以上まとめると、Lisp インタプリタは S 式を評価した時に、まず殆どの場合ある値を返し、大抵はある副作用を実行する。そして、そうでない場合はエラーを出力する、ということになる。

1.6.1 内部のリストの評価

もしあるリストの内部のリストが評価された場合、その外側のリストが評価される時には最初に内部のリストを評価した時に返った値を情報として使うことが出来る。このことから、何故内部のリストが先に評価されるかが説明出来る。即ち、内部の S 式が返す値を外部の S 式が用いるためである。

我々はこのプロセスを次のような例で実際に確かめることが出来る。次の式の直後にカーソルを持っていき、*C-x C-e* とタイプしてみよう。

```
(+ 2 (+ 3 3))
```

数字の 8 がエコー領域に表示されたはずだ。

この例で起こったことはこうだ。まず Lisp インタプリタは内部の S 式である `(+ 3 3)` を評価する。それに対して値 6 が返される。それによって外部の S 式があたかも `(+ 2 6)` であるように評価され、値 8 を返すことになる。これらを内部に含む評価すべき S 式はこれ以上存在しないので、インタプリタはこの値をエコー領域に表示する。

さて、ここまでくれば *C-x C-e* で引き起こされるコマンドの名前を理解することは簡単だ。このコマンドの名前は `eval-last-sexp` となっている。この中の `sexp` という文字は 'symbolic expression' (S 式) の略である。ということで、これは直前の S 式を評価する (evaluate last symbolic expression) ことを表している。

実験的に、次に挙げる例の S 式の直後の行の先頭や S 式の内部にカーソルを持っていって S 式を評価してみることも出来る。

これがその例である。

```
(+ 2 (+ 3 3))
```

もしカーソルを S 式の直後の空行の先頭に置いて *C-x C-e* とタイプしたなら、その場合もやはり 8 がエコー領域に表示されることだろう。では次に、カーソルを S 式の内部に持っていこう。最後の括弧のすぐ右側の括弧の後にカーソルを持っていって (つまりカーソルは最後の括弧の上に見えている) そこ

で評価したならば、エコー領域には 6 と表示される。これはコマンドが (+ 3 3) という式を評価したからである。

今度はカーソルを数字の直後に持っていこう。*C-x C-e* とするとその数字そのものが返される。Lisp の中では、数字を評価した場合、その値自身が返される。これが数字がシンボルと違うところである。もし + みたいなシンボルを先頭に持つリストの直後で S 式を評価したなら、返された値はその名前に付随する関数の定義の中の命令を計算機が実行した結果が返ってくる。が、もしシンボルそのものが評価されたなら別のことが起きる。これについては次のセクションで見ることにしよう。

1.7 変数

Lisp の中では、シンボルは、関数定義を持つことが出来るのと同様に、ある値を持つことが出来る。これら二つは全く別の物である。関数定義は計算機が従うべき命令の集まりであるのに対し、値は数値とか名前のように、何か変化し得るものである。(このようなシンボルが変数と呼ばれるのはこのためである。) どんな Lisp の S 式もシンボルの値になり得る。例えばシンボルや数値、リスト、文字列なんかが値となることが出来る。値としてのシンボルはしばしば変数 (*variable*) と呼ばれる。

シンボルは、関数定義と値の両方を同時に持つことが出来る。この二つは分離されているのだ。これは、ケンブリッジという名前がマサチューセッツの中のある都市そのものを表わすのと同様に、その都市名に付随するなんらかの情報、例えば、「大きなプログラミングセンター」であるといった情報をも持ち得るのと似ている。

このことを理解するもう一つの考え方を述べよう。それはシンボルを整理棚であると見做す方法である。関数定義はその引き出しの一つに入っており、値はまた別の引き出しに入っている... という具合である。値の入っている引き出しの中身は同じ棚の関数定義の入っている引き出しの中身に影響を与えることなく変えることが可能だし、逆も同様である。

変数 `fill-column` を例にとって、値を持つシンボルというものを説明してみよう。GNU Emacs の全てのバッファにおいて、このシンボルはある値にセットされている。大抵は 72 か 70 であるが、他の値を持つこともある。この値が何かを見るにはこのシンボル自身を評価してみれば良い。もしこの文章を GNU Emacs の Info で読んでいるなら、カーソルを次のシンボルの直後に持っていって *C-x C-e* とタイプするだけである。

```
fill-column
```

私が今 *C-x C-e* とタイプしてみたところ、Emacs はエコー領域に 72 と表示した。これが、私がこの文章を書いている時に `fill-column` にセットされている値である。あなたの Info のバッファではまた別の値が設定されているかもしれない。変数として評価されて返す値は、ある関数が命令を実行した時に値を返す時と全く同じように表示されることに注意しよう。Lisp インタプリタの立場からすると、どちらも返り値であることに違いはない。どんな種類の S 式から来ているかということは一旦その値が分ってしまえば重要ではないのだ。

シンボルはどんな値も持つことが出来る。あるいは専門用語を使っていうなら我々は変数に 72 といった数値や `"such as this"` といった文字列や、`(spruce pine oak)` をバインド (*bind*) (束縛とも言う) することが出来る。変数に関数定義をバインドすることすら可能である。

シンボルに値を設定するには幾つか方法がある。その方法の一つについては、Section 1.9 “変数の値の設定”, page 12, を参照のこと。

1.7.1 Error Message for a Symbol Without a Function

`fill-column` という単語には周りに括弧がついていないことに注意しよう。これは我々がこのシンボルを関数の名前としては使わなかったためである。

`fill-column` がリストの最初、もしくは唯一の要素であった場合、Lisp インタプリタはこれに付随する関数定義を見つけようとする。しかし、`fill-column` は関数定義を持ってはいない。例えば次を評価してみよう。

```
(fill-column)
```

You will create a *Backtrace* buffer that says:

```
---------- Buffer: *Backtrace* ----------
Debugger entered--Lisp error: (void-function fill-column)
  (fill-column)
  eval((fill-column))
  eval-last-sexp-1(nil)
  eval-last-sexp(nil)
  call-interactively(eval-last-sexp)
---------- Buffer: *Backtrace* ----------
```

(Remember, to quit the debugger and make the debugger window go away, type *q* in the *Backtrace* buffer.)

1.7.2 値のないシンボルに対するエラーメッセージ

もし、値を持たないシンボルを評価しようとしたなら、エラーメッセージが返されるはずだ。2 に 2 を加える例で実験してみよう。次の S 式のなかで最初の 2 の直前にある + の直後にカーソルを持っていって *C-x C-e* とタイプしてみよう。

```
(+ 2 2)
```

(未訳) In GNU Emacs 22, you will create a *Backtrace* buffer that says:

```
---------- Buffer: *Backtrace* ----------
Debugger entered--Lisp error: (void-variable +)
  eval(+)
  eval-last-sexp-1(nil)
  eval-last-sexp(nil)
  call-interactively(eval-last-sexp)
---------- Buffer: *Backtrace* ----------
```

(未訳) (Again, you can quit the debugger by typing *q* in the *Backtrace* buffer.)

このバックトレースは我々が見た最初のエラーメッセージとは違っている。さっきは 'Debugger entered--Lisp error: (void-function this)' というものだった。今回のケースでは関数は変数としての値を持っておらず、もう一方のケースではシンボル ('this' という単語) は関数定義を持たなかった。

この + を使った実験の中でやったことは Lisp インタプリタに + を評価させ、関数定義では無く変数の値を探させることである。そのためにカーソルを前のようにリストを囲む括弧の後では無く、シンボルのすぐ右隣に持っていったのだった。結果として Lisp インタプリタは直前の S 式、この場合は + そのもの、を評価したのだ。

+ は関数定義は持っていても値は持ってはいないので、エラーメッセージはシンボルの変数としての値が空だよと報告したのである。

1.8 引数

どうしたら情報が関数に渡されるかを見るために、これまで何回か使ってきた 2 に 2 を足す例をもう一度見てみよう。

```
(+ 2 2)
```

もしこの S 式を評価したなら数字の 4 がエコー領域に表示される。Lisp インタプリタがやったことは + の後に続く数字を加えることである。

+ によって加えられた数字は関数 + の 引数 (*argument*) と呼ばれる。これらの数字は関数に与えられる、もしくは渡される (*pass* される) 情報である。

「argument」という単語は数学での用法から来ているもので、二人の人が議論するという意味ではない。そうではなく関数、今の場合でいうと + という関数、に与えられる情報のことを指している。Lisp の中では関数に対する引数は関数に続くアトムもしくはリストである。これらのアトムやリストを評価して返された値が関数に渡されることになる。異なる関数は異なる数の引数を取り得る。全く引数を取らない関数もある。[1]

[1] 「argument」という単語が二つの異なる意味を持つようになった経緯を追ってみるのは、結構興味深いことである。一つは数学での意味であり、もう一つは日常使われている英語での意味である。Oxford English Dictionary によると、この単語は 'to make clear, prove' ('明確にする、証明する') という意味のラテン語から来ているということだ。その結果、一方では「証明として与えられる証拠」

1.8.1 引数のデータ型

関数に渡されるデータの型はその関数がどんな種類の情報を必要としているかによる。+ のような関数の引数は数値でなければならない。というのも+ は数値を加える命令だからである。他の関数は引数として他の種類のデータを取る。

例えば、concat という関数は二つ以上のテキストの文字列を連結ないしは合成して一つの文字列を作る。この場合の引数は文字列である。二つの文字文字列 abc, def を連結 (concatenate) すると、abcdef という一つの文字列が出来る。これは次の例で確かめられる。

```
(concat "abc" "def")
```

このS式を評価すると "abcdef" という値が返される。

substring のような関数は引数として文字列と数値の両方を取る。これは文字列である最初の引数の部分文字列を返す関数である。引数の数値は3つである。最初の引数は文字からなる文字列である。二番目と三番目の引数は部分文字列の初めと終りの場所を示す数値であり、文字列の最初からの (スペースや句読点も含めた) 文字数を指定する。

例えば次を評価してみよう。

```
(substring "The quick brown fox jumped." 16 19)
```

(未訳) you will see "fox" appear in the echo area. The arguments are the string and the two numbers.

substring に渡される文字列は、たとえスペースで区切られた複数の単語からなっていたとしても一つのアトムであることに注意しよう。Lisp は二つの二重引用符の間に狭まれる全てを空白も含めて文字列の一部と数える。substring という関数は一種の「アトム粉砕機 (atom smasher)」だと見做すことが出来る。というのもこいつは他の方法では分離できないアトムを取り込み、その一部を抽出するからである。しかしながら、substring は文字列である引数から部分文字列を抽出することが出来るだけで、他の種類の例えば数値とかシンボルであるアトムについてはそのようなことは出来ない。

1.8.2 引数には変数の値やリストも使える

引数は評価された時に値を返すシンボルであることも可能だ。例えば fill-column というシンボルそれ自身が評価された場合には、数値が返される。この数値は加法演算にも使える。

次のS式の後にカーソルを持っていって *C-x C-e* とタイプしてみよう。

```
(+ 2 fill-column)
```

返される値は fill-column を単独で評価した場合に得られる数に 2 を加えたものになるはずだ。私には 74 が返される。何故なら fill-column の値が 72 であるからだ。

たった今見た様に、引数は評価された時に値を返すシンボルであり得る。それに加えて、引数は評価された時に値を返すリストでもあっても良い。例えば次のS式の中では関数 concat の引数は "The ", "red foxes." そしてリスト (+ 2 fill-column) である。

```
(concat "The " (number-to-string (+ 2 fill-column)) " red foxes.")
```

もしこのS式を評価したならエコー領域には "The 74 red foxes." が表示されるはずだ。(最後の文字列を表示させるには 'The' という単語の後と 'red' という単語の前に空白を入れなければならないことに注意。) (訳註：74 の所は勿論場合に応じて変わる。)

1.8.3 可変な数の引数

例えば concat, +, * といった幾つかの関数では引数の数は固定されていない。(* は乗法を表わすシンボルである。) これは次に挙げる各々のS式を評価してみると分る。エコー領域に表示されるべき結果はこのテキストの中の '⇒' の後に書かれている。これは「を評価した結果は」と読み換えられる。

から「与えられた情報」という意味になり、それが Lisp での意味の方に派生していった。しかし、もう一方では「他人が主張するのと対立するような主張をする」という意味に派生し、それが議論することを表わす単語の意味になっていった。(ここでこの英単語が同時に二つの異なる定義を持っていることに注意しよう。これに対して Emacs Lisp では、シンボルは同時に二つの関数定義を持つことは出来ない。)

最初は引数がない場合である。

```
(+)       ⇒ 0

(*)       ⇒ 1
```

こちらは一つだけ引数を持っている。

```
(+ 3)     ⇒ 3

(* 3)     ⇒ 3
```

次では、各々三つずつ引数を持っている。

```
(+ 3 4 5) ⇒ 12

(* 3 4 5) ⇒ 60
```

1.8.4 関数に間違った型の引数を与えると

ある関数に間違った型の引数が与えられると、Lisp インタプリタはエラーメッセージを返す。例えば、`+` という関数は引数の値として数値を要求する。実験的に、これに数値ではなく `hello` というシンボルを与えてみよう。次の S 式の後にカーソルを持っていって *C-x C-e* とタイプしてみよう。

```
(+ 2 'hello)
```

そうするとエラーメッセージが表示される。この場合に起きたことはこうだ。`+` は 2 に `'hello` を評価して返される値を加えようとした。しかし返された値は `hello` というシンボルであり数値ではない。だが数値でないものを足すことは出来ない。そのために、`+` は加算を実行することが出来なかったのである。

（未訳）You will create and enter a `*Backtrace*` buffer that says:

```
---------- Buffer: *Backtrace* ----------
Debugger entered--Lisp error:
         (wrong-type-argument number-or-marker-p hello)
  +(2 hello)
  eval((+ 2 (quote hello)))
  eval-last-sexp-1(nil)
  eval-last-sexp(nil)
  call-interactively(eval-last-sexp)
---------- Buffer: *Backtrace* ----------
```

（未訳）As usual, the error message tries to be helpful and makes sense after you learn how to read it.[2]

エラーメッセージの最初の部分は極めて直接的だ。‘`Wrong type argument`’ (引数の型が違う) である。次に来るのはミステリアスな専門用語 ‘`number-or-marker-p`’ である。この言葉はあなたに `+` という関数が要求する引数のタイプを教えようとしている。

`number-or-marker-p` というシンボルの名前は、Lisp インタプリタが与えられた情報 (引数の値) が数値 (number) もしくはマーカ (marker) (バッファでの位置を表わす特殊なオブジェクト) であるかどうかを判定することを意味している。今の場合、実際に行うのは `+` にちゃんと加えるべき整数が与えられたかどうかのテストである。そして、マーカと呼ばれる特殊オブジェクトかどうかもテストされる。これは Emacs Lisp 固有のものである。(Emacs の中ではバッファ内での位置がマーカとして記録される。*C-@* あるいは *C-SPC* というコマンドによってマークがセットされるとその位置がマーカとして保持されるのである。このマークは数—バッファの始まりからその位置までの文字数—と考えることが出来る。) Emacs Lisp では `+` はマーカの位置を示す数を数値として加えるのにも使うことが出来るわけである。

`number-or-marker-p` の中の ‘`p`’ は Lisp プログラミングの初期の頃に始まったある試みの具体化である。‘`p`’ は「述語 (predicate)」を表わしている。初期の Lisp 研究者の間で用いられていた専門用語では「述語」はある属性が正しいかどうかを判定する関数のことを指していた。従って、‘`p`’は `number-or-marker-p` が、与えられた引数が数値 (number) もしくは マーカであるかどうかを判定する関数の名前であることを意味している。他の Lisp のシンボルで ‘`p`’ で終わるものに `zerop` がある。これは引数が 0 という値であるかどうかを判定する関数である。同様に `listp` は引数がリストかどうかを判定する関数である。

[2] `(quote hello)` is an expansion of the abbreviation `'hello`.

さて、いよいよエラーメッセージの最後の部分 hello であるが、これは + に渡された引数の値である。もし加法演算に正しい型のオブジェクトが与えられていたとするなら、渡される値は hello のようなシンボルではなく、例えば 37 のような数値であったはずである。もっとも、その場合は、エラーメッセージは出ないけれども。

1.8.5 関数 message

+ と同様、message という関数も引数の数は固定されていない。この関数はユーザにメッセージを送るものだが、今説明したように、いろいろと役に立つものである。

メッセージはエコー領域に表示される。例えば次のリストを評価することで、あなたのエコー領域にメッセージを表示させてみよう。

```
(message "This message appears in the echo area!")
```

二重引用符に挟まれた文字列全体が一つの引数になっており、その全体が表示される。(この例ではメッセージそれ自身が二重引用符の中に入っていることに注意しよう。これは、message という関数が返す値そのものがそうなっているからである。あなたが書くプログラムの中での普通の message の使い方ではエコー領域に表示されるのは副作用によるものである。それには、二重引用符は付いていない。このような例については Section 3.3.1 “インタラクティブな multiply-by-seven”, page 22, を参照のこと。)

しかしながら、もし ‘%s’ が二重引用符に挟まれた文字文字列に入っていたとしたら、関数 message は ‘%s’ をそのまま表示したりはせずに、文字列に続く引数を見にいく。まずは二番目の引数を評価し、その結果を ‘%s’ のある所に代入して表示する。

次の S 式の最後にカーソルを持っていって *C-x C-e* とタイプしてみることで、このことが確かめられる。

```
(message "The name of this buffer is: %s." (buffer-name))
```

Info の中では、"The name of this buffer is: *info*." という文がエコー領域に表示されたことだろう。buffer-name という関数は現在のバッファ名を文字列として返す。message はそれを %s の位置に挿入したのだ。

値を 10 進数として表示するには、‘%s’ の代わりに ‘%d’ という関数を同じように使う。例えば、現在の fill-column の値をエコー領域に表示させるには次を評価すれば良い。

```
(message "The value of fill-column is %d." fill-column)
```

(未訳) 私の使っているシステムでは、このリストを評価した時点ではエコー領域には"The value of fill-column is 72." と表示された [3]。

もし一つ以上の ‘%s’ が二重引用符で狭まれた文字列の中にあったとすると、最初の ‘%s’ の位置にはその文字列に続く最初の引数の値が表示され、二番目の ‘%s’ の位置には二番目の引数の値が表示され... という具合になる。

例として次を評価してみよう。

```
(message "There are %d %s in the office!"
         (- fill-column 14) "pink elephants")
```

かなり風変わりなメッセージが表示されたことと思う。私のシステムでは、"There are 58 pink elephants in the office!"。のように表示された。

ここでは (- fill-column 14) という S 式が評価され、その結果として返された値が ‘%d’ の位置に挿入される。そして二重引用符で狭まれた"pink elephants" という文字列が一つの引数として扱われて、‘%s’ の位置に置かれる。 (つまり、二重引用符で挟まれた文字列は、評価された時に、数値と同じ様にそれ自身を値として返すわけである。)

最後に、単なる数の計算ではなく、S 式の中に S 式を置き、‘%s’ で置き換えることで文章を作る例として、ちょっと複雑なものを挙げておこう。

```
(message "He saw %d %s"
         (- fill-column 32)
         (concat "red "
                 (substring
                  "The quick brown foxes jumped." 16 21)
                 " leaping."))
```

[3] Actually, you can use %s to print a number. It is non-specific. %d prints only the part of a number left of a decimal point, and not anything that is not a number.

この例では、message は 3 つの引数を取る。"He saw %d %s" という文字列と、(- fill-column 32) という S 式と、concat という関数で始まる S 式である。(- fill-column 32) を評価して得られた結果は '%d' の位置に挿入され、concat から始まる S 式を評価して返される値は '%s' の位置に挿入される。

私がこの式を評価したところ、エコー領域には "He saw 38 red foxes leaping." というメッセージが表示された。

1.9 変数の値の設定

変数に値を格納する方法は幾つかある。一つの方法は、set もしくは setq という関数を使うことである。また別の方法として、let を使うというのもある。(Section 3.6 "let", page 24, 参照). (このプロセスのことを専門用語では、変数に値をバインド (束縛) すると言う。)

この後のセクションでは set と setq の働きを説明するだけではなく、引数の渡され方の様子も述べることにする。

1.9.1 set の利用

シンボル flowers の値としてリスト '(rose violet daisy buttercup) をセットするために、次の S 式の後にカーソルを持っていって *C-x C-e* とタイプして、この式を評価してみよう。

```
(set 'flowers '(rose violet daisy buttercup))
```

すると (rose violet daisy buttercup) というリストがエコー領域に表示されたはずだ。これは関数 set によって返された (*return* された) 値である。その副作用としてシンボル flowers が、このリストにバインドされる。つまり、シンボル flowers は、変数とみなすことが出来て、その値がこのリストであるようになったということである。(因みにこのプロセスは、値をセットするという Lisp インタプリタにとっては副作用でしかないことが、我々人間にとっては主効果たりうるということの説明になっている。Lisp インタプリタはエラーが出ない限り必ず値を返すが、副作用の方は、そのように意図しないと出ないものなのである。)

さっきの set を使った S 式を評価した後では、flowers というシンボルを評価することが出来るようになっている。そして、たった今、我々がセットした値を返す。実際に次のシンボルの後にカーソルを持っていって *C-x C-e* とタイプしてみよう。

```
flowers
```

flowers を評価すると、エコー領域にリスト (rose violet daisy buttercup) が現れたはずだ。

もし間違えて頭に引用符を付けた 'flowers の方を評価したとすると、エコー領域に現れるのはこのシンボル flowers それ自身になる。引用符を付けたやつを次に挙げておくので、評価してみて欲しい。

```
'flowers
```

もう一つ注意すべきことは、set を利用する時は、set の引数には、それらを評価しようとする場合を除いて、必ず引用符を付ける必要があるということである。先の例の場合では、引数 flowers 及び (rose violet daisy buttercup) のどちらも評価したくはなかったので、両方に引用符を付けている。(set の一番目の引数に引用符を付けなかった場合、まずまっ先にその引数が評価される。これを flowers に対してやったとして、もしこのシンボルがまだ値を持っていなかったとすると、'Symbol's value as variable is void' というエラーメッセージが表示される。一方、もし flowers がある値を返したとすると、set はそいつにある値を設定しようとする。勿論こうなるのが正しい動作である状況もあるが、そういうことは稀である。)

1.9.2 setq の利用

実際問題として、大抵の場合は第一引数には引用符をつけることになるだろう。set と引用符付きの第一引数の組み合わせというのは極めてよく使われるために、それに対して名前がついている。それが特殊形式 setq である。この特殊形式はほぼ set と同じなのだが、第一引数に自動的に引用符を付けてくれる所だけが違う。従ってあなた自身で引用符を付ける必要はない。また、そのことに加えて、setq には、一つの S 式の中で複数の異なる変数に対して各々に異なる値をセットすることが出来る、という便利さもある。

setq を使って carnivores という変数に '(lion tiger leopard) というリストをセットするには次の S 式が使われる。

```
(setq carnivores '(lion tiger leopard))
```

これは set を使ったものと、第一引数に引用符が付いていないことを除き、全く同じである。(‘setq’ の q は quote を意味している。)

set を使うと上のS式は次のようになる。

```
(set 'carnivores '(lion tiger leopard))
```

また、setq は異なる変数の各々に異なる値をセットすることも出来ると書いた。この場合には一番目の引数には二番目の引数の値がバインドされ、三番目の引数には四番目の引数の値がバインドされ、という風になる。例えば次のS式を使うとシンボル tree には樹木のリストを、シンボル herbivores には草食動物のリストをセットすることが出来る。

```
(setq trees '(pine fir oak maple)
      herbivores '(gazelle antelope zebra))
```

(このS式は一行で書いても構わないのだが、そうなるとページの大きさには合わなってしまう。人間が読む場合には、こういう形式の方が読みやすいことがお分りいただけると思う。)

これまで何度か「セットする」という言葉を使ってきたけれども、set とか setq の働きにはもう一つの捉え方がある。それは set や setq はそのシンボルを特定のリストへのポイントにする、というものだ。このような考え方は大変一般的であり、この後の章で、少なくとも一つ、名前の中にそのポインタを持つシンボルに出逢うことになる。そういう名前が付けられたのはそのシンボルに、ある値、より詳しくはリスト、がセットされているためである。もう一つの言い方では、そのシンボルはそのリストへのポインタに設定されているという風に表現出来る。

1.9.3 カウント

次の例は、setq をカウンタとして利用する方法を説明してくれる。これは、プログラムの中である部分が何回繰り返されたかを数える場合に使うものである。最初に変数を 0 に設定し、その後、プログラムの中でその部分が現われるごとに一つずつ足していくのである。そのためにはカウンタとなる一つの変数に加えて二つのS式を用意する必要がある。初めにこのカウンタを 0 にセットするための setq 式と、評価されるごとにその変数を一ずつ増やすための setq 式である。

```
(setq counter 0)                ; これをイニシャライザと呼ぼう。

(setq counter (+ counter 1))    ; これはインクリメンタと呼ぶ。

counter                         ; これはカウンタ。
```

(‘;’ に続く文はコメントである。Section 3.2.1 “関数定義の変更”, page 21, 参照。)

もし最初のイニシャライザであるS式 (setq counter 0) を評価してから三番目の counter というS式を評価したなら、エコー領域には数値 0 が表示される。次に二番目のインクリメンタのS式 (setq counter (+ counter 1)) を評価すると、カウンタは一つ増える。従って、ここでもう一度 counter を評価するとエコー領域には数字 1 が表示される。その後も二番目のS式を評価するごとに値は増えていく。

インクリメンタ (setq counter (+ counter 1)) が評価されると Lisp インタプリタは最初にもっとも内側にあるリストを評価する。このリストを評価するために、まず counter, 次に 1 が評価される。変数 counter を評価すると、その時の値が返される。それが数 1 とともに + に渡されて足し合わされる。で、その和は内側のリストの値として setq に渡され、カウンタに新しい値が設定されることになる。以上のようにして、変数 counter の値が変更される。

1.10 まとめ

Lisp を学ぶことは、最初の道が急な坂であるような丘を登るようなものである。あなたが現在登っているのはもっとも急な部分である。先に進むに従って、より楽に進めるようになるはずだ。

まとめると

- Lisp のプログラムはリストもしくは単一のアトムからなるS式で出来ている。
- リストは 0 以上のアトムもしくはリストからなる。これらは空白文字で区切られており、括弧で囲まれている。リストの中身は空であっても良い。
- アトムは幾つかの文字からなるシンボルである。例えば forward-paragraph とか一つの文字からなる +、二重引用符に両側を挟まれた文字列からなる文字列、それに数値などがそうである。
- 数値は評価されるとそれ自身を返す。
- 二重引用符で挟まれた文字列を評価した時もそれ自身を返す。

- シンボルそのものを評価すると、その値が返される。
- リストを評価すると、Lisp インタプリタはまずそのリストの一番目のシンボルを見にいき、そのシンボルにバインドされている関数定義を見る。そしてその定義の中にある命令を実行する。
- 引用符’があると Lisp インタプリタはその後に続く S 式を書かれたままの形で返し、二重引用符が無い場合のように評価したりはしない。
- 引数は関数に渡される情報である。ある関数に渡される引数は、その関数を先頭に持つリストの残りの要素を評価したものにある。
- 関数は評価された時は (エラーが起きない限り) 常にある値を返す。また、それに加えて「副作用」と呼ばれる効果を生ずる。多くの場合、その関数を使う第一の目的はその副作用を生じさせることにある。

1.11 練習問題

幾つか単純な練習問題を挙げておく。

- 括弧に挟まれない適当なシンボルを評価することにより、エラーメッセージを発生させなさい。
- 括弧に挟まれた適当なシンボルを評価することにより、エラーメッセージを発生させなさい。
- 1 ではなく 2 ずつ増えるカウンタを作りなさい。
- 評価された時にエコー領域に何かメッセージを表示するような S 式を書きなさい。

2 実際の評価の仕方

Emacs Lisp での関数の定義の書き方を学ぶ前に、ちょっと時間を取って、今まで書いてきた様々な S 式を評価する方法について述べておこう。これらの S 式は関数を最初の (そしてしばしば唯一の) 要素として持つリストである。幾つかのバッファに関連する関数は、単純であるし、興味深くもあるので、これらの関数から始めようと思う。また、別のセクションで、同じくバッファに関連する他のコードについて、それらがどのように書かれているか見てみるつもりだ。

Emacs Lisp に対して、カーソルを動かすとかスクリーンをスクロールするなどの編集コマンドを与える場合、あなたは常に *(*最初の要素が関数であるような*)* ある S 式を評価している。これが *Emacs* の動作の基本である。

あなたが何かキーを押すと、そのことで Lisp インタプリタはある S 式を評価する。そういう風にしてあなたは求む結果を得るのである。たとえプレーンテキストをタイプしている時でも、Emacs Lisp の関数を評価させているのだ。この場合に呼ばれるのは `self-insert-command` であり、これは、単にあなたがタイプした文字を挿入するものである。キーを押すことで評価されるような関数はインタラクティブ (*interactive*) な関数、あるいは、コマンド (*command*) と呼ばれる。関数をインタラクティブにする方法は関数の定義の書き方の章で説明することにする。Section 3.3 “関数をインタラクティブにする”, page 21, 参照。

キーボードコマンドをタイプするのに加えて、我々はもう一つ S 式を評価する方法を学んだのだった。それはカーソルをリストの終端に持っていって *C-x C-e* とタイプするやり方である。他にも同様に S 式を評価する方法がある。それらについては他のセクションでその都度説明するつもりだ。

評価されるのに使われることに加えて、次の幾つかのセクションで出てくる関数は、それらそのものが元々重要である。これらの関数を学習することで、バッファとファイルの違いやバッファを切り替える方法、そしてその中での位置を定める方法についてよく理解出来るようになるだろう。

2.1 バッファの名前

`buffer-name` と `buffer-file-name` という二つの関数を見てみると、ファイルとバッファの違いがよく分るだろう。`(buffer-name)` という S 式を評価すると、バッファの名前がエコー領域に表示される。一方、`(buffer-file-name)` を評価すると、バッファに関連付けられているファイルの名前がエコー領域に表示される。普通は `(buffer-name)` で返される名前は、それに関連付けられているファイルの名前と一致しており、`(buffer-file-name)` はそのファイルのフルパスを返す。

ファイルとバッファは異なる種類の物である。ファイルというのは計算機に記録されている情報であり、それを消去しない限りはずっと残っているものである。一方でバッファは Emacs の内部の情報であり、その時々のセッション (訳註：つまり、Emacs を起動して終了するまで) が終わると (もしくはそのバッファを kill すると) 消え去ってしまうものである。普通はバッファはファイルからコピーした情報を保持している。このことを、バッファがそのファイルにビジット (*visit*) しているという。我々はこのコピーに対して編集などの仕事を行う。バッファに対する変更はそのバッファをセーブするまではファイルには反映されない。バッファをセーブすると、バッファの内容はファイルにコピーされて、そのまま記録されることになる。

もしこの文を GNU Emacs の Info で読んでいるのであれば、以下の S 式の直後にカーソルを持っていって *C-x C-e* とタイプすることで、各々の式を評価することが出来る。

```
(buffer-name)

(buffer-file-name)
```

私がこれらの式を Info で評価してみたところ、`(buffer-name)` を評価すると `"*info*"` という値が返され、`(buffer-file-name)` を評価すると `nil` が返ってきた。

他方、私がこれらの式を執筆時に評価してみたところ、`(buffer-name)` を評価すると `"introduction.texinfo"` という値が返され、`(buffer-file-name)` を評価すると `"/gnu/work/intro/introduction.texinfo"` が返ってきた。

前者がバッファの名前で、後者はそのファイル名である。(S 式の中の括弧は、Lisp インタプリタが `buffer-name` と `buffer-file-name` を関数として扱っていることを示している。もし括弧が無ければ、インタプリタはそのシンボルを変数として評価しようとしただろう。詳しくは Section 1.7 “変数”, page 7, を参照。)

（未訳）When I am writing, the name of my buffer is `"introduction.texinfo"`. The name of the file to which it points is `"/gnu/work/intro/introduction.texinfo"`.

（未訳）(In the expressions, the parentheses tell the Lisp interpreter to treat `buffer-name` and `buffer-file-name` as functions; without the parentheses, the interpreter would attempt to evaluate the symbols as variables. See Section 1.7 “Variables”, page 7.)

ファイルとバッファは違うものであるにも関わらず、しばしばバッファのことをファイルと言ったり、また、その逆を言ったりする人は多い。実際、大抵の人は「私はファイルを編集している」といい、「私はファイルにセーブしようとしているバッファを編集している」などとは言ったりはしない。それでも殆どの場合、状況から言いたいことは通じてしまう。しかしながら計算機のプログラムを扱う場合は、その違いを心に留めておくことが重要だ。計算機は人間のように賢くはないのである。

ところで ‘buffer’ という言葉は衝突の際の衝撃を吸収するクッションを表わす言葉から来ている。初期の計算機では、バッファはファイルと計算機の中央演算処理装置 (CPU) の間のクッションの役目をしていた。ファイルを保存するドラムやテープと CPU は互いに全く異なる装置であり、速度も全然違っている。バッファは各々が互いに効果的に作業出来るようにするものだった。その後、バッファは中継ぎとか、一時的な保管場所とかいう意味から変化して作業する場所を示すようになった。この変化は小さな港から巨大な町への変化と似ている。かつては単に船に積まれる貨物が一時的に保管される場所でしかなかった所が、ビジネスや文化の中心としての存在になってしまったのである。

全てのバッファが必ず何らかのファイルに関連付けられているわけではない。例えば `*scratch*` というバッファはどのファイルにもビジットしていない。同様に `*Help*` というバッファもどのファイルにも関連付けられてはいない。

（未訳）In the old days, when you lacked a `~/.emacs` file and started an Emacs session by typing the command `emacs` alone, without naming any files, Emacs started with the `*scratch*` buffer visible. Nowadays, you will see a splash screen. You can follow one of the commands suggested on the splash screen, visit a file, or press the spacebar to reach the `*scratch*` buffer.

もし `*scratch*` バッファに切り替えて `(buffer-name)` とタイプし、その直後にカーソルを置いて *C-x C-e* とタイプしてそのS式を評価したなら、`"*scratch*"` という名前が返り、エコー領域に表示される。つまり、バッファの名前は `"*scratch*"` である。しかし、同じことを `(buffer-file-name)` でやったなら、エコー領域には `nil` と表示される。nil は「無 (nothing)」を表わすラテン語の単語である。この場合は `*scratch*` バッファがどのファイルにも関連付けられていないことを意味している。(Lisp では `nil` は「偽 (false)」という意味や、空のリストである `()` の同意語としても使われる。)

ついでに言っておくと、もしあなたが、`*scratch*` バッファにいて、あるS式が返す値をエコー領域ではなく `*scratch*` バッファそのものに表示したい場合は、*C-x C-e* の代わりに *C-u C-x C-e* とタイプすれば良い。こうすると、値はS式のすぐ後に表示される。つまり `*scratch*` バッファには次のように表示される。

```
(buffer-name)"*scratch*"
```

これは Info の中では実行出来ない。というのも Info バッファは書き込み不可 (read-only) であり、勝手にバッファの中身を修正したりすることは出来ないからである。しかし、編集可能なバッファであれば、どのバッファであろうと実行出来る。コードや文書 (例えばこのマニュアルなんか) を書いている時には、この機能は大変便利である。

2.2 バッファの獲得

`buffer-name` という関数はバッファの名前を返す。従って、バッファそのものを取り出すには、別の関数が必要になる。これは `current-buffer` という関数である。コードの中でこの関数を使うと、バッファそのものが得られる。

名前と、その名前がついているオブジェクトないしは実体とは、互いに異なっている。あなたはあなたの名前とは違う。あなたは、他の人からその名前で呼ばれる一人の人間である。もしあなたが George と話したいと頼んだ時に、‘`G`’, ‘`e`’, ‘`o`’, ‘`r`’, ‘`g`’, ‘`e`’ という文字の書かれたカードを渡されたとしたら、あなたは面白いとは思うかもしれないが、本来の要求は満たされない。あなたは名前と話したいというのではなく、その名前を持つ人間と話したいのだから。バッファもそれと同じだ。スクラッチバッファの名前は `*scratch*` である。しかし、その名前はバッファそのものではない。ということで、バッファそのものを取り出すときは、`current-buffer` のような関数が必要なわけである。

しかしながら、事情はもう少し複雑である。例えば `current-buffer` をすぐ後で実験するように、S式の中で評価したとする。そうすると、バッファの中身ではなく、バッファの名前のみが表示され

る。Emacs がこのように動作するのには二つの理由がある。そのバッファは何千行もあるかもしれない—これは手軽に表示出来るというものではない。そして他のバッファで、名前は違っても中身は同じものがあるかもしれない。それらを区別するのは、大変大切なことである。

さて、これがこの関数を含む S 式である。

```
(current-buffer)
```

もし、ごく普通にこの S 式を評価したとすると、エコー領域には`#<buffer *info*>` のように表示される。このように特殊な様式で表示することで、バッファの名前ではなく、そのバッファ自身が返されたことを示しているのである。

ついでに言うと、プログラムの中には数値やシンボルをタイプすることは出来るが、表示されているバッファについてはそんなことは出来ない。バッファを取り込む唯一の方法は `current-buffer` のような関数を使うことなのである。

関連する関数に `other-buffer` というのがある。これは、今、現在いるバッファ、つまりカレントバッファを除いてもっとも最近選択したバッファを返す。もし直前に `*scratch*` バッファに行って帰ってきたとしたら、`other-buffer` は、このバッファを返す。

このことは次の S 式を評価することで確かめられる。

```
(other-buffer)
```

エコー領域には `#<buffer *scratch*>` もしくは、あなたが最近行って帰ってきたバッファの名前が表示されたはずだ。[1]

2.3 バッファ間の移動

`other-buffer` という関数はこれを引数とする関数から呼ばれた場合には実際にそのバッファの中身を提供する。これは `other-buffer` と `switch-to-buffer` を使って他のバッファに切り替えることで確かめられる。

が、その前に、`switch-to-buffer` という関数を簡単に紹介しておこう。Info バッファから `(buffer-name)` を評価するために `*scratch*` バッファに行って戻ってくる際に、多分あなたはまず *C-x b* とタイプし、そしてミニバッファにどのバッファに移るか尋ねるプロンプトが出た時に`*scratch*` とタイプしたのではないだろうか。この時のキーストローク *C-x b* は Lisp インタプリタに対してインタラクティブに Emacs Lisp の関数 `switch-to-buffer` を評価させているのである。前にも言ったように Emacs はこのようにして動作している。また別のキーストロークはまた別の関数を呼び出したり走らせたりする。例えば *C-f* は `forward-char` を呼び出すし、*M-e* は `forward-sentence` を呼び出す、といった感じである。

S 式の中で `switch-to-buffer` と書いて、そいつに移るべきバッファを与えると、*C-x b* と全く同じようにしてバッファを切り替えることが出来る。

```
(switch-to-buffer (other-buffer))
```

`switch-to-buffer` というシンボルはこのリストの最初の要素であるから、Lisp インタプリタは、こいつを関数として扱い、それに定義されている命令を実行する。しかし、その前に、内側に括弧に挟まれた `other-buffer` があることを発見し、こちらを先に解釈する。`other-buffer` は、このリストの最初の (そしてこの場合は唯一の) 要素であるから、Lisp インタプリタはこの関数を走らせる。するとこいつは別のバッファを返す。次にインタプリタは返ってきたバッファを `switch-to-buffer` に引数として渡して、この関数を走らせる。そうして Emacs は他のバッファに切り替わる。もし、この文書を Info で読んでいるなら早速試してみよう。上の S 式を評価するのである。(戻ってくるには、*C-x b RET* とタイプする。)[2]

---

[1] Actually, by default, if the buffer from which you just switched is visible to you in another window, `other-buffer` will choose the most recent buffer that you cannot see; this is a subtlety that I often forget.

[2] Remember, this expression will move you to your most recent other buffer that you cannot see. If you really want to go to your most recently selected buffer, even if you can still see it, you need to evaluate the following more complex expression:

```
(switch-to-buffer (other-buffer (current-buffer) t))
```

In this case, the first argument to `other-buffer` tells it which buffer to skip—the current one—and the second argument tells `other-buffer` it is OK to switch to a visible buffer. In regular use, `switch-to-buffer` takes you to an invisible window since you would most likely use *C-x o* (`other-window`) to go to another visible buffer.

後の方のセクションに出てくるプログラミングの例では `switch-to-buffer` よりも `set-buffer` という関数の方をより多く見かけるだろう。これは計算機のプログラムと人間の違いによる。人間は目を持ち今作業している計算機の端末の上にそのバッファを見たいと思うものである。これはいうまでもなく当たり前のことである。しかしながら、プログラムは目を持たない。プログラムがバッファを扱っている時には、バッファはスクリーンに見えていなくたって良いのである。

`switch-to-buffer` は人間のために設計されており、二つのことを行う。一つは Emacs が注意を向けているバッファを切り替えることであり、もう一つは画面に見えているバッファを切り替えることである。一方、`set-buffer` の方は一つの仕事しかしない。単に計算機が注意を向けているバッファを別のバッファに切り替えるだけである。画面に見えているウィンドウはそのままだ (勿論、普通はそのコマンドが終了するまでは何も起こらない。)

ここでまた別の専門用語を紹介しよう。呼び出す (*call*) という言葉である。最初のシンボルが関数であるリストを評価すると、その関数が呼び出される。(call される。) この時の呼び出すという単語の使い方は、「呼び出す」ことで実体としての関数が何かをやってくれることから来ている—本物の配管工を「呼び出す」と水洩れを直してくれるのと一緒である。

2.4 バッファのサイズとポイントの位置

最後に、簡単な関数を幾つか見てみよう。`buffer-size`, `point`, `point-min`, そして `point-max` である。これらはバッファのサイズとその中でのポイントの位置の情報を与えてくれるものである。

`buffer-size` という関数は現在のバッファの大きさを教えてくれる。つまり、そのバッファの中の文字の数を返すのである。

```
(buffer-size)
```

これをいつものように、S 式の最後にカーソルを持っていって *C-x C-e* とタイプすることで評価してみよう。

Emacs の中では現在のカーソルの位置はポイントと呼ばれる。`(point)` という S 式はバッファの始まりからポイントまでの文字数を返す。それによって現在のカーソルの場所が分る。

次の S 式をこのバッファの中でいつも通りに評価すればポイントまでの文字数が分る。

```
(point)
```

これを書いているときには、`point` の値は 121905 であった。この `point` という関数は、このマニュアルの後の方に出てくる幾つかの例の中では頻繁に使われる。

勿論ポイントの値はバッファの中での位置に依存している。次の場所でポイントを評価すれば、その値はさっきよりも大きくなっているはずである。

```
(point)
```

私の場合この場所でのポイントの値は 66043 であった。これは両方の S 式の間に (空白も含めて) 319 文字あることを示している。(Doubtless, you will see different numbers, since I will have edited this since I first evaluated point.)

`point-min` という関数はちょっと `point` に似ているが、こちらはカレントバッファ中のポイントの値の中で許される最小の値を返す。この値はナローイング (*narrowing*) が有効になっていない限り 1 という数値である。(ナローイングというのは、あなた自身もしくはプログラムの操作をバッファの特定部分に制限してしまうある仕組みである。詳しくは Chapter 6 “ナローイングとワイドニング”, page 52, を参照のこと。) 同様に、`point-max` というのはカレントバッファ中のポイントの値の中で許される最大の値を返す。

2.5 練習問題

あなたが作業しているファイルを一つ開いて、その中央辺りまで行きなさい。そして、そのバッファ名、ファイル名、長さ、及びファイルの中での位置を調べなさい。

3 関数定義の書き方

Lisp インタプリタがリストを評価する場合、まずリストの最初のシンボルが関数定義を持っているかどうか、別の言い方をするとシンボルから関数定義へのポインタがあるかを見にいく。もしそうであれば、計算機はその定義に含まれる命令を実行する。関数定義を持つシンボルのことを単に関数と言う。(ただし、正確には定義が関数なのであり、シンボルはそれを指しているだけである。)

全ての関数は C 言語で書かれた少数の プリミティブ (*primitive*) な関数を除いて、別の関数を使って定義されている。関数を書く場合にはそれを Emacs Lisp を使って書き、他の関数を部品として使うことになる。あなたが使う関数はそれ自身 (あなたが書いたものを含めて) Emacs Lisp で書かれたものかもしれないし、C で書かれたプリミティブなものかもしれない。プリミティブな関数も Emacs Lisp で書かれているのものと全く同じように使われ、同じように振舞う。これらが C で書かれているのは、十分な能力を持ち、C を走らせることが出来るどの計算機でも簡単に GNU Emacs が動くようにするためである。

再度強調しておくけれども、Emacs Lisp のコードを書く場合、C で書かれた関数を使う場合と Emacs Lisp で書かれた関数を使う場合とで区別をつける必要は全くない。違いは全く無視して構わない。私がこの区別のことを書いたのは、単に知っていた方が面白いからである。実際、調査でもしない限り、既に書かれている関数が Emacs Lisp で書かれているか C で書かれているかは判断出来ない。

3.1 マクロ defun

Lisp では `mark-whole-buffer` のようなシンボルには、その関数が呼ばれた時に計算機が何をすべきかを教えるコードが付随している。このコードのことを 関数定義 (*function definition*) といい、シンボル `defun` で始まる S 式を評価した時に作られる。(これは *define function* の略である。) `defun` は通常とは別の引数の評価の仕方をするために、特殊形式 (*special form*) と呼ばれる。

この後に続く幾つかのセクションでは、例えば `mark-whole-buffer` のように Emacs のソースコードの中から関数定義を見ていくことにする。が、このセクションではもっと単純な関数定義を説明し、それがどんなものなのかを、まず理解してもらうことにしよう。簡単な例にするために算数についての例にする。算数を使った例が嫌いな人達もいるとは思うが、あなたがそうだとしてもがっくりくる必要はない。この入門書の残りに出てくる殆どのコードには算数とか数学は出てこない。出てくる例は、大体が何がしかの意味でテキストに関わるものである。

関数定義は `defun` という単語に続けて、次のような最大五つの部分を加えたものである。

1. 関数定義が付けられるシンボルの名前。
2. 関数に渡される引数のリスト。もし一つも引数をつけない場合、ここは空リスト `()` にする。
3. 関数を説明する文章。(技術的には省略可能であるが、書くことが強く望まれている。)
4. 省略可能。この関数をインタラクティブにする、つまり *M-x* に続けて関数名をタイプするか、適当なキーないしはキーコードをタイプすることで使うことが出来るようにするための S 式を書く。
5. 計算機に何をすべきかを命令するコードを書く。即ち、関数定義の本体である。

この関数定義の五つの部分の各々にスロットを割当てて、次のようなテンプレートにして整理しておくと役に立つだろう。

```
(defun 関数名 (引数...)
  "オプションの説明文字列..."
  (interactive 引数情報) ; 省略可能
  本体...)
```

一つの例として、引数に 7 をかける関数のコードを挙げる。(この例はインタラクティブではない。Section 3.3 “関数をインタラクティブにする”, page 21, を見よ。)

```
(defun multiply-by-seven (number)
  "Multiply NUMBER by seven."
  (* 7 number))
```

この定義は括弧と `defun` というシンボルで始まり、関数の名前が続いている。

関数の名前の次にはこの関数に渡される引数のリストが来る。このリストは引数リスト (*argument list*) と呼ばれる。今の場合、このリストの要素は `number` というシンボル一つだけである。この関数が使われる時には、このシンボルは関数の引数として使われる値にバインドされる。

引数の名前として `number` という単語を選ぶ代わりに他の名前を選んでも良い。例えば `multiplicand` という単語でも良かった。私が ‘number’ という単語を選んだのは、このスロットに

どういう種類の値が要求されるかが分るからである。しかし、このスロットに置かれる値がこの関数の働きの中でどういう役割を果たすかを示すという意味で 'multiplicand' という単語を選ぶのも悪くなかった。`foogle` としても構わないが、これは良い選択だとは思えない。これでは何のことか分らない。どういう名前にするかはプログラマ次第だが、関数の意味を明確にするようなものが選ばれるべきである。

実の所、引数リストの中のシンボルにはどんな名前でも選ぶことが出来る。たとえ、他の関数の中で使われているシンボル名であっても良いのである。引数リストの中で使われる名前は、その特定の定義の中だけでしか使われないプライベートなものであるからだ。この定義の中では、他のどんな関数定義の中にある同じ名前のシンボルとも異なる実体を指すのである。例えば、あなたが家族の間では 'Shorty' というニックネームをつけられているとしよう。この場合、あなたの家族の誰かが 'Shorty' という場合には、それはあなたのことを指している。しかしあなたの家族以外の所、例えば映画の中などでは 'Shorty' というのは別の誰かを指している。引数リストの中の名前はその関数の中だけで通用するものなので、この関数の本体の中でそのシンボルの値をどう変えたとしても外での値には何の影響もないというわけだ。似たような効果が `let` を使った S 式の場合でも見られる。(Section 3.6 "`let`", page 24, 参照。)

引数リストの次には、関数についての説明である、説明文字列が来る。これは *`C-h f`* に続けてその関数の名前をタイプした時に表示される内容である。ついでに言っておくと、あなたが説明文字列を書く場合には、最初の一行は完結した一つの文にすべきである。というのも、`apropos` のような幾つかのコマンドは複数行に渡って書かれた説明文の文字列のうちの最初の一行だけしか表示しないからである。また、二行目も書く場合はこれをインデントすべきではない。何故なら、*`C-h f`* (`describe-function`) を使ったときに変な表示のされかたをするからである。説明文字列は省略可能ではあるが、大変役に立つものなので、関数を書く際にはなるべく書くようにすべきである。

さっきの例での三行目は関数定義の本体をなしている。 (普通の関数定義は勿論これよりもっと長い。) 今の場合、この本体は `(* 7 number)` というリストであり、*number* の値に 7 をかけることを定めている。(Emacs Lisp では `*` は掛け算を表わす関数である。同様に `+` は足し算を表わす。)

`multiply-by-seven` という関数を使う場合、引数 `number` はあなたが実際に使いたい値に評価される。次の例では `multiply-by-seven` の使い方を示している。ただし、まだこれを評価しないように！

```
(multiply-by-seven 3)
```

次のセクションで詳しく述べるが、今の例ではシンボル `number` に 3 という値が与えられた、ないしは「バインドされた」わけである。`number` は関数定義の括弧の中にあるが、`multiply-by-seven` という関数に渡される引数は括弧の外にあることに注意しておこう。関数定義の中に括弧があるのは、計算機が引数のリストが何処で終わり、残りの部分が何処から始まるかを知るためにあるのである。

この例を評価すると多分エラーメッセージが出るだろう。(試してみよう！) これは、我々は確かに関数定義を書いたのだが、まだその定義を計算機に伝えていない—つまり、まだ Emacs にこの関数定義をインストール (もしくはロード) していない—ためである。インストールの仕方については次のセクションで説明する。

3.2 関数定義のインストール

もし Emacs の Info の中でこれを読んでいるのなら、先に関数の定義を評価し、次に `(multiply-by-seven 3)` を評価することで、`multiply-by-seven` という関数を試してみることが出来る。この下に関数定義の写しを書いておくので、その直後にカーソルを持っていって *`C-x C-e`* とタイプしよう。そうすると、エコー領域に `multiply-by-seven` と表示される。(これは関数定義を評価した時に返す値はその関数の名前であることを示している。) 同時に、こうすることで関数定義がインストールされる。

```
(defun multiply-by-seven (number)
  "Multiply NUMBER by seven."
  (* 7 number))
```

この `defun` を評価することで、たった今 Emacs に `multiply-by-seven` がインストールされた。今ではこの関数は `forward-word` やあなたが編集に使う他の関数と全く同様、Emacs の一部である。(`multiply-by-seven` はあなたが Emacs を終了するまでインストールされたままになっている。このコードを Emacs を起動するたびに再ロードする方法については、Section 3.5 "コードをずっとインストールしておくには", page 23, を見よ。)

multiply-by-seven をインストールしたことでどうなったかは、次の例を評価することで見ることが出来る。カーソルを次の S 式の直後に持っていって *C-x C-e* とタイプしてよう。エコー領域には 21 という数字が表示されたはずだ。

```
(multiply-by-seven 3)
```

もし望むなら、*C-h f* (describe-function) に続けてこの関数の名前 multiply-by-seven をタイプすることで、この関数定義の説明を読むことも出来る。そうすると、*Help* というウィンドウに次のように表示される。

```
multiply-by-seven is a Lisp function.
(multiply-by-seven NUMBER)

Multiply NUMBER by seven.
```

(スクリーンを元の一つのウィンドウに戻すには、*C-x 1* とタイプすれば良い。)

3.2.1 関数定義の変更

multiply-by-seven の中のコードを変更したい場合は、単にそれを書き直すだけである。古いものを新しいバージョンに置き換えるには、その関数定義をもう一度評価すれば良い。これが Emacs の中のコードを修正する方法である。極めて単純だ。

例として、multiply-by-seven という関数を 7 をかけるのではなく、その数自身を 7 回加えるものに変更することが出来る。結果としては同じ答えが得られるけれども、途中の道筋は違っている。同時にコードの中にコメントを加えよう。これは Emacs には無視されるけれども、人間が読む場合には便利だなあとか解りやすいとか思うはずだ。今回は、これが二番目のバージョンであるというコメントを書いておくことにする。

```
(defun multiply-by-seven (number) ; 二番目のバージョン
  "Multiply NUMBER by seven."
  (+ number number number number number number number))
```

コメントはセミコロン ‘;’ の後に続いている。Lisp では、セミコロンの後に続く全てのものはコメントである。行の終わりがコメントの終わりになる。二行以上に渡ってコメントを書きたい場合は各々の行をセミコロンで始める。

Section 16.3 “.emacs の書き方”, page 145, 及び Section “Comments” in *The GNU Emacs Lisp Reference Manual*, にコメントについて、より詳しく説明されている。

今のバージョンの multiply-by-seven 関数をインストールするには最初のものをインストールした時と全く同じようにして評価してやれば良い。つまり、カーソルを最後の括弧のすぐ後に持ってきて *C-x C-e* とタイプすれば良いのである。

まとめると、Emacs でコードを書く方法は次の通りである。まずは関数を書き、インストールし、テストしてみる。で、不具合を修正したり拡張したりして再度インストールするという具合だ。

3.3 関数をインタラクティブにする

関数をインタラクティブなものにするには、特殊形式 interactive で始まるリストを説明文字列のすぐ後に置けばよい。インタラクティブな関数はユーザーが *M-x* に続けて関数の名前をタイプすることで呼び出すことが出来る。あるいは、その関数にバインドしたキーをタイプしてもよい。例えば、*C-n* とタイプすると next-line が呼ばれるし、*C-x h* とタイプすると mark-whole-buffer が呼ばれる。

面白いことに、インタラクティブな関数をインタラクティブに呼び出すと、返される値は自動的にはエコー領域に表示されない。これはインタラクティブな関数はしばしば、単語分や一行分前に進んだりするなどの副作用を目的に呼ぶのであって、返される値のことは気にしないからである。もしキーをタイプするごとにエコー領域に値が表示されたら、随分と鬱陶しいことだろう。

特殊形式 interactive を使いつつエコー領域に値を表示する方法について説明するために、multiply-by-seven のインタラクティブバージョンを作ってみよう。

これがそのコードである。

```
(defun multiply-by-seven (number) ; インタラクティブバージョン
  "Multiply NUMBER by seven."
  (interactive "p")
  (message "The result is %d" (* 7 number)))
```

このコードはカーソルをコードのすぐ後に持っていって C-x C-e とタイプすることでインストール出来る。ちゃんと、この関数の名前がエコー領域に表示されただろうか。次からこのコードを使うには、

まず *C-u* に続けて数をタイプし、その後に *M-x multiply-by-seven* とタイプして *RET* を押せば良い。エコー領域に、‘The result is ...’ という文に続いて計算結果が表示されるはずだ。

もっと一般的に言うと、このような関数は次の二通りの内どちらかの方法で呼び出すことが出来る。

1. まず渡されるべき数を含む前置引数をタイプする。ついで *M-x* と関数名をタイプする、例えば *C-u 3 M-x forward-sentence*、とする。
2. あるいは、その関数にバインドされたキーないしはキーコードをタイプする。例えば *C-u 3 M-e*。

今挙げた二つの例では、どちらの場合もポイントが 3 つの文だけ前方に移動する。(この関数を例として扱った理由は、multiply-by-seven はどのキーにもバインドされてはいないので、キーバインディングの例としては使えないからである。)

(コマンドをキーにバインドする方法については Section 16.7 “幾つかのキーバインディング”, page 148, を参照のこと。)

前置引数をインタラクティブな関数に渡すには、*M-3 M-e* のように *META* キーに続けて数をタイプする方法と、*C-u 3 M-e* のように *C-u* に続けて数をタイプする方法とがある。(もし *C-u* の後に何も数をタイプしなかった場合、デフォルトでは 4 が渡される。)

3.3.1 インタラクティブな multiply-by-seven

インタラクティブバージョンの multiply-by-seven の中の特殊形式 interactive の使い方と、関数 message を見てみよう。もう一度この関数の定義を書いておく。

```
(defun multiply-by-seven (number) ; インタラクティブバージョン
  "Multiply NUMBER by seven."
  (interactive "p")
  (message "The result is %d" (* 7 number)))
```

この関数の中では S 式 (interactive "p") は二つの要素からなるリストである。"p" は Emacs に対し、この関数に前置引数を渡し、その値を関数の引数として使うことを知らせるためのものである。

引数は数になる。これは number というシンボルが

```
(message "The result is %d" (* 7 number))
```

という行の中である数にバインドされるという意味である。例えば、前置引数が 5 だとすると、Lisp インタプリタはこの行を次のように評価することになる。

```
(message "The result is %d" (* 7 5))
```

(もし、この文章を GNU Emacs の中で読んでいるなら、この S 式をあなた自身で評価することが出来る。) 最初にインタプリタは内側にあるリストを評価する。今の場合は (* 7 5) である。これは値 35 を返す。次にその外側のリストが評価される。つまり、二番目及び残りの要素の値が message という関数に渡される。

前に見たように、message は特に一行のメッセージをユーザに送るための Emacs Lisp の関数であった。(Section 1.8.5 “関数 message”, page 11, 参照。) 手短に言うと、関数 message は最初の引数をエコー領域に表示するのだが、‘%d’, ‘%s’, 及び ‘%c’ だけは例外で、これらの制御文字列の内のどれかが現れた時は、残りの二番目以降の引数を順に見に行き、その値をこれらがある場所に挿入して表示する。

multiply-by-seven という関数では、制御文字列として ‘%d’ が使われ、これは数を要求する。今の場合ならその値は (* 7 5) を評価して返された 35 という数である。その結果、‘%d’ の場所には 35 が表示され、メッセージは ‘The result is 35’ となるわけである。

(ここで注意だが、multiply-by-seven という関数を呼んだ場合、メッセージは引用符無しで表示される。しかし message をそのまま呼んだ場合には、二重引用符で囲まれている。これは message が返す値そのものは message が先頭にある S 式を評価した場合にエコー領域に表示されるもので、これには引用符がついているのだが、これが他の関数の中に埋め込まれて使用された場合、表示されるものは message が副作用として出力するもので、これには二重引用符がついていないからである。)

3.4 interactive の他のオプション

上の例で multiply-by-seven は interactive の引数として"p"を使っている。この引数がある場合、Emacs はあなたが、*C-u* に続けて数をタイプするか、*Meta* キーに続けて数をタイプするかして、このコマンドに引数として数を渡すものと解釈する。Emacs はこのような interactive の引数を 20 種類以上用意している。大抵の場合、これらのオプションの内にあなたが望む通りの情報を関数に渡すも

のが一つか二つはあるはずだ。(Section “Code Characters for `interactive`” in *The GNU Emacs Lisp Reference Manual*, 参照。)

(未訳) Consider the function `zap-to-char`. Its interactive expression is

```
(interactive "p\ncZap to char: ")
```

(未訳) The first part of the argument to `interactive` is ‘`p`’, with which you are already familiar. This argument tells Emacs to interpret a ‘prefix’, as a number to be passed to the function. You can specify a prefix either by typing *`C-u`* followed by a number or by typing `META` followed by a number. The prefix is the number of specified characters. Thus, if your prefix is three and the specified character is ‘`x`’, then you will delete all the text up to and including the third next ‘`x`’. If you do not set a prefix, then you delete all the text up to and including the specified character, but no more.

(旧訳) あるいは、‘`B`’ であれば、Emacs はバッファの名前を聞いてきて、入力した値をその関数に渡す。もう少し詳しく言うと、Emacs はミニバッファにプロンプトを出して、ユーザーに名前を要求する。この時のプロンプトには ‘`B`’ に続く値が使われるので、例えば、`"BAppend to buffer: "` とか書いておく。Emacs はただ単にプロンプトを出すだけではなくて、判断出来る分だけの入力があれば、*`TAB`* キーを押すことで補完までしてくれる。

(未訳) The ‘`c`’ tells the function the name of the character to which to delete.

(未訳) More formally, a function with two or more arguments can have information passed to each argument by adding parts to the string that follows `interactive`. When you do this, the information is passed to each argument in the same order it is specified in the `interactive` list. In the string, each part is separated from the next part by a ‘`\n`’, which is a newline. For example, you can follow ‘`p`’ with a ‘`\n`’ and an ‘`cZap to char: `’. This causes Emacs to pass the value of the prefix argument (if there is one) and the character.

(旧訳) 関数に二つ以上の引数を渡したい場合、`interactive` に続けて複数の文字列を付け加えることで、各々の引数に情報を渡すことが出来る。この場合その情報は、各々の引数に `interactive` に書いたのと同じ順序で渡される。付け加える文字列は、各々の部分を ‘`\n`’, 即ち改行コードで区切る。例えば `"BAppend to buffer: "` に続けて ‘`\n`’ と ‘`r`’ を書いたりする。こうすると、Emacs はプロンプトでバッファ名を要求すると同時にポイントとマークの値もその関数に渡してくれる—三つの引数全てを渡してくれるのである。

(未訳) In this case, the function definition looks like the following, where `arg` and `char` are the symbols to which `interactive` binds the prefix argument and the specified character:

(旧訳) この場合の関数定義は次のような形式になる。`buffer`, `start`, 及び、`end` は `interactive` がバッファとその時のリージョンの始まりと終わりをバインドするシンボルである。

```
(defun 関数名 (arg char)
  "説明文字列..."
  (interactive "p\ncZap to char: ")
  関数の本体...)
```

(プロンプトのコロンの後の空白は、プロンプトを出す時の見栄えをよくするためのものである。`append-to-buffer` という関数はまさにこのようになっている。Section 4.4 “`append-to-buffer` の定義”, page 36, 参照。)

(未訳) When a function does not take arguments, `interactive` does not require any. Such a function contains the simple expression `(interactive)`. The `mark-whole-buffer` function is like this.

あるいは、もし上に挙げたような特定の文字による引数の与え方があなたの目的に合わない場合、あなた自身の引数をリストとして `interactive` に渡すことも可能である。

Section “Using `Interactive`” in *The GNU Emacs Lisp Reference Manual*, にこの上級テクニックについてのより詳しい解説がある。

3.5 コードをずっとインストールしておくには

関数定義を評価することで一旦その関数をインストールすると、Emacs を終了するまでその関数はインストールされたままになっている。一方、次に Emacs の新しいセッションを開始した時は、その関数定義を再度評価するまでその関数はインストールされない。

ある時点で、Emacs の新しいセッションを始める時に自動的にインストールしたいと思うかもしれない。方法は幾つかある。

- もし、自分自身で書いたコードがあれば、その関数定義のコードをあなたの初期化ファイル `.emacs` の中に書き込む。そうすると Emacs を起動した時にこの `.emacs` というファイルが自動的に評価されて、その中に書かれた全ての関数がインストールされる。Chapter 16 “`.emacs` ファイル”, page 143, 参照。
- あるいは、インストールしたい関数の定義を一つのファイル、もしくは関数ごとに複数のファイルに書いておき、`load` という関数を使って Emacs にそれを評価させてそれらの関数をインストールするという方法もある。Section 16.9 “ファイルのロード”, page 149, 参照。
- また、あなたのいるサイト全体でそのコードを使いたい場合は、普通はその関数を `site-init.el` 呼ばれるファイルに書いておく。このファイルは Emacs を作成する時に自動的にロードされる。こうすることで、あなたの計算機を使う全ての人がその関数を利用出来るようになる。(Emacs distribution の中の `INSTALL` ファイルを参照のこと。)

最後に、もしあなたが Emacs を使う全ての人が欲しくなるようなコードを書いたなら、そのコピーをネットワーク上にポストしたり、Free Software Foundation に送ることが出来る。(そうする場合には、どうかポストする前にコードに copy left の注意書きを添付して欲しい。) コピーを Free Software Foundation に送った場合、次の Emacs のリリース時にはそれを含めて配布されるかもしれない。大体において、ここ数年はこのような寄付によって Emacs が成長してきたのである。

3.6 `let`

`let` 式は、Lisp では多くの関数定義の中で必要となる特殊形式である。非常によく使われるものなので、このセクションの中で `let` について説明することにする。

`let` は、シンボルに値をバインドする際に、Lisp インタプリタがその関数以外の関数の中で使われている同じ名前の変数と混同しないようにするために使用される。

何故こんな関数が必要なのかということを納得するために、次の様な状況を考えよう。あなたは自分自身の家を持っていて、それを「家」と呼んでいるとする。例えば、「そろそろ家にもペンキ塗りが必要だなあ」という風に使うわけである。

しかし、あなたが友人宅を訪問し、そこのホストの人が「家」と言った場合、大抵それはあなたの家ではなく彼の家のことを指している。即ち、同じ「家」という言葉で言及されてはいても別のものなわけである。もし彼が彼自身の家のことを言っていて、あなたがあなた自身の家を思い浮かべていたなら混乱してしまうことであろう。同じことが Lisp についても言える。ある関数で使われている変数と同じ名前の変数が他の関数でも使われていて、しかもその二つが同じ値を持つことを期待されてはいない時である。The `let` special form prevents this kind of confusion.

`let` という特殊形式を使うことでこのような混乱を防ぐことが出来る。`let` は ローカル変数 (*local variable*) と呼ばれるものを発生させる。これは、その `let` 式の外にある同じ名前の変数からは隔離されている変数である。訪問先のホストの人が「家」という時は彼は彼自身の家を指しているのであって、あなたの家を指しているのではないのと似たようなものだ。 (引数リストの中のシンボルも同じような働きをする。Section 3.1 “特殊形式 `defun`”, page 19, を見よ。)

`let` 式によって発生したローカル変数の値が保持されているのは、その `let` 式の中だけである。(そして、その中の S 式はその `let` 式の中だけで呼ばれる。) 従って、ローカル変数はその `let` 式の外には全く影響を与えない。

`let` は一度に複数の変数を発生させることが出来る。また、`let` は各々の変数に初期値を設定する。あなたが指定すればその値になるし、そうしなければ `nil` が設定される。(専門用語では、このことを「その変数に値をバインドする」と言う。) `let` は変数を発生させ、それに値をバインドした後、本体のコードを実行し、本体の中の最後の S 式の値を返す。(実行 (Execute) するというのは、リストを評価するという意味の専門用語である。これはこの単語の「実質的な効果を与える (to give practical effect to)」という意味での使い方から来ている (*Oxford English Dictionary*)。あなたはある動作を引き起こすために S 式を評価しているのだから、この場合「実行」するというのは評価するのと同義であろう。)

3.6.1 `let` 式の構成部分

`let` という S 式は三つの部分からなるリストである。最初の部分はシンボル `let` である。二番目の部分は 変数リスト (*varlist*) と呼ばれるリストであり、各々の要素はシンボルそのものであるか、最初の

要素がシンボルである二つの要素からなるリストであるかのどちらかであり、三番目の部分は let 式の本体である。本体部分は大抵、一つないしは複数のリストからなる。

let 式のテンプレートは次のように書ける。

```
(let 変数リスト 本体...)
```

変数リストの中のシンボルは特殊形式 let 式によって初期値を設定される変数である。単独のシンボルそのものの場合は、初期値 nil が設定される。また、最初の要素がシンボルであるような二つの要素からなるリストの場合、その最初のシンボルに対して Lisp インタプリタが二番目の要素を評価した時に返される値が設定される。

というわけで、変数リストは (thread (needles 3)) という感じの式になる。この場合だと、let 式の中で Emacs はシンボル thread を初期値 nil に、シンボル needles を初期値 3 にバインドする。

let 式を書く場合にすることは、適切な S 式を let 式のテンプレートの中に置くことである。

もし、変数リストが二つ要素のリストからなる場合、といっても大抵はそうなのだが、let 式のテンプレートは次のようになる。

```
(let ((変数 値)
      (変数 値)
      ...)
      本体...)
```

3.6.2 let 式の例

次の S 式は二つの変数 zebra と tiger を発生させ、それに対して初期値を与えている。let 式の本体は関数 message を呼び出すリストである。

```
(let ((zebra 'stripes)
      (tiger 'fierce))
  (message "One kind of animal has %s and another is %s."
           zebra tiger))
```

ここで、変数リストは ((zebra 'stripes) (tiger 'fierce)) である。

二つの変数は zebra と tiger である。各々の変数は二つの要素からなるリストの最初の要素であり、その値はそのリストの二番目の要素になっている。Emacs は変数リストの中で zebra を stripes にバインドし、[1] また、tiger を fierce にバインドしている。この場合、値は両方とも引用符のついたシンボルである。値は勿論、他のリストや文字列であっても構わない。let 式の本体は変数を含むリストの後に来る。今の場合は本体は関数 message を使ったリストであり、エコー領域に文字列を表示する。

いつものようにカーソルを最後の括弧の直後に置いて *C-x C-e* とタイプすることで上の例を評価することが出来る。そうすると、エコー領域に次のように表示されるだろう。

```
"One kind of animal has stripes and another is fierce."
```

以前見たように、message 関数は最初の引数を '%s' を除いて表示する。今の場合、最初の '%s' の所には変数 zebra の値が表示され、二番目の '%s' の所には変数 tiger の値が表示される。

3.6.3 let 式の変数宣言の中で初期値を設定しなかった場合

もし、let 式の中で変数に特定の初期値をバインドしなかったとすると、それらの値は自動的に nil にバインドされる。次の例を見てみよう。

```
(let ((birch 3)
      pine
      fir
      (oak 'some))
  (message
   "Here are %d variables with %s, %s, and %s value."
   birch pine fir oak))
```

ここで、変数リストは ((birch 3) pine fir (oak 'some)) である。

この S 式をいつもの通りに評価したなら、エコー領域には次のように表示されるだろう。

[1] According to Jared Diamond in *Guns, Germs, and Steel*, "... zebras become impossibly dangerous as they grow older" but the claim here is that they do not become fierce like a tiger. (1997, W. W. Norton and Co., ISBN 0-393-03894-2, page 171)

```
"Here are 3 variables with nil, nil, and some value."
```

この場合には Emacs はシンボル `birch` に 3 という数をバインドし、シンボル `pine` と `fir` には `nil` をバインドし、更にシンボル `oak` には `some` をバインドしている。

`let` 式の最初の部分で、変数 `pine` と `fir` は単独のアトムであり、括弧で囲まれてはいないことに注意しよう。これは、これらの変数を `nil`, 即ち空リストにバインドするためである。しかし、`oak` の方は、`some` にバインドするので、リスト `(oak 'some)` の一部となっている。同様に、`birch` も 3 にバインドするので、この数と一緒のリストに入っている。(数値は評価された時にそれ自身の値を返すので、引用符は必要ない。また、message の中にこの数を表示する際は、'`%s`' ではなくて、'`%d`' を使っている。) そして、これら四つの変数がひとつのグループとしてリストの中に入り、`let` 式の本体と区別されているわけである。

3.7 特殊形式 `if`

`defun` と `let` に続く三つ目の特殊形式は、条件分岐 (conditional) `if` である。この特殊形式は計算機になんらかの判断をさせる時に使われる。`if` を使わなくても関数定義を書くことも出来るが、頻繁に使うものであるし、重要でもあるので、ここに含めることにする。これは例えば、`beginning-of-buffer` という関数の中で使われている。

`if` の背後にある基本的な考え方は「もし (if)、テストした結果が真ならば (then) S 式を評価する」というものだ。テストした結果が真でなければ、S 式は評価はされない。例えば、「もし、暖かくて太陽が出ていたなら、ビーチへ行こう。」というような判断をするようなものだ。

Lisp で書かれる `if` 式では、'then' という単語は使われない。テストと実行は `if` を最初の要素とするリストの二番目と三番目の要素である。にも関わらず、`if` 式のテスト部分は *if-part* と呼ばれ、二番目の引数は *then-part* と呼ばれることが多い。

また、`if` 式が書かれる場合、真か偽かのテストは普通、`if` と同じ行に書かれる。しかし、テストが真であった場合に実行される "then-part" は二行目以降に書かれる。こうすることで、`if` 式がより読みやすいものになる。

```
(if 真偽テスト
    テストが真である場合に実行する動作)
```

真偽テスト (true-or-false-test) の部分は List インタプリタにより評価される S 式である。

次にいつものようにして実行出来る例を挙げておく。テスト部分は 5 が 4 よりも大きいかどうかを判断するものだ。そうであれば、'`5 is greater than 4!`' というメッセージが表示される。

```
(if (> 5 4)                             ; if-part
    (message "5 is greater than 4!"))   ; then-part
```

(関数 `>` は最初の引数が二番目の引数よりも大きいかどうかを判断し、そうであれば真を返すものである。)

勿論、実際に使う場合には、`if` 式のテスト部分は常に `(> 5 4)` に固定されていたりはしない。その代わりに、少なくとも一つの変数が、前もって分らない値にバインドされる。(もし、前もって値が分っているなら、テストする必要などない！)

例えば、その値は関数定義の引数にバインドされたりする。次の関数定義では、関数に渡される値は動物の性格である。もし、`characteristic` にバインドされる値が `fierce` であれば、'`It's a tiger!`' というメッセージが表示される。そうでなければ、`nil` が返される。

```
(defun type-of-animal (characteristic)
  "Print message in echo area depending on CHARACTERISTIC.
If the CHARACTERISTIC is the symbol 'fierce',
then warn of a tiger."
  (if (equal characteristic 'fierce)
      (message "It's a tiger!")))
```

もし、これを GNU Emacs の中で読んでいるなら、上の関数定義を評価することで、Emacs にインストールすることが出来る。そうすると、次の二つの S 式を評価して結果を見ることも出来るようになる。

```
(type-of-animal 'fierce)

(type-of-animal 'zebra)
```

(type-of-animal 'fierce) を評価すると、エコー領域には "It's a tiger" と表示されるはずだ。一方、(type-of-animal 'zebra) を評価すると、nil が表示される。

3.7.1 関数 type-of-animal の詳細

type-of-animal 関数を詳しく見てみよう。

type-of-animal の関数定義は二つのテンプレートのスロットを埋めることで書かれている。一つは全体の関数定義のテンプレートで、もう一つは if 式のテンプレートである。

インタラクティブでない関数の関数定義のテンプレートは次の通りである。

```
(defun 関数名 (引数リスト)
  "説明文字列..."
  本体...)
```

このテンプレートに当てはめると、各部分は次のようになる。

```
(defun type-of-animal (characteristic)
  "Print message in echo area depending on CHARACTERISTIC.  If the
CHARACTERISTIC is the symbol ‘fierce’, then warn of a tiger."
  本体: if 式)
```

今の場合、関数の名前は type-of-animal である。これは引数を一つ取る。引数リストの後に、複数行に渡る説明文字列が続く。例の中にもちゃんと説明文字列を入れるのは、全ての関数定義に説明文字列を入れるのが良い習慣であるからである。関数定義の本体部分は if 式からなっている。

if 式のテンプレートは次の通りである。

```
(if 真偽テスト
    テストが真の場合に実行する動作)
```

type-of-animal 関数では、if に関する実際のコードは次のようになっている。

```
(if (equal characteristic 'fierce)
    (message "It's a tiger!")))
```

ここで、真偽テストの部分は次の S 式である。

```
(equal characteristic 'fierce)
```

Lisp では、equal は最初の引数と二番目の引数が等しいかどうかを判定する関数である。二番目の引数は引用符つきのシンボル 'fierce であり、一番目の引数は characteristic というシンボルの値—つまり、この関数に渡される引数の値である。

先に行った type-of-animal の最初の実行では、fierce が type-of-animal に渡した。fierce は fierce と等しいので、S 式 (equal characteristic 'fierce) は真の値を返す。この場合は、if は二番目の引数、つまり if 式の then-part である (message "It's tiger!") を評価する。

一方、type-of-animal の二番目の実行では zebra を引数として渡した。zebra は fierce ではないので、then-part は評価されず、if 式は nil を返す。

3.8 If–then–else 式

if 式は、オプションとして *else-part* と呼ばれる三番目の引き数を持つことが出来る。これは真偽テストが偽を返した場合のためのものである。この場合は if 式の二番目の引数である then-part は評価されず、代わりに三番目の引数である else-part が評価される。このことは、曇りの日なんかに、「もし、晴れて暖かかったらビーチに行こう。そうでなかったら、本でも読むか！」というふうに、代わりの選択肢を考えるのに当てはめてみれば納得出来るだろう。

“else” という単語は Lisp のコードの中には出てこない。単に、else-part は if 式の中の then-part の次にくる要素というだけである。実際に Lisp のコードを書く場合、else-part は新しく改行してから then-part よりは少ないインデントで書き始めるのが普通である。

```
(if 真偽テスト
    テストが真の場合に実行する動作)
  テストが偽の場合に実行する動作)
```

例えば次の `if` 式は、いつも通りに評価すると、‘`4 is not greater than 5!`’ というメッセージを表示するものである。

```
(if (> 4 5) ; if-part
    (message "5 is greater than 4!")  ; then-part
  (message "4 is not greater than 5!")) ; else-part
```

異なるレベルのインデントをすることで、then-part と else-part の区別がしやすくなることに注意しよう。(GNU Emacs は `if` 式を自動的に正しくインデントするコマンドを幾つか備えている。Section 1.1.3 “GNU Emacs によるリストのタイプの支援”, page 2, を参照のこと。)

`type-of-animal` の `if` 式に新しく else-part の部分を追加するだけで、この関数を拡張することが出来る。

次のバージョンの `type-of-animal` の関数定義を評価してインストールすれば、その後の、異なる引数を渡すような二つの S 式を評価して、結果を見ることも出来るようになる。

```
(defun type-of-animal (characteristic) ; 二番目のバージョン
  "Print message in echo area depending on CHARACTERISTIC.  If the
CHARACTERISTIC is the symbol ‘fierce’, then warn of a tiger; else say
it’s not fierce."
  (if (equal characteristic ’fierce)
      (message "It’s a tiger!")
    (message "It’s not fierce!")))

(type-of-animal ’fierce)

(type-of-animal ’zebra)
```

`(type-of-animal ’fierce)` を評価すると、エコー領域には `"It’s a tiger"` と表示されるはずだ。一方、`(type-of-animal ’zebra)` を評価すると、`"It’s not fierce!"` と表示される。

(勿論、*characteristic* が `ferocious` (獰猛) であったなら、`"It’s not fierce!"` と表示されるが、これは間違いである。コードを書く場合は、`if` によってこのような引数がテストされる可能性を考慮に入れて、それに応じてプログラムを書かなければならない。)

3.9 Lisp における真と偽

`if` 式の中の真偽テストについては、触れておかなければならない重要な側面がある。今までは、真とか偽とかいう言葉を、述語の値としてあたかも新しい Lisp のオブジェクトとして話してきた。しかし実際は「偽」というのは単に我々が親しんできた良き友 `nil` なのである。そして、その他は全て— どんなものであっても—「真」である。

真であるか否かテストされる S 式は、評価した結果が `nil` 以外の値であれば *true* と判断される。別の言い方をすると、テストの結果は返された値が 47 のような数値や、`"hello"` というような文字列、あるいは `flowers` というような (`nil` 以外の) シンボルや、リストであったり、はたまたバッファであったりする場合でも真であると解釈されるわけである。

このことを説明する前に、`nil` について説明せねばならないだろう。

Lisp ではシンボル `nil` は二つの意味を持つ。一つ目は空リストである。そして二つ目は偽であり、真偽テストの結果が間違いであった場合に返る値である。`nil` は空リストとして `()` のように書いても良いし、`nil` と書いてもよい。Lisp にとっては、`()` と `nil` は全く同じものである。しかし、人間にとっては `nil` は偽を表わすのに使い、`()` は空リストを表わすのに使うのが普通だろう。

Lisp では `nil`以外—即ち空リスト以外—の全ての値は真と解釈される。これは、評価した時に空リスト以外の何かを返すものは、`if` 式のテストで真と判断されることを意味する。例えば、もしテストの部分に数値を置いたとすると、Lisp では数値は評価された時には自分自身を返すように出来ているのだから、結果としてこの場合の `if` 式でのテストは真となる。テストの結果が偽となるのは S 式を評価して `nil`、即ち、空リストが返る時だけである。

次に挙げる二つの S 式を評価してみることで、このことを確かめることが出来る。

最初の例では、if 式のテストの結果、数字の 4 が評価され、その数自身が返される。結果として then-part が評価され、エコー領域に 'true' が表示される。二番目の例としては、nil は偽を表わすので 結果として else-part が評価され、エコー領域に 'false' と表示される。

```
(if 4
    'true
  'false)

(if nil
    'true
  'false)
```

ついでながら、もしテストで真を返すような適当な値が見つからなければ、Lisp インタプリタは真を表わすためにシンボル t を返す。例えば、(> 5 4) は評価された場合 t を返す。これもいつものようにして確認出来る。

```
(> 5 4)
```

一方、この関数は偽である場合は nil を返す。

```
(> 4 5)
```

3.10 save-excursion

関数 save-excursion は、この章で説明する 4 番目の、そして最後の特殊形式である。

編集用の Emacs Lisp プログラムでは、関数 save-excursion は大変よく使用されるものである。これは、ポイントとマークの位置をセーブし、関数の本体を実行し、そしてその際にポイントとマークの位置が変わったなら、それを元の位置に戻す。ポイントとマークが意外な場所に動いて、驚いたり、困惑したりするのを防ぐことがこの関数の主な目的である。

save-excursion について議論する前に GNU Emacs において、ポイントとマークがどういうものであるかを復習しておくのが良いだろう。*Point* は現在のカーソルの位置である。カーソルが何処にあろうと、それがポイントである。より正確に言うと、カーソルが文字の上にあるような端末では、ポイントは、その文字の左端の位置にある。Emacs Lisp ではポイントは整数である。バッファの最初の文字は 1 であり、二番目の文字は 2 であり... といった感じだ。関数 point は現在の位置を数として返す。各々のバッファがそれ自身のポイントの位置を持っている。

mark は、ポイントとはまた別の、バッファにおけるある位置を指し示すものである。その値は *C-SPC* (set-mark-command) で設定される。マークが設定されている場合、*C-x C-x* (exchange-point-and-mark) を使ってカーソルをマークにジャンプさせ、以前ポイントがあった場所にマークを設定することが出来る。更に、もし他のマークが既にあった場合、以前のマークはマークリングに保存される。多くのマークをこうして保存することが出来る。*C-u C-SPC* とタイプすることで、以前保存されたマークに順にさかのぼって行くことが出来る。

バッファの中で、ポイントとマークに挟まれた部分を リージョン (*region*) と呼ぶ。リージョンに関するコマンドは沢山ある。例えば、center-region、count-lines-region、kill-region、そして print-region なんかがそうである。

特殊形式 save-excursion はポイントとマークの位置を保存し、この特殊形式内の本体部分のコードが Emacs Lisp によって評価された後に、その位置を復元する。 こうして、例えばポイントがあるテキストの先頭にあり、あるコードがその位置をバッファの最後に移動してしまったとしても、本体部分の S 式が評価され終わった後に save-excursion がポイントを元の位置に戻してくれる。

Emacs ではユーザーが予期しない所でも、しばしば内部的にポイントの位置を動かしていることがある。例えば count-lines-region なんかもポイントの位置を移動している。予期しない、また (ユーザーの立場からみて) 不必要な移動によってユーザーが煩わされるの防ぐために、save-excursion は頻繁に使われる。save-excursion を使うのは家の中を住み心地の良いものにしておくことと一緒である。

家の中がきちんと整理されているかどうかを確認するために、save-excursion は、たとえ、その内部のコードがうまく動かない場合にも (専門用語を使ってより正確に言うと、「異常終了した場合にも」)、ちゃんとポイントとマークの位置を復元してくれる。この機能は大変役に立つ。

save-excursion は単にポイントとマークの値を記録するだけでなく、カントバッファも保持していて、それを復元してくれる。これは、バッファを変更するようなコードを書いても、save-excursion が元のバッファに戻してくれることを意味している。実際、append-to-buffer の中では、save-excursion はその目的で使われている。(Section 4.4 “append-to-buffer の定義”, page 36, 参照。)

3.10.1 save-excursion 式のテンプレート

save-excursion のテンプレートは単純である。

```
(save-excursion
  本体...)
```

関数の本体は一つないしは複数のS式であり、それらが順に Lisp インタプリタによって評価されていく。もし本体に二つ以上のS式があったとすると、save-excursion という関数自体の値としては、それらの内最後のS式の値が返される。他のS式は単に副作用として評価されるのである。save-excursion 自体は (ポイントとマークを復元するという) 副作用だけのために用いられる。

より詳しく書くと、save-excursion 式のテンプレートは次のように書ける。

```
(save-excursion
  本体の一番目の式
  本体の二番目の式
  本体の三番目の式
   ...
  本体の最後の式)
```

ここでS式は、シンボルそのものか、もしくはリストである。

Emacs Lisp のコードでは、save-excursion 式は、let 式の中でよく使われる。次のような感じだ。

```
(let 変数リスト
  (save-excursion
    本体...))
```

3.11 復習

ここまでの幾つかの章で、かなりの数の関数と特殊形式を紹介してきた。ここでそれらを簡単にまとめておくと同時に、そこでは触れなかった類似の関数なんかも紹介しておくことにする。

eval-last-sexp
: 現在のポイントの位置より前方の最後のS式を評価する。通常はS式が返す値はエコー領域に表示されるが、引数つきで呼び出された時だけはカレントバッファに表示する。普通は、*C-x C-e* にバインドされている。

defun
: 関数を定義する。この特殊形式は最大5つの部分からなる。関数の名前、関数に渡される引数のリスト、説明文字列、オプションのインタラクティブ宣言、そして関数定義の本体である。

  (未訳) For example, in an early version of Emacs, the function definition was as follows. (It is slightly more complex now that it seeks the first non-whitespace character rather than the first visible character.)

```
(defun back-to-indentation ()
  "Move point to first visible character on line."
  (interactive)
  (beginning-of-line 1)
  (skip-chars-forward " \t"))
```

interactive
: インタプリタに対してその関数がインタラクティブに使用出来ることを宣言する。この特殊形式の後に関数に渡す引数に関する情報として複数の部分からなる文字列が続くことがある。これらの部分の中にはインタプリタに対してその情報を求めるプロンプトを出すよう要求するものもある。この文字列の中では各々の部分が改行コード '\n' によって区切られている。

  よく使われるコード文字には、次のようなものがある。

  b
  : 現在存在するバッファ。

  f
  : 現在存在するファイル。

  p
  : 数値である 前置引数。('p' が小文字であることに注意。)

  r
  : 二つの数値引数としてのポイントとマークを、小さい方を先に並べたもの。このコード文字のみが二つの引数の列を特定する。(他は一つだけ。)

コード文字の完全なリストについては Section “Code Characters for ‘interactive’” in *The GNU Emacs Lisp Reference Manual*, を参照。

`let` 変数リストが `let` の本体部分のコードのみに対して使われることを宣言し、かつ、それらの変数に `nil` ないしは指定した初期値を与える。そして、`let` 式の本体部分にある S 式を評価し、その中の最後の S 式を評価した値を返す。`let` 式の本体の内部では、Lisp インタプリタはその `let` 式の外部にある同じ名前の変数は見ない。

例)

```
(let ((foo (buffer-name))
      (bar (buffer-size)))
  (message
   "This buffer is %s and has %d characters."
   foo bar))
```

`save-excursion`
この特殊形式の本体部分を評価する直前のポイントとマークの位置、及びカレントバッファを保存する。そして本体実行後、ポイントとマークの位置及びバッファを復元する。

例)

```
(message "We are %d characters into this buffer."
         (- (point)
            (save-excursion
              (goto-char (point-min)) (point))))
```

`if` この関数の最初の引数を評価する。そして、もしその値が真であれば、二番目の引数を評価し、そうでない場合、三番目の引数があればそれを評価する。

特殊形式 `if` は条件分岐 (*conditional*) と呼ばれる。Emacs Lisp には他にも条件分岐があるが、`if` はその中でも多分もっともよく使われるものであろう。

例)

```
(if (= 22 emacs-major-version)
    (message "This is version 22 Emacs")
  (message "This is not version 22 Emacs"))
```

`<`
`>`
`<=`
`>=` 関数 `<` は、最初の引数が二番目の引数よりも小さいかどうかをテストする。これと対をなす関数 `>` は、最初の引数が二番目よりも大きいかどうかをテストする。同様に、`<=` は最初の引数が二番目の引数以下であるかどうかを、`>=` は最初の引数が二番目の引数以上であるかどうかをテストする。どの場合でも引数は両方とも数値でなければいけない。

`=` （未訳）The = function tests whether two arguments, both numbers or markers, are equal.

`equal`
`eq` 二つのオブジェクトが同じであるかどうかをテストする。`equal` の方は、両者が同じ構造と中身を持ちさえすれば、真を返す。一方、`eq` の方は、引数が両方とも実際に同じオブジェクトである時のみ真を返す。

`string<`
`string-lessp`
`string=`
`string-equal`
The `string-lessp` function tests whether its first argument is smaller than the second argument. A shorter, alternative name for the same function (a `defalias`) is `string<`.

The arguments to `string-lessp` must be strings or symbols; the ordering is lexicographic, so case is significant. The print names of symbols are used instead of the symbols themselves.

An empty string, ‘`""`’, a string with no characters in it, is smaller than any string of characters.

`string-equal` provides the corresponding test for equality. Its shorter, alternative name is `string=`. There are no string test functions that correspond to `>`, `>=`, or `<=`.

`message` エコー領域にメッセージを表示する。このメッセージは一行のみでなければならない。最初の引数は '`%s`', '`%d`', あるいは '`%c`' を中に含む文字列であり、これらはその後に続く引数におきかえられる。'`%s`' の部分に入る引数は文字列かシンボルでなければならない。また、'`%d`' の部分に入る引数は、数でなければならない。'`%c`' の所に入る引数も数でなければならないが、こちらはそれに対応する ascii コードの文字が表示される。

`setq`
`set` 関数 `setq` は最初の引数に二番目の引数の値をセットする。最初の引数は `setq` により、自動的に引用符が付けられる。続けて引数のペアを書いた場合、それらについても同じことをする。もう一方の関数 `set` の方は、二つの引数しか取ることが出来ない。また、その引数は両方ともまず最初に評価され、その後、二番目の引数を評価して返された値を最初の引数を評価して返された値にセットする。

`buffer-name`
引数は持たず、バッファ名前を文字列として返す。

`buffer-file-name`
引数は持たず、バッファがビジットしているファイルの名前を返す。

`current-buffer`
Emacs がアクティブであるようなバッファの中身を返す。これはスクリーンに見えているバッファとは限らない。

`other-buffer`
(`other-buffer` に引数として渡したバッファ及び、カレントバッファを除いて) もっとも最近選択したバッファの中身を返す。

`switch-to-buffer`
Emacs にとってアクティブなバッファを選択し、カレントウィンドウに表示してユーザーが見えるようにする。普通は *`C-x b`* にバインドされている。

`set-buffer`
これからプログラムを走らせるバッファに Emacs を切り替える。ただし、表示しているウィンドウはそのままである。

`buffer-size`
カレントバッファの文字数を返す。

`point` 現在のカーソル位置の値をそのバッファの最初の位置からの文字数として返す。

`point-min`
カレントバッファの中で許される最小のポイントの値を返す。ナローイングが有効でない場合は 1 である。

`point-max`
カレントバッファの中で許される最大のポイントの値を返す。ナローイングが有効でない場合はバッファの最後の位置に一致する。

3.12 練習問題

- 引数の値を二倍するインタラクティブでない関数を書きなさい。また、それをインタラクティブにしなさい。
- 現在の `fill-column` の値が関数に渡される引数よりも大きいかどうか判定し、もしそうなら適当なメッセージを表示するような関数を書きなさい。

4 バッファに関する幾つかの関数

この章では、GNU Emacs で使われている関数を、幾つか詳しく見てみることにする。こういうことは、“walk-through” と呼ばれる。これらの関数は Lisp コードの例として扱うのであるが、最初の単純化された例を除いて、決して「机上」のものではない。実際に GNU Emacs で使われている現実のコードである。これらの関数定義から多くのことを学ぶことが出来る。この章で説明する関数は、皆バッファに関連したものである。他のものについては後で学ぶことにしよう。

4.1 情報の探し方

この walk-through の中では、新しく出て来た関数はその都度、時には細かく、時には簡単に、説明することにする。出て来た関数に興味を持った場合は、*C-h f* に続けてその関数名をタイプして *RET* を押せば、どんな Emacs Lisp 関数についても何時でも、完全な説明文を見ることが出来る。同様に変数についても *C-h v* に続けてその変数名 (と *RET*) を押せば、その変数の完全な説明文を見ることが出来る。

（未訳）Also, `describe-function` will tell you the location of the function definition.

(未訳) Put point into the name of the file that contains the function and press the `RET` key. In this case, `RET` means `push-button` rather than ‘return’ or ‘enter’. Emacs will take you directly to the function definition.

より一般的には、もしその関数の元のソースを見たくなったら、関数 `find-tag` を使ってその関数に飛ぶことが出来る。まず *M-.* とタイプし (即ち、*META* キーとピリオドキーを同時に押すか、または *ESC* キーを押してからピリオドキーを押し), 出てきたプロンプトに対してソースコードを見たい関数の名前、例えば `mark-whole-buffer` 等、をタイプする。すると、Emacs はバッファを切り替えて、その関数のソースコードをスクリーンに表示してくれる。元のバッファに戻るには、*C-x b RET* とタイプすれば良い。

To use the `find-tag` command, type *M-.* (i.e., press the period key while holding down the `META` key, or else type the `ESC` key and then type the period key), and then, at the prompt, type in the name of the function whose source code you want to see, such as `mark-whole-buffer`, and then type `RET`. Emacs will switch buffers and display the source code for the function on your screen. To switch back to your current buffer, type *C-x b RET*. (On some keyboards, the `META` key is labeled `ALT`.)

あなたの使っている Emacs のデフォルトの初期値いかんでは ‘タグテーブル’ を指定する必要があるかもしれない。これは `TAGS` と呼ばれるファイルである。（未訳）For example, if you are interested in Emacs sources, the tags table you will most likely want, if it has already been created for you, will be in a subdirectory of the `/usr/local/share/emacs/` directory; thus you would use the `M-x visit-tags-table` command and specify a pathname such as `/usr/local/share/emacs/22.1.1/lisp/TAGS`. If the tags table has not already been created, you will have to create it yourself. It will be in a file such as `/usr/local/src/emacs/src/TAGS`.

（未訳）To create a `TAGS` file in a specific directory, switch to that directory in Emacs using *M-x cd* command, or list the directory with *C-x d* (`dired`). Then run the compile command, with `etags *.el` as the command to execute:

```
M-x compile RET etags *.el RET
```

（未訳）For more information, see Section 12.5 “Create Your Own `TAGS` File”, page 110.

(旧訳) `emacs/src` ディレクトリにあるファイルを指定することになる場合が殆どであろう。その場合 `M-x visit-tags-table` コマンドを使って、などのようにパス名を指定する。Section “Tag Tables” in *The GNU Emacs Manual*, 参照。また、自分自身のものを作成する方法については Section 12.5 “自分自身の `TAGS` ファイルの作成”, page 110, を見よ。

（旧訳）Emacs Lisp に慣れてしまってからは、ソースコードを眺めるために頻繁に `find-tag` を使うことになるだろう。そして自分自身の `TAGS` テーブルを作ることになるはずだ。

ついでに付け加えておくと、Lisp コードを含むファイルは便宜上ライブラリ (`library`) と呼ばれる。このメタファは、一般的なライブラリではなく、例えば法律とか工学とかのような特定分野のライブラリから来ている。各々のライブラリないしはファイルには特定のトピックや動作に関する関数が含まれている。例えば、`abbrev.el` には省略等のショートカットキーを扱う関数が、また `help.el` にはオンラインヘルプを扱う関数が収められている、といった感じである。(時には複数のライブラリ

が一つの動作に関するコードを提供していることもある。電子メールを読むためのコードを提供する `rmail...` ファイルなんかがそうだ。) `The GNU Emacs Manual` を見れば、「*C-h p* コマンドによって標準的な Emacs Lisp ライブラリをトピックをキーワードにして検索することが出来る」と言った記述を見つけることが出来る。

4.2 簡略版 `beginning-of-buffer` の定義

まず `beginning-of-buffer` という関数から始めるのが良いであろう。この関数には既に慣れているだろうし、理解するのも易しいからだ。インタラクティブなコマンドとして使われるものであろうと、`beginning-of-buffer` はカーソルをバッファの先頭に移動させ、以前の位置にマークを置く。この関数は大抵は *M-<* にバインドされている。

この章では、この関数の短縮されたバージョンを説明することで、これがどういうふうに使われることが多いのかを示そう。この短縮版も上に書いたような動作をするが複雑なオプションは付いていない。完全版については別のセクションで説明することにする。(Section 5.3 “`beginning-of-buffer` の完全な定義”, page 47, 参照。)

コードを見る前に、どんな関数定義が含まれているかを考えてみよう。まず、関数をインタラクティブにするための、即ち *M-x beginning-of-buffer* とタイプしたり、あるいは *M-<* といったキーコードをタイプすることで呼び出せるようにするための S 式を含んでいなければならない。また、バッファの元の位置にマークするためのコードも必要だ。更にカーソルをバッファの先頭に移動するコードも入っていなくてはならない。

これがこの関数の短縮版の完全なテキストである。

```
(defun simplified-beginning-of-buffer ()
  "Move point to the beginning of the buffer;
leave mark at previous position."
  (interactive)
  (push-mark)
  (goto-char (point-min)))
```

他の関数定義と同様に、この定義も特殊形式 `defun` に続く五つの部分からなっている。

1. 名前: この場合は `simplified-beginning-of-buffer`。
2. 引数リスト: この場合は空リスト `()`
3. 説明文字列
4. インタラクティブにするための S 式
5. 本体

この関数定義では、引数リストは空である。これは、この関数が引数を必要としないことを意味する。(もっとも完全版の方の定義を見れば、そちらにはオプションの引数を渡すことが出来ることが分る。)

インタラクティブにするための S 式は、Emacs にこの関数がインタラクティブに使われることを伝えるための物である。今の場合 `interactive` は引数を持たない。というのも `simplified-beginning-of-buffer` は引数を必要としないからである。

関数の本体部分は次の二行からなる。

```
(push-mark)
(goto-char (point-min))
```

一行目は `(push-mark)` という S 式である。この S 式が Lisp インタプリタによって評価されると、カーソルが何処にあろうと、その位置にマークが設定される。マークの位置はマークリングに保存される。

二行目は `(goto-char (point-min))` である。この S 式はカーソルをバッファ内での最小ポイント、即ちバッファの先頭に移動する。(あるいは、もしナローイングがかかっていれば、アクセス出来る範囲内での最初の位置に移動する。Chapter 6 “ナローイングとワイドニング”, page 52, 参照。)

`push-mark` コマンドによって、`(goto-char (point-min))` でカーソルがバッファの先頭に飛ぶ前に、そのカーソルの位置にマークがセットされる。結果として、もし元の位置に戻ろうと思えば、*C-x C-x* によって元の位置に戻ることが出来る。

これがこの関数定義の全てである!

もし、このようなコードを読んでいて `goto-char` のようによく知らない関数に出くわしたとする。その場合には `describe-function` コマンドを使ってこれがどんな関数かを見ることが出来る。この

コマンドを使うには、まず *C-h f* とタイプし、次に知りたい関数の名前をタイプしてから *RET* を押せば良い。describe-function コマンドは *Help* ウィンドウにその関数の説明文字列を表示してくれる。例えば goto-char の説明文は次の通りである。

```
Set point to POSITION, a number or marker.
Beginning of buffer is position (point-min), end is (point-max).
```

(未訳) The function's one argument is the desired position.

(describe-function はプロンプトを出す時にデフォルトの値としてその時のカーソルの直前にあるシンボルの値を設定する。従って、カーソルをその関数の直後に持っていって *C-h f RET* とタイプすることも出来る。)

end-of-buffer 関数についても beginning-of-buffer とほぼ同じである。ただし、関数の本体部分で、(goto-char (point-min)) の部分が (goto-char (point-max)) となっている。

4.3 mark-whole-buffer の定義

関数 mark-whole-buffer も、関数 simplified-beginning-of-buffer と同じくらい簡単に理解出来る。ただし、今回は単純化したものではなく完全な関数を見ることにする。

mark-whole-buffer 関数は beginning-of-buffer 関数ほどには頻繁には使われない。しかし、同じくらい有用である。この関数はポイントをバッファの先頭に、マークをバッファの最後に置くことで、バッファ全体をリージョンとして指定する。また、大抵は *C-x h* にバインドされている。

GNU Emacs 22 における、この関数の完全なコードは次の通りである。

```
(defun mark-whole-buffer ()
  "Put point at beginning and mark at end of buffer.
You probably should not use this function in Lisp programs;
it is usually a mistake for a Lisp function to use any subroutine
that uses or sets the mark."
  (interactive)
  (push-mark (point))
  (push-mark (point-max) nil t)
  (goto-char (point-min)))
```

他の全ての関数と同様、mark-whole-buffer も関数定義のテンプレートにきちんと当てはまっている。このテンプレートは次のようなものだった。

```
(defun 関数名 (引数リスト)
  "説明文字列..."
  (インタラクティブ式...)
  本体...)
```

この関数の動作であるが、関数名は mark-whole-buffer である。引数リストが空リスト '()' になっているので、この関数は引数を必要としないことが分る。次に説明文字列が来ている。

次の行は (interactive) である。これをつけると Emacs はこの関数をインタラクティブに使えるものとみなす。ここら辺のことは前節の simplified-beginning-of-buffer 関数の所でも説明した通りである。

4.3.1 mark-whole-buffer の本体

関数 mark-whole-buffer の本体は次の三行からなる。

```
(push-mark (point))
(push-mark (point-max) nil t)
(goto-char (point-min))
```

この三行のうち最初の行は S 式 (push-mark (point)) である。

この行は関数 simplified-beginning-of-buffer の本体の最初の行の (push-mark) と全く同じ働きをする。どちらの場合にも Lisp インタプリタはカーソルの現在の位置にマークを設定する。

私には何故 mark-whole-buffer 関数では (push-mark (point)) と書かれ、beginning-of-buffer 関数では (push-mark) と書かれているか分らない。多分、このコードを書いた人は (push-mark) の引数は省略可能で、引数を受け取らない場合、デフォルトでは自動的に現在のポイントの位置にマークが設定されるということを知らなかったんだろうと思う。あるいはこの S 式が次の行と同じような構造になるように書いたのかもしれない。いずれにしてもこの行によって Emacs はポイントの位置を決定し、その位置にマークを設定する。

mark-whole-buffer の次の行は (push-mark (point-max)) である。この S 式は、バッファの中でポイントの数が最大の位置にマークを設定する。これは大抵はバッファの最後である。(もし、バッファがナローイングされていれば、バッファの中のアクセス可能な範囲での最後尾になる。ナローイングについての詳細は Chapter 6 “ナローイングとワイドニング”, page 52, 参照。) このマークが設定された時点で、現在のポイントの位置に設定されていた、以前のマークは解除される。しかし、Emacs は最近設定されたマークの位置と同様、その位置を記憶している。もし戻りたければ、*C-u C-SPC* を二度タイプすることで、その位置に戻ることが出来る。

In GNU Emacs 22, the (point-max) is slightly more complicated. The line reads

```
(push-mark (point-max) nil t)
```

(未訳) The expression works nearly the same as before. It sets a mark at the highest numbered place in the buffer that it can. However, in this version, push-mark has two additional arguments. The second argument to push-mark is nil. This tells the function it *should* display a message that says ‘Mark set’ when it pushes the mark. The third argument is t. This tells push-mark to activate the mark when Transient Mark mode is turned on. Transient Mark mode highlights the currently active region. It is often turned off.

さて、いよいよ最後の行の (goto-char (point-min)) である。これは beginning-of-buffer と全く同じように書かれている。この S 式はカーソルをバッファの中でポイントが最小の位置に移動させる。つまり、バッファの先頭 (もしくはアクセス可能な範囲での先頭) に移動させるのである。その結果、ポイントはバッファの先頭に移動し、バッファの最後尾にマークが設定される。このようにして、バッファ全体がリージョンになる。

4.4 append-to-buffer の定義

append-to-buffer コマンドも mark-whole-buffer コマンドと同じくらい単純な構造をしている。このコマンドがすることはカレントバッファのリージョン (つまり、バッファ中のポイントとマークある部分) を指定したバッファに複写することである。

append-to-buffer コマンドは、リージョンを複写するのに関数 insert-buffer-substring を使う。この関数は名前から察せられる通り、バッファのある部分から文字からなる文字列、即ち「部分文字列」を取り出して、それを他のバッファに挿入する。

関数 append-to-buffer の大部分は insert-buffer-substring がうまく動作するような状態に設定することに関するものである。このコードはテキストが写されるバッファとそのコピー元となるリージョンの両方の指定が必要となる。この関数の完全なコードは次の通りである。

(未訳) Here is the complete text of the function:

```
(defun append-to-buffer (buffer start end)
  "Append to specified buffer the text of the region.
It is inserted into that buffer before its point.

When calling from a program, give three arguments:
BUFFER (or buffer name), START and END.
START and END specify the portion of the current buffer to be copied."
  (interactive
   (list (read-buffer "Append to buffer: " (other-buffer
                                            (current-buffer) t))
         (region-beginning) (region-end)))
  (let ((oldbuf (current-buffer)))
    (save-excursion
      (let* ((append-to (get-buffer-create buffer))
             (windows (get-buffer-window-list append-to t t))
             point)
        (set-buffer append-to)
        (setq point (point))
        (barf-if-buffer-read-only)
        (insert-buffer-substring oldbuf start end)
        (dolist (window windows)
          (when (= (window-point window) point)
            (set-window-point window (point)))))))))
```

この関数は一連のテンプレートが埋められたものと見做すと理解しやすい。

最も外側のテンプレートは、関数定義のものである。今の場合、次のようになっている (幾つか既に埋めてあるスロットもある)。

```
(defun append-to-buffer (buffer start end)
  "説明文字列..."
  (interactive ...)
  本体...)
```

最初の一行は関数の名前とその三つの引数を含んでいる。引数はテキストが複写される行き先のバッファ及び、複写元のリージョンを指定する `start` と `end` である。

(旧訳) 関数の次の部分には簡潔で完全な説明が記されている。(未訳) The next part of the function is the documentation, which is clear and complete. As is conventional, the three arguments are written in upper case so you will notice them easily. Even better, they are described in the same order as in the argument list.

(未訳) Note that the documentation distinguishes between a buffer and its name. (The function can handle either.)

4.4.1 インタラクティブ式 append-to-buffer

`append-to-buffer` という関数はインタラクティブに使われる関数なので、`interactive` 式が必要である。(`interactive` について復習するには、Section 3.3 “関数をインタラクティブにする”, page 21, を見よ。)

```
(interactive
 (list (read-buffer
        "Append to buffer: "
        (other-buffer (current-buffer) t))
       (region-beginning)
       (region-end)))
```

(未訳) This expression is not one with letters standing for parts, as described earlier. Instead, it starts a list with these parts:

(未訳) The first part of the list is an expression to read the name of a buffer and return it as a string. That is `read-buffer`. The function requires a prompt as its first argument, ‘`"Append to buffer: "`’. Its second argument tells the command what value to provide if you don’t specify anything.

(未訳) In this case that second argument is an expression containing the function `other-buffer`, an exception, and a ‘`t`’, standing for true.

(未訳) The first argument to `other-buffer`, the exception, is yet another function, `current-buffer`. That is not going to be returned. The second argument is the symbol for true, `t`. that tells `other-buffer` that it may show visible buffers (except in this case, it will not show the current buffer, which makes sense).

The expression looks like this:

```
(other-buffer (current-buffer) t)
```

(未訳) The second and third arguments to the `list` expression are `(region-beginning)` and `(region-end)`. These two functions specify the beginning and end of the text to be appended.

(未訳) Originally, the command used the letters ‘`B`’ and ‘`r`’. The whole `interactive` expression looked like this:

```
(interactive "BAppend to buffer: \nr")
```

(未訳) But when that was done, the default value of the buffer switched to was invisible. That was not wanted.

(未訳) The prompt was separated from the second argument with a newline, ‘`\n`’. It was followed by an ‘`r`’ that told Emacs to bind the two arguments that follow the symbol `buffer` in the function’s argument list (that is, `start` and `end`) to the values of point and mark. That argument worked fine.)

(旧訳) 一つ目の部分は ‘`BAppend to buffer:`’ である。ここで、‘`B`’ は Emacs に対し、この関数に渡すバッファ名を尋ねるように要求している。これによって Emacs はミニバッファに ‘`B`’ に続く文字列—今の場合は ‘`Append to buffer: `’—からなるプロンプトを出して、ユ-ザーに名前を入力するよ

う促す。Emacs はこの関数の引数リストにある変数 buffer に、そこで指定されたバッファをバインドするのである。

（旧訳）改行コード '\n' は引数の最初の部分と二つ目の部分を分けるために用いられる。'\n' の次には 'r' が続いているが、これは関数の引数リストの中の buffer に続く二つのシンボル (つまり、start と end) にポイントとマークの値をバインドすることを示している。

4.4.2 append-to-buffer の本体

関数 append-to-buffer の本体部分は let から始まっている。

以前見たように (Section 3.6 "let", page 24, を参照) let 式の目的は、let 式内部だけで使われる変数を一つないしは複数作り、それに初期値を与えることである。これは、そのような変数は let 式外部の同じ名前の変数と混同されることがない、ということを意味している。

let 式の部分をアウトラインにして append-to-buffer のテンプレートを書いてみる。これによって let 式がどういうふうに関数に組み込まれているかが分る。

```
(defun append-to-buffer (buffer start end)
  "説明文字列..."
  (interactive "BAppend to buffer: \nr")
  (let ((変数 値))
       本体...)
```

let 式は三つの要素を持っている。

1. シンボル let。
2. 変数リストの中身。今の場合は二つの要素からなるリスト (*variable value*) が一つあるだけ。
3. let 式の本体。

関数 append-to-buffer の中では変数リストは次のようになっている。

```
(oldbuf (current-buffer))
```

let 式のこの部分では、変数 oldbuf が (current-buffer) を評価して返された値にバインドされる。変数 oldbuf は現在作業しているバッファの記録のために使われるものである。

変数リストの要素は一組の括弧に囲まれている。これによって Lisp インタプリタは変数リストと let 式の本体を区別出来るのである。その結果として変数リストの中の、二つの要素からなるリストであるような要素は更に一組の括弧によって囲まれることになる。ということで、この行は次のようになる。

```
(let ((oldbuf (current-buffer)))
  ... )
```

ひょっとすると oldbuf の前に二つの括弧があることに驚くかもしれないが、oldbuf の前の最初の括弧は変数リストの区切りを示すものであり、次の括弧は二つの要素からなるリスト (oldbuf (current-buffer)) の最初の括弧であることさえ分れば問題ないだろう。

4.4.3 append-to-buffer の中の save-excursion

append-to-buffer の中の let 式の本体は save-excursion 式からなっている。

関数 save-excursion はポイントとマークの位置を保存し、save-excursion の本体部分の S 式の実行が完了した時点でそれらを元の位置に戻す。更に save-excursion は元々のバッファが何であったかも憶えていて、そのバッファに戻してくれる。save-excursion は append-to-buffer の中でこのように使われている。

ついでだが、ここで Lisp の関数の書式としては、通常複数の行に広がって括弧で閉じられているようなものは全て、最初のシンボルよりも深くインデントされていることに注意しておいた方が良いだろう。今回の関数定義の中では次のように let 式は defun よりも深くインデントされている。

```
(defun ...
  ...
  ...
  (let...
    (save-excursion
      ...
```

このように書式を工夫することで、save-excursion の本体部分の二行が save-excursion に付随する括弧で囲まれていることが簡単に見てとれる。同様に、save-excursion それ自身が let に付随する括弧で囲まれていることもすぐに分る。

```
(let ((oldbuf (current-buffer)))
  (save-excursion
    ...
    (set-buffer ...)
    (insert-buffer-substring oldbuf start end)
    ...))
```

関数 save-excursion の使い方は、次のテンプレートのスロットが埋められているものだと思うとよく理解出来るだろう。

```
(save-excursion
  本体の最初の式
  本体の二番目の式
  ...
  本体の最後の式)
```

(旧訳) この関数では save-excursion には二つの S 式しか含まれていない。その本体は次の通りである。

(未訳) In this function, the body of the save-excursion contains only one expression, the let* expression. You know about a let function. The let* function is different. It has a '*' in its name. It enables Emacs to set each variable in its varlist in sequence, one after another.

(未訳) Its critical feature is that variables later in the varlist can make use of the values to which Emacs set variables earlier in the varlist. See "The let* expression", page 105.

(未訳) We will skip functions like let* and focus on two: the set-buffer function and the insert-buffer-substring function.

(未訳) In the old days, the set-buffer expression was simply

```
(set-buffer (get-buffer-create buffer))
```

(旧訳) このリストのもっとも内側にある S 式は (get-buffer-create buffer) というものである。この S 式は get-buffer-create 関数を使っている。これはその名前のバッファの内容を取り込むか、もし存在しない場合は新しくその名前を持つバッファを作成する。これは append-to-buffer 関数を使うことで、それ以前には存在してなかったバッファにもテキストを出力することが出来ることを示している。

(旧訳) get-buffer-create はまた set-buffer が不必要なエラーに遭遇するのを防いでもいる。つまり set-buffer を使う時は行き先のバッファが存在している必要があるのだ。もしありもしないバッファを指定したりすると Emacs はそこで仕事をやめてしまう。存在しないバッファである場合には get-buffer-create がそのバッファを作ってくれるので、set-buffer は常にバッファを得ることが出来るというわけである。

しかし今日では

```
(set-buffer append-to)
```

(未訳) append-to is bound to (get-buffer-create buffer) earlier on in the let* expression. That extra binding would not be necessary except for that append-to is used later in the varlist as an argument to get-buffer-window-list.

(未訳) The append-to-buffer function definition inserts text from the buffer in which you are currently to a named buffer. It happens that insert-buffer-substring copies text from another buffer to the current buffer, just the reverse—that is why the append-to-buffer definition starts out with a let that binds the local symbol oldbuf to the value returned by current-buffer.

(旧訳) 関数 append-to-buffer が評価されると、save-excursion の本体部分の二つの S 式が順に評価される。最後の S 式の値が save-excursion 関数の値として返される。もう一方の S 式は単に副作用だけのために評価される。

(旧訳) save-excursion の本体の最初の行は、現在のバッファを append-to-buffer の引数の最初の引数で指定されたものに切り替えるために関数 set-buffer を使っている。(バッファの切替は副作用である。前にも言ったように Lisp にとっての副作用こそが我々の主目的であることが多い。) 二行目がこの関数の本来の作業を行う。

（旧訳）関数 set-buffer は Emacs の注意をテキストをコピーしようとしているバッファの方に向けさせる。そのバッファから save-excursion によって元のバッファに帰ってくるのである。

（旧訳）append-to-buffer の最後の行はテキストを追加する働きをする。

The insert-buffer-substring expression looks like this:

```
(insert-buffer-substring oldbuf start end)
```

関数 insert-buffer-substring は最初の引数で指定されたバッファから文字列をコピーし、それを現在のバッファに挿入する。今の場合は insert-buffer-substring の引数は let で生成されバインドされた変数 oldbuf の値であり、これは append-to-buffer コマンドを実行した時点でのカレントバッファである。

insert-buffer-substring が作業を終えると、save-excursion が元のバッファに戻してくれる。そうして append-to-buffer の仕事は完了する。

骨組みだけ書くと、本体部分がやっていることは次の通りである。

```
(let (oldbufを current-bufferの値にバインド)
  (save-excursion                    ; バッファの状態の保存
    バッファ切替
    現在のバッファに oldbufから部分文字列を挿入)

  終わったら元のバッファに戻る
終了後は oldbufの局所的な意味は消え去る
```

以上まとめると、append-to-buffer は次のような働きをする。まずはカレントバッファの値を oldbuf という変数に保存する。次に別のバッファをユーザーから聞き出し、もし必要なら新規に作成して、そのバッファに移る。oldbuf の値を利用して元のバッファのリージョン内のテキストを取り出し新しいバッファに挿入する。そして最後に save-excursion を用いて元のバッファに戻る。

append-to-buffer を見ていく中で、かなり複雑な関数を探検したことと思う。let や save-excursion の使い方も見られたし、複数のバッファ間の行き来の仕方も見た。他の多くの関数定義の中でもこのようにして let, save-excursion, そして set-buffer を利用している。

4.5 復習

ここで、この章で議論した様々な関数について簡単にまとめておく。

describe-function
describe-variable
: 関数ないしは変数の説明文字列を表示する。通常は、*C-h f* と *C-h v* にバインドされている。

find-tag
: 関数のソースや変数を含むファイルを探して、そのバッファに移り、その関数や変数が定義されている位置にポイントを移動する。通常は *M-.* にバインドされている。(これは、META キーを押して終止符を押すという意味である。)

save-excursion
: ポイントとマークの位置を保存し、save-excursion の引数が評価された後で、それらの値を元に戻す。また、カレントバッファも憶えていて、そのバッファに戻してくれる。

push-mark
: マークを現在の位置に設定し、直前のマークの値を、マークリングに記録する。このマークは、バッファ内の位置を示し、たとえそのバッファにテキストが追加されたり削除されたりしても、その相対的な位置を保持してくれる。

goto-char
: ポイントを引数の値で指定された位置に移動する。引数は数値あるいはマーカ、もしくは (point-min) のように位置を表わす数値を返すような S 式でなければならない。

insert-buffer-substring
: この関数に引数として渡されたバッファのリージョンを現在のバッファに挿入する。

mark-whole-buffer
: バッファ全体をリージョンに設定する。普通は *C-x h* にバインドされている。

`set-buffer`
: Emacs が注目しているバッファを他のバッファに切り替える。が、ディスプレイしているウィンドウは変更しない。人間が他のバッファで作業したい時よりかはプログラムの中でよく使われるものである。

`get-buffer-create`
`get-buffer`
: ある名前を持つバッファを見つける。その名前のバッファが無い場合は新しく作る。`get-buffer` はもしその名前のバッファが無い場合には `nil` を返す。

4.6 練習問題

- あなた自身の `simplified-end-of-buffer` を書きなさい。そして、それが実際に動くことかどうかテストしてみなさい。
- (未訳) Use `if` and `get-buffer` to write a function that prints a message telling you whether a buffer exists.
- `find-tag` を利用して、`copy-to-buffer` 関数のソースを見つけなさい。

5 もう少し複雑な関数

この章では、前の章で学んだことを基礎により複雑な関数を見ていくことにする。`copy-to-buffer` 関数は、一つの定義の中で `save-excursion` を二回使う方法を示してくれる。一方、`insert-buffer` 関数の方は `interactive` 式の中での `*` の使い方と、`or` の使い方、そして、オブジェクトの名前と、その名前のついたオブジェクトとの違いについて教えてくれる。

5.1 `copy-to-buffer` の定義

`append-to-buffer` がどのように動作するかを理解した後であれば、`copy-to-buffer` の方も簡単に理解出来る。この関数はテキストを他のバッファにコピーする。ただし、そのバッファに付け加えるのではなく、以前あったテキストを置き換えてしまう。`copy-to-buffer` 関数のコードは `append-to-buffer` のコードとほぼ同じであるが、`erase-buffer` を使い、また `save-excursion` を二回使っている点が違う。(`append-to-buffer` の説明については Section 4.4 “append-to-buffer の定義”, page 36, を参照。)

`copy-to-buffer` の本体は次の通りである。

```
...
(interactive "BCopy to buffer: \nr")
(let ((oldbuf (current-buffer)))
  (with-current-buffer (get-buffer-create buffer)
    (barf-if-buffer-read-only)
    (erase-buffer)
    (save-excursion
      (insert-buffer-substring oldbuf start end)))))
```

(未訳) The `copy-to-buffer` function has a simpler `interactive` expression than `append-to-buffer`.

The definition then says

```
(with-current-buffer (get-buffer-create buffer) ...
```

(未訳) First, look at the earliest inner expression; that is evaluated first. That expression starts with `get-buffer-create buffer`. The function tells the computer to use the buffer with the name specified as the one to which you are copying, or if such a buffer does not exist, to create it. Then, the `with-current-buffer` function evaluates its body with that buffer temporarily current.

(未訳) (This demonstrates another way to shift the computer's attention but not the user's. The `append-to-buffer` function showed how to do the same with `save-excursion` and `set-buffer`. `with-current-buffer` is a newer, and arguably easier, mechanism.)

(未訳) The `barf-if-buffer-read-only` function sends you an error message saying the buffer is read-only if you cannot modify it.

(未訳) The next line has the `erase-buffer` function as its sole contents. That function erases the buffer.

(未訳) Finally, the last two lines contain the `save-excursion` expression with `insert-buffer-substring` as its body. The `insert-buffer-substring` expression copies the text from the buffer you are in (and you have not seen the computer shift its attention, so you don't know that that buffer is now called `oldbuf`).

(未訳) Incidentally, this is what is meant by 'replacement'. To replace text, Emacs erases the previous text and then inserts new text.

(旧訳) `append-to-buffer` の定義との違いが現れるのは、テキストがコピーされるバッファに移ってからである。つまり、この後 `copy-to-buffer` 関数では以前の内容を消去してしまう。(これが、「置き換え」と言った意味である。Emacs ではテキストを置き換える場合、以前の内容をまず消去し、次に新しいテキストを挿入する。) 以前の内容を消去した後、もう一度 `save-excursion` が使われる。そして新しいテキストが挿入される。

(旧訳) 何故 `save-excursion` が二回も使われるのか？ それを理解するために、もう一度この関数が何をやっているか見てみよう。

大ざっぱには、copy-to-buffer の本体は次のようになっている。

```
(let (oldbufを current-bufferの値にバインド)
      (with-the-buffer-you-are-copying-to
      (but-do-not-erase-or-copy-to-a-read-only-buffer)
      (erase-buffer)
      (save-excursion
        現在のバッファに oldbufから部分文字列を挿入)))
```

（旧訳）最初の save-excursion は Emacs にテキストのコピー元のバッファを返している。これは当たり前のことで、append-to-buffer でも同じような使われ方をしていたのだった。では何故二番目が必要なのか？ 理由は、insert-buffer-substring が常にポイントを挿入されたリージョンの最後に移動してしまうからである。二番目の save-excursion があるおかげで、Emacs は挿入されたテキストの最初の位置にポイントを移動する。大抵の状況では、ユーザーは挿入されたテキストの最初の位置にポイントがあるほうを好むものである。(勿論、copy-to-buffer 関数は終了時に元居たバッファに戻る—しかし、もしユーザが次にコピー先のバッファに移ったとすると、ポイントはそのテキストの先頭に移動してくれるというわけである。このように、二番目の save-excursion はちょっぴり気の効いた働きをしてくれるものなのだ。)

5.2 insert-buffer の定義

insert-buffer もまた、バッファに関する関数である。このコマンドは他のバッファをカレントバッファの中に挿入する。これは append-to-buffer や copy-to-buffer と逆である。これらの方は、カレントバッファからリージョンを取り出して他のバッファにコピーするのであった。

Here is a discussion based on the original code. The code was simplified in 2003 and is harder to understand.

(See Section 5.2.6 “New Body for insert-buffer”, page 46, to see a discussion of the new body.)

更に、このコードは *read-only* なバッファに関する interactive の使い方や、オブジェクトの名前と、その名前のついたオブジェクトそのものとの違いの重要性を理解させてくれる。これがそのコードである。

Here is the earlier code:

```
(defun insert-buffer (buffer)
  "Insert after point the contents of BUFFER.
Puts mark after the inserted text.
BUFFER may be a buffer or a buffer name."
  (interactive "*bInsert buffer: ")
  (or (bufferp buffer)
      (setq buffer (get-buffer buffer)))
  (let (start end newmark)
    (save-excursion
      (save-excursion
        (set-buffer buffer)
        (setq start (point-min) end (point-max)))
      (insert-buffer-substring buffer start end)
      (setq newmark (point)))
    (push-mark newmark)))
```

他の関数定義と同様に、テンプレートを使ってこの関数のアウトラインを見ることが出来る。

```
(defun insert-buffer (buffer)
  "説明文字列..."
  (interactive "*bInsert buffer: ")
  本体...)
```

5.2.1 insert-buffer のインタラクティブ式

insert-buffer ではインタラクティブ宣言の引数には二つの部分がある。一つはアスタリスク ‘*’ であり、もう一つは ‘bInsert buffer: ’ である。

書き込み不可のバッファ

アスタリスクはバッファが read-only—つまり変更出来ないバッファである場合のためのものである。もし、`insert-buffer` が read-only なバッファで呼ばれた場合、エコー領域にその旨を知らせるメッセージが表示され、ビープ音が鳴るか、画面が明滅する。「あなたはこのバッファに如何なるものも挿入してはいけませんよ」というわけである。アスタリスクの後には、次の引数と分離するために改行コードを付ける必要はない。

インタラクティブ式の中の‘b’

インタラクティブ式の次の引数は小文字の ‘b’ から始まっている。(これは `append-to-buffer` のコードの中のとは違うことに注意。こっちでは大文字の ‘B’ を使っていたのだった。Section 4.4 “`append-to-buffer` の定義”, page 36, 参照。) 小文字の ‘b’ は Lisp インタプリタに `insert-buffer` の引数が存在するバッファないしは、その名前でなければならないことを示している。(大文字の ‘B’ オプションの方は存在しないバッファであっても良い。) Emacs はプロンプトを出してバッファの名前を要求し、同時にデフォルトのバッファも提示してくれる。名前の補完もやってくれる。もし、ここで存在しないバッファを指定すると、“No match” というメッセージを受け取る。端末によってはビープ音がするかもしれない。

（未訳）The new and simplified code generates a list for `interactive`. It uses the `barf-if-buffer-read-only` and `read-buffer` functions with which we are already familiar and the `progn` special form with which we are not. (It will be described later.)

5.2.2 関数 `insert-buffer` の本体部分

関数 `insert-buffer` の本体は、主に二つの部分からなっている。一つは `or` 式であり、もう一つは `let` 式である。`or` 式を使う目的は、引数 `buffer` がバッファの名前ではなくちゃんとバッファそのものにバインドされていることを確認することである。`let` 式の本体は、他のバッファをカレントバッファに挿入するコードを含んでいる。

大まかに言って、二つの S 式は次のように `insert-buffer` 関数の中に組み込まれている。

```
(defun insert-buffer (buffer)
  "説明文字列..."
  (interactive "*bInsert buffer: ")
  (or ...
      ...
  (let (変数リスト)
      letの本体... )
```

引数 `buffer` がバッファの名前ではなくバッファそのものにバインドされているかを `or` 式がどのようにして確かめているかを理解するには、最初に関数 `or` を理解する必要がある。

が、その前にまずこの部分を `if` を使って書き直してみたい。そうすればより身近な方法で何がなされているかを見ることが出来るだろう。

5.2.3 `or` の代わりに `if` を使った `insert-buffer`

やるべき仕事は `buffer` の値がバッファそのものであり、バッファ名ではないことを確認することである。もし値が名前であったら、その名前のついたバッファを持ってこないといけない。

まずあなたが何かのコンファレンスに参加しており、案内係があなた方の名前が書かれたリストをもってあなたを探している状況を思い浮かべて欲しい。この状態では案内係はあなた自身ではなく、あなたの名前に「バインド」されている。しかし、案内係があなたを見つけてあなたの腕を取った時点で、案内係はあなた自身に「バインド」されたことになる。

Lisp ではこの状況を次のように説明出来る。

```
(if (not (お客をつかまえている))
    (お客を見つけて腕を取る))
```

我々はこれと同じことをバッファについてやりたいわけである—バッファそのものでない場合も、バッファを見つけて取って来たいのだ。

bufferp という (バッファの名前ではなく) バッファかどうかを判定してくれる述語を使うと、このコードを次のように書くことが出来る。

```
(if (not (bufferp buffer))              ; if-part
    (setq buffer (get-buffer buffer)))  ; then-part
```

コメントにもあるが、上の式で、if 式の真偽テストの部分は (not (bufferp buffer)) であり、一方 then-part の部分は (setq buffer (get-buffer buffer)) である。

このテストの中では bufferp 関数は引数がバッファであれば真の値を返す—しかしバッファの名前であれば偽の値を返す。(bufferp という関数の名前の最後は 'p' で終わっている。以前見たようにこの 'p' は、この関数が述語、つまり、ある性質を持つか否かを判定する関数であることを示すための工夫として付けられている。Section 1.8.4 "関数に間違った型の引数を与えると", page 10, を参照のこと。)

(bufferp buffer) 式の前には関数 not がある。即ち真偽テストは次の通りである。

```
(not (bufferp buffer))
```

not は引数が偽であれば真を返し、偽であれば真を返す関数である。従って、もし (bufferp buffer) が真を返せば、not 式が偽を返すし、偽を返せば真を返す。「真でない」というのは偽ということだし、「偽でない」というのは真だというわけである。

if は、このテストを使って次のように動作する。変数 buffer の値がバッファの名前ではなく実際にバッファだった場合、真偽テストの結果は偽である。すると、if 式は then-part を評価しない。これはまともな動作である。というのは、もし変数 buffer が本当にバッファであれば、これに対して特に何もする必要はないからである。

一方で、buffer の値がもしバッファそのものではなくバッファの名前であったとすると、真偽テストでは真が返り、結果として then-part が評価される。今の場合、then-part は (setq buffer (get-buffer buffer)) である。この S 式は、その名前のついた実際のバッファを得るために get-buffer 関数を使っている。そして、setq で、変数 buffer の値をバッファの名前から実際のバッファそのものに置き換えている。

5.2.4 本体部分の or

insert-buffer 関数で or 式を使う目的は引数 buffer がバッファの名前ではなくバッファそのものにバインドされているかどうかを確認することにある。前節で、どのようにして if を使ってこの作業をするかを見た。しかし、insert-buffer 関数では実際は or を使っている。この理由を理解するには、or 関数の働きを理解しなければならない。

関数 or はいくらでも引数を持つことが出来る。この関数は各引数を順に評価していき、nil でなかった最初の引数の値を返す。また、これが or の重要な特徴なのだが、この関数は引数で nil 以外の値を返すものがあった場合、その後の引数はもはや評価しない。

今の場合の or 式は次の通りである。

```
(or (bufferp buffer)
    (setq buffer (get-buffer buffer)))
```

or の最初の引数は S 式 (bufferp buffer) である。この式は、もし buffer がバッファそのものでありバッファの名前ではない場合は、真を返す。or 式では、この場合、or 式の値として真を返し、その後の引数は評価されない—我々にはこの方が都合が良い。というのもこの場合にはつまり buffer が真のバッファである場合には、この値について何もする必要がないからである。

一方で、もし (bufferp buffer) が nil であれば、それは buffer の値がバッファの名前であるということである。この場合、Lisp インタプリタは or 式の次の引数も評価する。これは (setq buffer (get-buffer buffer)) という式である。この式は値 nil を返す。これが変数 buffer にセットされた値である—そしてこの値はバッファそのものであり、バッファの名前ではない。

以上のことから、シンボル buffer は常に、あるバッファの名前ではなくバッファそのものにバインドされることになる。また、以上の手続きは全てが必要なことである。というのも、次の行の set-buffer 関数はバッファに対してしか働かないからである。

ついでだが、or を使った場合、案内係というのは次のように書ける。

```
(or (お客をつかまえている) (お客を見つけて腕を取る))
```

5.2.5 insert-buffer の中の let 式

insert-buffer 関数では、変数 buffer がバッファの名前ではなくバッファそのものを表わすことを確認した後には、 let 式が続く。ここでは三つの変数 start, end, newmark を指定している。これらの変数は let 式の内部でのみ使われ、let 式の評価が終わるまでの間は、一次的に Emacs 内の同じ名前の他の変数は隠されている。

let 式の本体は二つの save-excursion を使っている。まずは、内側にある方の save-excursion 式を細かく見ていこう。この式は次の通りである。

```
(save-excursion
  (set-buffer buffer)
  (setq start (point-min) end (point-max)))
```

S 式 (set-buffer buffer) は、Emacs が注目するバッファをテキストがコピー元のバッファに切り替える。このバッファの中で、point-min コマンド及び、point-max コマンドを使って変数 start と 変数 end に各々バッファの始まりと終わりの位置を設定する。setq を使って、一つの S 式で二つの変数を同時に設定する方法についてはもう既に説明したのだった。setq の最初の引数は二番目の値に、三番目の引数は四番目の引数の値にセットされる。

(未訳) After the body of the inner save-excursion is evaluated, the save-excursion restores the original buffer, but start and end remain set to the values of the beginning and end of the buffer from which the text will be copied.

外側の save-excursion 式は次の通りである。

```
(save-excursion
  (内側の save-excursion式
     (新しいバッファに移動し startと endをセット)
  (insert-buffer-substring buffer start end)
  (setq newmark (point)))
```

関数 insert-buffer-substring は buffer から、その中の start と end で指定されたリージョン内のテキストを、カレントバッファの中にコピーする。今の場合、コピー元のバッファ全体が start と end の間にあるために、そのバッファ全体が現在編集中のバッファにコピーされることになる。次に、この時点で挿入されたテキストの最後にあるポイントの値が、変数 newmark に記録される。

外側の save-excursion の本体が評価された後、ポイントとマークの位置は元の位置に戻される。

しかしながら、ポイントとマークの位置は新しく挿入されたテキストの最初と最後にある方が、何かと都合が良い。newmark 変数は挿入されたテキストの最後の位置を記録する。let 式の最後の行で (push-mark newmark) という S 式でマークをこの位置に設定している。(以前のマークの位置に移動することも可能である。この値はマークリングに保存されていて、*C-u C-SPC* で遡ることが出来る。) 一方、ポイントの方は初めから挿入されたテキストの最初の位置にある。これは insert 関数を呼び出す前のポイントの位置がここだからである。

完全な let 式は次の通りである。

```
(let (start end newmark)
  (save-excursion
    (save-excursion
      (set-buffer buffer)
      (setq start (point-min) end (point-max)))
    (insert-buffer-substring buffer start end)
    (setq newmark (point)))
  (push-mark newmark))
```

append-to-buffer 関数の時と同じように、insert-buffer 関数も let, save-excursion, そして set-buffer を使っている。それに加えて、この関数は or の使い方の一つも教えてくれている。これらの関数は皆大切な部品であり、これから何度でも出てくることだろう。

5.2.6 insert-buffer の新しい本体

(未訳) The body in the GNU Emacs 22 version is more confusing than the original.

（未訳）It consists of two expressions,

```
(push-mark
 (save-excursion
   (insert-buffer-substring (get-buffer buffer))
   (point)))

 nil
```

except, and this is what confuses novices, very important work is done inside the push-mark expression.

（未訳）The get-buffer function returns a buffer with the name provided. You will note that the function is *not* called get-buffer-create; it does not create a buffer if one does not already exist. The buffer returned by get-buffer, an existing buffer, is passed to insert-buffer-substring, which inserts the whole of the buffer (since you did not specify anything else).

（未訳）The location into which the buffer is inserted is recorded by push-mark. Then the function returns nil, the value of its last command. Put another way, the insert-buffer function exists only to produce a side effect, inserting another buffer, not to return any value.

5.3 beginning-of-buffer の完全な定義

beginning-of-buffer 関数の基本的な構造については、既に議論したことがあるのだった。(See Section 4.2 “簡略版 beginning-of-buffer の定義”, page 34.) このセクションではこの定義の複雑な部分を説明することにする。

以前説明した通り、引数無しで beginning-of-buffer を呼び出すと、その時点でのポイントの位置にマークを設定して、カーソルがバッファの先頭に移動する。しかしながら、もしこのコマンドが 1 から 10 までの数と一緒に呼び出された場合、この関数はバッファ全体を 10 としてこの数の割合の分を計算し、バッファの先頭からその割合の分だけ進んだ位置にカーソルを移動させる。というわけで、*M-<* でこのコマンドを呼び出せば、カーソルはバッファの先頭に移動するし、もし例えば *C-u 7 M-<* として呼び出せば、カーソルはバッファ全体の 70%分の位置にポイントを移動する。もし引数が 10 より大きかった場合はバッファの最後に位置に移動する。

beginning-of-buffer 関数は、引数無しで呼び出すことが出来る。つまり、引数は省略可能である。

5.3.1 省略可能な引数

特にそうでないと言われない限り、Lisp は、関数定義で引数が指定された関数はその引数のための値とともに呼ばれるものと思っている。もし、そうでなかった場合、エラーが出され、‘Wrong number of arguments’ というメッセージを受け取ることになる。

しかしながら、Lisp の特徴として省略可能引数というものがある。あるキーワード (*keyword*) を使うと Lisp インタプリタにその引数が省略可能だと伝えることが出来るのだ。そのキーワードとは &optional である。(‘optional’ の前の ‘&’ もこのキーワードの一部分である。) 関数定義の中でこのキーワード &optional の後に続く引数は、関数が呼ばれる際に必ずしも値を渡される必要はない。

というわけで、beginning-of-buffer の関数定義の最初の行は次のようになる。

```
(defun beginning-of-buffer (&optional arg)
```

また、関数定義全体のアウトラインは次のようになっている。

```
(defun beginning-of-buffer (&optional arg)
  "説明文字列..."
  (interactive "P")
  (or (is-the-argument-a-cons-cell arg)
      (and are-both-transient-mark-mode-and-mark-active-true)
      (push-mark))
  (let (determine-size-and-set-it)
  (goto-char
    (もし引数があれば
        どこへ行けばよいか調べる
      そうでなければ次の場所に
      (point-min))))
   do-nicety
```

この関数は simplified-beginning-of-buffer と似ている。違うのは、interactive 式に "P" が引数として使われていることと、引数が指定されている場合にカーソルを何処に移動するかを判断する goto-char 関数に続いて if-then-else 式が来ていることである。

（未訳）(Since I do not explain a cons cell for many more chapters, please consider ignoring the function consp. See Chapter 9 “How Lists are Implemented”, page 78, and Section “Cons Cell and List Types” in *The GNU Emacs Lisp Reference Manual*.)

interactive 式の中の "P" は、Emacs に、関数に前置引数を渡すように伝えるものである。前置引数というのは *META* キーに続いて数をタイプするか、*C-u* に押してから数をタイプすることで作ることが出来る。(もし *C-u* に続いて数を入力しない場合には、デフォルトとして 4 が渡される。)

if 式の真偽テストの部分は単純である。単に arg だけだ。もし、arg が nil 以外の値を持てば、即ち beginning-of-buffer が引数付きで呼ばれた場合は、真偽テストでは真が返り、if 式の then-part が評価される。一方、もし beginning-of-buffer が引数とともに呼ばれなかったなら、arg の値は nil になり、if 式の else-part の方が評価される。else-part は単に point-min であり、この場合には goto-char 式の全体は (goto-char (point-min)) になる。これは、以前の短縮版 beginning-of-buffer 関数で見た通りである。

5.3.2 引数付きで呼び出された場合の beginning-of-buffer

beginning-of-buffer が引数付きで呼び出された場合は、goto-char に渡す値を計算する、ある S 式が評価される。この S 式は、ちょっと見にはかなり複雑に見える。内部に if 式を含んでいるし、算数を沢山使っている。実際のコードは次の通りである。

```
(if (> (buffer-size) 10000)
    ;; バッファサイズが大きい場合のオーバーフローを避ける！
                          (* (prefix-numeric-value arg)
                             (/ size 10))
  (/
   (+ 10
      (*
       size (prefix-numeric-value arg))) 10)))
```

他の複雑そうな S 式と同様に、これもテンプレートに当てはめてみれば理解しやすくなる。この場合は、if-then-else 式のテンプレートに当てはめる。骨組みは次のようになる。

```
(if (バッファが大きいなら
    バッファのサイズを 10 で割り引数を掛ける
  そうでなければ別の計算をする
```

この中の if 式の真偽テストはバッファのサイズをチェックしている。その理由は、Emacs Lisp の version 18 は約 8,000,000 以下の数値しか扱わないのだが (そんなに大きな数は必要ない)、この計算に従うと、大きなバッファの場合は Emacs は大き過ぎる数を扱うことになってしまうからだ。コードのコメントの中に出てくる「オーバーフロー」とは数が大き過ぎることを意味している。

ということで、バッファが大きい場合と、そうでない場合の二つに分けている。

大きなバッファの場合

（旧訳）beginning-of-buffer の中で内側にある方の if 式はバッファのサイズが 10,000 文字を越えているかどうかをテストしている。そのために、> と buffer-size という関数を使っている。

（未訳）In the old days, the function buffer-size was used. Not only was that function called several times, it gave the size of the whole buffer, not the accessible part. The computation makes much more sense when it handles just the accessible part. (See Chapter 6 “Narrowing and Widening”, page 52, for more information on focusing attention to an ‘accessible’ part.)

その行は次の通りである。

```
(if (> size 10000)
```

もしバッファが大きい場合は、if 式の then-part が評価される。この部分を読みやすいよう書き直すと次のようになる。

```
(*
  (prefix-numeric-value arg)
  (/ size 10))
```

この S 式は乗法であり、二つの引数を関数 * に渡している。

最初の引数は (prefix-numeric-value arg) である。interactive 宣言で "P" が指定されている場合、引数は関数に「未処理の前置引数」(raw prefix argument) として渡され、数値としては渡されない。(リストの中では数値である。) 算数で使えるようにするには変換する必要があり、その仕事を prefix-numeric-value がやっているのである。

二番目の引数は (/ (buffer-size) 10) である。この式は数値としてのバッファの値を 10 で割る。これはバッファサイズの一割にどれだけの文字数が含まれているかを計算するものだ。(Lisp では / は除法に使われる。* が乗法に使われるのと同様である。)

乗法の式全体としては、この値に前置引数の値を掛けることになる—こんな感じだ。

```
(* 数値としての前置引数
   バッファの 10 分の 1 の文字数)
```

例えば、仮に前置引数が '7' だったとするなら、この 10 分の 1 の数に 7 を掛けることで、バッファの中の 70%の位置が得られることになる。

結局、バッファが大きい場合の goto-char 式全体は次のようになる。

```
(goto-char (* (prefix-numeric-value arg)
              (/ size 10)))
```

これでカーソルは望む場所に移動してくれる。

小さなバッファの場合

バッファが 10,000 文字よりも小さかった場合は、ちょっと違った計算方法が取られる。さっきの方法でも出来るのだから、こんなことをする必要はないと思うかもしれない。しかし、バッファが小さい場合、初めの方法ではカーソルを望む場所にきちんと移動してくれない。次に説明する二番目の方法の方がちゃんとした仕事をしてくれる。

コードは次の通りである。

```
(/ (+ 10 (* size (prefix-numeric-value arg))) 10))
```

このコードの中で何をしているかは、括弧の中にどのように関数が埋め込まれているかを見れば分る。各々の式をそれを含む括弧より深くインデントしてやれば、より見やすくなる。

```
(/
 (+ 10
    (*
     size
     (prefix-numeric-value arg)))
 10))
```

これらの括弧を見ると、もっとも内側にある関数は (prefix-numeric-value arg) であることが分る。これは未処理の前置引数を数値に変換するものである。この数値は次の式によってバッファサイズと掛け合わされる。

```
(* size (prefix-numeric-value arg))
```

この掛け算では、バッファのサイズの値よりも大きな数が生ずる—例えば引数が 7 なら 7 倍大きな数になる。これに 10 が加えられ、最後にこの大きな数が 10 で割られる。こうして得られる値は、バッファのサイズ中の目的とする割合よりも一文字分だけ大きな数である。

これら一連の操作の結果で得られた数が goto-char に渡され、カーソルがその位置に移動する。

5.3.3 完全版 beginning-of-buffer

ここで、beginning-of-buffer の完全なコードを挙げておく。

```
(defun beginning-of-buffer (&optional arg)
  "Move point to the beginning of the buffer;
leave mark at previous position.
With \\[universal-argument] prefix,
do not set mark at previous position.
With numeric arg N,
put point N/10 of the way from the beginning.

If the buffer is narrowed,
this command uses the beginning and size
of the accessible part of the buffer.
```

```
Don't use this command in Lisp programs!
\(goto-char (point-min)) is faster
and avoids clobbering the mark."
  (interactive "P")
  (or (consp arg)
      (and transient-mark-mode mark-active)
      (push-mark))
  (let ((size (- (point-max) (point-min))))
    (goto-char (if (and arg (not (consp arg)))
                   (+ (point-min)
                      (if (> size 10000)
                          ;; Avoid overflow for large buffer sizes!
                          (* (prefix-numeric-value arg)
                             (/ size 10))
                        (/ (+ 10 (* size (prefix-numeric-value arg)))
                           10)))
                 (point-min))))
  (if (and arg (not (consp arg))) (forward-line 1)))
```

二つの細かな点を除けば、これまでの議論でこの関数がどう動くかが分るだろう。細かな点の一つ目は、説明文字列についてであり、二つ目は、この関数の最後の行に関するものである。

で、一点目だが、説明文字列の中に、あるS式についてのリファレンスがある。

```
\\[universal-argument]
```

(旧訳) この式の最初の括弧の前に '\' が使われている。この '\' は Lisp インタプリタに対し、このS式は説明として表示されるものであって、S式として評価されるものではないことを示すものである。

次に二点目だが、`beginning-of-buffer` コマンドの最後の行は、この関数が引数とともに呼ばれた場合に、ポイントを次の行の最初に持ってくためのものである。

```
(if (and arg (not (consp arg))) (forward-line 1))
```

(旧訳) これによって、カーソルはバッファの中の適当な割合の位置の次の行の先頭に移動する。これが最良の位置であろう。カーソルは少くとも要求された割合よりかは常に大きい位置に移動するからだ。この動作はちょっと几帳面過ぎるかもしれないが、かと言ってそうしなければ、必ず文句を言う人がいるだろう。

5.4 復習

ここで、この章の幾つかのトピックを手短にまとめておく。

`or` 引数を順に評価していき、`nil` でない最初の値を返す。ただし、もしどの引数も `nil` を返した場合は `nil` を返す。一言で言うと、引数の中で真である最初の値を返す。つまり、ある引数が真であるかまたは (or) その他の引数のどれかが真であれば真の値を返す。

`and` 引数を順に評価していき、どれか一つでも `nil` があれば、`nil` を返し、どれも `nil` で無ければ最後の引数の値を返す。一言で言うと、全ての引数が真である場合のみ真である値を返す。つまり、ある引数が真であり、かつ (and) その他の引数も全て真である場合に真の引数を返す。

`&optional`
関数定義での引数が省略可能であることを示すために使うキーワード。これがあると、その引数がなくともその関数を評価することが出来る。

`prefix-numeric-value`
(未訳) Convert the 'raw prefix argument' produced by `(interactive "P")` to a numeric value.

`forward-line`
ポイントを次の行の先頭に移動する。1より大きい引数が与えられた場合はその数の行だけ進む。進めない場合は行けるところまで進み、行くはずではあったが進めなかった分の行数を値として返す。

`erase-buffer`
カレントバッファの内容を全て削除する。

`bufferp` 引数がバッファの場合、`t` を返す。そうでない場合は `nil` を返す。

5.5 optional 引数の練習問題

省略可能な引数を持ち、その引数の数が fill-column の値よりも大きいか小さいかをテストして結果を教えてくれるようなインタラクティブな関数を書きなさい。ただし、引数なしの場合は 56 をデフォルトとして使うようにしなさい。

6 ナローイングとワイドニング

ナローイングはバッファの特定部分にのみ注目し、その他の部分を間違って変更したりすることなく作業するための Emacs の機能である。ナローイングは普通は使えないようになっている。これは初心者を混乱させないためである。

ナローイングを使うと、バッファのその他の部分はあたかも存在しないかのように見えなくなる。これは、例えばバッファのある部分でのみ単語を置き換えたい場合なんかに有効である。つまり特定の部分にナローイングしてそこで置換を実行すると、その部分のみ置換が実行されて残りの部分は無視される。検索についても同様で、ナローイングされた部分でのみ検索を行い、外の部分は探さない。というわけで、文書のある部分のみ修正したい場合など、そのリージョンにナローイングをかけることで、間違って修正不要な箇所まで見てしまうのを防ぐことが出来るのである。

しかしながら、ナローイングするとバッファの残りの部分が見えなくなるので、意図せずナローイングしてしまった場合、ファイルのその部分を削除してしまったのではないかと不安に思う人も出てくるかもしれない。更に、undo コマンド (これは大抵 *C-x u* にバインドされている) でもナローイングは解除されない (それに、そうすべきではない)。従って、もし widen コマンドでまた見えるように出来るということを知らなかった場合、パニックに陥いるかもしれない。(Emacs version 18 では、widen のキーバインディングは C-x w だった。version 19 では *C-x n w* になっている。)

ナローイングは、人間だけではなく Lisp インタプリタにとっても便利なようになっている。 Emacs Lisp の関数はしばしば、バッファのある限定された部分に対してのみ働くように設計されている。逆に言うと、Emacs Lisp 関数はナローイングされたバッファ全体に対して働くようになっているとも言える。例えば what-line 関数は、もしバッファにナローイングがかかっていればそれを解除し、仕事が終わるとまたナローイングを元に戻す。一方、what-line 関数の中から呼び出される count-lines 関数は、まずナローイングを用いてバッファの目的とする部分に限定して仕事を行い、その後ナローイングを解除して元の状態に戻す。

6.1 特殊形式 save-restriction

Emacs Lisp では、save-restriction という特殊形式を使うことで、ナローイングのかかった状態を保持することが出来るようになっている。Lisp インタプリタが save-restriction 式に出逢った場合、まず save-restriction 式の本体部分を評価し、次にそのコードを実行することでナローイングの状況が変わった場合にそれを元に戻す。例えば、バッファがナローイングされていて、save-restriction の本体で、ナローイングが解除された場合、実行後に save-restriction はバッファにナローイングされたリージョンを返してくれる。what-line コマンドでは、バッファにどんなナローイングが設定されていても save-restriction に続く widen コマンドで解除してしまう。元のナローイングはこの関数が終了する直前に元に戻されるわけである。

save-restriction 式のテンプレートは単純である。

```
(save-restriction
  本体... )
```

save-restriction の本体は一つないしは複数の S 式であり、それらは Lisp インタプリタによって順に評価される。

最後に注意しておくことがある。それは save-excursion と save-restriction の両方を、片方のすぐ後にもう片方を続けて使う場合、save-excursion の方を外側にすることだ。これを逆にすると、save-excursion を呼び出してからナローイングが設定されたバッファに移動した場合にそのナローイングの状況を記録しそこねることがあるのだ。ということで、もし save-excursion と save-restriction を両方一緒に使うなら、次のように使うことになる。

```
(save-excursion
  (save-restriction
    本体...))
```

(未訳) In other circumstances, when not written together, the save-excursion and save-restriction special forms must be written in the order appropriate to the function.

For example,

```
(save-restriction
  (widen)
  (save-excursion
  本体...))
```

6.2 what-line

what-line コマンドは現在カーソルがある場所までの行数を教えてくれるものである。この関数を見ることで、save-restriction と save-excursion の使い方を知ることが出来る。以下がこの関数の完全なコードである。

```
(defun what-line ()
  "Print the current line number (in the buffer) of point."
  (interactive)
  (save-restriction
    (widen)
    (save-excursion
      (beginning-of-line)
      (message "Line %d"
               (1+ (count-lines 1 (point)))))))
```

(未訳) (In recent versions of GNU Emacs, the what-line function has been expanded to tell you your line number in a narrowed buffer as well as your line number in a widened buffer. The recent version is more complex than the version shown here. If you feel adventurous, you might want to look at it after figuring out how this version works. You will probably need to use *C-h f* (describe-function). The newer version uses a conditional to determine whether the buffer has been narrowed.

(未訳) (Also, it uses line-number-at-pos, which among other simple expressions, such as (goto-char (point-min)), moves point to the beginning of the current line with (forward-line 0) rather than beginning-of-line.)

関数 what-line は一行の説明文字列を持ち、あなたが予想した通り、インタラクティブな関数である。次の二つの行では save-restriction と widen を使っている。

特殊形式 save-restriction は、カレントバッファにどのようなナローイングがかかっていようと、それを記録して本体のコードが評価し終わった後にその状態に戻してくれる。

特殊形式 save-restriction の後には widen が続いている。この関数は、what-line が呼ばれた時にカレントバッファのナローイングを解除する。(ここでのナローイングは、save-restriction が憶えているナローイングである。) これによって行数を数えるコマンドは、バッファの最初からの行数を数えることが出来るのである。そうしない場合は、現在アクセス可能なリージョンの範囲内でしか行数をカウントしない。元のナローイングは特殊形式 save-restriction によってこの関数が終了する直前に元の状態に戻される。

widen の呼び出しに続いて save-excursion が使われている。これはカーソルの位置、(即ち、point の位置) とマークの位置を保存し、本体部分のコードの中の beginning-of-line 関数によって動かされたポイントの位置を元に戻す、という働きをしている。

((widen) 式が save-restriction と save-excursion の間にあることに注意しよう。これら二つの save- ... 式を続けて使う場合は save-excursion を外側に書かないといけないのであった。)

what-line 関数の最後の二つの行は、バッファ内の行数を数え、それをエコー領域に表示するものである。

```
(message "Line %d"
         (1+ (count-lines 1 (point)))))))
```

message 関数は Emacs の画面の最下行に一行メッセージを表示する関数である。最初の引数は二重引用符に挟まれていて、これは文字文字列を表示するものである。ただし、'%d', '%s', '%c' も入れることが出来て、これらはこの文字列の後に続く引数を表示する。'%d' は引数を 10 進数として表示するので、メッセージは例えば 'Line 243' と言った感じになる。

‘%d’ の場所に表示される数字は、この関数の最後の行で計算される。

```
(1+ (count-lines 1 (point)))
```

これがやっていることは 1 で示された位置、つまりバッファの最初の位置から (point) までの行数を数え、それに 1 を加えることである。(1+ という関数は引数に 1 を加える関数である。)1 を加えるのは、例えば二行目はその前に一行しかないからである。count-lines はカレント行の前の行までしか数えないので、こうしないと一行ずれてしまうのだ。

count-lines が仕事を終えてメッセージがエコー領域に表示された後は save-excursion によってポイントとマークの位置が元の状態に戻される。そして、save-restriction がナローイングの状態を元に戻す。

6.3 ナローイングの練習問題

カレントバッファの最初の 60 文字を表示するような関数を書きなさい。その際、例えばナローイングにより後半部分しかアクセス不能であり、最初の行が見えない状態であってもきちんと表示するようなものにしなさい。また、ポイント、マーク、及びナローイングを復元するようにすること。この練習問題のためには、save-restriction、widen、goto-char、point-min、buffer-substring、message 及び他の関数をうまく組み合わせて使う必要がある。

(未訳) (buffer-substring is a previously unmentioned function you will have to investigate yourself; or perhaps you will have to use buffer-substring-no-properties or filter-buffer-substring . . ., yet other functions. Text properties are a feature otherwise not discussed here. See Section “Text Properties” in *The GNU Emacs Lisp Reference Manual*.)

(未訳) Additionally, do you really need goto-char or point-min? Or can you write the function without them?

7 car, cdr, cons : 基本関数

Lisp では car、cdr、そして cons が基本的な関数である。cons 関数はリストを作るのに使われ、car 関数と cdr 関数はそれらを分解するのに使われる。

copy-region-as-kill 関数を “walk-through” する時に、cons と共に cdr の二つの変種である setcdr と nthcdr についても見るつもりだ。(Section 8.3 “copy-region-as-kill”, page 67).

cons 関数の名前は特に非合理的なものではない。この名前は ‘construct’ という単語を略したものである。一方、car と cdr の名前の由来は難解である。car は ‘Contents of the Address part of the Register’ というフレーズの頭文字から来ており、cdr (‘could-er’ と発音する) は ‘Contents of the Decrement part of the Register’ というフレーズから来ている。これらのフレーズは、Lisp が開発された頃の極めて初期のハードウェアの特定の部分に基づくものであるが、単に時代遅れであるというだけでなく、実に 25 年以上もの間、Lisp に関わる人々にとって全く見当はずれのものであった。だがしかし、これらの関数をもっと合理的な名前で呼ぼうとした学者も何人かいたにも関わらず、現在でもこの古い用語が使われている。特に Emacs Lisp のソースコードでもこの用語が使われているので、この入門書でもこれらを使うことにしよう。

7.1 car と cdr

リストの car とは、単にそのリストの最初の要素のことである。従って、リスト (rose violet daisy buttercup) の car は rose である。

もし、この文章を GNU Emacs の Info で読んでいるなら、次を評価してみることでこのことを確認出来る。

```
(car '(rose violet daisy buttercup))
```

この S 式を評価すると、エコー領域に rose と表示されたはずだ。

明らかに car には first という名前の方がふさわしいし、よくそのように提案されてもいる。

car は最初の要素を除いたりはしない。ただ単にそれが何かを伝えるだけである。あるリストに car を施した後も、そのリストは元のままである。専門用語では、car は「非破壊的 (non-destructive)」であると言ったりする。この特徴は、後で重要であることが分るだろう。

List の cdr はリストの残りの部分である。つまり、cdr 関数はリストの最初の要素を除いた部分を返す。従って、'(rose violet daisy buttercup) というリストの car が rose であるように、このリストの残りの部分、つまり cdr を施して返される値は (violet daisy buttercup) である。

このことも、次をいつも通り評価してみることで確かめられる。

```
(cdr '(rose violet daisy buttercup))
```

これを評価すると、エコー領域に (violet daisy buttercup) と表示されたはずだ。

car と同様、cdr もリストから要素を取り除いたりしない—ただ単に二番目以降の要素のリストを返すだけである。

ついでに言っておくと、上の例では花のリストに引用符が付いている。もしこれが付いていなければ Lisp インタプリタは rose を関数として呼び出して評価しようとするだろう。この例では、そういうことをしたいのではなかった。

明らかに cdr には rest という名前の方が、ふさわしい。

(ここでちょっと教訓。あなたが新しい関数に名前をつける場合、それについて極めて注意深く考えないといけない。というのも、あなたは多分、あなたが思うよりずっと長い間、その名前につきまとわれることになるからだ。この文章でこれらの名前を使っているのは、Emacs Lisp のソースコードでこれらを使っているからであり、そうしないとコードを読むのがつらいだろうからである。しかし、どうかあなた自身はなるべくこういった言葉使いを避けて欲しい。そうすれば後からやってきた人々から感謝されるだろう。)

car や cdr を (pine fir oak maple) といったシンボルからなるリストに施した時、car によって返される要素はシンボル pine であり、括弧は付いてはいない。pine はこのリストの最初の要素である。しかしながら、リストの cdr は (fir oak maple) というリストである。それは、次の式をいつものように評価してみればすぐに分る。

```
(car '(pine fir oak maple))

(cdr '(pine fir oak maple))
```

一方、リストのリストでは、そのリストの最初の要素がそれ自身リストである。この場合、car は最初のリストとしての要素を返す。例えば次のリストは三つのサブリストを含んでいる。各々、肉食動物、草食動物、海洋哺乳類のリストである。

```
(car '((lion tiger cheetah)
       (gazelle antelope zebra)
       (whale dolphin seal)))
```

この場合、最初の要素、つまり全体のリストの car は肉食動物のリスト (lion tiger cheetah) である。そしてリストの残りは ((gazelle antelope zebra) (whale dolphin seal)) である。

```
(cdr '((lion tiger cheetah)
       (gazelle antelope zebra)
       (whale dolphin seal)))
```

ここでもう一度、car と cdr が非破壊的であると言っておいたほうがいいだろう。これらの関数はリストを変化させないのだ。このことはこれらの関数を使うにあたって極めて重要になる。

また最初の章でアトムについて議論した時に、Lisp では「ある種のアトム、例えば配列 (array)、は複数の部分に分解出来る。しかし、そのメカニズムはリストを分解する時のメカニズムとは違う。リストの操作に関する限り、リストのアトムは分解不可能である。」と書いた。(Section 1.1.1 "Lisp Atoms", page 1). car 関数と cdr 関数はリストを分解するのに使われ、Lisp の基本と考えられている。しかし、これらの関数は配列を分解したり、部分を取り出したりするのには使えない。従って、配列はリストではなくアトムであると考えられるわけである。また、もう一つの基本的な関数である cons も、リストを構築することは出来るが、配列を作ることは出来ない。(配列は、配列を扱う特殊な関数を使って操作する。詳しくは Section "Arrays" in *The GNU Emacs Lisp Reference Manual*, を参照のこと。)

7.2 cons

cons は、リストを作る関数である。car や cdr とは逆の働きをするものだと言える。例えば、cons は三つの要素を持つリスト (fir oak maple) から四つの要素を持つリストを作ることが出来る。つまり、

```
(cons 'pine '(fir oak maple))
```

を評価すると、

```
(pine fir oak maple)
```

というリストがエコー領域に表示される。このように、cons はリストの先頭に新しい一つの要素を付け加える。リストに要素を追加するわけである。

(未訳) We often say that 'cons puts a new element at the beginning of a list; it attaches or pushes elements onto the list', but this phrasing can be misleading, since cons does not change an existing list, but creates a new one.

(未訳) Like car and cdr, cons is non-destructive.

cons には、まず要素を付け加えるべきリストが必要になる。[1] 何も無いところからスタートすることは出来ないのである。もし、リストを作りたいならば、少くともまず初めに空リストを用意する必要がある。次に挙げるのは、cons を連続して使って花のリストを作るものである。もし、この文章を Emacs の Info で読んでいるのなら、各々の S 式をいつものように評価することが出来る。表示される値が '⇒' の後に書かれている。これを「次のように評価される」と置き換えて読んで欲しい。

```
(cons 'buttercup ())
     ⇒ (buttercup)

(cons 'daisy '(buttercup))
     ⇒ (daisy buttercup)

(cons 'violet '(daisy buttercup))
     ⇒ (violet daisy buttercup)

(cons 'rose '(violet daisy buttercup))
     ⇒ (rose violet daisy buttercup)
```

[1] 実際は、アトムに要素を cons してドット対を作ることも出来る。が、ドット対のことはここでは説明しない。これについては次を参照せよ。Section "Dotted Pair Notation" in *The GNU Emacs Lisp Reference Manual*.

最初の例では、空リストが () という形で出ている。そして、buttercup とこの空リストから新しいリストが作られる。見ればわかると思うが、空リストは作成されたリストの要素としては出て来ない。(buttercup) というリストが見えるだけである。空リストは要素としては数えられない。というのも、空リストには要素が一つもないからである。一般的にいうと、空リストは見えないのである。

二番目の例では (cons 'daisy '(buttercup)) を評価することで daisy を buttercup の前に付け加えて、二つの要素を持つリストを作っている。三番目の例では daisy と buttercup の前に violet を加えて三つの要素を持つリストを作っている。

7.2.1 リストの長さを調べる: length

あるリストがどれだけ多くの要素を持つかを見るには、length という Lisp の関数を使えばよい。次のように感じである。

```
(length '(buttercup))
     ⇒ 1

(length '(daisy buttercup))
     ⇒ 2

(length (cons 'violet '(daisy buttercup)))
     ⇒ 3
```

三番目の例では、三つの要素を持つリストを作るために cons 関数が使われ、その結果出来たリストが length 関数に引数として渡されている。

length 関数を使って、空リストの中の要素の数を数えることも出来る。

```
(length ())
     ⇒ 0
```

予想した通りだと思うが、空リストの中の要素の数は零個と数えられる。

リストが無い場合にその長さを求めさせてみるのも面白い実験である。つまり、引数として空リストすら渡さずに length を呼び出すのである。

```
(length )
```

やってみれば分るが、これを評価すると、エラーメッセージが表示される。

```
Lisp error: (wrong-number-of-arguments length 0)
```

(旧訳) これは引数の数が間違っているという意味のメッセージである。今の場合なら、本当は一つの引数が渡されなければならない。例えば引数がリストなら、そのリストの長さをこの関数が測ることになるはずだったのである。(リストが沢山の要素を持っていたとしても、一つのリストは一つの引数であることに注意。)

(未訳) The part of the error message that says '`length`' is the name of the function.

7.3 nthcdr

nthcdr 関数は、cdr 関数に関連した関数である。これは、リストの cdr を繰り返し取るものである。

(pine fir oak maple) の cdr を取ると、(fir oak maple) が返される。もしこの返された値にもう一度同じ操作をすると、(oak maple) が返される。(勿論、リストに何回 cdr を施しても、元のリストは変化しない。cdr は対象リストを変化させはしないからである。従って、cdr の cdr を評価する必要がある。) これを続ければ、結果として空リストが返る。この場合、() ではなく nil が表示される。

復習のために、以下に cdr の繰り返しの例を書いておく。'⇒' の後に表示される結果が書いてある。

```
(cdr '(pine fir oak maple))
     ⇒(fir oak maple)

(cdr '(fir oak maple))
     ⇒ (oak maple)

(cdr '(oak maple))
     ⇒(maple)

(cdr '(maple))
     ⇒ nil
```

```
(cdr 'nil)
     ⇒ nil

(cdr ())
     ⇒ nil
```

複数の cdr を、間に表示される値を挟まずに続けて書くことも出来る。

```
(cdr (cdr '(pine fir oak maple)))
     ⇒ (oak maple)
```

この場合は、Lisp インタプリタは最も内側のリストを先に評価する。最も内側にあるリストには引用符が付いているので、それはそのままリストとして返され、内側の方の cdr に渡される。この内側の cdr はこのリストの二番目以降の要素からなるリストを返し、外側の cdr に渡す。これは、元のリストの三番目以降の要素からなるリストを返すというわけである。この例のように、cdr 関数を二回続けて使うと、元のリストの最初の二つの要素を除いたリストが返されることになる。

nthcdr 関数は、cdr 関数を繰り返して使うのと同じことをしてくれる。次の例では引数 2 がリストと一緒に nthcdr 関数に渡され、その結果最初の二つの要素を除いたリストが返されている。これはまさに cdr を二回繰り返してリストに施したのと同じ結果である。

```
(nthcdr 2 '(pine fir oak maple))
     ⇒ (oak maple)
```

元の 4 つの要素を持ったリストを使って、0、1、5 などを含むいろいろな引数を nthcdr に与えた場合に何が起こるかを見てみよう。

```
;; リストはそのまま。
(nthcdr 0 '(pine fir oak maple))
     ⇒ (pine fir oak maple)

;; 最初の要素を除いた残りのコピーを返す。
(nthcdr 1 '(pine fir oak maple))
     ⇒ (fir oak maple)

;; 三つの要素を除いた残りのコピーを返す。
(nthcdr 3 '(pine fir oak maple))
     ⇒ (maple)

;; 四つ全部を除いた残りを返す。
(nthcdr 4 '(pine fir oak maple))
     ⇒ nil

;; これも全部を除いた残りを返す。
(nthcdr 5 '(pine fir oak maple))
     ⇒ nil
```

7.4 nth

nthcdr も cdr と同様に元のリストを変化させないことに一言触れておくほうがいいだろう。この関数もまた非破壊的なのである。これは setcar 関数や setcdr 関数とはっきり違っているところである。

（未訳）Thus, if it were not defined in C for speed, the definition of nth would be:

```
(defun nth (n list)
  "Returns the Nth element of LIST.
N counts from zero.  If LIST is not that long, nil is returned."
  (car (nthcdr n list)))
```

（未訳）(Originally, nth was defined in Emacs Lisp in subr.el, but its definition was redone in C in the 1980s.)

（未訳）The nth function returns a single element of a list. This can be very convenient.

（未訳）Note that the elements are numbered from zero, not one. That is to say, the first element of a list, its CAR is the zeroth element. This is called ‘zero-based’ counting and often bothers people who are accustomed to the first element in a list being number one, which is ‘one-based’.

For example:

```
(nth 0 '("one" "two" "three"))
    ⇒ "one"

(nth 1 '("one" "two" "three"))
    ⇒ "two"
```

(未訳) It is worth mentioning that nth, like nthcdr and cdr, does not change the original list—the function is non-destructive. This is in sharp contrast to the setcar and setcdr functions.

7.5 setcar

この二つの名前から連想されるように、setcar と setcdr という関数は、リストの car や cdr に新しい値を持たせる。これらは car や cdr が値を変化させないのとは異なり、実際に元のリストの値を変化させてしまう。これがどういうことなのかを見る一つの方法は、実験してみることだろう。まずは setcar 関数から始めてみよう。

最初にリストを作り、setq を使ってある変数の値にそのリストをセットする。動物のリストでやってみよう。

```
(setq animals '(antelope giraffe lion tiger))
```

GNU Emacs の Info でこの文章を読んでいるなら、このS式をいつものように評価してみよう。カーソルをS式の後に持っていき、*C-x C-e* とタイプするのである。(私はこう書く時には実際にこの操作を試している。これは計算機環境にインタプリタが組み込まれていることの利点の一つである。Incidentally, when there is nothing on the line after the final parentheses, such as a comment, point can be on the next line. Thus, if your cursor is in the first column of the next line, you do not need to move it. Indeed, Emacs permits any amount of white space after the final parenthesis.)

変数 animals を評価してみれば、この変数が (antelope giraffe lion tiger) というリストにバインドされていることが確かめられる。

```
animals
     ⇒ (giraffe antelope tiger lion)
```

別の言い方をすれば、変数 animals はリスト (giraffe antelope tiger lion) を指しているとも言える。

次に、関数 setcar に変数 animals と引用符付きのシンボル hippopotamus の二つ引数を渡して評価してみよう。これは三つの要素を持つ (setcar animals 'hippopotamus) というリストを書いて、これをいつものように評価することで行うことが出来る。

```
(setcar animals 'hippopotamus)
```

このS式を評価した後で、変数 animals をもう一度評価してみよう。すると、動物のリストが変化していることが分る。

```
animals
     ⇒ (hippopotamus giraffe lion tiger)
```

リストの最初の要素 antelope が hippopotamus に変わっている。

以上のことから分るように、setcar は cons のように新しい要素を付け加えるのではなく、giraffe を hippopotamus に置き換えている。即ち、この関数はリストの最初の要素を変更するものなのだ。

7.6 setcdr

setcdr 関数も、setcar 関数と似ているのだが、こちらはリストの最初の要素ではなく、二番目以降の要素を置き換えるものである。

(To see how to change the last element of a list, look ahead to "The kill-new function", page 70, which uses the nthcdr and setcdr functions.)

実際どのように働くかを見るために、まず次のS式を評価して、変数の値を家畜のリストにセットしておく。

```
(setq domesticated-animals '(horse cow sheep goat))
```

この時点でこの変数を評価したなら、リスト (horse cow sheep goat) が返されるはずだ。

```
domesticated-animals
     ⇒ (horse cow sheep goat)
```

次に、setcdr に、このリストを値に持つ変数名とそのリストの cdr の部分を置き換えるリストの二つの引数を与えて、評価してみる。

```
(setcdr domesticated-animals '(cat dog))
```

このS式を評価すると、リスト (cat dog) がエコー領域に表示される。これがこの関数によって返される値である。我々にとって興味があるのは副作用の方であるが、こちらは変数 domesticated-animal を評価してみればどんなものか理解出来る。

```
domesticated-animals
     ⇒ (horse cat dog)
```

このリストが (horse cow sheep goat) から (horse cat dog) に変化しているのが分る。リストの cdr が (cow sheep goat) から (cat dog) に置き換わっているのである。

7.7 練習問題

四種類の鳥のリストを幾つかの cons を使ったS式を評価することで作りなさい。また、あるリストにそれ自身を cons した時に、何が起きるかを見なさい。四種類の鳥のリストの最初の要素を魚で置き換えなさい。更に、そのリストの残りの要素も他の魚で置き換えなさい。

8 テキストの切り取りと保存

GNU Emacs で 'kill' コマンドを使ってバッファからテキストをカットないしは切り取った (clip した) 場合、常にその情報はリストとして保持されており、それをヤンク (yank) コマンドで取り出すことが出来る。

(Emacs で特に何らかの実体を破壊しないプロセスに対して 'kill' という言葉を使うのは、不幸な歴史的出来事である。これには 'clip' という単語の方がずっとふさわしい。これこそ kill コマンドがやっていることに他ならないからである。このコマンドはバッファからテキストを 'clip' し、それをまた取り出すことが出来るような置場所に移しているのだ。私はしばしば Emacs のソースの中に出てくるあらゆる 'kill' を 'clip' に、あらゆる 'killed' を 'clipped' に置き換えたい衝動にかられる。)

テキストがバッファから切り取られた場合、そのテキストはあるリストに保存される。次にまたテキストを切り取りると、そのテキストもまたこのリストの中に続けて保存される。従ってリストは次のようになる。

```
("a piece of text" "last piece")
```

関数 `cons` を使えば、以下のように、このリストにテキストを加えることが出来る。

```
(cons "another piece"
      '("a piece of text" "previous piece"))
```

このS式を評価すると、次の三つの要素を持つリストがエコー領域に表示されるはずだ。

```
("another piece" "a piece of text" "last piece")
```

関数 `car` や `nthcdr` を使うと、この中からどの部分でも取り出すことが出来る。例えば次に挙げるコードでは `nthcdr 1 ...` が最初の要素を除いたリストを返し、`car` がその残りのリストの中の最初の要素—元々のリストでは二番目の要素—を返す。

```
(car (nthcdr 1 '("another piece"
                 "a piece of text"
                 "previous piece")))
     ⇒ "a piece of text"
```

勿論、実際に Emacs で使われている関数はこれよりももっと複雑である。テキストをカットしたり取り出したりするためのコードは Emacs があなたが取り出したいテキストがどれか分るように書かれていなければならない。それが一番目にあろうと二番目にあろうと三番目にあろうと、あるいはもっと他の場所にあってもである。また、もしリストの最後まで来た場合、そこで終わりとはせずに再度最初の要素に戻らなくてはなるまい。

テキストの部分を保存しているリストは *kill* リング (*kill ring*) と呼ばれる。この章では少しずつ kill リングについて話していくことにする。そのために、まず `zap-to-char` 関数がどのように働くかを見ることで、kill リングの使い方を見ていきたいと思う。山の頂上を目指す前にまず手頃な丘に登ってみるわけである。

実際にテキストがどのようにバッファから切り取られ、あるいは取り出されるのかについては、次の章で説明する。Chapter 10 “Yanking Text Back”, page 81, 参照。

8.1 zap-to-char

Let us look at the interactive `zap-to-char` function.

インタラクティブ関数 `zap-to-char` は、カーソルの位置 (つまりポイントの位置) から次に指定した文字が現れるまでの、その文字も含めた領域のテキストを取り去る。`zap-to-char` によって取り除かれたテキストは、kill リングに置かれる。これは `C-y` (`yank`) とタイプすることで取り出すことが出来る。もし、このコマンドが引数と共に呼び出されたなら、その回数だけ指定された文字が出てくるまでの間のテキストを取り去る。従って、もしカーソルが次の文の先頭にあり、特定の文字が 's' だったとすると、'Thus' が取り去られることになる。

Thus, if the cursor were at the beginning of this sentence and the character were 's', 'Thus' would be removed.

(訳註：日本語の文章では `zap-to-char` は使えないので、止むなく英文のままにした。)

またもし引数が 2 であれば、'`Thus, if the curs`' が取り去られる。ここでも最後の '`s`' が含まれていることに注目しよう。

もし指定した文字が現れなかった場合には単にエラーを返すだけである。(従って、テキストを取り除いたりもしない。)

どれだけのテキストを取り去るかを決定するために、どちらのバージョンの場合でも検索関数を使っている。検索はもっぱらテキストを扱うコードの中で使われる。そして、削除コマンドだけでなく検索関数にも注目するのは意味のあることである。

Here is the complete text of the version 22 implementation of the function:

```
(defun zap-to-char (arg char)
  "Kill up to and including ARG'th occurrence of CHAR.
Case is ignored if `case-fold-search' is non-nil in the current buffer.
Goes backward if ARG is negative; error if CHAR not found."
  (interactive "p\ncZap to char: ")
  (if (char-table-p translation-table-for-input)
      (setq char (or (aref translation-table-for-input char) char)))
  (kill-region (point) (progn
                         (search-forward (char-to-string char)
                                         nil nil arg)
                         (point))))
```

The documentation is thorough. You do need to know the jargon meaning of the word 'kill'.

8.1.1 `interactive` 式

`zap-to-char` コマンド内のインタラクティブ式は、次の通りである。

```
(interactive "p\ncZap to char: ")
```

(旧訳) 引用符の中の部分は、`"p\ncZap to char: "` であり、二つのことを指定している。一番目の部分は単なるアスタリスク '`*`' で、バッファが read-only だったら警告のためにエラーを出すためのものである。つまり、read-only なバッファで `zap-to-char` を使おうとしてもテキストを取り除くことは出来ず、代わりに "buffer is read-only" というメッセージを受け取ることになることを意味する。端末によってはビープ音もなる。

(旧訳) `p\ncZap to char: "` の二番目の部分は '`p`' である。この部分は改行 '`\n`' で終っている。'`p`' はこの関数の最初の引数として、処理された前置引数の値が渡されることを意味する。前置引数は *`C-u`* に続いてある数を、もしくは *`M-`* に続いてある数をタイプすることで渡される。もし、この関数が前置引数なしで呼び出された場合、1 が引数として渡される。

(旧訳) `*p\ncZap to char: "` の三番目の部分は、'`cZap to char: `' である。この部分の最初の小文字の '`c`' は `interactive` がプロンプトを出して、文字の入力を要求することを指示するものである。プロンプトは '`c`' の後に続くもので、この場合は文字列 '`Zap to char: `' である。(コロンの後の空白は、単に見栄えをよくするためのものである。)

結局これがすることは、正しい型の `zap-to-char` の引数を用意し、ユーザーに対してプロンプトを出すということである。

(未訳) In a read-only buffer, the `zap-to-char` function copies the text to the kill ring, but does not remove it. The echo area displays a message saying that the buffer is read-only. Also, the terminal may beep or blink at you.

8.1.2 `zap-to-char` の本体部分

`zap-to-char` 関数の本体部分は、現在のカーソルの位置から指定した文字までの間の領域のテキストを削除するコードを含んでいる。

その最初の部分は次の通りである。

```
(if (char-table-p translation-table-for-input)
    (setq char (or (aref translation-table-for-input char) char)))
(kill-region (point) (progn
                       (search-forward (char-to-string char) nil nil arg)
                       (point)))
```

(未訳) `char-table-p` is an hitherto unseen function. It determines whether its argument is a character table. When it is, it sets the character passed to `zap-to-char` to one of them, if that character exists, or to the character itself. (This becomes important for certain characters in non-European languages. The `aref` function extracts an element from an array. It is an array-specific function that is not described in this document. See Section "Arrays" in *The GNU Emacs Lisp Reference Manual*.)

`(point)` は現在のカーソルの位置である。

この後には、progn を使った式が続いている。progn の本体部分では、search-forward と point を呼び出している。

まず search-forward を学んでからの方が progn の働きを理解しやすいので、先にこちらを見て、次に progn について述べることにしよう。

8.1.3 関数 search-forward

search-forward 関数は、zap-to-char の中で、削除する範囲の端を定める文字 (以下 zapped-to-character と書く) の位置を定めるのに使われている。もし検索が成功すると、search-forward は目的の文字列の最後の文字のすぐ後にポイントを移動する。(今の場合、目的の文字列はちょうど一文字である。) もし検索が後方に向かうものであれば、search-forward は目的の文字列の最初の一文字のすぐ前にポイントを移動する。また、search-forward 自身は値として t を返す。(従って、ポイントの移動は副作用である。)

zap-to-char の中で search-forward は次のように使われている。

```
(search-forward (char-to-string char) nil nil arg)
```

関数 search-forward は四つの引数を取る。

1. 最初の引数は、検索しようとするターゲットである。これは‘"z"’のように文字列でなければならない。

   今回の場合は、たまたま zap-to-char に渡される引数は単独の文字である。計算機の設計上、Lisp インタプリタは単独の文字を一文字からなる文字列とは区別して扱う。計算機の中では単独の文字は一文字からなる文字列とは別の電気的なフォーマットを持っているのである。(単独の文字はたいてい計算機の中で 1 バイトで記録されている。(訳註：勿論日本語とかでは事情は別である。) しかし、文字列の方は長いものもあれば短いものもあるので、計算機の方でそれに備えなければならないのである。) search-forward は文字列を検索するものなので、zap-to-char 関数に引数として渡された文字を計算機の内部で別のフォーマットに変換しないといけない。でないと、search-forward は検索に失敗してしまう。この変換をするために、char-to-string という関数が使われている。
2. 二番目は、検索の範囲を指定するためのものである。これはバッファ内の位置として指定される。今の場合、検索はバッファの最後まで行うので、範囲を制限したりはしない。従って、二番目の引数は nil である。
3. 三番目は、この関数に検索が失敗し場合にどうするか—警告してエラーを表示するか、単に nil を返すか—を伝えるものである。 三番目の引数として nil が指定されていると、検索に失敗した場合にはエラーを出して警告する。
4. search-forward の四番目の引数は、繰り返しの回数—つまり検索対象の文字列が何回現れるのを調べるか—を指定するものである。この引数は省略可能であり、もしこの引数なしで関数が呼び出された場合、この引数として 1 が渡される。引数が負であれば、検索は後方に向かってなされる。

テンプレートで書くと、search-forward 式は次のようになる。

```
(search-forward "目的とする文字列"
                検索の範囲の限界
                検索に失敗した時の動作
                繰り返しの階数)
```

次に progn を見ることにしよう。

8.1.4 関数 progn

progn は引数を各々順番に評価していき、最後の式の値を返すような関数である。それ以前の S 式はその副作用のためだけに評価され、これらが返す値は無視される。

progn 式のテンプレートは極めて単純である。

```
(progn
  本体...)
```

zap-to-char の中では、progn 式は二つのことを行う。まず、ポイントをちょうど正しい位置に置くこと。次に、ポイントの位置を返して、kill-region がどこまでテキストを削除すればよいかを教えることである。

progn の最初の引数は search-forward である。search-forward は検索する文字列を見つけた場合、その文字列の最後の文字のすぐ後にポイントを移す。(今の場合、検索文字列は一文字である。) ただし、もし検索が後方方向なら最初の一文字の直前の位置にポイントを移す。ポイントの移動は副作用である。

二番目の、そして最後の progn の引数は (point) である。この S 式はポイントの位置を返す。今の場合なら、これは search-forward によって移動されたポイントの位置である。この値が progn 式の値として返され、kill-region に kill-region の二番目の引数として渡される。

8.1.5 zap-to-char についての総括

さて、以上で search-forward と progn の働きを見てきた。ここで zap-to-char 全体がどのように動作するかを見ることにしよう。

zap-to-char コマンドが与えられた時の kill-region の最初の引数はカーソルの位置—その時点でのポイントの値—である。また progn の中で検索コマンドによってポイントが zapped-to-character のすぐ後に移され、point によってその位置の値が返される。kill-region 関数はこれら二つの値の最初の値をリージョンの始まりに、二番目の値をリージョンの終わりに指定して、そのリージョンを削除する。

progn 関数が必要なのは、kill-region 関数の引数が二つであるからである。仮に search-forward 式と point 式が続けて二つの引数として書かれていたとしたら、うまく動作してはくれない。progn は kill-region にとって一つの引数であり、kill-region が第二引数として必要な一つの値を返すのである。

8.2 kill-region

zap-to-char は kill-region 関数を使っている。This function clips text from a region and copies that text to the kill ring, from which it may be retrieved.

(未訳) The Emacs 22 version of that function uses condition-case and copy-region-as-kill, both of which we will explain. condition-case is an important special form.

(未訳) In essence, the kill-region function calls condition-case, which takes three arguments. In this function, the first argument does nothing. The second argument contains the code that does the work when all goes well. The third argument contains the code that is called in the event of an error.

(未訳) We will go through the condition-case code in a moment. First, let us look at the definition of kill-region, with comments added:

```
(defun kill-region (beg end)
  "Kill (\"cut\") text between point and mark.
This deletes the text from the buffer and saves it in the kill ring.
The command \\[yank] can retrieve it from there. ... "

  ;; • Since order matters, pass point first.
  (interactive (list (point) (mark)))
  ;; • And tell us if we cannot cut the text.
  ;; 'unless' is an 'if' without a then-part.
  (unless (and beg end)
    (error "The mark is not set now, so there is no region"))

  ;; • 'condition-case' takes three arguments.
  ;;    If the first argument is nil, as it is here,
  ;;    information about the error signal is not
  ;;    stored for use by another function.
  (condition-case nil

      ;; • The second argument to 'condition-case' tells the
      ;;    Lisp interpreter what to do when all goes well.
```

```
    ;;    It starts with a ‘let’ function that extracts the string
    ;;    and tests whether it exists.  If so (that is what the
    ;;    ‘when’ checks), it calls an ‘if’ function that determines
    ;;    whether the previous command was another call to
    ;;    ‘kill-region’; if it was, then the new text is appended to
    ;;    the previous text; if not, then a different function,
    ;;    ‘kill-new’, is called.

    ;;    The ‘kill-append’ function concatenates the new string and
    ;;    the old.  The ‘kill-new’ function inserts text into a new
    ;;    item in the kill ring.

    ;;    ‘when’ is an ‘if’ without an else-part.  The second ‘when’
    ;;    again checks whether the current string exists; in
    ;;    addition, it checks whether the previous command was
    ;;    another call to ‘kill-region’.  If one or the other
    ;;    condition is true, then it sets the current command to
    ;;    be ‘kill-region’.
    (let ((string (filter-buffer-substring beg end t)))
      (when string                    ;STRING is nil if BEG = END
        ;; Add that string to the kill ring, one way or another.
        (if (eq last-command 'kill-region)
            ;;    — ‘yank-handler’ is an optional argument to
            ;;    ‘kill-region’ that tells the ‘kill-append’ and
            ;;    ‘kill-new’ functions how deal with properties
            ;;    added to the text, such as ‘bold’ or ‘italics’.
            (kill-append string (< end beg) yank-handler)
          (kill-new string nil yank-handler)))
      (when (or string (eq last-command 'kill-region))
        (setq this-command 'kill-region))
      nil)

;;  • The third argument to ‘condition-case’ tells the interpreter
;;    what to do with an error.
;;    The third argument has a conditions part and a body part.
;;    If the conditions are met (in this case,
;;              if text or buffer are read-only)
;;    then the body is executed.
;;    The first part of the third argument is the following:
((buffer-read-only text-read-only) ;; the if-part
 ;; ...  the then-part
 (copy-region-as-kill beg end)
 ;;    Next, also as part of the then-part, set this-command, so
 ;;    it will be set in an error
 (setq this-command 'kill-region)
 ;;    Finally, in the then-part, send a message if you may copy
 ;;    the text to the kill ring without signaling an error, but
 ;;    don't if you may not.
 (if kill-read-only-ok
     (progn (message "Read only text copied to kill ring") nil)
   (barf-if-buffer-read-only)
   ;; If the buffer isn't read-only, the text is.
   (signal 'text-read-only (list (current-buffer)))))
```

8.2.1 condition-case

As we have seen earlier (see Section 1.3 “Generate an Error Message”, page 3), when the Emacs Lisp interpreter has trouble evaluating an expression, it provides you with help; in the jargon, this is called “signaling an error”. Usually, the computer stops the program and shows you a message.

However, some programs undertake complicated actions. They should not simply stop on an error. In the `kill-region` function, the most likely error is that you will try to kill text that is read-only and cannot be removed. So the `kill-region` function contains code to handle this

circumstance. This code, which makes up the body of the `kill-region` function, is inside of a `condition-case` special form.

The template for `condition-case` looks like this:

```
(condition-case
  var
  bodyform
  error-handler...)
```

The second argument, *bodyform*, is straightforward. The `condition-case` special form causes the Lisp interpreter to evaluate the code in *bodyform*. If no error occurs, the special form returns the code's value and produces the side-effects, if any.

In short, the *bodyform* part of a `condition-case` expression determines what should happen when everything works correctly.

However, if an error occurs, among its other actions, the function generating the error signal will define one or more error condition names.

An error handler is the third argument to `condition case`. An error handler has two parts, a *condition-name* and a *body*. If the *condition-name* part of an error handler matches a condition name generated by an error, then the *body* part of the error handler is run.

As you will expect, the *condition-name* part of an error handler may be either a single condition name or a list of condition names.

Also, a complete `condition-case` expression may contain more than one error handler. When an error occurs, the first applicable handler is run.

Lastly, the first argument to the `condition-case` expression, the *var* argument, is sometimes bound to a variable that contains information about the error. However, if that argument is nil, as is the case in `kill-region`, that information is discarded.

In brief, in the `kill-region` function, the code `condition-case` works like this:

```
If no errors, run only this code
    but, if errors, run this other code.
```

8.2.2 Lisp macro

The part of the `condition-case` expression that is evaluated in the expectation that all goes well has a `when`. The code uses `when` to determine whether the `string` variable points to text that exists.

A `when` expression is simply a programmers' convenience. It is an `if` without the possibility of an else clause. In your mind, you can replace `when` with `if` and understand what goes on. That is what the Lisp interpreter does.

Technically speaking, `when` is a Lisp macro. A Lisp macro enables you to define new control constructs and other language features. It tells the interpreter how to compute another Lisp expression which will in turn compute the value. In this case, the 'other expression' is an `if` expression.

The `kill-region` function definition also has an `unless` macro; it is the converse of `when`. The `unless` macro is an `if` without a then clause

For more about Lisp macros, see Section "Macros" in *The GNU Emacs Lisp Reference Manual*. The C programming language also provides macros. These are different, but also useful.

Regarding the `when` macro, in the `condition-case` expression, when the string has content, then another conditional expression is executed. This is an `if` with both a then-part and an else-part.

```
(if (eq last-command 'kill-region)
    (kill-append string (< end beg) yank-handler)
  (kill-new string nil yank-handler))
```

The then-part is evaluated if the previous command was another call to `kill-region`; if not, the else-part is evaluated.

`yank-handler` is an optional argument to `kill-region` that tells the `kill-append` and `kill-new` functions how deal with properties added to the text, such as 'bold' or 'italics'.

last-command is a variable that comes with Emacs that we have not seen before. Normally, whenever a function is executed, Emacs sets the value of last-command to the previous command.

In this segment of the definition, the if expression checks whether the previous command was kill-region. If it was,

```
(kill-append string (< end beg) yank-handler)
```

concatenates a copy of the newly clipped text to the just previously clipped text in the kill ring.

8.3 copy-region-as-kill

The copy-region-as-kill function copies a region of text from a buffer and (via either kill-append or kill-new) saves it in the kill-ring.

(旧訳) copy-region-as-kill 関数はバッファからリージョンのテキストを複写して、kill-ring と呼ばれる変数の中に保存する。

もし、kill-region コマンドを前回の kill-region コマンドの直後に呼び出したとすると、Emacs は新しく複写したテキストを以前複写したテキストに付加 (append) する。これは、もしそのテキストをヤンクした場合、以前複写した分と今回複写した分を合わせた全体を戻すことになるということを意味する。一方、もし copy-region-as-kill に先立って他のコマンドを使用した場合には、複写したテキストは kill リングの新しい別の部分に、それまでに複写したものと区別されて保存される。

Here is the complete text of the version 22 copy-region-as-kill function:

```
(defun copy-region-as-kill (beg end)
  "Save the region as if killed, but don't kill it.
In Transient Mark mode, deactivate the mark.
If `interprogram-cut-function' is non-nil, also save the text for a window
system cut and paste."
  (interactive "r")
  (if (eq last-command 'kill-region)
      (kill-append (filter-buffer-substring beg end) (< end beg))
    (kill-new (filter-buffer-substring beg end)))
  (if transient-mark-mode
      (setq deactivate-mark t))
  nil)
```

これまで同様、この関数もまとまりのある幾つかの部分に分けることが出来る。

```
(defun copy-region-as-kill (argument-list)
  "documentation..."
  (interactive "r")
  body...)
```

The arguments are beg and end and the function is interactive with "r", so the two arguments must refer to the beginning and end of the region. If you have been reading through this document from the beginning, understanding these parts of a function is almost becoming routine.

(旧訳) 引数は beg と end であり、"r" によってインタラクティブ宣言がなされている。つまり、二つの引数は各々リージョンの始まりと終わりを示すものである。もしこの文書をきちんと最初から読んできたなら、この部分を理解するのは殆どルーティーンワークにすぎないだろう。

The documentation is somewhat confusing unless you remember that the word 'kill' has a meaning different from usual. The 'Transient Mark' and interprogram-cut-function comments explain certain side-effects.

After you once set a mark, a buffer always contains a region. If you wish, you can use Transient Mark mode to highlight the region temporarily. (No one wants to highlight the region all the time, so Transient Mark mode highlights it only at appropriate times. Many people turn off Transient Mark mode, so the region is never highlighted.)

Also, a windowing system allows you to copy, cut, and paste among different programs. In the X windowing system, for example, the interprogram-cut-function function is x-select-text, which works with the windowing system's equivalent of the Emacs kill ring.

The body of the copy-region-as-kill function starts with an if clause. What this clause does is distinguish between two different situations: whether or not this command is executed

immediately after a previous `kill-region` command. In the first case, the new region is appended to the previously copied text. Otherwise, it is inserted into the beginning of the kill ring as a separate piece of text from the previous piece.

（旧訳）この関数の本体は、`if` 式で始まっている。この式でやっているのは二つの状況を区別することである。つまり、このコマンドが `kill-region` コマンドが呼び出された直後に実行されたかどうかの判断である。もし直後に実行されたなら、新しく複写されるリージョンは以前複写された部分に付加される。そうでない場合は、kill リングの先頭に、以前複写されたものとは区別されて保存される。

The last two lines of the function prevent the region from lighting up if Transient Mark mode is turned on.

（旧訳）この関数定義の最後の二行は、二つの `setq` 式である。一つは変数 `this-command` を `kill-region` に、もう一つは変数 `kill-ring-yank-pointer` を kill リングに設定する。

この `copy-region-as-kill` の本体部分は詳しく見る価値がある。

8.3.1 `copy-region-as-kill` の本体部分

The `copy-region-as-kill` function works in much the same way as the `kill-region` function. Both are written so that two or more kills in a row combine their text into a single entry. If you yank back the text from the kill ring, you get it all in one piece. Moreover, kills that kill forward from the current position of the cursor are added to the end of the previously copied text and commands that copy text backwards add it to the beginning of the previously copied text. This way, the words in the text stay in the proper order.

（旧訳）`copy-region-as-kill` 関数は、二行以上の kill を一つの項目につなげるように書かれている。もし、それらを kill リングからヤンクして取り出すと、一箇所にまとめて取り出される。更に、現在のカーソルの位置から前方 (つまりテキストの終わりの方) に向かって kill すると、その部分は以前 kill したテキストの後に付け加えられ、後方 (つまりテキストの始まりの方) に向かって kill すると、以前 kill したテキストの前に付け加えられる。こうしてテキストの部分は正しい順序に保たれることになる。

Like `kill-region`, the `copy-region-as-kill` function makes use of the `last-command` variable that keeps track of the previous Emacs command.

（旧訳）この関数は、現在、及びその一つ前に Emacs が実行したコマンドを憶えておくために二つの変数 `this-command` と `last-command` を使っている。

Normally, whenever a function is executed, Emacs sets the value of `this-command` to the function being executed (which in this case would be `copy-region-as-kill`). At the same time, Emacs sets the value of `last-command` to the previous value of `this-command`.

（旧訳）通常、Emacs はどんな関数を実行している場合でも `this-command` の値を現在実行中のコマンドの値 (今の場合なら `copy-region-as-kill`) にセットしている。しかしながら、`copy-region-as-kill` の場合は違う。この関数は `this-command` の値を `kill-region` にセットする。これは `copy-region-as-kill` を呼び出している関数の名前である。

In the first part of the body of the `copy-region-as-kill` function, an `if` expression determines whether the value of `last-command` is `kill-region`. If so, the then-part of the `if` expression is evaluated; it uses the `kill-append` function to concatenate the text copied at this call to the function with the text already in the first element (the CAR) of the kill ring. On the other hand, if the value of `last-command` is not `kill-region`, then the `copy-region-as-kill` function attaches a new element to the kill ring using the `kill-new` function.

（旧訳）`copy-region-as-kill` の本体部分の最初のパートでは、`if` 式で `last-command` が `kill-region` かどうかを判定している。もしそうであれば、`kill-append` 関数を使って今回複写したテキストを kill リングの最初の要素 (つまり CAR) のテキストと結合する。一方、もしも `last-command` の値が `kill-region` で無ければ `copy-region-as-kill` 関数は kill リングに新しい要素を付け加えることになる。

The `if` expression reads as follows; it uses `eq`: `if` 式は次の通りである。これは `eq` を使っている。

```
(if (eq last-command 'kill-region)
    ;; then-part
    (kill-append  (filter-buffer-substring beg end) (< end beg))
  ;; else-part
  (kill-new  (filter-buffer-substring beg end)))
```

(未訳) (The `filter-buffer-substring` function returns a filtered substring of the buffer, if any. Optionally—the arguments are not here, so neither is done—the function may delete the initial text or return the text without its properties; this function is a replacement for the older `buffer-substring` function, which came before text properties were implemented.)

`eq` 関数は、最初の引数が二番目の引数と同じ Lisp オブジェクトかどうかを判定するものである。`eq` 関数は `equal` 関数と、等しいかどうかの判定をするという点では似ているが、異なる名前を持つものが、計算機の内部で実際に同じオブジェクトかどうかを判定するのかそうでないか、という点で異なっている。`eq` の方は同じオブジェクトの場合でないと真を返さないが、`equal` 関数の方は二つの式が同じ構造でありかつ同じ内容でありさえすれば真を返す。

(訳註：ここはちょっと分り難いかもしれない。詳しくは Section "Equality Predicates" in *The GNU Emacs Lisp Reference Manual*, を参照。なお、Chapter 9 "リストはどのように実装されているか", page 78, も参考になる。)

(未訳) If the previous command was `kill-region`, then the Emacs Lisp interpreter calls the `kill-append` function

関数 kill-append

`kill-append` 関数は次のようなものである。

```
(defun kill-append (string before-p &optional yank-handler)
  "Append STRING to the end of the latest kill in the kill ring.
If BEFORE-P is non-nil, prepend STRING to the kill.
... "
  (let* ((cur (car kill-ring)))
    (kill-new (if before-p (concat string cur) (concat cur string))
              (or (= (length cur) 0)
                  (equal yank-handler
                         (get-text-property 0 'yank-handler cur)))
              yank-handler)))
```

The `kill-append` function is fairly straightforward. It uses the `kill-new` function, which we will discuss in more detail in a moment.

(Also, the function provides an optional argument called `yank-handler`; when invoked, this argument tells the function how to deal with properties added to the text, such as 'bold' or 'italics'.)

It has a `let*` function to set the value of the first element of the kill ring to `cur`. (I do not know why the function does not use `let` instead; only one value is set in the expression. Perhaps this is a bug that produces no problems?)

Consider the conditional that is one of the two arguments to `kill-new`. It uses `concat` to concatenate the new text to the CAR of the kill ring. Whether it prepends or appends the text depends on the results of an `if` expression:

(旧訳) この関数をパートごとに見ていこう。`setcar` 関数は `concat` を使って新しいテキストを kill リングの CAR と結合している。新しい方のテキストを前に持ってくるか、後に持ってくるかは `if` 式の結果による。

```
(if before-p                            ; if-part
    (concat string cur)                 ; then-part
  (concat cur string))                  ; else-part
```

もし、今回切り取ったリージョンが直前のコマンドで切り取ったテキストよりも前にあれば、今回切り取ったものは以前保存したものの前につなげるべきであるし、逆に、今回切り取ったテキストが以前切り取ったテキストの後に続くものなら、以前保存したものの後につなげるべきである。`if` 式は今回保存するテキストを以前保存しておいたテキストの前につけるか後につけるかを `before-p` という述語で判断している。

シンボル before-p は kill-append の引数の名前の一つである。kill-append 関数が評価されると、これは実際の引数を評価して返された値にバインドされる。今の場合であれば、(< end beg) の値がバインドされる。このS式は、今回のコマンドで切り取られたテキストが直前のコマンドで切り取られたテキストの前に属しているか、それとも後に属しているかを、直接判定しているわけではない。この式がやっているのは変数 end の値が変数 beg の値よりも小さいかどうかの判定だけである。もし小さければ、それはユーザがおそらくはバッファの先頭に向かっているということを意味する。というわけで、この場合は以前のテキストの前に今回のテキストを追加するようになっている。一方、もし end の値の方が beg の値よりも大きかった場合、今回のテキストは以前のテキストの後尾に追加される。

新しく保存したテキストが以前保存したテキストの前に追加される場合は、その新しいテキストを前にして古いテキストと結合される。

```
(concat string cur)
```

逆に、テキストが後に追加される場合は、新しいテキストを古いテキストの後にして結合する。

```
(concat cur string))
```

この関数の動作を理解するためには、concat 関数の復習から始めなければならないだろう。concat 関数は二つのテキスト文字列を一つにつなげるものである。結果も文字列になる。例えば次のような感じである。

```
(concat "abc" "def")
     ⇒ "abcdef"

(concat "new "
        (car '("first element" "second element")))
     ⇒ "new first element"

(concat (car
        '("first element" "second element")) " modified")
     ⇒ "first element modified"
```

以上で kill-append 関数が理解出来るようになる。これは、kill リングの中身を修正するものである。kill リングはリストであり、各々の要素は保存されたテキストである。 kill-append 関数は kill-new 関数を使い、この関数は setcar 関数を使う。

The kill-new 関数

(未訳) The kill-new function looks like this:

```
(defun kill-new (string &optional replace yank-handler)
  "Make STRING the latest kill in the kill ring.
Set `kill-ring-yank-pointer' to point to it.

If `interprogram-cut-function' is non-nil, apply it to STRING.
Optional second argument REPLACE non-nil means that STRING will replace
the front of the kill ring, rather than being added to the list.
..."
  (if (> (length string) 0)
      (if yank-handler
          (put-text-property 0 (length string)
                             'yank-handler yank-handler string))
    (if yank-handler
        (signal 'args-out-of-range
                (list string "yank-handler specified for empty string"))))
  (if (fboundp 'menu-bar-update-yank-menu)
      (menu-bar-update-yank-menu string (and replace (car kill-ring))))
  (if (and replace kill-ring)
      (setcar kill-ring string)
    (push string kill-ring)
    (if (> (length kill-ring) kill-ring-max)
        (setcdr (nthcdr (1- kill-ring-max) kill-ring) nil)))
  (setq kill-ring-yank-pointer kill-ring)
  (if interprogram-cut-function
      (funcall interprogram-cut-function string (not replace))))
```

(Notice that the function is not interactive.)

As usual, we can look at this function in parts.

The function definition has an optional `yank-handler` argument, which when invoked tells the function how to deal with properties added to the text, such as 'bold' or 'italics'. We will skip that.

(未訳) The first line of the documentation makes sense:

```
Make STRING the latest kill in the kill ring.
```

Let's skip over the rest of the documentation for the moment.

Also, let's skip over the initial `if` expression and those lines of code involving `menu-bar-update-yank-menu`. We will explain them below.

The critical lines are these:

```
(if (and replace kill-ring)
    ;; then
    (setcar kill-ring string)
  ;; else
(push string kill-ring)
  (setq kill-ring (cons string kill-ring))
  (if (> (length kill-ring) kill-ring-max)
      ;; avoid overly long kill ring
      (setcdr (nthcdr (1- kill-ring-max) kill-ring) nil)))
(setq kill-ring-yank-pointer kill-ring)
(if interprogram-cut-function
    (funcall interprogram-cut-function string (not replace))))
```

The conditional test is `(and replace kill-ring)`. This will be true when two conditions are met: the kill ring has something in it, and the `replace` variable is true.

When the `kill-append` function sets `replace` to be true and when the kill ring has at least one item in it, the `setcar` expression is executed:

```
(setcar kill-ring string)
```

The `setcar` function actually changes the first element of the `kill-ring` list to the value of `string`. It replaces the first element.

On the other hand, if the kill ring is empty, or replace is false, the else-part of the condition is executed:

```
(push string kill-ring)
```

`push` puts its first argument onto the second. It is similar to the older

```
(setq kill-ring (cons string kill-ring))
```

or the newer

```
(add-to-list kill-ring string)
```

When it is false, the expression first constructs a new version of the kill ring by prepending `string` to the existing kill ring as a new element (that is what the `push` does). Then it executes a second `if` clause. This second `if` clause keeps the kill ring from growing too long.

Let's look at these two expressions in order.

The `push` line of the else-part sets the new value of the kill ring to what results from adding the string being killed to the old kill ring.

We can see how this works with an example.

First,

```
(setq example-list '("here is a clause" "another clause"))
```

After evaluating this expression with *C-x C-e*, you can evaluate `example-list` and see what it returns:

```
example-list
     ⇒ ("here is a clause" "another clause")
```

Now, we can add a new element on to this list by evaluating the following expression:

```
(push "a third clause" example-list)
```

When we evaluate `example-list`, we find its value is:

```
example-list
     ⇒ ("a third clause" "here is a clause" "another clause")
```

Thus, the third clause is added to the list by `push`.

Now for the second part of the `if` clause. This expression keeps the kill ring from growing too long. It looks like this:

```
(if (> (length kill-ring) kill-ring-max)
    (setcdr (nthcdr (1- kill-ring-max) kill-ring) nil))
```

The code checks whether the length of the kill ring is greater than the maximum permitted length. This is the value of `kill-ring-max` (which is 60, by default). If the length of the kill ring is too long, then this code sets the last element of the kill ring to `nil`. It does this by using two functions, `nthcdr` and `setcdr`.

We looked at `setcdr` earlier (see Section 7.6 “`setcdr`”, page 59). It sets the CDR of a list, just as `setcar` sets the CAR of a list. In this case, however, `setcdr` will not be setting the CDR of the whole kill ring; the `nthcdr` function is used to cause it to set the CDR of the next to last element of the kill ring—this means that since the CDR of the next to last element is the last element of the kill ring, it will set the last element of the kill ring.

The `nthcdr` function works by repeatedly taking the CDR of a list—it takes the CDR of the CDR of the CDR ... It does this N times and returns the results. (See Section 7.3 “`nthcdr`”, page 57.)

Thus, if we had a four element list that was supposed to be three elements long, we could set the CDR of the next to last element to `nil`, and thereby shorten the list. (If you set the last element to some other value than `nil`, which you could do, then you would not have shortened the list. See Section 7.6 “`setcdr`”, page 59.)

You can see shortening by evaluating the following three expressions in turn. First set the value of `trees` to `(maple oak pine birch)`, then set the CDR of its second CDR to `nil` and then find the value of `trees`:

```
(setq trees '(maple oak pine birch))
     ⇒ (maple oak pine birch)

(setcdr (nthcdr 2 trees) nil)
     ⇒ nil

trees
     ⇒ (maple oak pine)
```

(The value returned by the `setcdr` expression is `nil` since that is what the CDR is set to.)

To repeat, in `kill-new`, the `nthcdr` function takes the CDR a number of times that is one less than the maximum permitted size of the kill ring and `setcdr` sets the CDR of that element (which will be the rest of the elements in the kill ring) to `nil`. This prevents the kill ring from growing too long.

The next to last expression in the `kill-new` function is

```
(setq kill-ring-yank-pointer kill-ring)
```

The `kill-ring-yank-pointer` is a global variable that is set to be the `kill-ring`.

Even though the `kill-ring-yank-pointer` is called a ‘`pointer`’, it is a variable just like the kill ring. However, the name has been chosen to help humans understand how the variable is used.

Now, to return to an early expression in the body of the function:

```
(if (fboundp 'menu-bar-update-yank-menu)
     (menu-bar-update-yank-menu string (and replace (car kill-ring))))
```

It starts with an `if` expression

In this case, the expression tests first to see whether `menu-bar-update-yank-menu` exists as a function, and if so, calls it. The `fboundp` function returns true if the symbol it is testing

has a function definition that 'is not void'. If the symbol's function definition were void, we would receive an error message, as we did when we created errors intentionally (see Section 1.3 "Generate an Error Message", page 3).

The then-part contains an expression whose first element is the function `and`.

The `and` special form evaluates each of its arguments until one of the arguments returns a value of `nil`, in which case the `and` expression returns `nil`; however, if none of the arguments returns a value of `nil`, the value resulting from evaluating the last argument is returned. (Since such a value is not `nil`, it is considered true in Emacs Lisp.) In other words, an `and` expression returns a true value only if all its arguments are true. (See Section 5.4 "Second Buffer Related Review", page 50.)

The expression determines whether the second argument to `menu-bar-update-yank-menu` is true or not.

`menu-bar-update-yank-menu` is one of the functions that make it possible to use the 'Select and Paste' menu in the Edit item of a menu bar; using a mouse, you can look at the various pieces of text you have saved and select one piece to paste.

The last expression in the `kill-new` function adds the newly copied string to whatever facility exists for copying and pasting among different programs running in a windowing system. In the X Windowing system, for example, the `x-select-text` function takes the string and stores it in memory operated by X. You can paste the string in another program, such as an Xterm.

The expression looks like this:

```
(if interprogram-cut-function
    (funcall interprogram-cut-function string (not replace))))
```

If an `interprogram-cut-function` exists, then Emacs executes `funcall`, which in turn calls its first argument as a function and passes the remaining arguments to it. (Incidentally, as far as I can see, this `if` expression could be replaced by an `and` expression similar to the one in the first part of the function.)

We are not going to discuss windowing systems and other programs further, but merely note that this is a mechanism that enables GNU Emacs to work easily and well with other programs.

This code for placing text in the kill ring, either concatenated with an existing element or as a new element, leads us to the code for bringing back text that has been cut out of the buffer—the yank commands. However, before discussing the yank commands, it is better to learn how lists are implemented in a computer. This will make clear such mysteries as the use of the term 'pointer'. But before that, we will digress into C.

（旧訳）この辺りの話をするにはやはり、バッファから切り取ったテキストを取り戻すためのコマンド—ヤンクコマンドについて話さなければなるまい。が、その話をする前に、リストが計算機の内部でどのように扱われるかを説明しておいた方が良いだろう。そうすることで、「ポインタ」といったミステリアスな言葉を明確に理解出来るようになるだろう。

8.4 `delete-region`: ちょっと脱線して C の話を

`copy-region-as-kill` 関数 (see Section 8.3 "`copy-region-as-kill`", page 67) は `filter-buffer-substring` 関数を使っている。そして、これはこれでまた別の関数 `delete-and-extract-region` を利用している。It removes the contents of a region and you cannot get them back.

これまでに議論してきたコードとは異なり、`delete-and-extract-region` は Emacs Lisp では書かれていない。これは C で書かれており、GNU Emacs system のプリミティブの一つである。これは大変単純なものなので、ちょっとの間 Lisp から脱線して、これについて説明することにしよう。

他の多くの Emacs のプリミティブと同様、`delete-region-extract-region` も C のマクロのインスタンス、つまりコードのテンプレートとしてのマクロとして書かれている。このマクロの最初の部分は次のようである。

```
DEFUN ("delete-and-extract-region", Fdelete_and_extract_region,
       Sdelete_and_extract_region, 2, 2, 0,
       doc: /* Delete the text between START and END and return it.  */)
       (Lisp_Object start, Lisp_Object end)
{
  validate_region (&start, &end);
  if (XINT (start) == XINT (end))
    return empty_unibyte_string;
  return del_range_1 (XINT (start), XINT (end), 1, 1);
}
```

マクロの記述の細かい点に立ち入ることはしないが、このマクロが DEFUN という単語から始まっていることを指摘しておく。ここで DEFUN という単語を選んだのは、このコードが Lisp における defun と同じ目的を持っているためである。単語 DEFUN の後に続く括弧の中に 7 つの部分がある。

- 最初の部分は、この関数に Lisp の中で与えられる関数の名前である。この場合は delete-region である。
- 二番目は、この関数の C での名前 Fdelete_region である。便宜上、'F' から始まっている。C では名前にハイフンを使わないので、代わりに下線 (underscore) が使われている。
- 三番目は、内部でこの関数の情報を記録しておくための C の constant な構造体の名前である。これは、C での関数名であるが、'F' ではなく 'S' で始まっている。
- 四番目と五番目は、この関数が引数として取れる最小の数と最大の数を指定している。今の場合はちょうど 2 つの引数を取る。
- 六番目は Lisp の関数定義のインタラクティブ宣言に続く引数と同じようなものである。文字と、大抵はそれに続くプロンプトが書かれる。今の場合、その文字は "r" である。これは、この関数の引数がそのバッファのリージョンの最初と最後の位置であることを示している。この場合はプロンプトはない。
- 七番目は、説明文字列である。これも Emacs Lisp の関数定義の場合とほぼ同じであるが、全ての改行が明示的に '\n' に続いてバックスラッシュと復帰コードを書くことで示されている。

次に形式的なそのオブジェクトの種類の宣言と一緒にパラメータが来て、そして、このマクロの本体部分と呼ばれる部分が来る。delete-region の場合は、本体は次の三行からなる。

```
validate_region (&start, &end);
if (XINT (start) == XINT (end))
  return empty_unibyte_string;
return del_range_1 (XINT (start), XINT (end), 1, 1);
```

最初の関数 validate_region は、リージョンの始まりと終わりとして渡された値が正しい型で、適切な範囲かどうかをチェックしている。二番目の関数、del_range は、実際にテキストを削除する関数である。もしこの関数がエラーを起こさず仕事を終えたら、三行目がそのことを示すために Qnil を返す。

del_range は複雑な関数なので、立ち入らないことにする。これはバッファを更新するなどの働きをする。しかしながら、del_range に渡される二つの引数には注目しておいた方がいいだろう。XINT(b) と XINT(e) の二つである。

C 言語として見る限り、b と e は削除するリージョンの始まりと終わりを表わす二つの整数である。[1].

(旧訳) しかしながら、他の Emacs Lisp の数と同様、本来の数として使用されるのは 32 ビット中の最初の 24 ビットだけである。残りの 8 ビットは情報の種類を保持したりするなど、他の目的のために使われる。(ある種の機械では、最初のほんの 6 ビットしか使わなかったりもする。) 今の場合は、この 8 ビットはこれらの数がバッファ内での位置を表すためのものであることを示すために用いられる。このような目的のために使われる数は、*tag* と呼ばれる。このように 32 ビット整数の中に 8 ビットのタグを埋め込むことで、そうしないよりも Emacs がずっと速く動作するようにすることが出来る。一方で、数を 24 ビットに制限してしまうために、Emacs のバッファは約 8 メガバイトに制限されてしまう。(バッファの最大サイズは、コンパイルする前に emacs/src/config.h の中で VALBITS と GCTYPEBITS を定義することでぐっと増やすことが出来る。Emacs の配布に含まれる emacs/etc/FAQ を参照のこと。)

[1] More precisely, and requiring more expert knowledge to understand, the two integers are of type 'Lisp_Object', which can also be a C union instead of an integer type.

（未訳）In early versions of Emacs, these two numbers were thirty-two bits long, but the code is slowly being generalized to handle other lengths. Three of the available bits are used to specify the type of information; the remaining bits are used as ‘content’.

‘XINT’ is a C macro that extracts the relevant number from the longer collection of bits; the three other bits are discarded.

（旧訳）‘XINT’ は、32 ビットの Lisp オブジェクトから 24 ビットの数を取り出す C のマクロである。他の目的に使われる 8 ビットの部分は破棄される。

The command in delete-and-extract-region looks like this:

```
del_range_1 (XINT (start), XINT (end), 1, 1);
```

これは、開始位置 b から 終了位置 e までの間を消去することになる。

Lisp を書く人の立場から見れば、Emacs はおしなべて大変単純な構造をしている。しかし、そのような動作をさせるために、裏では極めて複雑な処理をやっているというわけである。

8.5 defvar を用いた変数の初期化

（旧訳）delete-region 関数と違って、copy-region-as-kill 関数は Emacs Lisp で書かれている。これはバッファのリージョンを複写して kill-ring と呼ばれる変数に保存するものである。このセクションではこの変数がどのようにして生成され初期化されるのかを説明する。

(ここでも、kill-ring という言葉は誤用と言えるだろう。バッファから切り取られたテキストは取り戻すことが出来るのである。これは、死体の輪なんかではなく、復活出来るテキストの輪なのだ。)

Emacs Lisp では、kill-ring のような変数は defvar という特殊形式によって生成され、初期値を与えられる。この名前は “define variable” から来ている。

defvar という特殊形式は、変数に値を設定する点で setq と似ている。setq と違う点は二つある。一つ目は、その変数がまだ値を持っていない場合にのみ値を設定するということである。既に値がある場合には、defvar はその値を上書きしたりはしない。二つ目は defvar には説明文字列があるということである。

（未訳）(There is a related macro, defcustom, designed for variables that people customize. It has more features than defvar. (See Section 16.2 “Setting Variables with defcustom”, page 144.)

現在の変数の値は、どんな値であれ、describe-variable 関数を使って見ることが出来る。これは普通は *C-h v* をタイプすることで呼び出せる。*C-h v* とタイプして、プロンプトが出た所で kill-ring (と改行) を入力すると、現在の kill リングの中身がどうなっているか見ることが出来る。これは、滅茶苦茶多いこともある！ 逆に、もし今回 Emacs を起動してからこの文書を読むだけで他には何もしていなければ、中には何もないはずだ。また、*Help* バッファの最後に kill-ring の説明がなされていることも見てとれる。

```
Documentation:
List of killed text sequences.
Since the kill ring is supposed to interact nicely with cut-and-paste
facilities offered by window systems, use of this variable should
interact nicely with ‘interprogram-cut-function’ and
‘interprogram-paste-function’.  The functions ‘kill-new’,
‘kill-append’, and ‘current-kill’ are supposed to implement this
interaction; you may want to use them instead of manipulating the kill
ring directly.
```

この kill リングは defvar を使って次のように定義されている。

```
(defvar kill-ring nil
  "List of killed text sequences.
..."）
```

この変数定義の中では、この変数は初期値 nil を与えられている。これはもっともなことである。何もセーブしないうちから、ヤンク (yank) コマンドで何かを取り出したいとは思わないだろう。説明文字列は defun の説明文字列と同じように書かれている。defun のものと同様、最初の行はそれだけで完全な文になっているべきだ。何故なら apropos のような幾つかのコマンドは、最初の一行しか表示しないからである。また、その後に続く行はインデントすべきではない。そうしないと C-h v (describe-variable) を使って見た場合に見栄えがよくない。

8.5.1 defvar and an asterisk

In the past, Emacs used the `defvar` special form both for internal variables that you would not expect a user to change and for variables that you do expect a user to change. Although you can still use `defvar` for user customizable variables, please use `defcustom` instead, since it provides a path into the Customization commands. (See Section 16.2 "Specifying Variables using `defcustom`", page 144.)

When you specified a variable using the `defvar` special form, you could distinguish a variable that a user might want to change from others by typing an asterisk, '`*`', in the first column of its documentation string. For example:

```
(defvar shell-command-default-error-buffer nil
  "*Buffer name for ‘shell-command’ ... error output.
... ")
```

You could (and still can) use the `set-variable` command to change the value of `shell-command-default-error-buffer` temporarily. However, options set using `set-variable` are set only for the duration of your editing session. The new values are not saved between sessions. Each time Emacs starts, it reads the original value, unless you change the value within your `.emacs` file, either by setting it manually or by using `customize`. See Chapter 16 "Your `.emacs` File", page 143.

For me, the major use of the `set-variable` command is to suggest variables that I might want to set in my `.emacs` file. There are now more than 700 such variables, far too many to remember readily. Fortunately, you can press `TAB` after calling the `M-x set-variable` command to see the list of variables. (See Section "Examining and Setting Variables" in *The GNU Emacs Manual*.)

8.6 復習

この辺りで導入した関数についての簡単なまとめを載せておく。

`car`
`cdr` `car` はリストの最初の要素を返す。`cdr` はリストの二番目以降の要素からなるリストを返す。

例)

```
(car '(1 2 3 4 5 6 7))
     ⇒ 1
(cdr '(1 2 3 4 5 6 7))
     ⇒ (2 3 4 5 6 7)
```

`cons` `cons` は最初の引数を二番目の引数の前に加えることで新たなリストを作成する。

例)

```
(cons 1 '(2 3 4))
     ⇒ (1 2 3 4)
```

`funcall` `funcall` evaluates its first argument as a function. It passes its remaining arguments to its first argument.

`nthcdr` リストに対して 'n' 回 `cdr` を取った結果を返す。The n^{th} `cdr`. 言わば '残り' の '残り' である。

例)

```
(nthcdr 3 '(1 2 3 4 5 6 7))
     ⇒ (4 5 6 7)
```

`setcar`
`setcdr` `setcar` はリストの最初の要素を置き換える。`setcdr` はリストの二番目以降の要素を置き換える。

例)

```
(setq triple '(1 2 3))

(setcar triple '37)

triple
     ⇒ (37 2 3)

(setcdr triple '("foo" "bar"))

triple
     ⇒ (37 "foo" "bar")
```

`progn` 引数を順に評価していき、最後の値を返す。

例)

```
(progn 1 2 3 4)
     ⇒ 4
```

`save-restriction`
カレントバッファのナローイングの情報をどんな状態であれ保存し、引数を評価し終った後、その状態を復元する。

`search-forward`
文字列を検索し、もし文字列が見つかれば、そこにポイントを移動する。With a regular expression, use the similar `re-search-forward`. (See Chapter 12 "Regular Expression Searches", page 101, for an explanation of regular expression patterns and searches.)

引数は次の四つ

1. 検索文字列。
2. (省略可能) 検索範囲。
3. (省略可能) 検索に失敗した時に `nil` を返すかエラーメッセージを出すかを指定。
4. (省略可能) 検索を何回実行するかを指定。負数の場合は後方検索になる。

`kill-region`
`delete-region`
`copy-region-as-kill`
`kill-region` はバッファのポイントとマークの間のテキストを切り取ってそれを kill リングに保存する。このテキストはヤンクによって取り出すことが出来る。

`copy-region-as-kill` はポイントとマークの間のテキストを kill リングに複写する。これはヤンクによって取り出すことが出来る。この関数はバッファからテキストを切り取ったり削除したりはしない。

`delete-and-extract-region` はポイントとマークの間のテキストをバッファから削除して捨ててしまう。復活させることは出来ない。(This is not an interactive command.)

8.7 検索についての練習問題

- 文字列を検索するインタラクティブな関数を書きなさい。もし文字列の検索に成功すればその直後にポイントを移動し、「見つけた！」というメッセージを表示するようにしなさい。(この関数の名前に `search-forward` という名前を使ってはいけない。そうしてしまうと、元の Emacs で定義された現在の `search-forward` を上書きしてしまう。例えば `test-search` という名前を使いなさい。
- 現在の kill リングの三番目の内容をエコー領域に表示するような関数を書きなさい。その際、もし kill リングに三番目の要素が無い場合にも適切なメッセージを表示するようにしなさい。

9 リストはどのように実装されているか

Lisp では、アトムは直接的な方法で記録される。たとえ現実の実装方法が直接的ではなかったとしても、理論上は直接的である。例えば ‘rose’ というアトムは四つの隣接する文字 ‘r’, ‘o’, ‘s’, ‘e’ として記録される。一方、リストはもっと異なる方法で記録される。メカニズムとしては同じくらい単純なのだが、そのアイディアに慣れるのにはちょっと時間がかかるかもしれない。リストは、ポインタのペアの列として記録される。この列の中で、各々のペアの一番目のポインタは、あるアトムか、もしくは他のリストを指している。そして、二番目のポインタは次のペアか、シンボル nil を指している。nil の場合はリストの終わりということである。

ポインタそれ自身は、指されている対象の極めて単純な電気的な住所だと言える。従って、リストも電気的な住所の列として記録されていると言える。

例えば、(rose violet buttercup) というリストは ‘rose’, ‘violet’, ‘buttercup’ という三つの要素を持っている。計算機の中では、‘rose’ の電気的な住所は、ある計算機のメモリのセグメントに ‘violet’ というアトムの電気的な住所を示すアドレスに並んで記されている。そしてこっちの住所 (‘violet’ の場所を示すもの) は ‘buttercup’ というアトムの電気的な住所を示すアドレスに並んで記されている。

このように書くと何やら複雑に思えるかもしれないが、図に書いてみれば簡単に理解出来るだろう。



上の図の中で、各々の箱は Lisp オブジェクトを保持する計算機のメモリのワードを表現している。これは普通はメモリアドレスの形をしている。これらの箱、即ち住所は、ペアになっている。各々の矢印は、そのアドレスを持つ住所を指している。その場所には、アトムか他のアドレスのペアがある。最初の箱は ‘rose’ の電気的な住所であり、矢印は ‘rose’ を指している。二番目の箱は次の箱のペアを指しており、こっちのペアの一番目の箱は ‘violet’ の住所、そして二番目の箱はまた次のペアのアドレス、という具合になっている。そして最後の箱は、シンボル nil を指している。これは、リストの最後を示している。

setq などの関数によってある変数がこのリストにセットされた場合、このリストの先頭の箱のアドレスがこの変数にセットされる。従って、

```
(setq bouquet '(rose violet buttercup))
```

を評価すると、以下のような状況になる。



この場合、シンボル bouquet が最初のペアのアドレスを持つことになる。

同じリストをちょっと箱の書き方を変えた別のイラストで描いてみると、次のようになる。



(Symbols consist of more than pairs of addresses, but the structure of a symbol is made up of addresses. Indeed, the symbol bouquet consists of a group of address-boxes, one of which is the address of the printed word ‘bouquet’, a second of which is the address of a function definition attached to the symbol, if any, a third of which is the address of the first pair of address-boxes for the list (rose violet buttercup), and so on. Here we are showing that the symbol's third address-box points to the first pair of address-boxes for the list.)

もし、シンボルがあるリストの CDR にセットされたなら、そのリスト自身は変化しない。そのシンボルは単にそのリストのアドレスを持つようになるだけである。(専門用語では、CAR や CDR は「非破壊的 (non-destructive)」 であると言う。) 従って、次の S 式を評価すると、

```
(setq flowers (cdr bouquet))
```

以下のような状況が生ずる。



`flowers` の値は `(violet buttercup)` であるが、これはシンボル `flowers` がアドレスボックスのペアのアドレスを持ったということである。このペアの最初の箱には `violet` のアドレスがあり、二番目の箱には `buttercup` のアドレスが入っている。

アドレスボックスのペアはコンスセル (*cons cell*) もしくはドット対 (*dotted pair*) と呼ばれる。Section “Cons Cell and List Type ” in *The GNU Emacs Lisp Reference Manual*, 及び Section “Dotted Pair Notation” in *The GNU Emacs Lisp Reference Manual*, にコンスセルとドット対についてのより詳しい情報がある。

関数 `cons` は、上に述べたようなアドレスの列の先頭に新しいアドレスペアを付け加える。例えば、

```
(setq bouquet (cons 'lily bouquet))
```

を評価すると



という状況になる。この操作ではシンボル `flowers` の値を変更しているわけではない。これは次を評価することで確かめられる。

```
(eq (cdr (cdr bouquet)) flowers)
```

これは、`t` を返すはずだ。

再設定されない限り、`flowers` はずっと `(violet buttercup)` という値を持ったままである。つまり、最初のアドレスが `violet` であるコンスセルのアドレスを持っているということだ。また上の操作は、既に存在しているコンスセルを変えたりはしない。それらは全てそのままである。

従って、Lisp でリストの CDR を得るには、単にその列の隣りのコンスセルのアドレスを得ればよく、また CAR を得るには、単にそのリストの最初の要素のアドレスを得ればよい。更に、リストに新しい要素を `cons` するには、リストの先頭に新しいコンスセルを加えればよい。以上がこの話題に関する全てである。Lisp の裏側は驚くほど明解で単純のなのだ。

では、コンスセルの列の最後のアドレスはどこに関連づけられているのか？ これは `nil`, 即ち、空リストである。

まとめると、Lisp の変数にある値にセットされる時は、その変数には、その変数が見にいくリストのアドレスが与えられるというわけである。

9.1 Symbols as a Chest of Drawers

In an earlier section, I suggested that you might imagine a symbol as being a chest of drawers. The function definition is put in one drawer, the value in another, and so on. What is put in the

drawer holding the value can be changed without affecting the contents of the drawer holding the function definition, and vice-verse.

Actually, what is put in each drawer is the address of the value or function definition. It is as if you found an old chest in the attic, and in one of its drawers you found a map giving you directions to where the buried treasure lies.

(In addition to its name, symbol definition, and variable value, a symbol has a 'drawer' for a *property list* which can be used to record other information. Property lists are not discussed here; see Section "Property Lists" in *The GNU Emacs Lisp Reference Manual*.)

Here is a fanciful representation:



9.2 練習問題

`flowers` に `violet` と `buttercup` をセットしなさい。また、このリストに更に二つの花を cons し、新しく出来たリストを `more-flowers` にセットしなさい。`flowers` の CAR に魚の名前をセットしなさい。この時、`more-flowers` のリストの中身はどうなるか。

10 テキストのヤンク

GNU Emacs で 'kill' コマンドでバッファからテキストを切り取った場合、そのテキストは常に、ヤンク (yank) コマンドで取り戻すことが出来る。バッファから切り取られたテキストは kill リングに置かれており、ヤンクコマンドは、この kill リングの中から該当する内容を取り出してバッファに挿入する (これは、元のバッファとは限らない)。

単なる *C-y* (yank) コマンドは、kill リングの最初の要素をカレントバッファに挿入する。もし、*C-y* コマンドのすぐ後に続けて *M-y* とタイプすると、その最初の要素が二番目の要素と取り換えられる。続けて何回も *M-y* をタイプすると、その回数だけ最初の要素を入れ替えていき、一回りするとまた最初に戻ってそれを繰り返す。(そういうわけで kill リングは「リスト」ではなく「リング」と呼ばれるのである。しかし、実際にテキストを保存しているデータの型はリストに他ならない。リストをリングとして扱う方法についてのより詳しい話は Appendix B “kill リングの扱い”, page 165, を参照。)

10.1 Kill リングについての概観

kill リングはテキスト形式の文字列のリストである。例えば次のような形をしている。

```
("some text" "a different piece of text" "yet more text")
```

もし私の kill リングが上のものだったとして、ここで私が *C-y* と押すと、'some text' という文字列が私が現在いるバッファのカーソルの位置に挿入される。

yank コマンドはまた、テキストを複写するのにも使われる。コピーされるテキストはバッファから切り取られるのではなく、一旦 kill リングに置かれ、その後ヤンクされることで挿入されることになる。

kill リングのテキストを取り出すには三つの関数が使われる。通常 *C-y* にバインドされている yank, 同じく通常 *M-y* にバインドされている yank-pop、そして、この二つの関数によって使われる rotate-yank-pointer である。

These functions refer to the kill ring through a variable called the kill-ring-yank-pointer. Indeed, the insertion code for both the yank and yank-pop functions is:

```
(insert (car kill-ring-yank-pointer))
```

(未訳) (Well, no more. In GNU Emacs 22, the function has been replaced by insert-for-yank which calls insert-for-yank-1 repetitively for each yank-handler segment. In turn, insert-for-yank-1 strips text properties from the inserted text according to yank-excluded-properties. Otherwise, it is just like insert. We will stick with plain insert since it is easier to understand.)

yank や yank-pop の働きを理解するには、まず変数 kill-ring-yank-pointer と、関数 rotate-yank-pointer を理解する必要がある。

10.2 変数 kill-ring-yank-pointer

kill-ring-yank-pointer は、kill-ring と同じく変数である。一般に変数は、あるものの値にバインドされることで、それを指し示す働きをする。

従って、もし kill リングの値が

```
("some text" "a different piece of text" "yet more text")
```

であり、また kill-ring-yank-pointer が二番目の位置を指していたとすると、kill-ring-yank-pointer の値は

```
("a different piece of text" "yet more text")
```

である。前章で説明したように (Chapter 9 “リストはどのように実装されているか”, page 78,)、計算機は kill-ring と kill-ring-yank-pointer とに指されたテキストを各々別々に保持しているわけではない。“a different piece of text” と “yet more text” の二つの言葉は、二重に複製されているのではないのである。その代わり二つの Lisp 変数は同じテキストの集まりを指している。図に表わすと次の通りである。



変数 kill-ring と 変数 kill-ring-yank-pointer は両方ともポインタである。しかし、kill リングそのものは、実際に要素を組み合わせたものとして説明されることが多い。つまり、kill-ring はリストの計算機内での位置を指しているというよりは、リストそのものの意味で使われる。それに対し、kill-ring-yank-pointer の方は、リストへのポインタという意味で使われることが多い。

全く同じことについて、こういうふうに二つの言い方を使うと最初は混乱するように思えるかもしれない。しかしよく考えてみれば合理的である。kill リングは一般に、Emacs のバッファから最近切り取られたテキストのデータからなっていると捉えられている。一方 kill-ring-yank-pointer の方は、kill リングの中でその最初の要素 (つまり CAR) がこれから挿入するデータであるような部分、を指し示すために使われる。

（旧訳、削除？）rotate-yank-pointer 関数は、kill-ring-yank-pointer が指し示す要素を交換するものである。より詳しく言うと、ポインタが kill リングの最後の要素を越えて更に次の要素に移ろうとする際に、自動的にポインタを kill リングの最初の要素に移動してくれるのだ。このようにしてリストがあたかもリングであるように扱えるのである。rotate-yank-pointer 関数それ自身は単純であるが、その中身には多くの細かい説明が必要である。この関数及び、これよりずっと単純な yank, yank-pop 関数については次の所で説明する。Appendix B “kill リングの扱い”, page 165.

10.3 yank と nthcdr についての練習問題

- *C-h* v (describe-variable) を使って、あなたの kill リングの値を見なさい。幾つか kill リングの要素を増やしてから、もう一度 kill リングの中身を見なさい。*M-*y (yank-pop) を使って kill リングの中を動き回りなさい。あなたの kill リングの中にはどのくらい多くの要素がありますか？ kill-ring-max の値を見なさい。kill リングはもう一杯ですか？それともまだテキストのブロックを追加出来ますか？
- nthcdr と car を使って、各々リストの一番目、二番目、三番目、及び四番目の要素を返すような四つの S 式を作りなさい。

11 ループと再帰

Emacs Lisp は、一つないしは複数の S 式を繰り返し評価させるのに、主として二つの方法を持っている。一つは `while` ループを利用するもの、もう一つは再帰 (*recursion*) を使うものである。

繰り返しは非常に役に立つものである。例えば四つの文の分だけ先に移動したい場合には、一つの文だけ移動するプログラムを書いて、そのプロセスを四回繰り返せば良い。計算機には飽きるとか疲れるとかいった感情はないので、このような繰り返しをいくらやらせたところで、人間のように過度の繰り返しをさせてしまってミスを犯しやすくなるといった弊害は無い。

(未訳) People mostly write Emacs Lisp functions using `while` loops and their kin; but you can use recursion, which provides a very powerful way to think about and then to solve problems[1].

11.1 while

特殊形式 `while` は、最初の引数を評価して返された値が真か偽かをテストする。ここまでは `if` の時と同じである。しかし、次に Lisp インタプリタがやることはちょっと違っている。

`while` 式においても、返された値が偽の場合は、Lisp インタプリタは残りの S 式 (この S 式の本体部分) をスキップし、それを評価したりはしない。しかし、返された値が真の場合は、Lisp インタプリタは本体部分を評価した後、もう一度 `while` の最初の引数が真か偽かをテストする。もし、返された値がまたもや真であったなら、Lisp インタプリタは再度本体部分を実行する。

`while` 式のテンプレートは次の通りである。

```
(while 真偽テスト
  本体...)
```

`while` 式の真偽テストの部分が真を返している限り、本体部分は繰り返し評価される。このプロセスはループと呼ばれる。これは、Lisp インタプリタが何回も同じことを繰り返す様子が、飛行機の宙返りと似ているためである。真偽テストの結果が偽の場合は、Lisp インタプリタはそこで `while` 式の残りの部分を評価するのを止めて「ループを抜ける」。

当たり前のことだが、もし `while` 式の最初の引数が常に真を返すとすると、その後の本体部分は何回でも無限に評価され続ける。逆にもし真になり得ないとすると、本体部分の式は決して評価されることはない。`while` 式のループを書く工程とは、本体部分の S 式の評価を繰り返したい回数だけ真偽テストが真の値を返し、そしてその後、偽を返すような仕掛けを作ることである。

`while` 式を評価した場合に返される値は真偽テストが返す値である。従って面白いことに、エラーなく評価された `while` ループは、たとえ 100 回ループを繰り返した後でも一度もループしなかった場合でも、必ず `nil` を返す。正しく評価された `while` 式は決して真の値を返すことはないわけだ！ このことは、`while` 式は常に副作用を目的として評価されることを意味する。即ち、`while` ループの中の本体部分の S 式を評価するために評価されるのである。これは理にかなっている。単にループを繰り返すことが目的なのではなく、ループの中の S 式を繰り返し評価した時に生じる結果こそが望むものなのだから。

11.1.1 while ループとリスト

`while` ループをコントロールする一般的な方法は、リストが要素を持っているかどうかをテストすることである。もし要素が一つもなければ、それは空リストであり、結果として空リスト `()` が返される。これは、`nil` もしくは偽 (false) の同義語である。一方、要素を持つリストの場合は、それらの要素そのものが返される。Lisp は `nil` 以外のものは全て真と見做すので、この場合は `while` ループでは真が返されたことになる。

---

[1] You can write recursive functions to be frugal or wasteful of mental or computer resources; as it happens, methods that people find easy—that are frugal of 'mental resources'—sometimes use considerable computer resources. Emacs was designed to run on machines that we now consider limited and its default settings are conservative. You may want to increase the values of `max-specpdl-size` and `max-lisp-eval-depth`. In my `.emacs` file, I set them to 15 and 30 times their default value.

例えば、次のS式を評価すると、変数 empty-list の値を nil にセットすることが出来る。

```
(setq empty-list ())
```

この setq 式を評価した後、変数 empty-list をいつものようにカーソルをこのシンボルのすぐ後に持っていって C-x C-e とタイプすことで評価してみよう。エコー領域に nil が表示されるはずだ。

```
empty-list
```

一方、変数を要素を持つリストにセットした場合は、その変数を評価するとそのリストが表示される。これは次の二つのS式を評価することで確かめられる。

```
(setq animals '(gazelle giraffe lion tiger))

animals
```

ということで、リスト animals に要素があるかをテストに使用するような while ループを作るには、ループの最初の部分を次のように書けばよい。

```
(while animals
       ...
```

while が最初の引数をテストすると、変数 animals が評価される。これはリストを返す。このリストが要素を持っている限り、while はテストの結果が真だと判断する。しかし、リストが空になると、結果は偽だと判断される。

while ループが無限に続くことを避けるためには、最終的にリストが空になるような何らかの仕掛けが必要である。よく使われるテクニックは、while 式の本体部分の形式の一つでそのリストの値をそのリストの cdr で置き換えるようにしておくことである。cdr 関数が評価されるごとにそのリストは短くなっていき、ついには空リストが残されるというわけである。そうなると while ループのテストは偽を返し、while の引数はもはや評価されず、ループが終了する。

例えば、動物のリストが変数 animals にバインドされていた時は、次のS式を使ってその変数を元のリストの CDR にセットすることが出来る。

```
(setq animals (cdr animals))
```

もし一つ前のS式を評価してあったなら、このS式を評価するとエコー領域に (gazelle lion tiger) が表示されるはずだ。もう一度このS式を評価すると、今度は (lion tiger) が表示される。更に評価すると、(tiger) が表示され、そこでまた評価すると、やっと空リストになって、nil が表示される。

cdr 関数を繰り返し使うことで最終的に真偽テストで偽を返す while ループのテンプレートは次のようになる。

```
(while リストが空かどうかのテスト
  本体...
  リストに自分自身の cdr をセット)
```

このテストと cdr の利用を組み合わせて、リストの各要素を各々一行ごとに表示するような関数を作ることが出来る。これを次に説明することにしよう。

11.1.2 リストを使ったループの例: print-elements-of-list

リストを使った while ループのことを理解するには print-elements-of-list 関数を見ると良い。

（旧訳）この関数は、出力を表示するために幾つかの行を必要とする。エコー領域は一行しかないので、今まで説明して来たように Info の中で評価するようなやり方ではその動作をうまく描写出来ない。If you are reading this in a recent instance of GNU Emacs, you can evaluate the following expression inside of Info, as usual.

代わりに必要なS式を *scratch* バッファにコピーして、そこでそれらを評価する必要がある。

コピーするには、まず対象となるリージョンの最初を *C-SPC* (set-mark-command) でマークしてからカーソルをリージョンの最後に移動し、*M-w* (copy-region-as-kill) でそのリージョンをコピーする。そして*scratch* バッファで *C-y* (yank) することで、そのS式を取り出せばよい。

scratch バッファにそのS式をコピーしたら、今度はそのS式を評価するのだが、最後の (print-elements-of-list animal) は *C-u C-x C-e* とタイプして評価する必要がある。つまり eval-last-sexp に前置引数を与えるのである。すると、評価した結果はエコー領域ではなく *scratch* バッファに表示される。(単に *C-x C-e* とするだけだと、返された値はエコー領域に ^Jgiraffe^J^Jlion^J^Jtiger^Jnil のように表示されてしまう。ここで出て来る '^J' は改行を表している。従って *scratch* バッファでは各々の単語は一行ごとに表示されることになるわけである。このことは、今 Info バッファで次のS式を評価してみても確かめられる。)

In a recent instance of GNU Emacs, you can evaluate these expressions directly in the Info buffer, and the echo area will grow to show the results.

```
(setq animals '(gazelle giraffe lion tiger))

(defun print-elements-of-list (list)
  "Print each element of LIST on a line of its own."
  (while list
    (print (car list))
    (setq list (cdr list))))

(print-elements-of-list animals)
```

これらの三つのS式を順に *scratch* バッファで評価していくと、そのバッファの中で次のように表示されるはずだ。

```
gazelle

giraffe

lion

tiger
nil
```

リストの各要素が各々一行ごとに表示され (これは print の仕事である)、そして最後にこの関数自体が返した値が表示される。この関数の中の最後のS式は while ループであり、while ループは常に nil を返すので、nil がリストの最後の要素の後に表示される。

11.1.3 増加するカウンタを使ったループ

ループは止まるべき時に止まってくれないことには役に立たない。リストを使ったループの制御以外の一般的な方法としては、最初の引数として、正しい回数だけ繰り返すと偽を返すテストを書くという方法がある。これは、ループがカウンタ—つまりループを繰り返した回数を数えるS式—を持つということである。

真偽テストとしては、(< count desired-number) といったS式が使える。これは count の値が期待する繰り返しの数 desired-number よりも小さければ、t という真の値を返し、desired-number 以上であれば、偽の値 nil を返す。カウンタを増加させるS式は (setq count (1+ count)) といったごく簡単な setq 式でよい。ここで 1+ は Emacs Lisp の組み込み関数で、引数に 1 を加えるものである。((1+ count) は (+ count 1) としても同じだが、この方が人間にとって読みやすいであろう。)

ということで、増加するカウンタを使った while ループのテンプレートは次のようになる。

```
カウンタを初期値に戻す
(while (< count desired-number)         ; 真偽テスト
  本体...
  (setq count (1+ count)))              ; インクリメンタ
```

この場合 count の初期値を定めなければならないことに注意しよう。通常は 1 にセットする。

増加カウンタの例

あなたは浜辺で遊んでおり、小石で三角形を作ろうと思ったとしよう。まず、一行目に 1 つの小石を置き、二行目に 2 つ置き、三行目に 3 つ置き..., という具合に続けていくのだ。図で描くと次のようになる。



(約 2500 年前、ピタゴラス一派はこのような問題を考えて数論の初歩を発展させていった。)

ここで、七行の三角形を作るにはいくつの小石が必要かを知りたいと思ったとする。

当然、ここであなたがしなければならないのは 1 から 7 までの数を加えることである。これを行うには二通りの方法がある。小さい数から始めて 1、2、3、4、... という数列を加えて行くか、大きい方から始めて 7、6、5、4、... という数列の和を取るかである。どちらの場合にも while ループを書く

時の共通の方法を説明してくれるので、この下から登るのと、上から降りるのと両方の例を作ってみようと思う。最初の例として 1 に 2、3、4 を加えていくものから始めよう。

和を取る数のリストが短い場合は、一度に全部足してしまうのがもっとも簡単な方法である。しかしながら、もし前もって数のリストがどれくらいの長さになるか分らなかったり、あるいは極めて長いリストの場合にも対処したい場合には、複雑なプロセスを一度にやるのではなく、単純なプロセスを沢山繰り返して和を求めるような方法を考える必要がある。

例えば、小石を一度に全部加える代わりに、まず最初の行の小石の数 1 に二行目の小石の数 2 を加え、次にその合計に三行目の小石の数 3 を加え、今度はその和に四行目の数 4 を加え、という操作を続けるのである。

このプロセスで大切な特徴は、繰り返し行う操作は単純ですむということである。今の場合、各々のステップでやっていることは、二つの数を加えるということだけである。その時点での行の小石の数と、それまでの合計は既に分っている。この二数の和を取るプロセスは、最後の行の数をそれまでの合計に加えるまで何回でも繰り返される。もっと複雑なループでは、繰り返し行う操作はそれ程単純ではない。しかしそれでも全てを一度にやるよりはずっと簡単なのである。

関数定義の各部分

前節での分析は、我々の関数定義の骨組みを与えてくれる。まず、小石の全体の数を表わす変数が要る。これは `total` としていいだろう。これがこの関数の返す値である。

次に、この関数には三角形の全行数を表わす引数が要る。これは `number-of-rows` として良いだろう。

最後に、カウンタとして使う変数が必要である。これは `counter` としても構わないのだが、それよりも `row-number` とする方が良い。何故なら、このカウンタがやることは行を数えることであり、プログラムは出来るだけ分りやすく書くべきものだからである。

Lisp インタプリタがこの関数内の S 式の評価を開始すると、まず `total` の値が零にセットされる。これはまだ何も加えていないからである。次に、最初の行の小石の数が足される。そして二行目の数を加え、三行目の数を加え、というふうに続き、最後の行の小石を加えたところで終了する。

`total` も `row-number` も共にこの関数内部でしか使われない。従って、`let` を使って局所変数として宣言し、初期値を与えればよい。明らかに `total` の初期値は 0 であり、また `row-number` の初期値は 1 である。ということで、`let` 式は次のようになる。

```
(let ((total 0)
      (row-number 1))
  本体...)
```

内部変数が宣言されて初期値にバインドされたなら、`while` ループを開始する。テストの部分の S 式は `row-number` が `number-of-rows` 以下である場合その時のみ `t` を返すようなものでなければならない。(`row-number` と `number-of-rows` が一致する場合も含めないと最後の行の小石の数が加えられないことに注意。)

Lisp には `<=` という、最初の引数が二番目の引数以下の場合に真を返し、そうでない場合は偽を返す関数がある。これを使うと、`while` 式のテスト部分の S 式は次のように書ける。

```
(<= row-number number-of-rows)
```

全体の小石の数は、各々の行の小石の数をそれまでの合計に繰り返し加えていくことで計算される。ある行の小石の数はその行の番号と一致するので、全体の数は行の番号を加えていくことで求まることになる。(言うまでもないが、状況が複雑な場合には、ある行の小石の数とその行の番号とはもっと複雑な関係で結ばれている。そういう場合は行番号は他の適当な S 式で置き換えることになる。)

```
(setq total (+ total row-number))
```

この S 式では `total` の新しい値としてそれまでの合計に現在の行の小石の数を加えたものをセットしている。

`total` の値をセットしたら、次のループに移る場合のために条件を整えておかなければならない。それには、カウンタ用の変数 `row-number` の値に 1 を加えれば良い。`row-number` の値が 1 増やされると、次のループの `while` 式の先頭の真偽テストで、この新しい値が `number-of-row` の値以下であるかが判定される。もしそうであれば、変数 `row-number` の新しい値が前回のループでの `total` の値に加えられる。

Emacs Lisp の組み込み関数 `1+` は数に 1 を加えてくれる。従って、`row-number` 変数は次の S 式で 1 増加させることが出来る。

```
(setq row-number (1+ row-number))
```

各部分の総合

これまでで関数定義の各々の部分を書いたので、ここでそれらを一つにまとめることにしよう。

まず while 式の中身は次の通り。

```
(while (<= row-number number-of-rows)   ; 真偽テスト
  (setq total (+ total row-number))
  (setq row-number (1+ row-number)))    ; インクリメンタ
```

これに let 式の変数リストを加えればほぼ関数定義の本体部分は完成するが、最後にほんの少しだけ加えるべき要素がある。

それは変数 total それ自身を while 式の後に書くことである。そうしないと関数全体が返す値は let 式の本体部分の最後の S 式が返す値、即ち while 式が返す値になるので、常に nil が返されてしまうからである。

このことはぱっと見ただけでは気がつかないかもしれない。値を 1 増加させる式が最後だから良いように思えるかもしれないが、これはあくまで while 式の一部なのである。シンボル while から始まるリストの最後の要素なのだ。更に while ループ全体は let 式の本体部分の中のリストなのである。

大ざっぱに書くと関数は次のような形をしている。

```
(defun 関数名 (引数リスト)
  "説明文字列..."
  (let (変数リスト)
    (while (真偽テスト)
      while の本体... )
    ... )                          ; ここに最後の式が来る。
```

このように定義された関数が返す値は let 式が返す値になる。というのも let は defun 以外の他のどのリストにも含まれてはいないからである。だから、もし while が let 式の本体の最後の S 式であれば、この関数は常に nil を返してしまうことになる。これは我々が望む結果ではない！ これを回避するには、単にシンボル total を let で始まるリストの最後に置けば良い。この式はこのリストの他の S 式が評価された後に評価される。これによって、全体として正しい値が返されることになる。

let で始まるリスト全体を一行で表示してみると、理解しやすいだろう。こう書くと変数リスト *varlist* と while 式が let 式の二番目と三番目の要素であることがよく分る。そして最後の要素が total になるわけである。

```
(let (変数リスト) (while (真偽テスト) while の本体... ) total)
```

以上を一つにまとめると、triangle 関数の定義は次のようになる。

```
(defun triangle (number-of-rows)    ; インクリメンタを使った
                                    ;  バージョン。
  "Add up the number of pebbles in a triangle.
The first row has one pebble, the second row two pebbles,
the third row three pebbles, and so on.
The argument is NUMBER-OF-ROWS."
  (let ((total 0)
        (row-number 1))
    (while (<= row-number number-of-rows)
      (setq total (+ total row-number))
      (setq row-number (1+ row-number)))
    total))
```

これを評価して triangle をインストールした後、実際に試してみよう。二つの例を挙げる。

```
(triangle 4)

(triangle 7)
```

最初の 4 つの数の和は 10 になり、最初の 7 つの数の和は 28 になる。

11.1.4 減少するカウンタを使ったループ

while ループのテスト部分のもう一つの一般的な書き方としては、カウンタが零を越えるかどうかを判定するというものがある。カウンタが零より大きい間はループを繰り返すが、零以下になると止まるというわけである。このように動作させるためにはカウンタは零以上の数から始めて繰り返しの部分が評価されるごとに小さくなるようにしなければならない。

テスト部分は (> counter 0) のようになる。これは counter の値が零よりも大きければ真として t を返し、零以下なら偽として nil を返すというものである。数を次第に小さくしていくには、(setq

`counter (1- counter))` という単純な `setq` 式を使う。ここで、`1-` は引数を 1 減らす Emacs Lisp の組み込み関数である。

ということで、`while` ループのテンプレートは次のようになる。

```
(while (> counter 0)                    ; 真偽テスト
  本体...
  (setq counter (1- counter)))          ; デクリメンタ
```

減少するカウンタを使った例

減少するカウンタを使ったループを説明するために、`triangle` 関数を、このようなカウンタを使って書き直してみることにする。

数え方は、この関数の前回のバージョンの逆である。今回は、例えば三行からなる三角形の小石の数を求めるために、まず三行目の 3 つの小石を加え、次にその前の二行目の 2 個の小石の数を加え、更にその合計に 1 行目の小石の数 1 を加えるという操作をすることになる。

同様にして、七行からなる三角形の小石の数を求めるには、七行目の小石の数 7 にその前の行の小石の数 6 を加え、次にその合計に その前の行に小石の数 5 を加え、とやっていくことになる。前回の例と同じく、各々の足し算ではそれまでの合計と現在の行の小石の数の二つの数を加えているだけである。このプロセスは、足す小石の数がなくなるまで繰り返される。

最初に加える小石の数も分る。最後の行の小石の数は全体の行の数に等しいからである。三角形が七行からなっていれば、最終行の小石の数は 7 である。同様に、次々と足していく小石の数も分る。それは、前回足した数から 1 を引いたものである。

関数定義の各部分

まず三つの変数が必要である。三角形の行の総数、各々の行の小石の数、そして、小石の総数だ。この最後の数が今回求めようとしている数である。これら三つの変数を各々 `number-of-rows`, `number-of-pebbles-in-row`, `total` と名付ることにする。

`total` も `number-of-pebbles-in-row` もこの関数の内部でしか使われないので、`let` を用いて宣言される。`total` の初期値は勿論零である。一方、`number-of-pebbles-in-row` の初期値は三角形の行の数に等しくなるべきである。これは、もっとも長い行から小石の数を数えていくためである。

従って `let` 式の始めの部分は次のようになる。

```
(let ((total 0)
      (number-of-pebbles-in-row number-of-rows))
  本体...)
```

小石全体の数は、各々の行の小石の数を順番に数えていくことで求められる。つまり、繰り返し次の S 式を評価していけばよい。

```
(setq total (+ total number-of-pebbles-in-row))
```

`number-of-pebbles-in-row` は `total` に加えられた後、次のループに備えて一つだけ値を減らさなければならない。というのも次に加えられる行は一段上の行であり、現在の長さよりも一つ分短いからである。これは次の S 式を評価することでなされる。

The number of pebbles in a preceding row is one less than the number of pebbles in a row, so the built-in Emacs Lisp function `1-` can be used to compute the number of pebbles in the preceding row. This can be done with the following expression:

```
(setq number-of-pebbles-in-row
      (1- number-of-pebbles-in-row))
```

最後に、小石が無くなった時点で `while` ループを抜けなければならないが、そのための条件部は次のような簡単なものでよい。

```
(while (> number-of-pebbles-in-row 0)
```

各部分の総合

(未訳) We can put these expressions together to create a function definition that works. However, on examination, we find that one of the local variables is unneeded!

関数定義は次のようになる。

```
;;; デクリメンタを使った最初のバージョン。
(defun triangle (number-of-rows)
  "Add up the number of pebbles in a triangle."
  (let ((total 0)
        (number-of-pebbles-in-row number-of-rows))
    (while (> number-of-pebbles-in-row 0)
      (setq total (+ total number-of-pebbles-in-row))
      (setq number-of-pebbles-in-row
            (1- number-of-pebbles-in-row)))
    total))
```

これはこれで、うまく動作する。

しかしながら、実は変数 number-of-pebbles-in-row は必要でない！

triangle 関数が評価されると、シンボル number-of-rows は初期値を与えられて、ある数にバインドされる。この数は関数の本体の中であたかも局所変数であるかのごとく変化させることが可能で、この関数の外でのこの変数の値には何の影響も与える心配はない。これは Lisp の大変便利な特徴であるが、このことから関数内の number-of-pebbles-in-row を全て変数 number-of-rows で置き換えても良いことが分る。

ということで、以下に、この関数の少し整理したバージョンを挙げる。

```
(defun triangle (number)                ; 二番目のバージョン。
  "Return sum of numbers 1 through NUMBER inclusive."
  (let ((total 0))
    (while (> number 0)
      (setq total (+ total number))
      (setq number (1- number)))
    total))
```

まとめてみよう。きちんと書かれた while ループは以下の三つの部分からなる。

1. ループを正しい回数繰り返した後、偽を返すテスト
2. 繰り返し評価された後、最後に望む値を返すような S 式
3. 正しい回数だけループを繰り返した後にテストが偽を返すよう、真偽テストに渡される値を変化させる S 式

11.2 Save your time: dolist and dotimes

In addition to while, both dolist and dotimes provide for looping. Sometimes these are quicker to write than the equivalent while loop. Both are Lisp macros. (See Section “Macros” in *The GNU Emacs Lisp Reference Manual*.)

dolist works like a while loop that ‘CDRs down a list’: dolist automatically shortens the list each time it loops—takes the CDR of the list—and binds the CAR of each shorter version of the list to the first of its arguments.

dotimes loops a specific number of times: you specify the number.

The dolist Macro

Suppose, for example, you want to reverse a list, so that “first” “second” “third” becomes “third” “second” “first”.

In practice, you would use the reverse function, like this:

```
(setq animals '(gazelle giraffe lion tiger))

(reverse animals)
```

Here is how you could reverse the list using a while loop:

```
(setq animals '(gazelle giraffe lion tiger))

(defun reverse-list-with-while (list)
  "Using while, reverse the order of LIST."
  (let (value)  ; make sure list starts empty
    (while list
      (setq value (cons (car list) value))
      (setq list (cdr list)))
    value))

(reverse-list-with-while animals)
```

And here is how you could use the `dolist` macro:

```
(setq animals '(gazelle giraffe lion tiger))

(defun reverse-list-with-dolist (list)
  "Using dolist, reverse the order of LIST."
  (let (value)  ; make sure list starts empty
    (dolist (element list value)
      (setq value (cons element value)))))

(reverse-list-with-dolist animals)
```

In Info, you can place your cursor after the closing parenthesis of each expression and type *C-x C-e*; in each case, you should see

```
(tiger lion giraffe gazelle)
```

in the echo area.

For this example, the existing `reverse` function is obviously best. The `while` loop is just like our first example (see Section 11.1.1 "A `while` Loop and a List", page 83). The `while` first checks whether the list has elements; if so, it constructs a new list by adding the first element of the list to the existing list (which in the first iteration of the loop is `nil`). Since the second element is prepended in front of the first element, and the third element is prepended in front of the second element, the list is reversed.

In the expression using a `while` loop, the `(setq list (cdr list))` expression shortens the list, so the `while` loop eventually stops. In addition, it provides the `cons` expression with a new first element by creating a new and shorter list at each repetition of the loop.

The `dolist` expression does very much the same as the `while` expression, except that the `dolist` macro does some of the work you have to do when writing a `while` expression.

Like a `while` loop, a `dolist` loops. What is different is that it automatically shortens the list each time it loops—it 'CDRs down the list' on its own—and it automatically binds the CAR of each shorter version of the list to the first of its arguments.

In the example, the CAR of each shorter version of the list is referred to using the symbol '`element`', the list itself is called '`list`', and the value returned is called '`value`'. The remainder of the `dolist` expression is the body.

The `dolist` expression binds the CAR of each shorter version of the list to `element` and then evaluates the body of the expression; and repeats the loop. The result is returned in `value`.

The `dotimes` Macro

The `dotimes` macro is similar to `dolist`, except that it loops a specific number of times.

The first argument to `dotimes` is assigned the numbers 0, 1, 2 and so forth each time around the loop, and the value of the third argument is returned. You need to provide the value of the second argument, which is how many times the macro loops.

For example, the following binds the numbers from 0 up to, but not including, the number 3 to the first argument, *number*, and then constructs a list of the three numbers. (The first number is 0, the second number is 1, and the third number is 2; this makes a total of three numbers in all, starting with zero as the first number.)

```
(let (value)      ; otherwise a value is a void variable
  (dotimes (number 3 value)
    (setq value (cons number value))))

⇒ (2 1 0)
```

`dotimes` returns `value`, so the way to use `dotimes` is to operate on some expression *number* number of times and then return the result, either as a list or an atom.

Here is an example of a `defun` that uses `dotimes` to add up the number of pebbles in a triangle.

```
(defun triangle-using-dotimes (number-of-rows)
  "Using dotimes, add up the number of pebbles in a triangle."
(let ((total 0))  ; otherwise a total is a void variable
  (dotimes (number number-of-rows total)
    (setq total (+ total (1+ number))))))

(triangle-using-dotimes 4)
```

11.3 再帰

再帰関数とは、自分自身を評価するようなコードを含んでいるものである。関数が自分自身を評価した場合、それはまた自分自身を評価するコードに出くわす。結果としてその関数はまた自分自身を評価し . . . となってこれがずっと続く。再帰関数は自分自身止める条件を与えられない限り、永久に自分自身を繰り返し呼び出し続けることになる。

(未訳) Eventually, if the program is written correctly, the 'slightly different arguments' will become sufficiently different from the first arguments that the final instance will stop.

11.3.1 Building Robots: Extending the Metaphor

(未訳) It is sometimes helpful to think of a running program as a robot that does a job. In doing its job, a recursive function calls on a second robot to help it. The second robot is identical to the first in every way, except that the second robot helps the first and has been passed different arguments than the first.

(未訳) In a recursive function, the second robot may call a third; and the third may call a fourth, and so on. Each of these is a different entity; but all are clones.

(未訳) Since each robot has slightly different instructions—the arguments will differ from one robot to the next—the last robot should know when to stop.

(未訳) Let's expand on the metaphor in which a computer program is a robot.

(未訳) A function definition provides the blueprints for a robot. When you install a function definition, that is, when you evaluate a `defun` macro, you install the necessary equipment to build robots. It is as if you were in a factory, setting up an assembly line. Robots with the same name are built according to the same blueprints. So they have, as it were, the same 'model number', but a different 'serial number'.

(未訳) We often say that a recursive function 'calls itself'. What we mean is that the instructions in a recursive function cause the Lisp interpreter to run a different function that has the same name and does the same job as the first, but with different arguments.

(未訳) It is important that the arguments differ from one instance to the next; otherwise, the process will never stop.

11.3.2 The Parts of a Recursive Definition

再帰関数は、典型的には次のような三つの部分からなる条件分岐部を含んでいる。

1. この関数がもう一度呼び出されるかどうかを決定する真偽テスト。ここでは *do-again-test* と呼ぶ。
2. この関数の名前

3. 条件分岐部が正しい回数だけ繰り返しを行った後に偽を返すようにするためのS式。ここでは *next-step-expression* と呼ぶ。

再帰関数は、他の種類の関数に比べて最も簡単な形に書ける。実際、再帰関数を使い始めると、しばしば奇妙な程単純な形になってしまい、不可解な感じがすることが多いようである。再帰関数の定義を読むためには、ある種のコツが必要で、最初は難しく思えても、慣れると単純であることが分ってくる。初めて自転車に乗るときと同じである。

There are several different common recursive patterns. A very simple pattern looks like this:
(旧訳) 再帰関数のテンプレートは次のようになる。

```
(defun 再帰関数名 (変数リスト)
  "説明文字列..."
  本体...
  (if do-again-test
    (再帰関数名
         next-step-expression)))
```

Each time a recursive function is evaluated, a new instance of it is created and told what to do. The arguments tell the instance what to do.

引数が next-step-expression の値にバインドされ、そしてその値が do-again-test で使われる。

next-step-expression は関数をもう繰り返す必要がなくなった場合に do-again-test が偽を返すように設計されている。

The value returned by the next-step-expression is passed to the new instance of the function, which evaluates it (or some transmogrification of it) to determine whether to continue or stop. The next-step-expression is designed so that the do-again-test returns false when the function should no longer be repeated.

The do-again-test is sometimes called the *stop condition*, since it stops the repetitions when it tests false.

do-again-test は 停止条件 (*stop-condition*) と呼ばれることもある。これはテストが偽の場合に繰り返しが止まるからである。

11.3.3 List を使った再帰

先に挙げた、 `while` ループを使ったリストの要素を表示する関数の例は、再帰的に書くことも可能である。次に、変数 `animals` の値をあるリストにセットするS式も含めて、そのコードを書いてみることにしよう。

この例は `*scratch*` バッファにコピーして、各々のS式をそのバッファで評価してやらなければならない。そして `(print-elements-recursively animals)` を評価する際は *C-u C-x C-e* を使う必要がある。でないと、Lisp インタプリタは結果をエコー領域に一行分だけしか表示してくれない。

また、カーソルを `print-elements-recursively` 関数の最後の閉じ括弧の直後の、コメントの手前の位置に持っていって評価しないといけない。そうしないと、Lisp インタプリタはコメントまで評価しようとしてしまう。

```
(setq animals '(giraffe gazelle lion tiger))

(defun print-elements-recursively (list)
  "Print each element of LIST on a line of its own.
Uses recursion."
  (print (car list))                  ; 本体
  (if list                            ; do-again-test
      (print-elements-recursively     ; 再帰呼び出し
       (cdr list))))                  ; next-step-expression

(print-elements-recursively animals)
```

(旧訳) `print-elements-recursively` 関数は最初にリストの一番目の引数、即ち、リストの CAR を表示する。そしてもしリストが空でなければ、この関数内で自分自身を呼び出す。ただし引数としては、全体のリストではなく二番目以降の要素からなるリスト、即ち、そのリストの CDR を渡す。

Put another way, if the list is not empty, the function invokes another instance of code that is similar to the initial code, but is a different thread of execution, with different arguments than the first instance.

Put in yet another way, if the list is not empty, the first robot assembles a second robot and tells it what to do; the second robot is a different individual from the first, but is the same model.

この時の評価では、この関数は引数として受け取ったリストの最初の要素 (これは元々のリストでは二番目の要素であたる) を表示する。そして `if` 式が評価され、それが真であれば、この関数は自分自身を今回受け取った引数の CDR (これは、元々のリストの CDR の CDR である) を引数として再度自分自身を呼び出す。

Note that although we say that the function 'calls itself', what we mean is that the Lisp interpreter assembles and instructs a new instance of the program. The new instance is a clone of the first, but is a separate individual.

関数が自分自身を呼び出す度に、引数として渡されるリストは元々のリストに比べて短くなっていき、結果として最後には空リストとともに呼び出すことになる。`print` 関数は空リストを `nil` として表示する。そして次に条件分岐の部分が `list` の値をテストする。`list` は `nil` なので、`if` 式は偽を返し、もはや then-part は実行しない。関数全体としては `nil` が返されるので、この関数を評価すると最後に二回 `nil` が現れる。

Eventually, the function invokes itself on an empty list. It creates a new instance whose argument is `nil`. The conditional expression tests the value of `list`. Since the value of `list` is `nil`, the `when` expression tests false so the then-part is not evaluated. The function as a whole then returns `nil`.

`(print-elements-recursively animals)` を `*scratch*` バッファで評価すると、次のような結果が表示されるはずだ。

```
gazelle

giraffe

lion

tiger

nil
nil
```

(最初の `nil` は空リストの値が表示されたものであり、二番目の `nil` は関数全体の値である。)

11.3.4 カウンタの代わりに再帰を使う

前節で説明した `triangle` 関数もまた再帰的に書ける。これは次のようになる。

```
(defun triangle-recursively (number)
  "Return the sum of the numbers 1 through NUMBER inclusive.
Uses recursion."
  (if (= number 1)                    ; do-again-test
      1                               ; then-part
    (+ number                         ; else-part
       (triangle-recursively          ; 再帰呼び出し
        (1- number)))))               ; next-step-expression

(triangle-recursively 7)
```

これを評価することでこの関数をインストールし、試しに `(triangle-recursively 7)` を評価してみよう。(カーソルを関数定義の直後の、コメントの手前の位置に持っていって評価することを忘れずに。) The function evaluates to 28.

この関数がどのように動作するかを確かめるため、この関数に 1、2、3、4 等の様々な引数を与えた場合に何が起きるかを考えてみよう。

まず引数の値が 1 だとどうなるか？

この関数では説明文字列の後に `if` 式がくる。これは `number` の値が 1 かどうかテストするものである。もし 1 であれば、Emacs は `if` 式の then-part を評価する。この場合は関数の値として 1 が返される。(一行からなる三角形の中には小石は 1 つしかない。)

では引数の値が 2 であればどうだろう。この場合は Emacs は `if` 式の else-part を評価する。

今の場合、else-part は足し算と `triangle-recursively` の再帰呼び出し、そしてデクリメントからなっている。具体的には次の通り。

```
(+ number (triangle-recursively (1- number)))
```

Emacs がこの S 式を評価する時は、まず最も内側の S 式から評価していき、順に他の部分を評価していく。詳しく書くと次のようなステップを踏むことになる。

Step 1 最も内側の S 式の評価。
今の場合、最も内側の S 式は `(1- number)` なので、Emacs は `number` を 2 から 1 に減らす。

Step 2 `triangle-recursively` 関数の評価。
（旧訳）この関数が自分自身の内部に含まれていることとは関係なく、Emacs は Step 1 の結果をこの関数の引数として渡す。

The Lisp interpreter creates an individual instance of `triangle-recursively`. It does not matter that this function is contained within itself. Emacs passes the result Step 1 as the argument used by this instance of the `triangle-recursively` function

今の場合、Emacs は `triangle-recursively` を引数 1 とともに評価する。さっき見たように、この場合この関数は 1 を返す。

Step 3 `number` の値の評価。
ここでいう変数 `number` は `+` で始まるリストの二番目の要素。その値は 2 である。

Step 4 `+` 式の評価。
`+` 式は二つの引数を受け取る。一つ目は `number` を評価して返された値 (Step 3) であり、二つ目は `triangle-recursively` を評価して返された値 (Step 2) である。
足し算の結果は 2 と 1 の和であり、3 が返される。これは正しい結果である。二行からなる三角形の中には小石は 3 個含まれる。

引数 **3** か **4** の場合

`triangle-recursively` が引数 3 とともに呼び出されたとする。

Step 1 *do-again-test* の評価。
まずは `if` 式が評価される。これは do-again-test であり、偽が返される。従って、`if` 式の else-part が評価される。(この例では、テストの結果が真の時ではなく偽の時に自分自身を再帰呼び出しすることに注意しよう。)

Step 2 *else-part* のもっとも内側の S 式の評価。
else-part の最も内側の S 式が評価され、3 が 2 にデクリメントされる。これが next-step-expression である。

Step 3 `triangle-recursively` 関数の評価。
数値 2 が `triangle-recursively` 関数に渡される。
前節で説明した通り、`triangle-recursively` は引数 2 とともに評価されると 3 を返すのであった。

Step 4 足し算の評価。
足し算の式では 3 がこの時の `number` の値 3 に加えられる。

全体として、この関数が返す値は 6 になる。

以上で `triangle-recursively` に引数 3 を与えるとどうなるかが分った。もはや引数が 4 の場合に何が起きるかは明らかであろう。次のような感じだ。

再帰呼び出しで

```
(triangle-recursively (1- 4))
```

が評価され、結果として

```
(triangle-recursively 3)
```

の値を返す。これは 6 であり、この値に三行目の足し算で 4 が加えられる。
全体としてこの関数が返す値は 10 になる。

（旧訳）`triangle-recursively` が評価されるごとに、より小さい引数とともに自分自身を評価し、その状況が、引数がもはや再帰呼び出しを起こさない程小さくなるまで続けられるというわけである。

Each time `triangle-recursively` is evaluated, it evaluates a version of itself—a different instance of itself—with a smaller argument, until the argument is small enough so that it does not evaluate itself.

Note that this particular design for a recursive function requires that operations be deferred.

Before `(triangle-recursively 7)` can calculate its answer, it must call `(triangle-recursively 6)`; and before `(triangle-recursively 6)` can calculate its answer, it must call `(triangle-recursively 5)`; and so on. That is to say, the calculation that `(triangle-recursively 7)` makes must be deferred until `(triangle-recursively 6)` makes its calculation; and `(triangle-recursively 6)` must defer until `(triangle-recursively 5)` completes; and so on.

If each of these instances of `triangle-recursively` are thought of as different robots, the first robot must wait for the second to complete its job, which must wait until the third completes, and so on.

There is a way around this kind of waiting, which we will discuss in Section 11.3.7 “Recursion without Deferments”, page 97.

11.3.5 `cond` を使った再帰の例

以前説明したバージョンの `triangle-recursively` は特殊形式 `if` を用いて書かれていた。これは `cond` と呼ばれる特殊形式を用いても書くことが出来る。特殊形式 `cond` の名前は `conditional` という単語の短縮形から来ている。

特殊形式 `cond` は Emacs Lisp では `if` ほど頻繁に使われているとは言えないが、ここで説明する価値がある程度には使われている。

`cond` 式のテンプレートは次の通りである。

```
(cond
 本体...)
```

ここで 本体 はリストの列である。

本体の中身をもっと詳しく書くと次のような感じになる。

```
(cond
 ((最初の真偽テスト 最初の結果部)
  (二番目の 二番目の結果部)
  (三番目の 三番目の結果部)
  ...)
```

Lisp インタプリタが `cond` 式を評価する時は、まず `cond` の本体の S 式の列の最初の S 式の最初の要素 (CAR つまり真偽テストの部分) から評価する。

もし、真偽テストが `nil` を返したなら、その式の残りの部分 (これを結果部 (consequent) と呼ぼう) はスキップされて、次の S 式の真偽テストが評価される。こうして、もしある S 式で真偽テストが `nil` 以外の値を返したなら、その S 式の結果部が評価される。結果部は一つでも複数でも構わない。複数の場合は各々の式が順に評価されていき、最後の値が返される。もしその S 式が結果部を持たなければ、真偽テストの結果が返される。(訳註：そして、真偽テストが真であった S 式以降は無視される。)

どの S 式の真偽テストも偽を返した場合は `cond` 式は `nil` を返す。

`cond` を使って書くと、`triangle` 関数は次のようになる。

```
(defun triangle-using-cond (number)
  (cond ((<= number 0) 0)
        ((= number 1) 1)
        ((> number 1)
         (+ number (triangle-using-cond (1- number))))))
```

この例では、`cond` は number が 0 以下の場合は 0 を返し、1 の場合は 1 を返し、1 より大きい場合は `(+ number (triangle-using-cond (1- number)))` が評価される。

11.3.6 Recursive Patterns

Here are three common recursive patterns. Each involves a list. Recursion does not need to involve lists, but Lisp is designed for lists and this provides a sense of its primal capabilities.

Recursive Pattern: *every*

In the `every` recursive pattern, an action is performed on every element of a list.

The basic pattern is:

- If a list be empty, return `nil`.
- Else, act on the beginning of the list (the CAR of the list)
  - through a recursive call by the function on the rest (the CDR) of the list,
  - and, optionally, combine the acted-on element, using `cons`, with the results of acting on the rest.

Here is example:

```
(defun square-each (numbers-list)
  "Square each of a NUMBERS LIST, recursively."
  (if (not numbers-list)                ; do-again-test
      nil
    (cons
     (* (car numbers-list) (car numbers-list))
     (square-each (cdr numbers-list))))) ; next-step-expression

(square-each '(1 2 3))
    ⇒ (1 4 9)
```

If `numbers-list` is empty, do nothing. But if it has content, construct a list combining the square of the first number in the list with the result of the recursive call.

(The example follows the pattern exactly: `nil` is returned if the numbers' list is empty. In practice, you would write the conditional so it carries out the action when the numbers' list is not empty.)

The `print-elements-recursively` function (see Section 11.3.3 "Recursion with a List", page 92) is another example of an `every` pattern, except in this case, rather than bring the results together using `cons`, we print each element of output.

The `print-elements-recursively` function looks like this:

```
(setq animals '(gazelle giraffe lion tiger))

(defun print-elements-recursively (list)
  "Print each element of LIST on a line of its own.
Uses recursion."
  (when list                            ; do-again-test
        (print (car list))              ; body
        (print-elements-recursively     ; recursive call
         (cdr list))))                  ; next-step-expression

(print-elements-recursively animals)
```

The pattern for `print-elements-recursively` is:

- When the list is empty, do nothing.
- But when the list has at least one element,
  - act on the beginning of the list (the CAR of the list),
  - and make a recursive call on the rest (the CDR) of the list.

Recursive Pattern: *accumulate*

Another recursive pattern is called the `accumulate` pattern. In the `accumulate` recursive pattern, an action is performed on every element of a list and the result of that action is accumulated with the results of performing the action on the other elements.

This is very like the 'every' pattern using `cons`, except that `cons` is not used, but some other combiner.

The pattern is:

- If a list be empty, return zero or some other constant.
- Else, act on the beginning of the list (the CAR of the list),
  - and combine that acted-on element, using `+` or some other combining function, with
  - a recursive call by the function on the rest (the CDR) of the list.

Here is an example:

```
(defun add-elements (numbers-list)
  "Add the elements of NUMBERS-LIST together."
  (if (not numbers-list)
      0
    (+ (car numbers-list) (add-elements (cdr numbers-list)))))

(add-elements '(1 2 3 4))
    ⇒ 10
```

See Section 14.9.2 “Making a List of Files”, page 130, for an example of the accumulate pattern.

Recursive Pattern: *keep*

A third recursive pattern is called the `keep` pattern. In the `keep` recursive pattern, each element of a list is tested; the element is acted on and the results are kept only if the element meets a criterion.

Again, this is very like the ‘every’ pattern, except the element is skipped unless it meets a criterion.

The pattern has three parts:

- If a list be empty, return `nil`.
- Else, if the beginning of the list (the CAR of the list) passes a test
  - act on that element and combine it, using `cons` with
  - a recursive call by the function on the rest (the CDR) of the list.
- Otherwise, if the beginning of the list (the CAR of the list) fails the test
  - skip on that element,
  - and, recursively call the function on the rest (the CDR) of the list.

Here is an example that uses `cond`:

```
(defun keep-three-letter-words (word-list)
  "Keep three letter words in WORD-LIST."
  (cond
   ;; First do-again-test: stop-condition
   ((not word-list) nil)

   ;; Second do-again-test: when to act
   ((eq 3 (length (symbol-name (car word-list))))
    ;; combine acted-on element with recursive call on shorter list
    (cons (car word-list) (keep-three-letter-words (cdr word-list))))

   ;; Third do-again-test: when to skip element;
   ;;   recursively call shorter list with next-step expression
   (t (keep-three-letter-words (cdr word-list)))))

(keep-three-letter-words '(one two three four five six))
    ⇒ (one two six)
```

It goes without saying that you need not use `nil` as the test for when to stop; and you can, of course, combine these patterns.

11.3.7 Recursion without Deferments

Let’s consider again what happens with the `triangle-recursively` function. We will find that the intermediate calculations are deferred until all can be done.

Here is the function definition:

```
(defun triangle-recursively (number)
  "Return the sum of the numbers 1 through NUMBER inclusive.
Uses recursion."
  (if (= number 1)                    ; do-again-test
      1                               ; then-part
    (+ number                         ; else-part
       (triangle-recursively          ; recursive call
        (1- number)))))               ; next-step-expression
```

What happens when we call this function with a argument of 7?

The first instance of the `triangle-recursively` function adds the number 7 to the value returned by a second instance of `triangle-recursively`, an instance that has been passed an argument of 6. That is to say, the first calculation is:

```
(+ 7 (triangle-recursively 6))
```

The first instance of `triangle-recursively`—you may want to think of it as a little robot—cannot complete its job. It must hand off the calculation for `(triangle-recursively 6)` to a second instance of the program, to a second robot. This second individual is completely different from the first one; it is, in the jargon, a 'different instantiation'. Or, put another way, it is a different robot. It is the same model as the first; it calculates triangle numbers recursively; but it has a different serial number.

And what does `(triangle-recursively 6)` return? It returns the number 6 added to the value returned by evaluating `triangle-recursively` with an argument of 5. Using the robot metaphor, it asks yet another robot to help it.

Now the total is:

```
(+ 7 6 (triangle-recursively 5))
```

And what happens next?

```
(+ 7 6 5 (triangle-recursively 4))
```

Each time `triangle-recursively` is called, except for the last time, it creates another instance of the program—another robot—and asks it to make a calculation.

Eventually, the full addition is set up and performed:

```
(+ 7 6 5 4 3 2 1)
```

This design for the function defers the calculation of the first step until the second can be done, and defers that until the third can be done, and so on. Each deferment means the computer must remember what is being waited on. This is not a problem when there are only a few steps, as in this example. But it can be a problem when there are more steps.

11.3.8 No Deferment Solution

The solution to the problem of deferred operations is to write in a manner that does not defer operations[2]. This requires writing to a different pattern, often one that involves writing two function definitions, an 'initialization' function and a 'helper' function.

The 'initialization' function sets up the job; the 'helper' function does the work.

Here are the two function definitions for adding up numbers. They are so simple, I find them hard to understand.

```
(defun triangle-initialization (number)
  "Return the sum of the numbers 1 through NUMBER inclusive.
This is the 'initialization' component of a two function
duo that uses recursion."
  (triangle-recursive-helper 0 0 number))
```

[2] The phrase *tail recursive* is used to describe such a process, one that uses 'constant space'.

```
(defun triangle-recursive-helper (sum counter number)
  "Return SUM, using COUNTER, through NUMBER inclusive.
This is the 'helper' component of a two function duo
that uses recursion."
  (if (> counter number)
      sum
    (triangle-recursive-helper (+ sum counter)  ; sum
                               (1+ counter)     ; counter
                               number)))        ; number
```

Install both function definitions by evaluating them, then call `triangle-initialization` with 2 rows:

```
(triangle-initialization 2)
    ⇒ 3
```

The 'initialization' function calls the first instance of the 'helper' function with three arguments: zero, zero, and a number which is the number of rows in the triangle.

The first two arguments passed to the 'helper' function are initialization values. These values are changed when `triangle-recursive-helper` invokes new instances.[3]

Let's see what happens when we have a triangle that has one row. (This triangle will have one pebble in it!)

`triangle-initialization` will call its helper with the arguments `0 0 1`. That function will run the conditional test whether `(> counter number)`:

```
(> 0 1)
```

and find that the result is false, so it will invoke the else-part of the `if` clause:

```
(triangle-recursive-helper
 (+ sum counter)  ; sum plus counter ⇒ sum
 (1+ counter)     ; increment counter ⇒ counter
 number)          ; number stays the same
```

which will first compute:

```
(triangle-recursive-helper (+ 0 0)  ; sum
                           (1+ 0)   ; counter
                           1)       ; number
```

which is:

```
(triangle-recursive-helper 0 1 1)
```

Again, `(> counter number)` will be false, so again, the Lisp interpreter will evaluate `triangle-recursive-helper`, creating a new instance with new arguments.

This new instance will be;

```
(triangle-recursive-helper
 (+ sum counter)  ; sum plus counter ⇒ sum
 (1+ counter)     ; increment counter ⇒ counter
 number)          ; number stays the same
```

which is:

```
(triangle-recursive-helper 1 2 1)
```

In this case, the `(> counter number)` test will be true! So the instance will return the value of the sum, which will be 1, as expected.

Now, let's pass `triangle-initialization` an argument of 2, to find out how many pebbles there are in a triangle with two rows.

That function calls `(triangle-recursive-helper 0 0 2)`.

---

[3] The jargon is mildly confusing: `triangle-recursive-helper` uses a process that is iterative in a procedure that is recursive. The process is called iterative because the computer need only record the three values, `sum`, `counter`, and `number`; the procedure is recursive because the function 'calls itself'. On the other hand, both the process and the procedure used by `triangle-recursively` are called recursive. The word 'recursive' has different meanings in the two contexts.

In stages, the instances called will be:

```
                                sum counter number
(triangle-recursive-helper 0    1       2)

(triangle-recursive-helper 1    2       2)

(triangle-recursive-helper 3    3       2)
```

When the last instance is called, the `(> counter number)` test will be true, so the instance will return the value of `sum`, which will be 3.

This kind of pattern helps when you are writing functions that can use many resources in a computer.

11.4 ループについての練習問題

- 各々の行の値が行番号の自乗であるような場合に、`triangle` 関数と同様な関数を書きなさい。ただし、`while` ループを使うこと。
- `triangle` 関数と同様だが、各々の行の値を足すのでは無く、掛けていくような関数を書きなさい。
- 上の二つの関数を再帰的な関数に書き直しなさい。また、`cond` を使って書き直しなさい。
- Texinfo モードのために、パラグラフの中に含まれる全ての '`@dfn`' に対する索引の項目をそのパラグラフの最初に作成するような関数を書きなさい。(Texinfo ファイルでは、'`@dfn`' が定義の印になっている。本書は Texinfo で書かれている。)

  （未訳）Many of the functions you will need are described in two of the previous chapters, Chapter 8 “Cutting and Storing Text”, page 61, and Chapter 10 “Yanking Text Back”, page 81. If you use `forward-paragraph` to put the index entry at the beginning of the paragraph, you will have to use *`C-h f`* (`describe-function`) to find out how to make the command go backwards.

  （未訳）For more information, see “Indicating Definitions, Commands, etc.” in *Texinfo, The GNU Documentation Format*.

12 正規表現の検索

GNU Emacs の中では、正規表現の検索が徹底的に活用されている。例えば、forward-sentence とか forward-paragraph といった関数を調べてみれば、こういった検索についてよく理解出来るだろう。

正規表現の検索は Section “Regular Expression Search” in *The GNU Emacs Manual*, の中や Section “Regular Expressions” in *The GNU Emacs Lisp Reference Manual*, の中で説明されている。この章を書く際にも、私は読者が少なくともこれらをある程度は知っていることを想定している。大事な点は、正規表現を使うことで具体的な文字列そのものだけではなく、パターンをも検索出来るということである。例えば forward-sentence のコードは文 (sentence) の終わりを示すパターンを検索し、その場所にポイントを移動する。

実際に forward-sentence 関数のコードを見る前に、文の終わりを示すパターンがどんなものであるべきかを考えておいた方が良いだろう。このパターンについては次のセクションで議論することにする。その次に、正規表現の検索を行う関数である re-search-forward の説明をする。forward-sentence 関数の説明はその後である。この章の最後の節では、forward-paragraph 関数の説明をする。forward-paragraph は複雑な関数なので、幾つか新しい特徴を紹介することになる。

12.1 sentence-end の正規表現

シンボル sentence-end は文末 (訳註：ここでは勿論英語の文章を想定している。) を示すあるパターンにバインドされている。この正規表現はどうあるべきだろうか？

明らかに、文末には終止符か疑問符、もしくは感嘆符が来る。実際、この三つの文字の中の一つで終了する文節だけが文末と見做されるべきである。これはパターンの中に次の文字集合が含まれるということだ。

```
[.?!]
```

しかしながら、forward-sentence が単に終止符や疑問符、感嘆符に移動するというだけではまずい。というのも、これらの文字は文中にも使われることがあるからである。例えば終止符は略語の後にも使われる。従って、他の情報も必要になる。

慣習的に、普通文末には二つの空白を打つが、文中の終止符、疑問符、感嘆符のの後には一つの空白しか打たない。従って、終止符、疑問符、感嘆符に続いて二つの空白というのが文末の良い目印になるだろう。ただし、ファイルの中では二つの空白はタブや行末であっても良い。つまり、正規表現の中には、これら三つのどれかというものが含まれる。

この部分は次のように書ける。

```
\\($\\| \\|  \\)
       ^   ^^
      TAB  SPC
```

ここで ‘$’ は行末を表わす。それと、タブと二つの空白がこの表現の中に入っていることを指摘しておく。両方とも、この表現の中に実際の文字が挿入されている。

括弧や縦棒の前には二つのバックスラッシュ ‘\\’ が必要になる。最初のバックスラッシュは、Emacs の中でその後に続くバックスラッシュを quote するためのものであり、二番目のバックスラッシュは、その後に続く括弧や縦棒が特殊文字であることを示すものである。

また、文の後には次のように一つ以上の復帰コードが続くこともある。

```
[
]*
```

タブや空白と同様、復帰コードも実際のコードを埋め込むことで正規表現の中に挿入される。末尾のアスタリスクは、RET が零回以上繰り返すことを示す。

ただ、文末が必ずしも終止符や疑問符、ないしは感嘆符に続いて空白で終っているとは限らない。閉引用符や何らかの閉括弧が空白の前に来るかもしれない。実際の所、このような記号が空白の前に二つ以上続くこともある。これらのために、次のような表現が必要になる。

```
[]\"')}]*
```

この表現の中で、最初の ‘]’ が最初に来ていることに注意しよう。(訳註：‘[’ と ‘]’ で狭んで定める文字集合の中に ‘]’ を含めるには、このように文字集合の最初に ‘]’ を記述する。) また、二番目の文字は ‘"’ である。前の ‘\’ は Emacs にこれが特殊文字ではなく文字列の一部だと伝えるためのものである。残りの三文字は、‘’’、‘)’、‘}’ そのものを表わす。これらの文字が零回以上現れるということになる。

これら全てを合せたものが、あるべき文末の正規表現を形成している。そして、実際に sentence-end を評価してみると、次のような値が返される。

```
sentence-end
     ⇒ "[.?!][]\"')}]*\\($\\|     \\|  \\)[
]*"
```

(未訳) (Well, not in GNU Emacs 22; that is because of an effort to make the process simpler and to handle more glyphs and languages. When the value of sentence-end is nil, then use the value defined by the function sentence-end. (Here is a use of the difference between a value and a function in Emacs Lisp.) The function returns a value constructed from the variables sentence-end-base, sentence-end-double-space, sentence-end-without-period, and sentence-end-without-space. The critical variable is sentence-end-base; its global value is similar to the one described above but it also contains two additional quotation marks. These have differing degrees of curliness. The sentence-end-without-period variable, when true, tells Emacs that a sentence may end without a period, such as text in Thai.)

12.2 関数 re-search-forward

re-search-forward 関数は、search-forward 関数と非常によく似ている。(後者については、Section 8.1.3 “関数 search-forward”, page 63, 参照。)

re-search-forward は正規表現を検索するためのものである。もし検索が成功すれば、ただちに目的とする文字の後にポイントを移動する。後方検索の場合は目的の文字の直後に移動する。検索成功時には re-search-forward は t を返す。(訳註：Emacs version 19 ではポイントの位置を返す。) (従って、ポイントの移動は「副作用」である。)

search-forward と同じく re-search-forward 関数も四つの引数を持つ。

1. 最初の引数は、検索する正規表現である。正規表現は引用符に囲まれた文字列でなければならない。
2. 二番目の引数は省略可能であり、関数が検索する範囲を制限するために用いる。これはバッファの中の位置として指定される。
3. 三番目の引数も省略可能で、検索に失敗した場合の挙動を決めるためのものである。もし引数が nil なら失敗時にはエラーが返され、メッセージが表示される。他の値の場合は失敗時には nil が返り、成功時には t が返される。(訳註：Emacs version 19 ではポイントの位置が返される。)
4. 四番目の引数は繰り返しの回数である。負の引数を与えると、後方検索になる。

re-search-forward のテンプレートは次の通りである。

```
(re-search-forward "正規表現"
                   検索範囲の限界
                   検索失敗時の動作
                   繰り返しの回数)
```

二番目から四番目までの引数は省略可能である。しかし、最後の二つの片方ないしは両方に値を渡したい場合は、それ以前の全ての引数を与えなければならない。そうしないと Lisp インタプリタはどの引数を何処へ渡すかを間違えてしまう。

forward-sentence 関数では、正規表現は変数 sentence-end の値である。つまり、次の通りである。

```
"[.?!][]\"')}]*\\($\\|  \\|  \\)[
]*"
```

検索の限界はパラグラフの終わりまでである (文がパラグラフを越えて続くことはないので)。検索に失敗した場合は nil が返される。また、繰り返しの回数は forward-sentence の引数として与えられる。

12.3 forward-sentence

カーソルを文の前方に移動するコマンドは、Emacs Lisp での正規表現検索の使い方をストレートに説明してくれる。この関数は、実際以上に長くて複雑そうに見えるが、それは、前方に検索するだけではなく後方にも検索出来るようになっていたり、オプションとして複数の文を移動することが出来るようになっているためである。この関数は通常は *M-e* というキーにバインドされている。

以下が forward-sentence のコードである。

```
(defun forward-sentence (&optional arg)
  "Move forward to next `sentence-end'.  With argument, repeat.
With negative argument, move backward repeatedly to `sentence-beginning'.

The variable `sentence-end' is a regular expression that matches ends of
sentences.  Also, every paragraph boundary terminates sentences as well."
  (interactive "p")
  (or arg (setq arg 1))
  (let ((opoint (point))
        (sentence-end (sentence-end)))
    (while (< arg 0)
      (let ((pos (point))
            (par-beg (save-excursion (start-of-paragraph-text) (point))))
       (if (and (re-search-backward sentence-end par-beg t)
                (or (< (match-end 0) pos)
                    (re-search-backward sentence-end par-beg t)))
           (goto-char (match-end 0))
         (goto-char par-beg)))
      (setq arg (1+ arg)))
    (while (> arg 0)
      (let ((par-end (save-excursion (end-of-paragraph-text) (point))))
       (if (re-search-forward sentence-end par-end t)
           (skip-chars-backward " \t\n")
         (goto-char par-end)))
      (setq arg (1- arg)))
    (constrain-to-field nil opoint t)))
```

ぱっと見ただけだと、この関数は長く感じてしまう。まずは骨組みを見て、それから肉の部分を見ていくのが賢明だろう。骨組みを見るには、桁が左にあるものから見て行けば良い。

```
(defun forward-sentence (&optional arg)
  "説明文字列..."
  (interactive "p")
  (or arg (setq arg 1))
  (let ((opoint (point)) (sentence-end (sentence-end)))
    (while (< arg 0)
      (let ((pos (point))
            (par-beg (save-excursion (start-of-paragraph-text) (point))))
       rest-of-body-of-while-loop-when-going-backwards
    (while (> arg 0)
      (let ((par-end (save-excursion (end-of-paragraph-text) (point))))
       rest-of-body-of-while-loop-when-going-forwards
    handle-forms-and-equivalent
```

こう書き直すとぐっと解りやすくなる。この関数定義は説明文字列と interactive 式、そして or 式と while ループからなっているのである。

では順に各々の部分を見ていくことにしよう。

まず、説明文字列が過不足なくかつ理解しやすく書かれていることに注意しよう。

この関数は interactive "p" 宣言を持っている。これは、(もし与えられたなら) 処理された前置引数が引数としてこの関数に渡されることを意味する。(これは数値である。) もしこの関数が引数を渡されなければ (引数は省略可能である) その場合、引数 arg は 1 にバインドされる。また、forward-sentence が非インタラクティブに引数無しで呼ばれた場合には、arg は nil にバインドされる。

前置引数を扱うのは or 式である。この式では、arg の値がある値にバインドされている場合はそのままにしておき、もし nil にバインドされていた場合は 1 にセットしている。

When forward-sentence is called non-interactively without an argument, arg is bound to nil. The or expression handles this. What it does is either leave the value of arg as it is, but only if arg is bound to a value; or it sets the value of arg to 1, in the case when arg is bound to nil.

Next is a let. That specifies the values of two local variables, point and sentence-end. The local value of point, from before the search, is used in the constrain-to-field function which handles forms and equivalents. The sentence-end variable is set by the

while ループ

or 式の後には、二つの while ループが続く。最初の while ループには、前置引数が負の値ならば真を返すような真偽テストが含まれている。これは後方検索のためのものである。このループの本体は二番目の while ループの本体とそっくりであるが全く同じではない。取り敢えずこちらの方はとばして、二番目のループの方に集中することにしよう。

二番目の while ループはポイントを前方に移動するものである。骨組みは次の通りである。

```
(while (> arg 0)            ; 真偽テスト
  (let 変数リスト
    (if (真偽テスト)
        then-part
      else-part
  (setq arg (1- arg))))     ; while ループのデクリメンタ
```

この while ループはデクリメントタイプの物である (Section 11.1.4 “減少カウンタを使ったループ”, page 87, 参照)。この中にはカウンタ (今の場合は変数 arg) が零よりも大きい間は真を返すような真偽テストが含まれている。そして、ループを繰り返すごとにカウンタの値を 1 減らすようなデクリメンタが含まれている。

もし forward-sentence に前置引数が与えられなかったなら、といってもこれが普通の使い方だが、その場合 while ループは一度だけ繰り返す。これは arg の値が 1 だからである。

while ループの本体は let 式からなる。これは局所変数を作成する。また、その本体として if 式を持っている。

while ループの本体は次のようである。

```
(let ((par-end
       (save-excursion (end-of-paragraph-text) (point))))
  (if (re-search-forward sentence-end par-end t)
      (skip-chars-backward " \t\n")
    (goto-char par-end)))
```

ここで let 式は局所変数 par-end を生成、バインドしている。この後見るように、この局所変数は正規表現検索に限界ないしは制限を与えるために用いられている。もしこの検索でパラグラフ内に適切な文末が見つからなければパラグラフの終端で検索をやめる。

が、その前にまずどうやって par-end がパラグラフの終端の値にバインドされるかを見てみよう。ここでは let 式を使って、次の S 式を Lisp インタプリタが評価した際に返される値に par-end の値をセットしている。

```
(save-excursion (end-of-paragraph-text) (point))
```

この式では (end-of-paragraph-text) によってポイントがパラグラフ終端に移動し、(point) によってそのポイントの値が返される。そして、save-excursion によって元の位置にポイントが戻されるというわけである。このようにして let は par-end に save-excursion 式が返す値、つまりパラグラフの終端の位置をバインドする。(end-of-paragraph-text 関数は forward-paragraph を使っている。これについては後で簡単に触れる。)

Emacs は次に let 式の本体を評価する。これは次のような if 式である。

```
(if (re-search-forward sentence-end par-end t) ; if-part
    (skip-chars-backward " \t\n")              ; then-part
  (goto-char par-end)))                        ; else-part
```

if は最初の引数が真かどうかテストし、もし真なら then-part を評価し、そうでなければ else-part を評価する。今の場合 if の真偽テストは正規表現検索である。

forward-sentence のような関数の実際の動作は奇妙に感じられるかもしれない。しかし Lisp ではこのような操作が行われるのは、ごく一般的なことである。

正規表現の検索

re-search-forward 関数は文末、つまり正規表現 sentence-end で定義されたパターンを検索する。もしパターンが見つかったなら—即ち文末が見つかったなら—その時は re-search-forward 関数は二つのことを行う。

1. re-search-forward 関数は副作用を実行する。即ち、ポイントを見つけた文末まで移動する。
2. re-search-forward 関数は真の値を返す。これは if によって返される値であり、検索が成功したことを意味する。

副作用としてのポイントの移動は if 関数が検索成功の結果として値を返すよりも前の時点に行われる。

if 関数が検索に成功した re-search-forward から呼び出されて真の値を返す際には if は then-part、即ち (skip-chars-backward "\t\n") の評価も行う。この S 式はタブや改行などを含む全ての種類の空白文字を越えて、表示される文字 (printed character) の所まで前に戻り、その文字の直後にポイントを置く。ポイントは既に文末のパターンの所まで移動しているので、この動作で文が目に見える文字で終わっている部分の直後に来ることになる。通常はピリオドだろう。

一方、もし re-search-forward 関数が文末パターンを見つけられなかった場合には、関数は偽を返す。この場合は if は三番目の引数を評価する。これは (goto-char par-end) である。これはパラグラフの終わりにまでポイントを移動する関数である。

正規表現検索は極めて便利なものであり、forward-sentence—その中では検索は if 式のテストになっている—で説明されたパターンは手軽に使えるものである。あなたもこのパターンを取り入れたコードを見たり書いたりするであろう。

12.4 forward-paragraph：関数の金脈

forward-paragraph 関数はポイントをパラグラフの終わりまで移動する。これは通常 *M-}* にバインドされており、例えば let*、match-beginning、looking-at のようなそれ自身も重要であるような他の幾つかの関数を利用している。

forward-paragraph 関数の定義は forward-sentence 関数の定義に比べてかなり長い。これは各々の行が fill-prefix (行詰め接頭辞) で始まるようなパラグラフも相手にしなければならないためである。

Fill prefix は 各々の行の先頭に繰り返し現れる文字列からなる。例えば Lisp コードでは便宜上パラグラフの各々の行が ‘;;; ’ から始まる。またテキストモードでは四つの空白文字インデントされたパラグラフの fill prefix としてよく使われる。(Fill prefix についてのより詳しい情報は Section “Fill Prefix” in *The GNU Emacs Manual*, を参照せよ。)

Fill prefix があるということは、forward-paragraph 関数は、中の行が左端から始まっているようなパラグラフの終わりを見つけるだけではなく、そのバッファの全て、ないしは多くの行がある fill prefix で始まっているよう場合にもパラグラフの終わりを見つけなければならないということを意味する。

更に、時には fill prefix があっても無視した方が良い場合だってある。特に空行でパラグラフが区切られているような場合などがそうだ。これにより、更に複雑さが増す。

ここでは forward-paragraph 関数を全て書き出すのはやめて、その中の一部だけを見ることにする。準備なしに読もうとすると、ちょっと臆してしまうかもしれない。

この関数のアウトラインは次のようである。

```
(defun forward-paragraph (&optional arg)
  "documentation..."
  (interactive "p")
  (or arg (setq arg 1))
  (let*
      変数リスト
    (while (and (< arg 0) (not (bobp)))        ; 後方に戻る場合のコード
      ...
    (while (and (> arg 0) (not (eobp)))        ; 前方に進む場合のコード
      ...
```

関数の最初の部分はいつもの通りである。引数のリストには一つ省略可能な引数があるだけである。次に説明文字列が続く。

インタラクティブ宣言の中の小文字の ‘p’ はもし前置引数があれば、それを処理してから関数に渡すことを意味する。これは数値であり、いくつ分のパラグラフを移動するかを表わす。次の行の or 式は関数に一つも引数が与えられなかった場合を扱うための物である。これはこの関数がインタラクティブではなく他のコードから呼び出された場合に起きる。これについては以前説明した。(Section 12.3 “forward-sentence”, page 102, 参照。) ここまでは、今まで慣れ親しんできた部分である。

let* 式

forward-paragraph の次の行は let* 式で始まる。これは以前に出てた式とは異なる。このシンボルは let* であって let ではない。

特殊形式 let* は基本的に let と同じなのだが、Emacs が変数を順にセットしていくため、変数リストの中で後に出てくる変数がそれ以前に出てきた変数の値を参照することが出来る、という点のみが異なっている。

(Section 4.4.3 “save-excursion in append-to-buffer”, page 38.)

この関数の let* 式の中では Emacs は二つの変数 fill-prefix-regexp と paragraph-separate をバインドしているのだが、paragraph-separate がバインドされる値は fill-prefix-regexp の値に依存しているのである。

The variable parsep appears twice, first, to remove instances of ‘^’, and second, to handle fill prefixes.

The variable opoint is just the value of point. As you can guess, it is used in a constrain-to-field expression, just as in forward-sentence.

シンボル fill-prefix-regexp は次のリストを評価した値にセットされる。

```
(and fill-prefix
     (not (equal fill-prefix ""))
     (not paragraph-ignore-fill-prefix)
     (regexp-quote fill-prefix))
```

これは最初の要素が関数 and であるような S 式である。

関数 and は各々の引数をそのどれかが nil を返すまで評価していく。どれかが nil を返した場合は and は nil を返す。しかし、もしどの引数も nil を返さなければ、最後の引数を評価して返された値を返す。(この場合、その値は nil ではないので、Lisp では真と見なされる。) 別の言い方をすれば、and は引数全てが真である場合にのみ真を返すわけである。

今の場合なら、fill-prefix-regexp は後に続く四つの S 式を評価して全て真 (即ち、非 nil) が返された場合にのみ、非 nil の値を返す。そうでない場合は fill-prefix-regexp は nil にバインドされる。

fill-prefix
: この変数を評価すると、もしあれば fill prefix の値が返される。fill prefix が無い場合は nil が返る。

(not (equal fill-prefix "")
: この S 式は fill prefix があった場合にそれが空文字列、つまり文字を一つも含まない文字列かどうかを判定する。空文字列は fill prefix としての役には立たない。

(not paragraph-ignore-fill-prefix)
: この S 式は、もし変数 paragraph-ignore-fill-prefix が t 等の真の値にセットされている場合に nil を返す。

(regexp-quote fill-prefix)
: これは and 関数の最後の引数になる。もし and の全ての引数が真であれば、and の値としてはこの S 式を評価して返された値が返されることになる。そしてそれが fill-prefix-regexp の値になる。

この and 式を評価して真が返された場合、fill-prefix-regexp は regexp-quote によって修正された fill-prefix の値にバインドされる。regexp-quote は、文字列を読み取り、それのみにマッチし、その他の文字列にはマッチしないような正規表現を返す。結局、fill prefix が存在する場合、fill-prefix-regexp はその fill prefix の値にちょうどマッチする値にセットされ、そうでなければ nil にセットされる。

The next two local variables in the let* expression are designed to remove instances of ‘^’ from parstart and parsep, the local variables which indicate the paragraph start and the paragraph separator. The next expression sets parsep again. That is to handle fill prefixes.

（旧訳）let* 式の二番目の局所変数は paragraph-separate である。これは次の S 式を評価して返された値にバインドされる。

This is the setting that requires the definition call let* rather than let. The true-or-false-test for the if depends on whether the variable fill-prefix-regexp evaluates to nil or some other value.

If fill-prefix-regexp does not have a value, Emacs evaluates the else-part of the if expression and binds parsep to its local value. (parsep is a regular expression that matches what separates paragraphs.)

(旧訳)もし fill-prefix-regexp が値を持たなかったなら、(訳註: つまり nil であれば) Emacs は if 式の else-part を評価し、paragraph-separate を現在の局所的な値にバインドする。(paragraph-separate はパラグラフの区切りにマッチする正規表現である。)

But if fill-prefix-regexp does have a value, Emacs evaluates the then-part of the if expression and binds parsep to a regular expression that includes the fill-prefix-regexp as part of the pattern.

(旧訳) しかし、もし fill-prefix-regexp が値を持てば、(訳註: 真の値を持てば) Emacs は if 式の then-part を評価し、paragraph-separate の値を fill-prefix-regexp をパターンの一部として含むような正規表現にバインドする。

Specifically, parsep is set to the original value of the paragraph separate regular expression concatenated with an alternative expression that consists of the fill-prefix-regexp followed by optional whitespace to the end of the line. The whitespace is defined by "[\t]*$".) The '\\|' defines this portion of the regexp as an alternative to parsep.

(旧訳) より詳しく言うと paragraph-separate は、元々の paragraph-separate の値を fill-prefix-regexp に空行を加えた表現を連結した値にセットされることになる。'^' は fill-prefix-regexp が行頭に来なければならないことを意味し、その後に空白が来ても良いことが、"[\t]*$" で定義されている。'\\|' は、この正規表現か元の paragraph-separate かどちらかがマッチしなければならないことを示すものである。

(未訳) According to a comment in the code, the next local variable, sp-parstart, is used for searching, and then the final two, start and found-start, are set to nil.

Now we get into the body of the let*. The first part of the body of the let* deals with the case when the function is given a negative argument and is therefore moving backwards. We will skip this section.

(旧訳) では let* 式の本体部分に入ろう。本体の最初の部分では、この関数に負の引数が与えられた場合、即ち後方に戻る場合を扱っている。この部分は省略することにする。

前方に移動する場合の while ループ

The second part of the body of the let* deals with forward motion. It is a while loop that repeats itself so long as the value of arg is greater than zero. In the most common use of the function, the value of the argument is 1, so the body of the while loop is evaluated exactly once, and the cursor moves forward one paragraph.

(旧訳) let* 式の本体の二番目の部分は前方に進む場合を扱っている。これは arg が零よりも大きい間は繰り返すような while ループである。この関数を使う場合、大抵この引数は 1 である。従って while ループの本体はちょうど一度だけ実行され、それによりカーソルはパラグラフ一つ分だけ移動する。

This part handles three situations: when point is between paragraphs, when there is a fill prefix and when there is no fill prefix.

while ループは次のような形をしている。

```
;; going forwards and not at the end of the buffer
(while (and (> arg 0) (not (eobp)))

  ;; パラグラフとパラグラフの間に居る場合
  ;; Move forward over separator lines...
  (while (and (not (eobp))
              (progn (move-to-left-margin) (not (eobp)))
              (looking-at parsep))
    (forward-line 1))
  ;;  This decrements the loop
  (unless (eobp) (setq arg (1- arg)))
  ;; ... and one more line.
  (forward-line 1)
```

```
(if fill-prefix-regexp
    ;; There is a fill prefix; it overrides parstart;
    ;; we go forward line by line
    (while (and (not (eobp))
                (progn (move-to-left-margin) (not (eobp)))
                (not (looking-at parsep))
                (looking-at fill-prefix-regexp))
      (forward-line 1))

  ;; There is no fill prefix;
  ;; we go forward character by character
  (while (and (re-search-forward sp-parstart nil 1)
              (progn (setq start (match-beginning 0))
                     (goto-char start)
                     (not (eobp)))
              (progn (move-to-left-margin)
                     (not (looking-at parsep)))
              (or (not (looking-at parstart))
                  (and use-hard-newlines
                       (not (get-text-property (1- start) 'hard)))))
    (forward-char 1))

  ;; and if there is no fill prefix and if we are not at the end,
  ;;     go to whatever was found in the regular expression search
  ;;     for sp-parstart
  (if (< (point) (point-max))
      (goto-char start))))
```

We can see that this is a decrementing counter `while` loop, using the expression `(setq arg (1- arg))` as the decrementer. That expression is not far from the `while`, but is hidden in another Lisp macro, an `unless` macro. Unless we are at the end of the buffer—that is what the `eobp` function determines; it is an abbreviation of '`End Of Buffer P`'—we decrease the value of `arg` by one.

（旧訳）分ることは、これは減少カウンタの `while` ループであり、デクリメンタとして `(setq (1- arg))` というS式を使っているということである。

（未訳）(If we are at the end of the buffer, we cannot go forward any more and the next loop of the `while` expression will test false since the test is an `and` with `(not (eobp))`. The `not` function means exactly as you expect; it is another name for `null`, a function that returns true when its argument is false.)

（未訳）Interestingly, the loop count is not decremented until we leave the space between paragraphs, unless we come to the end of buffer or stop seeing the local value of the paragraph separator.

（未訳）That second `while` also has a `(move-to-left-margin)` expression. The function is self-explanatory. It is inside a `progn` expression and not the last element of its body, so it is only invoked for its side effect, which is to move point to the left margin of the current line.

The `looking-at` function is also self-explanatory; it returns true if the text after point matches the regular expression given as its argument.

（旧訳）`looking-at` は、ポイントの後に続くテキストが `looking-at` の引数として与えられた正規表現にマッチする場合に真を返す関数である。

The rest of the body of the loop looks difficult at first, but makes sense as you come to understand it.

First consider what happens if there is a fill prefix:

```
(if fill-prefix-regexp
    ;; There is a fill prefix; it overrides parstart;
    ;; we go forward line by line
    (while (and (not (eobp))
                (progn (move-to-left-margin) (not (eobp)))
                (not (looking-at parsep))
                (looking-at fill-prefix-regexp))
      (forward-line 1))
```

This expression moves point forward line by line so long as four conditions are true:

1. ポイントはバッファの最後ではない。
2. We can move to the left margin of the text and are not at the end of the buffer.
3. ポイントに続くテキストはパラグラフの区切りではない。
4. ポイントに続くパターンは fill prefix を表わす正規表現である。

最後の条件はちょっと戸惑うかもしれないが、ポイントが forward-paragraph 関数によって既に行頭に移動していることを思い出せば納得出来るだろう。つまり、テキストに fill prefix があれば、looking-at 関数がそれを見ることになるのである。

fill prefix がない場合に何が起こるかを考えてみる。

```
(while (and (re-search-forward sp-parstart nil 1)
            (progn (setq start (match-beginning 0))
                   (goto-char start)
                   (not (eobp)))
            (progn (move-to-left-margin)
                   (not (looking-at parsep)))
            (or (not (looking-at parstart))
                (and use-hard-newlines
                     (not (get-text-property (1- start) 'hard)))))
  (forward-char 1))
```

This while loop has us searching forward for sp-parstart, which is the combination of possible whitespace with a the local value of the start of a paragraph or of a paragraph separator. (The latter two are within an expression starting \(?: so that they are not referenced by the match-beginning function.)

The two expressions,

```
(setq start (match-beginning 0))
(goto-char start)
```

mean go to the start of the text matched by the regular expression search.

The (match-beginning 0) expression is new. It returns a number specifying the location of the start of the text that was matched by the last search.

（旧訳）この中で、これまでで慣れ親しんでいない部分は match-beginning の使い方だけである。この関数自体も初めて出てくるものだ。match-beginning 関数は、最後に行った正規表現検索でマッチしたテキストの始まりの位置を表わす数値を返す関数である。

The match-beginning function is used here because of a characteristic of a forward search: a successful forward search, regardless of whether it is a plain search or a regular expression search, moves point to the end of the text that is found. In this case, a successful search moves point to the end of the pattern for sp-parstart.

（旧訳）ここで match-beginning 関数が使われているのは前方検索の性質のためである。つまり、前方検索が成功した場合、通常の検索か正規表現の検索かどうかにかかわらず、ポイントを見つけたテキストの位置にまで移動してしまう。今の場合なら、検索に成功した場合はポイントは paragraph-start のパターンの終わりにまで移動するのだが、これは現在のパラグラフの終わりではなく、次のパラグラフの始まりである。

However, we want to put point at the end of the current paragraph, not somewhere else. Indeed, since the search possibly includes the paragraph separator, point may end up at the beginning of the next one unless we use an expression that includes match-beginning.

（旧訳）しかしながら、我々の目的はポイントを次のパラグラフの先頭ではなく現在のパラグラフの終わりにまで移動することである。この二つの位置は、パラグラフとパラグラフの間に幾つかの空行がある場合などでは、当然異なる。

When given an argument of 0, match-beginning returns the position that is the start of the text matched by the most recent search. In this case, the most recent search looks for sp-parstart. The (match-beginning 0) expression returns the beginning position of that pattern, rather than the end position of that pattern.

（旧訳）引数 0 で呼ばれた場合、match-beginning は最も最近、正規表現の検索に成功した位置を返す。今の場合、最も最近の正規表現の検索は paragraph-start を探すものなので、match-beginning

はそのパターンの開始位置を返す。(終了位置ではない。) この開始位置は現在のパラグラフの最後である。

(Incidentally, when passed a positive number as an argument, the `match-beginning` function returns the location of point at that parenthesized expression in the last search unless that parenthesized expression begins with `\(?:`. I don't know why `\(?:` appears here since the argument is 0.)

(旧訳) (ついでにいうと、引数として正の数が渡された場合、`match-beginning` 関数は最後の正規表現の中の括弧でくくられた部分の表現の開始位置を返す。これは便利な機能である。)

The last expression when there is no fill prefix is

```
(if (< (point) (point-max))
    (goto-char start))))
```

This says that if there is no fill prefix and if we are not at the end, point should move to the beginning of whatever was found by the regular expression search for `sp-parstart`.

`forward-paragraph` 関数の完全な定義は、上に挙げた前方に進むコードだけでなく、後方に戻るコードも含んでいる。

もし、この文書を GNU Emacs の中で読んでいるなら、この関数全体のコードを見たい場合には *`C-h f`* (`describe-function`) とタイプし、プロンプトが出たら関数名をタイプすれば良い。(未訳) This gives you the function documentation and the name of the library containing the function's source. Place point over the name of the library and press the RET key; you will be taken directly to the source. (Be sure to install your sources! Without them, you are like a person who tries to drive a car with his eyes shut!)

12.5 自分自身の TAGS ファイルの作成

簡単にソースの場所までジャンプ出来るように、あなた自身の `TAGS` ファイルを作成することが出来る。例えば、もしあなたの `~/emacs` ディレクトリに沢山のファイルがあったとすると—何を隠そう、私もこの場所に 137 個の `.el` ファイルがあり、その内 17 個をロードしているのだが—そのディレクトリに `TAGS` ファイルを作ることで、`grep` やその他の道具で関数名を検索するよりはずっと簡単に、特定の関数の位置にジャンプすることが出来るようになる。

If the `find-tag` function first asks you for the name of a `TAGS` table, give it the name of a `TAGS` file such as `/usr/local/src/emacs/src/TAGS`. (The exact path to your `TAGS` file depends on how your copy of Emacs was installed. I just told you the location that provides both my C and my Emacs Lisp sources.)

`TAGS` ファイルは、Emacs の配布に含まれる `etags` プログラムを使って作成出来る。普通、`etags` は Emacs が構築された時に一緒にコンパイルされインストールされる。(ただし、`etags` は Emacs Lisp 関数や Emacs の一部分ではない。これは C のプログラムである。)

You can also create your own `TAGS` file for directories that lack one.

You often need to build and install tags tables yourself. They are not built automatically. A tags table is called a `TAGS` file; the name is in upper case letters.

You can create a `TAGS` file by calling the `etags` program that comes as a part of the Emacs distribution. Usually, `etags` is compiled and installed when Emacs is built. (`etags` is not an Emacs Lisp function or a part of Emacs; it is a C program.)

`TAGS` ファイルを作成するには、まずこのファイルを作成したいディレクトリに移動する。Emacs の中だと、*`M-x cd`* コマンドを使うか、そのディレクトリにあるファイルをビジットするか、あるいは *`C-x d`* (`dired`) を使うことで移動することが出来る。そして

```
M-x compile RET etags *.el RET
```

とタイプしてやれば良い。

For example, if you have a large number of files in your `~/emacs` directory, as I do—I have 137 `.el` files in it, of which I load 12—you can create a `TAGS` file for the Emacs Lisp files in that directory.

etags プログラムは普通のシェルで使える全てのワイルドカードを理解出来る。例えば、もし二つのディレクトリに対して一つの TAGS ファイルを作りたい場合は、二番目のディレクトリを../elisp/だとして、次のようにタイプすればよい。

```
M-x compile RET etags *.el ../elisp/*.el RET
```

また、etags が受け付けるオプションのリストを見たい場合には

```
M-x compile RET etags --help RET
```

とタイプする。

The etags program handles more than 20 languages, including Emacs Lisp, Common Lisp, Scheme, C, C++, Ada, Fortran, HTML, Java, LaTeX, Pascal, Perl, PostScript, Python, TeX, Texinfo, makefiles, and most assemblers. The program has no switches for specifying the language; it recognizes the language in an input file according to its file name and contents.

（旧訳）etags プログラムは Emacs Lisp, Common Lisp, Scheme, C, Fortran, Pascal, LaTeX, そして大抵のアセンブラを扱うことが出来る。このプログラムには言語を特定するためのスィッチはない。その代わりにファイル名や中身からそのファイルの言語を認識するのである。

また etags は、自分自身でコードを書いたり、既に書いた関数を後から参照したりするのにも大変便利である。新しい関数を書いたら、ときおり etags プログラムを走らせよう。そうすることでそれらの関数が TAGS ファイルに付け加わる。

（未訳）If you think an appropriate TAGS file already exists for what you want, but do not know where it is, you can use the locate program to attempt to find it.

（未訳）Type *M-x locate RET TAGS RET* and Emacs will list for you the full path names of all your TAGS files. On my system, this command lists 34 TAGS files. On the other hand, a 'plain vanilla' system I recently installed did not contain any TAGS files.

（未訳）If the tags table you want has been created, you can use the M-x visit-tags-table command to specify it. Otherwise, you will need to create the tag table yourself and then use M-x visit-tags-table.

Building Tags in the Emacs sources

（未訳）The GNU Emacs sources come with a Makefile that contains a sophisticated etags command that creates, collects, and merges tags tables from all over the Emacs sources and puts the information into one TAGS file in the src/ directory. (The src/ directory is below the top level of your Emacs directory.)

（未訳）To build this TAGS file, go to the top level of your Emacs source directory and run the compile command make tags:

```
M-x compile RET make tags RET
```

(The make tags command works well with the GNU Emacs sources, as well as with some other source packages.)

For more information, see Section "Tag Tables" in *The GNU Emacs Manual*.

12.6 復習

ここでは最近導入した関数の簡単なまとめを載せておく。

while　本体部分の S 式を、本体の最初の要素が真を返す間だけ繰り返し評価する。最後は nil を返す。(つまり、この式は副作用のためだけに評価される。)

例)

```
(let ((foo 2))
  (while (> foo 0)
    (insert (format "foo is %d.\n" foo))
    (setq foo (1- foo))))

 ⇒      foo is 2.
        foo is 1.
        nil
```

(insert 関数は引数をポイントに挿入する。また format 関数は引数を message 関数が整形するのと同様に整形して出来たストリングを返す。\n は改行である。

`re-search-forward`
: パターンを検索し、発見した場合はその直後にポイントを移動する。

  `search-forward` と同様、四つの引数を取る。

  1. 検索パターンを表わす正規表現。
  2. 省略可能。検索の限界。
  3. 省略可能。検索が失敗した場合にどうするか。`nil` を返すかエラーメッセージを出すか。
  4. 省略可能。検索を何回繰り返すか。負の数の場合は後方に検索する。

`let*`
: 幾つかの変数を特定の値に局所的にバインドし、残りの引数を評価する。値としては最後の引数の値を返す。局所変数をバインドする際、より先にバインドした変数があれば、その値を利用出来る。

  例)

```
(let* ((foo 7)
       (bar (* 3 foo)))
  (message "‘bar’ is %d." bar))
     ⇒ ‘bar’ is 21.
```

`match-beginning`
: 最後の正規表現で見つかったテキストの始まりの位置を返す。

`looking-at`
: ポイントに続くテキストが引数の正規表現にマッチした場合に真として `t` を返す。

`eobp`
: ポイントがそのバッファのアクセス可能な最大位置にある場合に真として `t` を返す。アクセス可能な範囲というのは、もしナローイングがかかっていない場合はバッファの最後の位置であり、ナローイングがかかっている場合は、その部分の最後である。

12.7 `re-search-forward` についての練習問題

- 二行以上続く空行にマッチする正規表現を検索する関数を書きなさい。
- ‘the the’ のように二度続く単語を検索する関数を書きなさい。二つの同じ部分からなる文字列にマッチする regexp (regular expression = 正規表現) の書き方については Section “Syntax of Regular Expression” in *The GNU Emacs Manual*, を参考にしなさい。regexp の書き方はいく通りもある。その内の幾つかは残りのものよりも良い。私が利用している関数は、幾つかの正規表現と一緒に次の所に記されている。Appendix A “関数 `the-the`”, page 164.

13 カウント：繰り返しと正規表現

繰り返しと正規表現の検索は、Emacs Lisp でコードを書く際によく使われる大変強力な道具である。この章では、正規表現の検索の利用の仕方を、while ループないしは再帰を利用した単語を数えるコマンドの作成を通して説明していくことにする。

標準的な Emacs の配布にはリージョン内の行数を数える関数が含まれている。しかし同じことを単語について行う関数はない。

文章の種類によっては単語の数を数えなければならないようなことがある。エッセイを書く場合は 800 単語以内に制限した方がよいであろうし、小説を書く場合など、一日 1000 単語は書くぞと決心することもあるだろう。個人的には Emacs に単語を数えるコマンドがないのは変だと思う。多分 Emacs を使う人は大抵コードとかドキュメントを書いていて、単語の数を数える必要などないのだろう。あるいは Operating system 附属の単語数を数えるコマンドである wc しか使わなかったりするのかもしれない。はたまた編集者の都合に合わせて、文書の中の単語の数を文字数を 5 で割ったものとして数えている人もいるかもしれない。何はともあれ、以下で単語数を数えるコマンドを紹介することにする。

（未訳）There are many ways to implement a command to count words. Here are some examples, which you may wish to compare with the standard Emacs command, count-words-region.

13.1 関数 count-words-example

単語数を数えるコマンドとしては、行、パラグラフ、あるいはリージョンやバッファなどの中に含まれる単語の数を数えるものが考えられる。コマンドとしてはどんなことまで出来ればよいだろうか。コマンドをバッファ全体の単語を数えるように設計することも出来るが、Emacs の伝統からいって、もっと柔軟性を持たせた方がよいだろう。例えばバッファ全体ではなく、あるセクションの中の単語の数を教えるようなコマンドが欲しい場合も出てくるかもしれない。従って、リージョンの中の単語数を数えるように設計した方が合理的だろう。一度、このような count-words-region というコマンドを作ってしまえば、*C-x h* (mark-whole-buffer) を使ってバッファ全体をマークすることで、バッファ全体の単語を数えることだって出来る。

単語を数えることは、明らかに繰り返しを伴う動作だ。リージョンの始まりからスタートして最初の単語を数え、次に二番目に行き、三番目に行き、というふうにしてリージョンの最後に来るまで続けるわけである。これは単語を数えるという行為が、もともと再帰や while ループに適しているということを意味する。

まずは、while ループを使って、次に再帰を使って、単語を数えるコマンドを実際に作ってみることにする。これらのコマンドは勿論インタラクティブであるべきである。

これまでにも何度も見たように、インタラクティブな関数の定義のテンプレートは次のようになっている。

```
(defun 関数名 (引数リスト)
  "説明文字列..."
  (インタラクティブ式...)
  本体...)
```

このスロットを埋めればよい。

関数の名前はそれ自体で意味が通じ、かつ既にある関数 count-lines-region の名前と似たものであるべきである。これは名前を覚えやすくするためだ。count-words-region なんかが良いだろう。

この関数はリージョン内の単語の数を数える。従って引数リストには、リージョンの始まりと終わりの二つの位置各々にバインドされるシンボルが含まれていなければならない。これらの二つの位置は ‘beginning’ と ‘end’ と呼べばいいだろう。説明文字列の最初の行は一つの文であるべきである。というのは apropos のようなコマンドで表示される部分は一行目が全てであるからである。インタラクティブ式は ‘(interactive "r")’ という形になる。こうすることで、リージョンの始まりと終わりの位置をこの関数の引数リストに渡すことが出来る。ここまではルーティーンワークである。

関数の本体は、三つの仕事をするように書かれなければならない。一番目は while ループが単語を数える時の条件を設定すること、二番目は while ループを走らせること、そして三番目はユーザにメッセージを送ることである。

ユーザが count-words-region を呼び出した時点では、ポイントはリージョンの始まり、もしくは終わりにある。しかし、カウントは常にリージョンの始まりの位置から開始しなければならない。従って、そうでない場合はまずその位置までポイントを移動しておいて欲しい。(goto-char beginning)

を実行することで、このことが保証される。勿論、関数が仕事を終えたらポイントの位置は元の位置に戻っていて欲しい。このために、本体部分は `save-excursion` 式で囲む必要がある。

本体の中心部分は、あるS式で次の単語に一つずつジャンプし、もう一つのS式でジャンプの回数を数える `while` ループからなる。`while` ループの真偽テストはポイントがもう先へは進めなくなるまでは真を返し、リージョンの最後まできたら偽を返すようなものでなければならない。

ポイントを単語ごとに移動するS式としては `(forward-word 1)` を使ってもよい。しかし、正規表現の検索を使うなら、Emacs が何を単語と見做すかを見る方が簡単である。

単語のパターンを検索する正規表現の検索では、ポイントはマッチしたパターンの最後の文字の位置に移動する。従って、続けて検索が成功している間はポイントは単語を一つずつ移動していくことになる。

現実問題として、正規表現の検索では、単語だけではなく単語間の空白や句読点をジャンプするようになっていて欲しい。実際、単語間の空白を越えて進まないような正規表現では二単語以上先には進めなくなってしまう。つまり、正規表現の中には単語だけでなく、その後に続く空白や句読点等も含まれていなければならないわけである。(単語はバッファの最後に来るかもしれないので、後に必ずしも空白がくるとは限らない。従って、空白等の部分の正規表現はオプションになる。)

というわけで、我々が求めている正規表現は、まず幾つかの単語構成文字が続き、オプションとして単語の構成要素以外の文字が一つ以上の続くようなパターンである。このような正規表現は次のように書ける。

```
\w+\W*
```

どの文字が単語構成文字であり、どの文字がそうでないかは、そのバッファのシンタックステーブルによる。(詳しくは Section 14.2 “単語やシンボルは何から構成されているか”, page 122, あるいは Section “The Syntax Table” in *The GNU Emacs Manual*, や Section “Syntax Tables” in *The GNU Emacs Lisp Reference Manual*, を参照。)

検索のためのS式は次の通りである。

```
(re-search-forward "\\w+\\W*")
```

(‘`w`’ や ‘`W`’ の前にはバックスラッシュが二つずつ組になっていることに注意しよう。単独のバックスラッシュは Emacs インタプリタにとって特殊な意味を持っている。つまり、後に続く文字が普通とは異なる意味に解釈されるのである。例えば、‘`\n`’ という二つの文字は、バックスラッシュに続く ‘`n`’ ではなく、改行コードを表わす。バックスラッシュが二つ続いて初めて通常の「特殊でない」バックスラッシュを表わすのである。)

(訳註：つまり、バックスラッシュが二つ要るのは、それが一度 Emacs の Lisp リーダによって一つのバックスラッシュに解釈された後の結果が正規表現として解釈されるためである。ここは混乱しやすいので注意しよう。)

いくつ単語があったかを数えるためのカウンタも必要である。この変数は最初は 0 にセットされ、while ループを一つ繰り返す度に一つずつ増えていく。この一つ増やすためのS式は単に

```
(setq count (1+ count))
```

だけでよい。さて、最後にユーザに対してリージョンにいくつ単語があったかを知らせたい。このようにユーザに情報を表示する目的には、`message` 関数が用意されている。今の場合、メッセージは単語の数がいくつあった場合でもきちんとした文章になっていなければならない。例えば、“there are 1 words in the region” というふうになっては困る。単語が一つなのに複数形が使われるのは文法的に正しくないからである。この問題は、条件分岐式を用いて単語数によってメッセージを変えるようにすれば回避出来る。可能性は三つある。一つはリージョン内には全く単語が無い場合、もう一つは一個だけある場合、そして、複数の単語がある場合である。このような場合には `cond` を使うとよいのであった。

以上を総合すると、次のような関数定義になる。

```
;;; 最初のバージョン: バグあり！
(defun count-words-region (beginning end)
  "Print number of words in the region.
Words are defined as at least one word-constituent
character followed by at least one character that
is not a word-constituent.  The buffer's syntax
table determines which characters these are."
  (interactive "r")
  (message "Counting words in region ... ")
```

```
;;; 1. 適切な状況の設定。
  (save-excursion
    (goto-char beginning)
    (let ((count 0))

;;; 2.  while loop を走らせる。
      (while (< (point) end)
        (re-search-forward "\\w+\\W*")
        (setq count (1+ count)))

;;; 3. ユーザにメッセージを送る。
      (cond ((zerop count)
             (message
              "The region does NOT have any words."))
            ((= 1 count)
             (message
              "The region has 1 word."))
            (t
             (message
              "The region has %d words." count))))))
```

が、以下に書かれている通り、この関数は全ての状況できちんと動作するわけではない。

13.1.1 count-words-region の空白文字に関するバグ

前節で説明した count-words-region コマンドには、二つのバグ、というか二つの現れ方をする一つのバグがある。まず、たとえリージョンがある文章の途中にある空白文字しか含まない場合でも、count-words-region は単語が一つあると言ってくる。また、バッファないしはアクセス可能な部分の最後にある空白文字しか含まない場合は、

```
Search failed: "\\w+\\W*"
```

というエラーメッセージが返ってしまう。この文を GNU Emacs の Info で読んでいるなら、このバグを実際に自分で確かめることが出来る。

まずはこの関数をいつも通り評価してインストールしよう。

お望みなら、次の式を評価することでこのコマンドのためのキーバインディングもインストール出来る。

```
(global-set-key "\C-c=" 'count-words-example)
```

最初のテストをするために、次の行の最初と最後にマークとポイントを設定して *C-c =* (あるいは *C-c =* というキーバインディングをインストールしていなければ *M-x count-words-region*) とタイプしよう。

```
one   two  three
```

Emacs はリージョンには三つの単語があると正しく答えるはずだ。

次に、マークを行頭にセットして、ポイントをちょうど ‘one’ という単語の前に置いて同じテストをしてみよう。*C-c =* (あるいは *M-x count-words-region*) とタイプした場合、Emacs はリージョンには一つも単語がないと答えるべきである。リージョンには行頭の空白文字だけしか含まれないからだ。しかし実際は Emacs はリージョンには一つの単語があると答えてくる！

三番目に、サンプルの行を *scratch* バッファの最後にコピーして行の終わりに幾つか空白文字をタイプしてから、マークを ‘three’ の直後に、ポイントを行末に設定する。(即ち、行末をバッファの最後にする。) ここでさっきと同じく *C-c =* (あるいは *M-x count-words-region*) とタイプしよう。空白文字しかないのだから、今回も Emacs は単語は一つもないと答えるべきである。しかし、Emacs は ‘Search failed’ というエラーメッセージを表示する。

この二つのバグは共通する問題から生じたものだ。

最初のバグを考えてみよう。コマンドを実行すると、空白だけしかないはずの行頭部分に一つの単語が含まれていると言ってきたのであった。これは何故だろうか？ M-x count-words-region コマンドは、まずポイントをリージョンの最初に移動させる。while テストは point の値が end の値よりも小さいかどうかテストする。これは正しいので、結果として最初の単語を見つける正規表現の検索が実行され、最初の単語が見つかる。ポイントは単語の後ろに移され、count は 1 にセットされる。while ループはもう一度実行されるが、今度はポイントの値が end の値よりも大きくなるので、ループはそこで終了する。そしてこの関数は、リージョン内の単語は一つだと言うメッセージを表示すると

いうわけである。手短にいうと、正規表現の検索では、単語がリージョンの外にあるにも関わらず、単語を発見してしまうのである。

二番目のバグの場合、リージョン内にはバッファの最後の空白文字だけしかなかった。Emacs は‘Search failed’と言う。これは何故か？ まず最初の while ループでは真偽テストは真になるので、検索が実行される。しかし、バッファの最後には単語は一つもないので、検索は失敗するというわけである。

どちらの現象も、検索がリージョンを越えて実行される所に問題があった。

解決法は、検索をリージョンに制限することである—これはかなり単純な動作に思えるが、実際に考えてみると、思った程簡単にはいかないことが分る。

今まで見てきたように、re-search-forward 関数は最初の引数を検索パターンとする。しかし、この最初の必須の引数の他に、三つの省略可能な引き数を取ることが出来る。省略可能である二番目の引数は検索の限界である。オプションである三番目の引数は、もし t なら検索に失敗した場合にエラーを出さずに nil を返す。オプションである四番目の引数には繰り返しの回数を指定する。(Emacs では、関数定義の説明文字列を見るには *C-h f* に続けて関数名をタイプして *RET* キーを押せばよい。)

count-words-region の定義内では、変数 end にリージョンの最後の値がこの関数の引数として渡され保持される。そこで、正規表現の検索の引数として、この end の値を与えてみよう。

```
(re-search-forward "\\w+\\W*" end)
```

しかし、もし count-words-region の値にこの変更だけしか加えずに、新しいバージョンの関数を、空白だけの場所でテストすると、‘Search failed’というメッセージが返される。

これはどうしてかというと、リージョン内には単語を構成する文字が無いために、検索をそこに制限すると、検索が失敗してしまうためである。そのためにエラーメッセージが返される。しかし、この場合にエラーメッセージは受け取りたくない。期待されるのは "The region does NOT have any words." というメッセージである。

この問題を回避するには re-search-forward に三番目の引数として t を与えれば良い。すると関数はエラーメッセージの代わりに nil を返す。

しかし、この変更を加えてからもう一度テストしてみると、“Counting words in region ... ”というメッセージが表示された後いくら待っても何も表示されない。結局 *C-g* (keyboard-quit) とタイプするまでずっとこのメッセージを見続けることになる。

どうしてこうなるかだが､検索はリージョンに制限されているのでさっきと同じように失敗する。これはリージョンには単語を構成する文字が含まれないのだから期待通りである。結果として re-search-forward 式は nil を返す。そして他のことは何もしない。特に､ポイントを動したりもしない。これは検索に成功した場合の副作用だからだ。re-search-forward が nil を返した後､while ループの次のＳ式が評価される。これはカウンタを一つ増やす。そして、次のループに移る。re-search-forward 式でポイントが移動しないために、ポイントの値は end の値よりも小さいままであり、真偽テストは真になる。そしてループが繰り返される。これがずっと続く ...

検索が失敗した場合は while ループの真偽テストが偽を返すようにするために、またもや count-words-region の定義に変更を加える必要がある。即ち、カウンタを一つ増やす前に行う真偽テストでは二つの条件が満たされなければならないのだ。一つはポイントがリージョン内にあること、もう一つは検索によって単語が見つかることである。

最初の条件も二番目の条件も同時に真にならなければいけないので、次のようにリージョンテストを行うＳ式と検索を行うＳ式の二つのＳ式を and 関数で連結して while ループの真偽テストに埋め込むことにしよう。

```
(and (< (point) end) (re-search-forward "\\w+\\W*" end t))
```

(and 関数についての詳細は、 “The kill-new 関数”, page 70, を参照。)

re-search-forward 式は、もし検索が成功した場合は真を返し、副作用としてポイントを移動する。結果として、単語が見つかった場合にはポイントはリージョンを越えてしまうことがある。検索が失敗した場合やポイントがリージョンの終わりに来てしまった場合は、真偽テストは偽を返し、while ループが終了して、メッセージが表示される。

これらの最終的な変更を加えると、count-words-region は (少なくとも私が見た範囲では！) バグ無しに動作するようになる。コードは次の通りである。

```
;;; 最終バージョン: while
(defun count-words-region (beginning end)
  "Print number of words in the region."
  (interactive "r")
  (message "Counting words in region ... ")

;;; 1. 適切な状況の設定
  (save-excursion
    (let ((count 0))
      (goto-char beginning)

;;; 2.  while loop を走らせる。
      (while (and (< (point) end)
                  (re-search-forward "\\w+\\W*" end t))
        (setq count (1+ count)))

;;; 3. ユーザにメッセージを送る
      (cond ((zerop count)
             (message
              "The region does NOT have any words."))
            ((= 1 count)
             (message
              "The region has 1 word."))
            (t
             (message
              "The region has %d words." count))))))
```

13.2 再帰を使った単語数のカウント

単語を数える関数は `while` ループではなく再帰を使っても書くことが出来る。実際にどのようになるかを見てみよう。

まず、`count-words-region` 関数は三つの仕事をするということをはっきり認識しておこう。初めに、カウントを行う際の適当な条件を整え、次にリージョン内の単語を数え、最後に単語の数を知らせるメッセージをユーザに対し表示するというわけである。

これらの仕事を全て一つの再帰関数でやらせようとすると、全ての再帰呼び出しでメッセージを受け取ってしまうことになる。例えばリージョンに 13 の単語があった場合、13 個のメッセージが並んでしまうのだ。これでは困る。そこで、これらの仕事を二つの関数に分けてやらせることにする。一つを再帰関数にして、それをもう一方で呼び出すのである。片方は条件を整え、メッセージを表示するものであり、もう片方は、単語を数えるものである。

まずはメッセージを表示する方の関数から始めよう。今回も関数の名前は `count-words-region` を使うことにする。

こちらはユーザが呼び出す方の関数であり、インタラクティブなものになる。実の所、これはこの関数の前回のバージョンにそっくりである。ただし、リージョン内の単語を数えるために、途中で `recursive-count-words` を呼び出している所だけが違う。

前回の関数を参考にすれば、簡単にこの関数のテンプレートを作ることが出来る。

```
;; 再帰バージョン; 正規表現検索を利用
(defun count-words-region (beginning end)
  "説明文字列..."
  (インタラクティブ式...)

;;; 1. 適切な状況の設定
  (状況説明のメーセージ)
  (初期設定のための関数...

;;; 2. 単語数のカウント
     再帰呼び出し

;;; 3. ユーザにメッセージを送る。
     単語数を伝えるメッセージ))
```

この定義は非常に素直に書かれているが、再帰呼び出しで返されたカウントを単語の数を表示するメッセージに渡す所が、幾分ややこしいかもしれない。これは、ちょっと考えれば、`let` 関数を使えばよいと気付く。つまり、`let` 式の変数リストの変数に、再帰関数から返されたリージョン内の単語数をバインドすればよいのである。その後、`cond` 式を利用してその値をユーザに表示する。

`let` 式の中での変数のバインドというと、何か関数の二義的な仕事に思われることも多い。しかし、今の場合には、関数の主な仕事だと思われる単語を数えるということが、`let` 式の中で行われることになる。

`let` を使うと、関数定義は次のようになる。

```
(defun count-words-region (beginning end)
  "Print number of words in the region."
  (interactive "r")

;;; 1. 適切な状況の設定
  (message "Counting words in region ... ")
  (save-excursion
    (goto-char beginning)

;;; 2. 単語数のカウント
    (let ((count (recursive-count-words end)))

;;; 3. ユーザに対しメッセージを送る
      (cond ((zerop count)
             (message
              "The region does NOT have any words."))
            ((= 1 count)
             (message
              "The region has 1 word."))
            (t
             (message
              "The region has %d words." count))))))
```

次に、再帰関数の方に移ろう。

再帰関数は少なくとも三つの部分を持たねばならない。‘do-again-test’、‘next-step-expression’、そして、再帰呼び出しである。

do-again-test は関数が自分自身をもう一度呼ぶかどうかを決定するものである。ここではリージョン内の単語を数え、その中の単語の分だけ前に移動するのであるから、do-again-test ではポイントがまだリージョンの内部にあるかどうかをチェックすれば良い。つまり、ポイントの値を調べて、それがリージョンの終わりの値よりも小さいか大きいかを比べるわけである。ポイントの値を調べるには `point` 関数を使えばよい。リージョンの終わりの値は、勿論、この再帰関数の引数として渡す必要がある。

更に do-again-test では検索で単語が見つかったかどうかもテストしなければならない。無かった場合は再帰呼び出しをすべきではない。

next-step-expression は再帰関数が自分自身の呼び出しを止めるべき時に止めるように値を変化させるものである。より正確に言うと、next-step-expression は適切な時に do-again-test が再帰呼び出しを繰り返すのを止めるように、値を変化させていく。今の場合なら、next-step-expression はポイントを単語ごとに移動させていく S 式になる。

再帰関数の三つ目の部分は、再帰呼び出しである。

また、どこかにこの関数の実際の「仕事」をする部分、つまり単語数を数える部分を書く必要がある。これが一番重要な部分だ！

が、ともかく、再帰関数のアウトラインは出来た。

```
(defun recursive-count-words (region-end)
  "説明文字列..."
   do-again-test
   next-step-expression
   再帰呼び出し)
```

このスロットを埋めていけばよい。まずは、一番単純な場合から始めよう。もしポイントがリージョンの終わりもしくはそれ以降の位置にあった場合は、リージョン内には単語はない。従って、関数は 0 を返すべきである。同様に検索が失敗した場合も単語はないので、0 を返すべきである。

一方、もしポイントがリージョンの中にあって検索が成功した場合には、関数は自分自身を再度呼び出すべきである。

というわけで、do-again-test は次のようになる。

```
(and (< (point) region-end)
     (re-search-forward "\\w+\\W*" region-end t))
```

（旧訳）検索をするS式も do-again-test の一部であることに注意しよう。この関数は検索が成功すれば `t` を返し、失敗すれば `nil` を返す。(Section 13.1.1 “`count-words-region` の空白文字に関するバグ”, page 115, を見れば、`re-search-forward` がどのように動作するかが分る。)

do-again-test は `if` 式の真偽テストである。明らかに、もし真偽テストが真だった場合には、`if` 式はこの関数を再度呼び出さなければならない。が、もし偽だった場合には、else-part によって零が返されるべきである。何故なら、単語が見つからなかったということは、ポイントがリージョンの外だったり検索に失敗したということだからだ。

しかし、再帰呼び出しについて考える前に、next-step-expression について考える必要があるのだった。これは今の場合何か？ 興味深いことに、これは do-again-test の検索部分なのである。

do-again-test で `t` や `nil` を返すことに加えて、`re-search-forward` は検索成功時には副作用としてポイントを前方に移動する。これこそが、ポイントがリージョン内を移動しきった場合に、再帰関数が自分自身を呼び出すのを止めるようにポイントの値を変化させていく動作なのである。というわけで、`re-search-forward` 式が next-step-expression になっているのである。

従って `recursive-count-words` のアウトラインは次のようになる。

```
(if do-again-test-and-next-step-combined
    ;; then
    再帰呼び出しをしてカウントを返す
  ;; else
  ゼロを返す)
```

これにカウントを数えさせるような機構を組み込むには、どうしたらよいだろうか。

If you are not used to writing recursive functions, a question like this can be troublesome. But it can and should be approached systematically.

next-step-expression では一単語ごとに前方に進み、従って再帰呼び出しは一つ単語を進むごとに行われるわけだから、カウントの仕組は `recursive-count-words` が呼ばれるごとに数を一つ増やすようなものでなければならない。

幾つかの場合を考えてみよう。

- リージョン内に二つの単語があった場合、この関数は、最初の単語を数える時に返した値にリージョンの残りの単語数、今の場合なら 1、を足した数に 1 を加えた値を返さなければならない。
- リージョン内に一つだけ単語があった場合、この関数は、最初の単語を数える時に返した値にリージョンの残りの単語数、今の場合なら 0、を足した数に 1 を加えた値を返さなければならない。
- リージョン内に一つも単語がなければ、この関数は 0 を返すべきである。

以上のことから、`if` 式の else-part では、単語が一つもなかった場合のために 0 を返せばよいことが分る。これは、`if` 式の then-part では残りの単語の合計の値に 1 加えた値を返すべきだということを意味する。

実際のS式は次のようになる。ここで、`1+` は引数に 1 加える関数である。

```
(1+ (recursive-count-words region-end))
```

以上から `recursive-count-words` 関数全体は次のようになる。

```
(defun recursive-count-words (region-end)
  "説明文字列..."

  ;;; 1. do-again-test
  (if (and (< (point) region-end)
           (re-search-forward "\\w+\\W*" region-end t))

  ;;; 2. then-part: 再帰呼び出し
      (1+ (recursive-count-words region-end))

  ;;; 3. else-part
      0))
```

実際にどのような動作をするかを見てみよう。

もし、リージョンに単語が一つもなければ、`if` 式の else-part が評価され、結果としてこの関数は 0 を返す。

もし、リージョンに単語が一つあれば、ポイントの値は `region-end` の値よりも小さく、検索は成功する。この場合、`if` 式の真偽テストは真を返し、then-part が評価され、カウント式が評価される。この式が返す値が関数全体の返す値になるのだが、これは、再帰呼び出しで返された値に 1 加えた値である。

ところで、この時同時に next-step-expression によってポイントがリージョン内の最初の単語 (といっても今はこの一個だけだが) を越えて移動する。つまり、(`recursive-count-words region-end`) が二回目に評価された時には、ポイントの位置はリージョンの終わり以降の位置にあることになる。従って、今度は `recursive-count-words` は 0 を返す。これにさっきの 1 が加えられ、結果として元の `recursive-count-words` は 0 足す 1、つまり、1 を返すことになる。これは正しい数である。

ここまでくれば後はもうお分りだと思うが、リージョンの中に単語が二つある場合には、最初に呼び出された `recursive-count-words` は次に残りの単語を含むリージョンで呼び出された `recursive-count-words` の値に 1 加えた値、つまり、1 + 1 = 2 という値を返す。これも正しい数である。

同様にして、もしリージョンの中に単語が三つある場合も、最初に呼び出された `recursive-count-words` は残りのリージョンで呼び出された `recursive-count-words` の値に 1 加えた値を返す。このような感じで、単語が沢山ある場合も正しい数を返してくれる。

説明文字列も完全につけると、二つの関数は次のようになる。

再帰関数:

```
(defun recursive-count-words (region-end)
  "Number of words between point and REGION-END."

;;; 1. do-again-test
  (if (and (< (point) region-end)
           (re-search-forward "\\w+\\W*" region-end t))

;;; 2. then-part: 再帰呼び出し
      (1+ (recursive-count-words region-end))

;;; 3. else-part
    0))
```

外側の関数 (wrapper):

```
;;; 再帰バージョン
(defun count-words-region (beginning end)
  "Print number of words in the region.

Words are defined as at least one word-constituent
character followed by at least one character that is
not a word-constituent.  The buffer's syntax table
determines which characters these are."
  (interactive "r")
  (message "Counting words in region ... ")
  (save-excursion
    (goto-char beginning)
    (let ((count (recursive-count-words end)))
      (cond ((zerop count)
             (message
              "The region does NOT have any words."))
            ((= 1 count)
             (message "The region has 1 word."))
            (t
             (message
              "The region has %d words." count))))))
```

13.3 練習問題：句読点のカウント

`while` ループを使って、リージョン内の句読点—終止符、カンマ、セミコロン、コロン、感嘆符、疑問符—の数を数える関数を書きなさい。同じ関数を再帰関数を使って書きなさい。

14 defun 内の単語のカウント

次の計画の目標は、関数定義の中の単語の数を数えることである。当たり前のことだが、これは count-words-region の使い方をちょっと工夫すれば出来てしまう。Chapter 13 “カウント：繰り返しと正規表現”, page 113, を参照のこと。例えば、ある一つの定義の中の単語数を数えたければ、*C-M-h* (mark-defun) コマンドを使って定義部分をマークしてから count-words-region を呼び出せばよい。

しかしながら、ここではもっと大きなことをやってみたいと思う。Emacs のソースの中の全ての定義の中の単語とシンボルの数を数えて、そこにどれだけの関数があり、各々がどのくらいの長さかをグラフにして出力するとか、40 個から 49 個までの単語とシンボルを関数がどれだけあるか、50 個から 59 個までではどうか、といったことを調べるのである。私はしばしば典型的な関数というのがどのくらいの長さかを知りたくなる。これは、そういったことを教えてくれるものである。

はっきり言って、このヒストグラムを書く計画は人をひるませる類のものである。しかし、これをいくつもの細かいステップに分けて、各々を一つずつ見ていくことにすれば、それほど恐れるほどのものではない。そこで、どんなステップに分けるべきかを書いてみることにする。

- まず最初に、一つの定義の中に単語がどれだけあるかを数える関数を書く。ここでは、単語と同じくシンボルをどう扱うかが問題になる。
- 二番目に各々の関数の中の単語数をリストにする関数を書く。この関数では count-words-in-defun を使うことが出来る。
- 三番目に、各ファイルの中の各関数の中の単語数をリストにする関数を書く。これをやろうとすると、必然的に、自動的に様々なファイルを見つけてそれらに移り、それに含まれる関数定義の中の単語数を数えるということをしなければならなくなる。
- 四番目に、三番目のステップで作成した数のリストをグラフとして出力するのに適した形に変換する関数を書く。
- 五番目に結果をグラフとして表示する関数を書く。

これはかなりの大計画である。しかし、各々のステップをゆっくりと進んでいけば、それ程困難なものではない。

14.1 何を数えればよいか？

関数定義の中の単語数を数えるにはどうしたらよいか、を最初に考え始めた時に、まず疑問に思うこと (あるいは、考えるべきこと) は、我々は何を数えればよいかということである。Lisp の関数定義に関して単語のことを話す場合、実際には大抵シンボルのことを言っている。例えば、次の multiply-by-seven 関数は、defun、multiply-by-seven、number、*、そして 7という 5 個のシンボルを含んでいる。これに加えて説明文字列の中に ‘Multiply’、‘NUMBER’、‘by’、そして ‘seven’ という単語が含まれている。‘number’ は繰り返して使われているので、関数定義の中には合計 10 個の単語とシンボルが含まれていることになる。

```
(defun multiply-by-seven (number)
  "Multiply NUMBER by seven."
  (* 7 number))
```

ところが、もし multiply-by-seven の定義を *C-M-h* (mark-defun) でマークして、そこで count-words-region を呼び出してみると、10 個ではなく 11 個の単語があるという答えが帰ってくる。何かがおかしい！

実は、問題は二重になっている。count-words-region は ‘*’ を単語とは数えないが、逆に、一つのシンボル multiply-by-seven を三つの単語だと数えてしまうのである。これはハイフンが一つの単語内でのつながりを示すものとしてではなく、単語間の間の空白と同じように扱われるためのである。従って、‘multiply-by-seven’ は ‘multiply by seven’ と書かれているように扱われることになる。

このような混乱の原因は、count-words-region の定義内で一つの単語ずつ移動する際に使っている正規表現にある。標準的な count-words-region のバージョンで使われている正規表現は

```
"\\w+\\W*"
```

である。この正規表現は一つ以上単語構成文字が続いた後に 0 個以上の非単語構成文字が続くというパターンである。「単語構成文字」によって何が意味されるかという問題は、構文 (syntax) の問題になる。これには一つのセクションを割当てて論じる価値がある。

14.2 単語やシンボルは何から構成されているか

Emacs では、各々の文字はある構文カテゴリ (*syntax categories*) に属するものとして扱われる。例えば正規表現 '\\w+' は一つ以上の「単語構成文字」(*word constituent*) が続くパターンを表している。単語構成文字というのは、ある一つの構文カテゴリーの要素のことである。他の構文カテゴリーの要素は、例えば終止符やカンマ等の句読点文字のクラス、スペースやタブ等の空白文字のクラスを含んでいる。(より詳しいことについては Section "The Syntax Table" in *The GNU Emacs Manual*, 及び Section "Syntax Tables" in *The GNU Emacs Lisp Reference Manual*, を参照のこと。)

構文テーブルとはどの文字がどのカテゴリーに属するかを定めるものである。普通、ハイフンは「単語構成文字」には分類されない。そうではなく「シンボルの名前ではあるが、単語ではないものの一部をなす文字のクラス」に分類される。これは `count-words-region` 関数がハイフンを単語間の空白文字と同じ扱いをすることを意味する。これが `count-words-region` が '`multiply-by-seven`' を三つの単語だと数える理由である。

Emacs に `multiply-by-seven` を一つの単語だと数えさせるには二つの方法がある。一つは構文テーブルを書き換える方法、もう一つは正規表現を書き直す方法である。

Emacs が各々のモードに対して持つ構文テーブルを書き換えることで、ハイフンを単語構成文字だと再定義することが出来る。この動作は我々の目的に殆ど合うのだが、ハイフンだけが単語の中には現れずシンボルの中には出てくる文字というわけではない。似たような文字は他にもある。

代わりに、`count-words` の定義中の正規表現の方を書き直してシンボルを含むようにすることも出来る。こちらの方法の方がより簡明である。ただし、実際にやることは少々トリッキーだ。

最初の部分は十分に単純である。パターンとしては「少なくとも一つ以上続く、単語ないしはシンボルの構成要素」にマッチするもの、つまり、

```
"\\(\\w\\|\\s_\\)+"
```

になる。'\\(' は '\\w' と '\\s_' のいずれかを表わす正規表現のグループの開始を示す部分である。対象となる二つの部分は '\\|' で区切られている。'\\w' は任意の単語構成文字にマッチし、'\\s_' はシンボル名の一部になり得るが、単語構成文字ではないような任意の文字にマッチする。後に続く '+' は、このグループに属する文字、即ち単語かシンボルの構成文字が少くとも一回はマッチしなければならないことを意味する。

しかしながら、正規表現の二番目の部分はもっと難しい。欲しいものは、一番目の正規表現に続けて「単語の一部にもシンボルの一部にもならない文字が一つ以上続いてもかまわない」ことを表わす表現である。まず思い浮んだのは次のようなものである。

```
"\\(\\W\\|\\S_\\)*"
```

上の大文字の 'W' と 'S' は各々単語、あるいはシンボルの構成文字ではないような文字にマッチする。しかし、この表現では、単語構成文字ではないか、もしくはシンボル構成文字ではない文字に一致してしまう。これではどんな文字にもマッチしてしまう。

次に私は、テストしているリージョン内の全ての単語やシンボルの後には空白文字 (スペース、タブ、もしくは改行) があることに気がついた。そこで、単語かシンボルの構成文字が一つ以上続くというパターンの後に、一つ以上の空白文字が続くというパターンを置いてみた。しかし、これも失敗した。通常は単語やシンボルは空白で終わるのだが、実際のコードでは、シンボルの後に括弧が来たり、単語の後に句読点が来たりすることだってある。というわけで、結局、単語かシンボルの構成文字の後に 0 個以上の空白文字以外の文字が続き、その後に 0 個以上の空白文字が来る、というパターンにすることにした。

次がその正規表現である。

```
"\\(\\w\\|\\s_\\)+[^ \t\n]*[ \t\n]*"
```

14.3 関数 count-words-in-defun

以前見たように、`count-words-region` 関数を書く方法は幾つかあった。が、今回 `count-words-in-defun` を書く際には、この内の一つの方法だけを採用することにする。

方法としては `while` ループを使う方法が理解しやすいだろうから、こちらを採用することにする。`count-words-in-defun` は、より複雑なプログラムの一部になるので、インタラクティブである必要も、メッセージを出す必要もなく、ただカウントを返しさえすればよい。これらのことを考慮すると、定義は少し単純になる。

一方、count-words-in-defun は関数定義を含むバッファの中で使われる。従って、現在ポイントが関数定義内にあるか判定し、もしそうであればその定義内の単語を数えるというふうにするのが合理的であろう。こうすると、ちょっとコードが複雑にはなるが、関数に引数を与える手間は省ける。

以上のことを考慮すると、テンプレートは次のようになる。

```
(defun count-words-in-defun ()
  "説明文字列..."
  (初期設定...
     (while ループ...)
   カウントを返す)
```

いつも通り、やるべきことはこの中の空きスロットを埋めていくことである。

まずは初期設定から。

この関数は、関数定義が含まれるバッファの中で呼び出されることを想定されている。現在のポイントは関数定義の中にあるか、外にあるかどちらかである。count-words-in-defun が動作してくれるためには、ポイントが関数定義の先頭に移動し、カウンタがゼロから始まり、ポイントが関数定義の最後に来たらループが終了するようになっていてくれなければならない。

beginning-of-defun 関数は後方に向かって行頭の ‘(’ などの開き括弧を検索し、そこにポイントを移動するか、検索の限界まで移動する。実際には、beginning-of-defun はポイントを現在ポイントが含まれている関数定義もしくはポイント以前のポイントに最も近い関数定義の開始位置、あるいはバッファの先頭まで移動することになる。従って、beginning-of-defun を使うことで望みの位置までポイントを移動することが出来る。

while ループでは、数えた単語やシンボルの数を保持しておくカウンタが必要である。let 式によって、この目的のための変数を作り、その値をゼロに初期化することが出来る。

end-of-defun 関数は beginning-of-defun と同じような働きをするのだが、ポイントを関数定義の終了位置に移動する点だけが異なっている。end-of-defun は関数定義の終了位置を決定する S 式の一部として使うことが出来る。

ということで、count-words-in-defun の初期設定部分はあっさり書けてしまう。まずは、関数定義の最初にポイントを移動し、次にカウンタのための局所変数を用意し、最後に while ループが止まるべき所で止まれるように関数定義の終了位置を記録しておくのである。

コードは次の通りである。

```
(beginning-of-defun)
(let ((count 0)
      (end (save-excursion (end-of-defun) (point))))
```

このコードは単純である。ちょっとややこしいのは end に関するところだろう。これには、関数定義の終了位置がバインドされる。その際、save-excursion 式の中で一時的に end-of-defun で関数定義の終了位置に移動した後にポイントの位置を返すという方法を用いている。

さて、count-words-in-defun の初期設定に続く二番目の部分は while ループである。

このループでは、ポイントを単語やシンボルごとに移動する S 式、及びジャンプの回数を数える S 式が必要である。また、while ループの真偽テストでは、ポイントがまだジャンプすべきなら真を返し、定義の終了位置まで到達したなら偽を返すようなものであるべきである。目的のための正規表現は既に再定義してしまっているので、(Section 14.2 “単語やシンボルは何から構成されているか”, page 122, 参照) ループは簡単に書ける。

```
(while (and (< (point) end)
            (re-search-forward
             "\\(\\w\\|\\s_\\)+[^ \t\n]*[ \t\n]*" end t))
  (setq count (1+ count)))
```

関数定義の三番目の部分は単語やシンボルの数を返す部分である。この部分は let 式の本体部分の最後の部分だが、極めて単純に局所変数 count を書いておくだけでよい。これを評価すると数が返るわけである。

以上をまとめると、count-words-in-defun は次のようになる。

```
(defun count-words-in-defun ()
  "Return the number of words and symbols in a defun."
  (beginning-of-defun)
  (let ((count 0)
        (end (save-excursion (end-of-defun) (point))))
```

```
    (while
        (and (< (point) end)
             (re-search-forward
              "\\(\\w\\|\\s_\\)+[^ \t\n]*[ \t\n]*"
              end t))
      (setq count (1+ count)))
    count))
```

これをテストするにはどうしたらよいだろうか。この関数はインタラクティブではないが、ちょっとS式をかぶせることで簡単にインタラクティブにすることが出来る。これには count-words-region の再帰関数版とほぼ同じコードが使える。

```
;;; インタラクティブバージョン
(defun count-words-defun ()
  "Number of words and symbols in a function definition."
  (interactive)
  (message
   "Counting words and symbols in function definition ... ")
  (let ((count (count-words-in-defun)))
    (cond
     ((zerop count)
      (message
       "The definition does NOT have any words or symbols."))
     ((= 1 count)
      (message
       "The definition has 1 word or symbol."))
     (t
      (message
       "The definition has %d words or symbols." count)))))
```

便宜上 *C-c =* というキーバインディングをもう一度使うことにしよう。

```
(global-set-key "\C-c=" 'count-words-defun)
```

以上で、count-words-defun をテストするための準備が整った。まずは、count-words-in-defun 及び count-words-defun を両方ともインストールして、キーバインディングの設定もしてしまおう。そして、カーソルを次の定義の中に移動して実験してみる。

```
(defun multiply-by-seven (number)
  "Multiply NUMBER by seven."
  (* 7 number))
    ⇒ 10
```

成功だ！ この定義の中には確かに 10 個の単語とシンボルがある。

次の問題は、一つのファイルの中の幾つかの定義にある単語とシンボルの数を数えることである。

14.4 一つのファイルにある複数の defun を数える

simple.el のようなファイルの中には 80 以上の関数定義が含まれていたりする。我々の最終的な目標は沢山のファイルについての統計を取ることであるが、その最初のステップとして、まずは一つのファイルについての統計を取ることを目標にしよう。

情報は数の列の形で与えられ、各々の数は関数定義の長さになる。これらの数はリストの中に保持しておくことが出来る。

一つのファイルについての情報は最終的には多くのファイルについての情報の形に統合されることになる。従って、ここで作成する一つのファイル内の関数定義の長さを数える関数は単に「長さ」のリストを返すだけでよく、特にメッセージとかを表示する必要はない。

単語を数えるコマンドには、単語ごとにポイントを前方に進めるS式とジャンプの回数を数えるS式が含まれていた。定義の長さを測る関数も、同じように設計することが出来る。この場合は定義ごとにポイントを進めるS式と長さのリストを作成するようなS式が含まれることになる。

問題をこのように言い替えてしまえば、関数定義を書くのは簡単なことである。明らかに、カウントはファイルの先頭から始めなければならない。従って、最初のコマンドは (goto-char (point-min)) である。次に while ループに入る。ここでは、ループの真偽テストは次の関数定義を探す正規表現に取れる—検索が成功している間はポイントを進め、本体を評価するわけである。本体内には長さのリストを作成するS式が必要である。これには cons というリストを構成するコマンドが使える。やるべきことの殆どは、以上で終わっている。

部分的にコードを書くと次のようになる。

```
(goto-char (point-min))
(while (re-search-forward "^(defun" nil t)
  (setq lengths-list
        (cons (count-words-in-defun) lengths-list)))
```

この他にやることは、関数定義を含むファイルを見つけることである。

今までの例では、この Info file を使うか、*scratch* バッファなどの他のバッファに一旦戻って、また帰ってくるということしかしていなかった。

ファイルを見つける (find) ことは、この文書では初めて出てくるプロセスである。

14.5 ファイルを見つける

To find a file in Emacs, you use the *C-x C-f* (find-file) command. This command is almost, but not quite right for the lengths problem.

まずは、find-file のソースを見てみよう。

```
(defun find-file (filename)
  "Edit file FILENAME.
Switch to a buffer visiting file FILENAME,
creating one if none already exists."
  (interactive "FFind file: ")
  (switch-to-buffer (find-file-noselect filename)))
```

(The most recent version of the find-file function definition permits you to specify optional wildcards to visit multiple files; that makes the definition more complex and we will not discuss it here, since it is not relevant. You can see its source using either *M-.* (find-tag) or *C-h f* (describe-function).)

定義には、簡潔な説明文字列が付いており、インタラクティブ式では対話的に用いた時のプロンプトが指定されている。で、定義の本体を見ると、find-file-noselect 及び switch-to-buffer という二つの関数が使われている。

C-h f (describe-function コマンド) で表示される説明によると、find-file-noselect は指定されたファイルをバッファに読み込み、そのバッファを返す。しかしながらバッファは選択されない、つまり Emacs は注意をそのバッファには向けない。(あるいは find-file-noselect を名前のついたバッファに対し使った場合は、あなたの注意も引かない。)

この仕事は switch-to-buffer がやってくれる。この関数は、Emacs が注目するバッファを指定するものである。更にこの関数は、ウィンドウに表示されているバッファを新しいバッファに切り替える。バッファの切り替えについては、また別の場所で議論することにしよう。(Section 2.3 “バッファ間の移動”, page 17, 参照。)

今やろうとしているヒストグラム計画では、定義の長さを調べる際にいちいち一つ一つのファイルをスクリーンに表示する必要はない。というわけで switch-to-buffer ではなく set-buffer を使うことにしよう。これも Emacs が注目するバッファを切り替えるのだが、スクリーンに表示するバッファはそのままである。従って、我々の目的のためには find-file は使えず、そのためのコードを書くことになる。

といっても、やることは簡単だ。単に find-file-noselect と set-buffer を使えばよいのである。

14.6 lengths-list-file についての詳細

lengths-list-file 関数の核心部分は、defun から defun へ移動していく関数を含む while ループと、各々の defun の中に含まれる単語やシンボルの数を数える関数である。そしてその周辺に、例えば、ファイルを見つけたり、ポイントが必ずファイルの先頭部分からスタートするようにしたりする、といった他の様々な仕事をする関数が来る。結局、定義は次のようになる。

```
(defun lengths-list-file (filename)
  "Return list of definitions' lengths within FILE.
The returned list is a list of numbers.
Each number is the number of words or
symbols in one function definition."
```

```
(message "Working on `%s' ... " filename)
(save-excursion
  (let ((buffer (find-file-noselect filename))
        (lengths-list))
    (set-buffer buffer)
    (setq buffer-read-only t)
    (widen)
    (goto-char (point-min))
    (while (re-search-forward "^(defun" nil t)
      (setq lengths-list
            (cons (count-words-in-defun) lengths-list)))
    (kill-buffer buffer)
    lengths-list)))
```

この関数は一つの引数を取る。これは作業対象となるファイルの名前である。説明文字列は四行あるが、インタラクティブ宣言はされていない。本体の一行目では、使った人が計算機が壊れたのではないかと心配しないように、最初に何をやっているかを表示するようにしている。

次の行で `save-excursion` が使われているので、Emacs は仕事が終わった後、ちゃんと元のバッファに注意を戻してくれる。こうすると、この関数を他の関数の中に埋め込んでいる場合などにも、ポイントを元の位置に戻してくれるので便利である。

`let` 式の変数リストの所で、Emacs はファイルを見つけて局所変数 `buffer` をそのファイルを中身とするバッファにバインドする。同時に Emacs は `lengths-list` を局所変数として生成する。

次に Emacs は注意をそのバッファに向ける。

次の行では Emacs はバッファを書き込み不可にしている。理論上は、この行は不要である。関数定義内の単語やシンボルの数を数える関数の中で、バッファを書き換えたりするようなものはないし、たとえそのバッファが変更されたとしても保存されたりはしない。こういう警戒をするのは、これらの関数が Emacs のソース上で作業するために、万が一にでもファイルを修正してしまったりすると非常に不都合であるという理由のためである。言うまでもないが、私自身は実験が失敗して私の Emacs のソースファイルが修正されるなんていう事態に会わない限り、この行が必要だと思うことはないだろう。

次に、バッファがナローイングされている場合には、それを広げるということをやっている。これは普通は必要ない—Emacs はそのファイルに対応するバッファが無い場合は新規にバッファを開くからだ。しかし、既にある場合には Emacs はそのバッファを返す。この場合、もしそのバッファがナローイングされていたなら、それを解除する必要がある。真にユーザーフレンドリーな関数にしたい場合には、ナローイングやポイントの位置なんかを保存しておくべきだろうが、ここでは何もしないことにする。

`(goto-char (point-min))` 式でポイントをバッファの先頭に移動する。

そして `while` ループが来る。ここで、この関数の仕事が実行されることになる。このループでは、Emacs は各々の定義の長さを調べ、その長さのリストを作っていく。

あるバッファでの作業が終わると Emacs はそのバッファを kill する。これは、Emacs 内部でのスペースの節約のためである。私が使っている Emacs 19 のバージョンには 300 以上ものソースファイルが含まれており、これらに `lengths-list-file` が適用される。もし Emacs がこれら全てを読み込んで一つも kill しなかったら、私の計算機は仮想記憶を使い切ってしまうだろう。

終りまで来ると、`let` 式の中の最後の S 式である `lengths-list` という変数が評価される。この値が関数全体の値となる。

この関数をいつも通りインストールして試してみることが出来る。インストールが終わったら、カーソルを次の S 式の直後に持っていってから *C-x C-e* (`eval-last-sexp`) とタイプして、評価してみよう。

```
(lengths-list-file
 "/usr/local/share/emacs/22.1.1/lisp/emacs-lisp/debug.el")
```

(多分、ファイルのパス名を変更する必要があるだろう。上に挙げたものは、この Info ファイルのあるディレクトリと Emacs のソースがあるディレクトリが`/usr/local/emacs/info` と `/usr/local/emacs/lisp` のように隣同士にある場合だけである。変更する場合は、この S 式を一旦 `*scratch*` バッファにコピーしてから、それを修正して評価する。)

(Also, to see the full length of the list, rather than a truncated version, you may have to evaluate the following:

```
(custom-set-variables '(eval-expression-print-length nil))
```

(See Section 16.2 “Specifying Variables using `defcustom`”, page 144. Then evaluate the `lengths-list-file` expression.)

The lengths' list for `debug.el` takes less than a second to produce and looks like this in GNU Emacs 22:

私が使っているバージョンの Emacs では、`debug.el` に対する長さのリストを生成するのに 7 秒かかり、結果は次のようになった。

```
(83 113 105 144 289 22 30 97 48 89 25 52 52 88 28 29 77 49 43 290 232 587)
```

(Using my old machine, the version 19 lengths' list for `debug.el` took seven seconds to produce and looked like this:

```
(75 41 80 62 20 45 44 68 45 12 34 235)
```

(The newer version of `debug.el` contains more defuns than the earlier one; and my new machine is much faster than the old one.)

ファイルの中の最後の定義の長さは、リストの最初に現れることに注意しよう。

Note that the length of the last definition in the file is first in the list.

14.7 異なるファイルの中の定義を数える

前節では、各ファイルの中に含まれる各関数の長さのリストを返すような関数を作成したのだった。今度は、ファイルのリストが与えられた時に、それらのファイルの中の関数の長さのマスターリストを返すような関数を定義してみたい。

リストの中の各ファイルに対する作業は繰り返しの動作なので、`while` ループや再帰を使って行うことが出来る。

`while` ループを使った方法はルーティーンワークである。関数に渡す引き数はファイルのリストになる。以前見たように (Section 11.1.1 “ループの例”, page 83, 参照)、このリストが要素を含んでいる時のみループの本体を実行し、要素が無くなったら抜けるようにすることが出来るのであった。これがうまく動作するためには、本体部分で、本体が一回評価されるごとにこのリストを短くしていき、結果として最後にはリストが空になるように、S 式を書いておく必要がある。このためには、本体が評価されるごとに、リストにそのリストの CDR の値をセットするという技法を用いるのが普通である。

テンプレートは次のようになる。

```
(while リストが空かどうかのテスト
  本体...
  リストを自分自身の cdr にセット)
```

さて、`while` ループは常に (真偽テストの結果として) `nil` を返し、本体内の S 式の値を返したりすることはないのだった。(従って、ループの本体の S 式は副作用として評価される。) しかしながら、長さのリストをセットする S 式は本体の一部である—にもかかわらず、その関数全体の値として返して欲しいのもこの値である。そこで、`while` ループを `let` 式で包んで、`let` 式の最後の要素が長さのリストの値を含むようにする。(“増加するカウンタを使ったループの例”, page 85, を参照。)

以上のことを考えれば、目的の関数が殆ど書けてしまう。

```
;;; while ループ を使う。
(defun lengths-list-many-files (list-of-files)
  "Return list of lengths of defuns in LIST-OF-FILES."
  (let (lengths-list)

;;; 真偽テスト
    (while list-of-files
      (setq lengths-list
            (append
             lengths-list

;;; 長さのリストの生成
             (lengths-list-file
              (expand-file-name (car list-of-files)))))
```

```
;;; ファイルのリストを短くする。
      (setq list-of-files (cdr list-of-files)))

;;; リストの長さの最終的な値を返す。
    lengths-list))
```

expand-file-name は組み込み関数であり、ファイル名を絶対パスも含めた省略無しの形に戻すものである。従って、例えば

Thus, if expand-file-name is called on debug.el when Emacs is visiting the /usr/local/share/emacs/22.1.1/lisp/emacs-lisp/ directory,

```
debug.el
```

は

```
/usr/local/share/emacs/22.1.1/lisp/emacs-lisp/debug.el
```

The only other new element of this function definition is the as yet unstudied function append, which merits a short section for itself.

14.7.1 関数 append

append 関数は、あるリストを、もう一つのリストに追加するものである。例えば、

```
(append '(1 2 3 4) '(5 6 7 8))
```

の結果は次のようになる。

```
(1 2 3 4 5 6 7 8)
```

lengths-list-file によって作成された二つの長さのリストを一つにまとめる際は、このような形になって欲しいのだった。cons を使った場合と比較してみよう。

```
(cons '(1 2 3 4) '(5 6 7 8))
```

こっちだと、cons の最初の引数が出来たリストの最初の要素になってしまう。

```
((1 2 3 4) 5 6 7 8)
```

14.8 異なるファイルの定義を再帰を使って数える

while ループではなく再帰を使っても各々のファイルのリストに対して作業することが出来る。再帰を使った lengths-list-many-files は短くて単純な形をしている。

再帰関数は、普通は 'do-again-test'、'next-step-expression'、そして再帰呼び出しの部分からなっている。'do-again-test' では、この関数が自分自身をもう一度呼び出すかどうかを決定する。今の場合は list-of-files がまだ残りの要素を持っているかどうかを調べることになる。'next-step-expression' では、list-of-files をそれ自身の CDR で置き換える。結果として最後にはリストは空になる。実際の完全なコードは、この説明よりも短い！

```
(defun recursive-lengths-list-many-files (list-of-files)
  "Return list of lengths of each defun in LIST-OF-FILES."
  (if list-of-files                     ; do-again-test
      (append
       (lengths-list-file
        (expand-file-name (car list-of-files)))
       (recursive-lengths-list-many-files
        (cdr list-of-files)))))
```

一言で言うと、この関数は list-of-files の最初のファイルについての長さのリストを、list-of-files の残りを引数に自分自身を呼び出した結果に追加している。

実際に、各ファイルに対して lengths-list-file を走らせながら recursive-lengths-list-many-files をテストしてみるには、次のようにする。

まず、まだやっていなければ recursive-lengths-list-many-files と lengths-list-files をインストールし (訳註：count-words-in-defun もインストールする必要がある)、その後、次に挙げる S 式を評価する。ただし、ファイルのパス名は変更する必要があるかもしれない。以下の式では、Info ファイルと Emacs のソースファイルが通常の位置に置いてある場合に有効である。これを変更したい場合は、これらの式を *scratch* バッファにコピーして、それらを編集した後、評価すればよい。

結果は '⇒' の後に示されている。(これらの結果は Emacs version 21.1.1 についてのものである。他のバージョンの Emacs については、また別の結果が出ることだろう。)

```
(cd "/usr/local/share/emacs/22.1.1/")

(lengths-list-file "./lisp/macros.el")
     ⇒ (283 263 480 90)

(lengths-list-file "./lisp/mail/mailalias.el")
     ⇒ (38 32 29 95 178 180 321 218 324)

(lengths-list-file "./lisp/makesum.el")
     ⇒ (85 181)

  (recursive-lengths-list-many-files
   '("./lisp/macros.el"
     "./lisp/mail/mailalias.el"
     "./lisp/makesum.el"))
       ⇒ (283 263 480 90 38 32 29 95 178 180 321 218 324 85 181)
```

このように recursive-lengths-list-many-files は期待した結果を返してくれるはずだ。

次のステップは、結果をグラフに表示するためのデータのリストを準備することである。

14.9 データをグラフに表示するための準備

recursive-lengths-list-many-files 関数は、数のリストを返す。各々の数は関数定義の長さの記録である。我々がこれからやらねばならないのは、このデータをグラフの表示に適した形の数値のデータに変換することである。新しく出来るリストからは、10 より少ない単語やシンボルしか含まない関数定義がどれだけあるかとか、10 から 19 や、20 から 29 までではどうか等ということが分るようになる。

手短に言うと、recursive-lengths-list-many-files 関数が生成したリストを見ていって、各々の範囲に入る関数がどれだけあるかを数えて、それらの数のリストを作ろうというのである。

これまでの経験から、長さのリストを 'CDR' しつつ各々の値を見ていき、それがどの範囲に入るのかを調べてその範囲についてのカウンタを増やす関数を書くことは、特に難しくはないものと察しがつくだろう。

しかしながら、実際に関数を書き下す前に、長さのリストをまずソートして、少ない方から大きい方に並べることによって得られるメリットについて考えるべきである。まず、ソートすることで、各々の範囲に属する関数の数を数えるのが楽になる。これは隣同士の数は同じ範囲に属するか隣同士の範囲に属するかどちらかになるからである。また、リストをソートしてしまえば最大の数と最小の数を簡単に見つけることが出来る。またそこから、後で必要となる最大と最小の差も決定出来ることになる。

14.9.1 リストのソート

Emacs は sort と呼ばれるリストをソートするための関数を持っている。sort 関数は二つの引数を持つ。ソートされるべきリストと、二つの要素の大小を比較する際の述語 (predicate) である。

以前説明したように (Section 1.8.4 "関数に間違ったタイプの引数を与えると", page 10, 参照)、述語とは、ある性質が真か偽かを判断する関数のことである。sort 関数は、リストの要素を述語が使用する性質に従って並べ換える。これは、数値以外のリストも、数値以外の基準で—例えばアルファベットの順番で— sort を利用して並べ換えることが出来ることを示している。

数値で比較する際には < 関数が使われる。例えば、

```
(sort '(4 8 21 17 33 7 21 7) '<)
```

の結果は次のようになる。

```
(4 7 7 8 17 21 21 33)
```

(この例では、引数が sort に渡される際に評価されないように、どちらのシンボルにも引用符が付いていることに注意しよう。)

recursive-lengths-list-many-files 関数によって返されたリストをソートするのは簡単である。

```
(sort
 (recursive-lengths-list-many-files
  '("./lisp/macros.el"
    "./lisp/mailalias.el"
    "./lisp/makesum.el"))
 '<)
```

とするだけだ。結果は次のようになる。

```
(29 32 38 85 90 95 178 180 181 218 263 283 321 324 480)
```

(この例では sort の最初の引数には引用符がついていない。これは、sort に渡される前にこのS式を評価して、リストを生成する必要があるからである。)

14.9.2 ファイルのリストの作成

recursive-lengths-list-many-files 関数は引数としてファイルのリストを必要とする。これまで実験した例では、これらのリストは手で作っていた。しかし、Emacs Lisp のソースディレクトリは大変大きいので、これらを一々手で書いているわけにはいかない。そこで、代わりに directory-files 関数を作って、このようなリストを作成する必要がある。

We did not have to write a function like this for older versions of GNU Emacs, since they placed all the '.el' files in one directory. Instead, we were able to use the directory-files function, which lists the names of files that match a specified pattern within a single directory.

However, recent versions of Emacs place Emacs Lisp files in sub-directories of the top level lisp directory. This re-arrangement eases navigation. For example, all the mail related files are in a lisp sub-directory called mail. But at the same time, this arrangement forces us to create a file listing function that descends into the sub-directories.

We can create this function, called files-in-below-directory, using familiar functions such as car, nthcdr, and substring in conjunction with an existing function called directory-files-and-attributes. This latter function not only lists all the filenames in a directory, including the names of sub-directories, but also their attributes.

To restate our goal: to create a function that will enable us to feed filenames to recursive-lengths-list-many-files as a list that looks like this (but with more elements):

```
("./lisp/macros.el"
 "./lisp/mail/rmail.el"
 "./lisp/makesum.el")
```

The directory-files-and-attributes function returns a list of lists. Each of the lists within the main list consists of 13 elements. The first element is a string that contains the name of the file—which, in GNU/Linux, may be a 'directory file', that is to say, a file with the special attributes of a directory. The second element of the list is t for a directory, a string for symbolic link (the string is the name linked to), or nil.

For example, the first '.el' file in the lisp/ directory is abbrev.el. Its name is /usr/local/share/emacs/22.1.1/lisp/abbrev.el and it is not a directory or a symbolic link.

This is how directory-files-and-attributes lists that file and its attributes:

```
("abbrev.el"
nil
1
1000
100
(20615 27034 579989 697000)
(17905 55681 0 0)
(20615 26327 734791 805000)
13188
"-rw-r--r--"
t
2971624
773)
```

On the other hand, `mail/` is a directory within the `lisp/` directory. The beginning of its listing looks like this:

```
("mail"
t
...
)
```

(To learn about the different attributes, look at the documentation of `file-attributes`. Bear in mind that the `file-attributes` function does not list the filename, so its first element is `directory-files-and-attributes`'s second element.)

We will want our new function, `files-in-below-directory`, to list the '`.el`' files in the directory it is told to check, and in any directories below that directory.

This gives us a hint on how to construct `files-in-below-directory`: within a directory, the function should add '`.el`' filenames to a list; and if, within a directory, the function comes upon a sub-directory, it should go into that sub-directory and repeat its actions.

However, we should note that every directory contains a name that refers to itself, called `.`, ("dot") and a name that refers to its parent directory, called `..` ("double dot"). (In `/`, the root directory, `..` refers to itself, since `/` has no parent.) Clearly, we do not want our `files-in-below-directory` function to enter those directories, since they always lead us, directly or indirectly, to the current directory.

Consequently, our `files-in-below-directory` function must do several tasks:

- Check to see whether it is looking at a filename that ends in '`.el`'; and if so, add its name to a list.
- Check to see whether it is looking at a filename that is the name of a directory; and if so,
  - Check to see whether it is looking at `.` or `..`; and if so skip it.
  - Or else, go into that directory and repeat the process.

Let's write a function definition to do these tasks. We will use a `while` loop to move from one filename to another within a directory, checking what needs to be done; and we will use a recursive call to repeat the actions on each sub-directory. The recursive pattern is 'accumulate' (see "Recursive Pattern: *accumulate*", page 96), using `append` as the combiner.

Here is the function:

```
(defun files-in-below-directory (directory)
  "List the .el files in DIRECTORY and in its sub-directories."
  ;; Although the function will be used non-interactively,
  ;; it will be easier to test if we make it interactive.
  ;; The directory will have a name such as
  ;;  "/usr/local/share/emacs/22.1.1/lisp/"
  (interactive "DDirectory name: ")
  (let (el-files-list
        (current-directory-list
         (directory-files-and-attributes directory t)))
    ;; while we are in the current directory
    (while current-directory-list
      (cond
       ;; check to see whether filename ends in '.el'
       ;; and if so, append its name to a list.
       ((equal ".el" (substring (car (car current-directory-list)) -3))
        (setq el-files-list
              (cons (car (car current-directory-list)) el-files-list)))
       ;; check whether filename is that of a directory
       ((eq t (car (cdr (car current-directory-list))))
        ;; decide whether to skip or recurse
        (if
            (equal "."
                   (substring (car (car current-directory-list)) -1))
            ;; then do nothing since filename is that of
            ;;   current directory or parent, "." or ".."
            ()
```

```
            ;; else descend into the directory and repeat the process
            (setq el-files-list
                  (append
                   (files-in-below-directory
                    (car (car current-directory-list)))
                   el-files-list)))))
          ;; move to the next filename in the list; this also
          ;; shortens the list so the while loop eventually comes to an end
          (setq current-directory-list (cdr current-directory-list)))
        ;; return the filenames
        el-files-list))
```

files-in-below-directory directory-files 関数は、一つの引数（ディレクトリ名）を取る。

従って、例えば私のシステムで

```
(length
 (files-in-below-directory "/usr/local/share/emacs/22.1.1/lisp/"))
```

とやると、私の Lisp のソースディレクトリには 1031 の ‘.el’ ファイルがあることが分る。

recursive-lengths-list-many-files が返すリストをソートするための S 式は次のようになる。

```
(sort
 (files-in-below-directory "/usr/local/share/emacs/22.1.1/lisp/")
 'string-lessp)
```

14.9.3 Counting function definitions

我々の取り敢えずの目標は、10 未満の単語やシンボルしか含まない関数定義の数はどれだけか、10 以上、20 未満ではどうか、20 以上、30 未満ではどうか、といったことを調べることである。

ソートされた数のリストを使うと、これは簡単である。まずは、10 未満の要素がどれだけあるかを数え、ついで、その次の要素から 20 未満の要素がどれだけか数え、また次の数から 30 未満の要素がどれだけか数える、というふうに続けていく。10、20、30、40 等の数は、その範囲の数の最大よりも大きい数になる。これらの数からなるリストは、top-of-ranges リストと呼べばよいだろう。

しようと思えば、このようなリストを自動的に生成することも可能である。が、今回は手で書いた方が早いだろう。次のような感じである。

```
(defvar top-of-ranges
 '(10  20  30  40  50
   60  70  80  90 100
  110 120 130 140 150
  160 170 180 190 200
  210 220 230 240 250
  260 270 280 290 300)
 "List specifying ranges for ‘defuns-per-range’.")
```

範囲を変更するには、このリストを編集すればよい。

次に、この各々の範囲に属する定義の数のリストを作る関数を書く必要がある。明らかに、この関数は引数として sorted-lengths と top-of-ranges リストを取ることになる。

defuns-per-range 関数は、二つの作業を何回も繰り返すことになる。一つは現在の top-of-range の値によって特定される範囲の定義の数を数えること、もう一つはその範囲の数を数え終わったら次に大きな top-of-ranges の値に移ることである。これらの動作は繰り返しなので、while ループを使うことが出来る。片方のループで現在の top-of-range の値で決まる範囲の定義の数をカウントし、もう片方のループでは順に top-of-range の値を選択していく。

各々の範囲について、sorted-lengths リストの中の幾つかのエントリがカウントされる。従って、sorted-lengths リストについてのループは top-of-ranges リストのループの中に置かれることになる。大きなギヤのなかの小さなギヤみたいな感じだ。

内部のループでは、該当する範囲の定義の数がカウントされる。これは、今までに何回も見たような単純なループである。(Section 11.1.3 “増加するカウンタを使ったループ”, page 85, を参照。) ループの真偽テストは sorted-lengths リストの中の数が現在の top-of-range の値よりも小さいかどうかを見ることになる。もしそうなら、カウンタを一つ増やして、次の sorted-lengths のエントリに移動する。

結局、内部のループは次のようになる。

```
(while 長さの要素が top-of-range より小さい
  (setq number-within-range (1+ number-within-range))
  (setq sorted-lengths (cdr sorted-lengths)))
```

外部のループは top-of-ranges リストの最小値から始まって、順に大きな値に移っていくことになる。そのためには、次のようにすればよい。

```
(while top-of-ranges
  ループの本体...
  (setq top-of-ranges (cdr top-of-ranges)))
```

これらを合わせると、二つのループは次のようになる。

```
(while top-of-ranges

  ;; 現在の範囲にある要素の数のカウント
  (while 長さの要素が top-of-range より小さい
    (setq number-within-range (1+ number-within-range))
    (setq sorted-lengths (cdr sorted-lengths)))

  ;; 次の範囲に移動
  (setq top-of-ranges (cdr top-of-ranges)))
```

更に、一回の外部ループごとに、Emacs にその範囲に属する定義の数を記録させる必要がある。(リストの中の number-within-range の値である。この目的には、cons が使える。(Section 7.2 “cons”, page 56, を参照。)

cons 関数はほぼうまく動作するのだが、一つ難点がある。出来るリストでは最初に大きい方の範囲に入る定義の数がきて、最後に小さい方の範囲の数がきてしまうのだ。これは、cons が新しい要素をリストの先頭に加えていくことと、上の二つのループは小さい方から大きい方へ長さのリストを作成していくために、defuns-per-range-list が最大の数で終わることからの当然の帰結である。しかし、グラフを表示する際には小さい値の方を先に表示したい。この問題を解決するには、defuns-per-range-list の順序を逆にしてしまえばよい。これは、nreverse というリストの順序を逆にする関数を使うとあっさり解決する。

例えば、

```
(nreverse '(1 2 3 4))
```

とすると、

```
(4 3 2 1)
```

が返る。

注意して欲しいのは、nreverse は「破壊的」であるということである。これは、作用させたリストを変更してしまうことを意味している。(訳註：これは逆の順のリストに設定されるということではなくて、文字通り破壊されてしまうということである。) 今の場合、元の defuns-per-range-list は必要ないので、これが破壊されても何の問題もない。(一方、reverse 関数は元のリストを逆に並べ換えた新しいリストを返す。この場合は元のリストは変化しない。)

以上を全て組み合わせると、defuns-per-range は次のようになる。

```
(defun defuns-per-range (sorted-lengths top-of-ranges)
  "SORTED-LENGTHS defuns in each TOP-OF-RANGES range."
  (let ((top-of-range (car top-of-ranges))
        (number-within-range 0)
        defuns-per-range-list)

    ;; 外部のループ
    (while top-of-ranges

      ;; 内部のループ
      (while (and
              ;; 数値引数として数が必要
              (car sorted-lengths)
              (< (car sorted-lengths) top-of-range))
```

```
        ;; 現在の範囲に入る関数の数を数える
        (setq number-within-range (1+ number-within-range))
        (setq sorted-lengths (cdr sorted-lengths)))

      ;; 内部のループは抜けるが、外部のループには入ったまま

      (setq defuns-per-range-list
            (cons number-within-range defuns-per-range-list))
      (setq number-within-range 0)      ; カウンタを 0 にリセット

      ;; 次の範囲に移動
      (setq top-of-ranges (cdr top-of-ranges))
      ;; 次の範囲のトップを特定
      (setq top-of-range (car top-of-ranges)))

    ;; 外部のループを抜けて最も大きいの範囲に属する関数定義の数
    ;; を数える
    (setq defuns-per-range-list
          (cons
           (length sorted-lengths)
           defuns-per-range-list))

    ;; 昇順で並ぶ関数定義の長さのリストを返す。
    (nreverse defuns-per-range-list)))
```

この関数は次のちょっとした点を除いては、非常に単純である。内部のループの真偽テストは、

```
(and (car sorted-lengths)
     (< (car sorted-lengths) top-of-range))
```

であって、

```
(< (car sorted-lengths) top-of-range)
```

ではない。このテストの目的は `sorted-lengths` リストの最初の要素がその時点での top-of-range の値よりも小さいかどうかを決定することである。

後に挙げた単純な方のテストでも `sorted-lengths` リストが `nil` になるまではうまく動作する。しかし、`nil` になると、`(car sorted-lengths)` 式は `nil` を返す。`<` 関数は数値と `nil`、即ち空リストとを比較できないため、Emacs はここでエラーを出し、関数はそこで実行を止めてしまう。

`sorted-lengths` リストはカウンタがリストの最後まで辿りつけば常に `nil` になる。従って、この `defuns-per-range` 関数の真偽テストの単純なバージョンの方は常に失敗することになる。

この問題は、`(car sorted-lengths)` 式と `and` 式を組み合わせることで解決することが出来る。`(car sorted-lengths)` 式はリストが最低一つでも要素を持てば、non-`nil` を返す。そしてリストが空になった時だけ `nil` を返す。`and` 式は最初に `(car sorted-lengths)` 式を評価し、もしそれが `nil` なら `<` 式を評価する前に偽を返す。しかし、もし `(car sorted-lengths)` 式が non-`nil` な値を返せば、`<` 式も評価し、その値を `and` 式全体の値を返す。

こうしてエラーが回避出来ることになる。(For information about `and`, see “The `kill-new` function”, page 70.)

（旧訳）`and` についての詳細は、Section 12.4 “`forward-paragraph`：関数の金脈”, page 105, を参照のこと。

次に `defuns-per-range` についての簡単なテストを載せておく。最初に (短縮した) リストを `top-of-ranges` にバインドする S 式を評価し、次に `sorted-lengths` リストをバインドする S 式を評価し、最後に `defuns-per-range` 関数を評価してみよう。

```
;; (後で使うものよりかは短いリスト)
(setq top-of-ranges
 '(110 120 130 140 150
   160 170 180 190 200))

(setq sorted-lengths
      '(85 86 110 116 122 129 154 176 179 200 265 300 300))

(defuns-per-range sorted-lengths top-of-ranges)
```

次のようなリストが返されるはずである。

```
(2 2 2 0 0 1 0 2 0 0 4)
```

実際、`sorted-lengths` リストには、110 未満の二つの要素が二つ、110 から 119 までの要素も二つ、120 から 129 までも二つ、といった感じになっている。また、200 以上の値の要素は四つある。

15 グラフを描く準備

我々の目標は、Emacs のソースの中の様々な長さの関数定義の数をグラフにして表示することであった。

実際には、グラフを生成するには gnuplot のようなプログラムを使っていることだと思う。(gnuplot は GNU Emacs と相性がよい。) しかしながら、今回は一からプログラムを書いていくことにする。そしてその過程で、今まで学んできた事柄を再度確認しつつ、新しいことも学んでいくことにしよう。

この章では、まずグラフを表示する単純な関数を書いてみる。最初の定義は雛型 (*prototype*) であり、グラフを作成するという未知の領域を偵察するために取り敢えず書いてみるといった類のものである。我々はドラゴンを発見するかもしれないし、あるいはそれが単なる神話であることが分るかもしれない。ともかく地理感覚が掴めてしまえば、自信もつくし、関数を拡張して軸に自動的にラベルをつけることも出来るようになるだろう。

Emacs はどんな種類のターミナルでも動作するよう柔軟に設計されている。その中にはキャラクターだけしか表示出来ないターミナルも含まれているので、グラフは「タイプライタ」の文字から出来ている必要がある。取り敢えずはアスタリスクを使うのがよいだろう。後から関数を拡張して、この文字をユーザーが選択出来るようにすることも可能だ。

この関数を、graph-body-print と呼ぶことにしよう。これは引数として numbers-list を取る。現段階ではグラフのラベルは出力せず、本体部分だけを表示することにする。

graph-body-print 関数は、numbers-list の各々の値に対して、アスタリスクを垂直に並べて表示する。それぞれの高さは、numbers-list の各要素の値によって決まる。

アスタリスクを垂直に並べる動作は繰り返しである。従って、while ループか再帰を使って書くことが出来る。

最初の困難は、如何にしてアスタリスクを縦に並べたものを表示するかである。普通、Emacs 上でタイプすると、文字はスクリーン上に水平に、行ごとに表示されていく。解決への道は二通りある。一つは自分で垂直に挿入するような関数を書くこと、もう一つは元々 Emacs にそのような関数がないか探すことである。

Emacs に特定の機能を持ったものがないか探す場合には、*M-x apropos* コマンドを使うことが出来る。このコマンドは、*C-h a* (command-apropos) コマンドとほぼ同様なのだが、command-apropos の方はコマンドだけしか検索しない。*M-x apropos* の方は、インタラクティブでないものも含めて正規表現にマッチするものは全てリストしてくれる。我々が探しているものは縦の文字列 (column) を表示 (print) したり挿入 (insert) したりするコマンドである。可能性としては、そのような関数は 'print' とか 'insert' とか 'column' といった文字列を含んでいそうである。そこで、単純に *M-x apropos RET print\|insert\|column RET* として、結果を見てみる。私のシステムでは、このコマンドの実行には暫く時間がかかり、結局 79 個の関数及び変数のリストが表示された。このリストを探してみると、我々の仕事に役立ちそうなのは、insert-rectangle だけである。実際、これが我々の求めていた関数である。説明を読んでみよう。

```
insert-rectangle:
Insert text of RECTANGLE with upper left corner at point.
RECTANGLE's first line is inserted at point,
its second line is inserted at a point vertically under point, etc.
RECTANGLE should be a list of strings.
```

(日本語訳)
RECTANGLE のテキストを左上がポイントに来るような位置に挿入する。
つまり RECTANGLE の最初の行がポイントの位置に挿入され、
二行目はポイントの真下の位置に挿入され、というふうになる。
RECTANGLE は文字列のリストでなければならない。

簡単なテストを行って、これが本当に求めるものかを確かめてみよう。

以下が、insert-rectangle 式の直後にカーソルを持っていって、*C-u C-x C-e* (eval-last-sexp) とタイプしてみた結果である。この関数は、'"first"', '"second"' そして 'third"' をポイントの下に表示する。関数全体としては、nil が返る。

```
(insert-rectangle '("first" "second" "third"))first
                                              second
                                              thirdnil
```

(訳註：広く配布されている Mule 2.3 (Emacs version 19.28) では、この関数にバグがある。これは、`lisp/rect.el` の最後の方にある `move-to-column-strictly` の関数定義の中で、四行目の `clm` を `(progn (if force (indent-to column)) column)` で置換えることで修正出来る。)

勿論、我々は `insert-rectangle` 式そのものをグラフを描くバッファに挿入したいのではない。そうではなく、我々のプログラムからこの関数を呼び出したいのである。ただその際、ポイントがバッファの中で、ちゃんと `insert-rectangle` 関数で縦の文字列を挿入すべき位置にあるかどうかを確かめなければならない。

もしこの文章を Info の中で読んでいるなら、まず `*scratch*` などの別バッファに移り、ポイントをバッファのどこかに置きつつ typing *M-:* とタイプしよう。(訳註：1998 年現在、日本でまだ使用が多いと思われる Emacs 19.28 ベースの Mule では *M-ESC* である。同様に多い Emacs 19.34 ベースのものでは *M-:* で良い。*M-ESC* はプレフィクスキーとして使われるようになった。) 続けてミニバッファで `insert-rectangle` 式をタイプしてやれば、このような動作が可能であることが分る。ここでは、S 式を評価するのはミニバッファの中だったのだが、その際のポイントの値としては、`*scratch*` バッファの中のポイントの値が使われたのであった。(*M-:* は `eval-expression` のキーバインディングである。)

実際にやってみると、ポイントが直前に入力した行の最後にある状態で終わること—つまり、この関数は副作用として、ポイントを移動することが分る。このコマンドを続けて実行すると、次の挿入は前回の位置から下方及び右方向に移動した所から行われる。これでは困る。棒グラフを作成するには、縦の列が隣同士に並ばなければならない。

というわけで、棒グラフを挿入する `while` ループの各々のサイクルで、ポイントの位置を適切な位置に再配置する必要があることが分った。また、その位置は縦の列の一番上であって底ではない。更に、このグラフを表示する際に、個々の縦の列の長さが全て揃うことはまずない。つまり、各々の縦棒のてっぺんの高さは、前のものと異なるのが普通である。従って、単に前と同じ行にポイントを移動していくだけでは駄目である。そうではなくて...

我々はグラフをアスタリスクで表示しようとしていたのであった。アスタリスクの数は `numbers-list` の現在の要素で指定される。このアスタリスクを要素とするリストを、`insert-rectangle` を呼び出すたびに作成する必要がある。もし、このリストが単に必要な数のアスタリスクだけからなっていたとすると、ポイントの位置をグラフの基準線から正しい行数だけ上の位置におかなければならない。しかし、これは難しい。

その代わり、もし `insert-rectangle` に対して常に一定の長さのリストを渡すことが出来れば、ポイントも常に同じ行の一つ右の桁に移動すればよくなる。しかし、この場合には `insert-rectangle` に渡されるリストのいくつかの要素はアスタリスクではなく空白になる。例えば、もしグラフの最大の高さが 5 であり、現在の高さが 3 であったとすると、`insert-rectangle` は、次のような引数を必要とする。

```
(" " " " "*" "*" "*")
```

この最後の方法は、縦の列の高さを決定出来さえすれば、それほど難しくはない。縦の列の高さを特定するには二つの方法がある。一つは我々が綺麗に表示されるような高さを勝手に指定してしまうというもの、もう一つは、数のリストを検索して、最大のものをグラフの最大の高さとして使うというものである。後者の方法が難しい場合は、前者を採用するのが楽である。が、Emacs には、初めから引数の中で最大のものを決定する関数が組み込まれている。この関数を使おう。関数の名前は `max` で、全ての引数の中で最大のものを返す。引数は数でなければならない。従って、例えば、

```
(max  3 4 6 5 7 3)
```

は 7 を返す。(これに対して `min` という関数は、引数の中で最小のものを返す。)

しかしながら、単に `numbers-list` の上で `max` を呼び出すことは出来ない。`max` は数のリストではなく、数を引数として要求するからだ。従って、

```
(max  '(3 4 6 5 7 3))
```

とやると、次のようなエラーメッセージが返される。

```
Wrong type of argument:  number-or-marker-p, (3 4 6 5 7 3)
```

従って、リストの中身を引数として関数に渡すような関数が必要になる。`apply` と呼ばれる関数がこの役目を果してくれる。この関数はその最初の引数である関数に、残りの引数を適用 (apply) する。この時、最後の引数はリストでなければならない。

例えば

```
(apply 'max 3 4 7 3 '(4 8 5))
```

は 8 を返す。

(ついでにいうと、私はこの関数をこの文書のようなものを使わずに見つけだす方法を知らない。他の search-forward や insert-rectangle といった関数ならば、名前の一部分を推測して apropos を使えばよい。けれども apply については、名前の元となるメタファ—最初の引数に残りを 'apply' する—は明らかであるにしても、初心者が apropos や他の助けを借りてこのような特定な単語に辿り着けるとは思わない。勿論、私が間違っているのかもしれない。結局は、関数はそれを作った人によって名付けられるものなのである。)

apply の二番目以降の引数は省略可能である。従って、apply を使って、関数にリストの要素を引数として渡して呼び出すことも出来る。例えば、次は 8 を返す。

```
(apply 'max '(4 8 5))
```

この最後の例が我々の目的にあっている。recursive-lengths-list-many-files 関数は、数のリストを返すが、これを max に適用することが出来るわけである。(max にソートされたリストを渡すことも出来るが、この場合はソートされていてもされていなくても関係ない。)

というわけで、グラフの最大の高さを求める操作は次の通りである。

```
(setq max-graph-height (apply 'max numbers-list))
```

では、グラフの縦の列を描くための文字列のリストをどうやって作ればよいかという問題に戻ろう。この関数はグラフの最大の高さと個々の縦の列の中のアスタリスクの数から、insert-rectangle コマンドが挿入する文字列のリストを返す必要がある。

各々の縦の列は空白とアスタリスクからなる。高さとアスタリスクの数が分れば、空白の数はその差として求められる。そして空白とアスタリスクの数が分れば、二つの while ループを使ってリストを作ることが出来る。

```
;;; 最初のバージョン
(defun column-of-graph (max-graph-height actual-height)
  "Return list of strings that is one column of a graph."
  (let ((insert-list nil)
        (number-of-top-blanks
         (- max-graph-height actual-height)))

    ;; アスタリスクを詰める
    (while (> actual-height 0)
      (setq insert-list (cons "*" insert-list))
      (setq actual-height (1- actual-height)))

    ;; 空白を詰める
    (while (> number-of-top-blanks 0)
      (setq insert-list (cons " " insert-list))
      (setq number-of-top-blanks
            (1- number-of-top-blanks)))

    ;; リスト全体を返す
    insert-list))
```

この関数をインストールしてから次の S 式を評価すると、求めるリストが得られるはずだ。

```
(column-of-graph 5 3)
```

を評価すると、

```
(" " " " "*" "*" "*")
```

が返されるというわけである。

前にも書いたが、column-of-graph には大きな欠陥がある。空白とグラフ本体の印のために用いられる記号はスペースとアスタリスクに「ハードコード」されている。これは雛型としてはよい。が、あなたや他のユーザは他の記号を使いたいと思うことも多いだろう。例えば、グラフ関数をテストしてみる際には、insert-rectangle 関数が呼ばれた時にポイントの位置がきちんと移動されているかどうかを見るために、スペースの代わりに終止符を使いたいと思うだろう。また、アスタリスクの代わりに '+' 等の記号を使ったりしたいと思うこともあるに違いない。更に、グラフの桁数をディスプレイの一桁の幅よりも広く取りたいと思うこともあるだろう。そういうわけでプログラムはもっと柔軟である

べきである。そのための方法としては、スペースとアスタリスクを graph-blank と graph-symbol という二つの変数で置き換えて、これらの変数を別に定義することが考えられる。

また、説明文字列も解りやすいとは言えない。以上のことを考慮すると、次のような関数に辿り着く。

```
(defvar graph-symbol "*"
  "String used as symbol in graph, usually an asterisk.")

(defvar graph-blank " "
  "String used as blank in graph, usually a blank space.
graph-blank must be the same number of columns wide
as graph-symbol.")
```

(defvar の説明については、Section 8.5 “defvar を用いた変数の初期化”, page 75, を参照。)

```
;;; 二番目のバージョン
(defun column-of-graph (max-graph-height actual-height)
  "Return list of MAX-GRAPH-HEIGHT strings;
ACTUAL-HEIGHT are graph-symbols.
The graph-symbols are contiguous entries at the end
of the list.
The list will be inserted as one column of a graph.
The strings are either graph-blank or graph-symbol."

  (let ((insert-list nil)
        (number-of-top-blanks
         (- max-graph-height actual-height)))

    ;; graph-symbols を詰める
    (while (> actual-height 0)
      (setq insert-list (cons graph-symbol insert-list))
      (setq actual-height (1- actual-height)))

    ;; graph-blanks を詰める
    (while (> number-of-top-blanks 0)
      (setq insert-list (cons graph-blank insert-list))
      (setq number-of-top-blanks
            (1- number-of-top-blanks)))

    ;; リスト全体を返す
    insert-list))
```

もし望むなら、column-of-graph も書き換えて、オプションとして棒グラフだけでなく、線グラフも書けるようにも出来る。これはそんなに難しいことではない。線グラフをどう描くかだが、例えば棒グラフにおいて、一番上の点から下は全て空白にしてしまえば、それはもう棒グラフとは呼べないだろう。線グラフのための縦の文字列を作るには、まずは値よりも一つだけ少ない数の空白のリストを作り、ついで cons を用いてリストにグラフ記号を追加し、最後に、上の余白部分の数の空白を付け足せばよい。

こういう関数を実際に書くのも簡単だが、今の所は必要ないので、書かないでおくことにする。が、ともかく書くことは出来るし、またこの関数を一度書いてしまえば、column-of-graph で使うことも出来る。ここで大切なことは、他の部分の書き換えは殆どしなくても良いということである。即ち、拡張しようと思えば簡単に出来るわけだ。

さて、やっと、はじめて実際のグラフを書いてみる所まで来た。ここではグラフの本体だけを表示し、縦軸や横軸のラベルは表示しない。そこで、この関数を graph-body-print と呼ぶことにする。

15.1 関数 graph-body-print

前節までの準備の後では、graph-body-print 関数はあっというまに出来てしまう。この関数は、数値のリストから各々の縦の列の中のアスタリスクの数を決定し、一桁おきにアスタリスクを用いたグラフを表示する。これは繰り返しの動作なので、減少 while ループか、再帰を使って書くことが出来る。このセクションでは、while ループを用いて定義を書いてみよう。

column-of-graph 関数は、引数としてグラフの高さを必要とする。この値を定めたなら、局所変数として記録しておくべきである。

これらのことから、このバージョンにおいては次のような while ループのテンプレートが出来る。

```
(defun graph-body-print (numbers-list)
  "説明文字列..."
  (let ((height  ...
         ...))

    (while numbers-list
      縦の列を挿入し、ポイントを移動
      (setq numbers-list (cdr numbers-list)))))
```

この中の空きスロットを埋めていくことになる。

当然、グラフの高さの決定には (apply 'max numbers-list) 式を使うことが出来る。

while ループは numbers-list の各要素に対して一回ずつ回る。このリストは (setq numbers-list (cdr numbers-list)) 式によって短くなっていき、各時点でのリストの CAR が column-of-graph に引数として渡される。

この while ループの各サイクルで、column-of-graph によって返されたリストが insert-rectangle 関数に渡される。insert-rectangle 関数はポイントを挿入された矩形の右下のポイントに移動するので、それを矩形が挿入される前の位置に戻してから、次の位置に水平方向に移動してやる必要がある。そこで次の insert-rectangle が呼ばれるわけである。

単独の空白とアスタリスクを使った場合などのように、もし挿入される棒グラフが一桁の幅ならば、移動のためのコマンドは単に (forward-char 1) となる。しかし、この幅はもっと大きくなるかもしれない。従って、(forward-char symbol-width) と書く方が良い。symbol-width は graph-blank の長さであり、(length graph-blank) という式で求めることが出来る。symbol-width という変数をグラフの幅にバインドする一番良い位置は let 式の変数リストの中である。

以上のことを総合すると次のような関数定義になる。

```
(defun graph-body-print (numbers-list)
  "Print a bar graph of the NUMBERS-LIST.
The numbers-list consists of the Y-axis values."

  (let ((height (apply 'max numbers-list))
        (symbol-width (length graph-blank))
        from-position)

    (while numbers-list
      (setq from-position (point))
      (insert-rectangle
       (column-of-graph height (car numbers-list)))
      (goto-char from-position)
      (forward-char symbol-width)
      ;; 各桁ごとのグラフの描写
      (sit-for 0)
      (setq numbers-list (cdr numbers-list)))
    ;; X 軸のラベルのためにポイントを移動
    (forward-line height)
    (insert "\n")
))
```

この関数では、ひとつ予期していなかった S 式が出てくる。それは while ループの中の (sit-for 0) 式である。この式を使うことで、グラフを表示する過程を見るのが楽しくなる。この式は Emacs に 0 時間待ってから画面を再描画させるものである。これを上の位置に置くと棒グラフが一つずつ描かれていくことになる。逆に置かなければ、関数が仕事を全て終了するまでグラフは描かれない。

次のようにすれば、この graph-body-print を短いリストに対してテストしてみることが出来る。

1. まず graph-symbol、graph-blank、column-of-graph そして graph-body-print をインストールする。
2. 次の S 式をコピーする。

```
(graph-body-print '(1 2 3 4 6 4 3 5 7 6 5 2 3))
```

3. *scratch* バッファに移り、グラフを表示させたい位置にカーソルを置く。
4. *M-:* (eval-expression) とタイプする。(訳註：Emacs 19.28 ベースの Mule では、*M-ESC*.)
5. *C-y* (yank) を使って graph-body-print 式をミニバッファに yank する。
6. *RET* を押して、graph-body-print 式を評価する。

Emacs は次のようなグラフを表示するはずである。

15.2 関数 recursive-graph-body-print

graph-body-print 関数は、再帰を使って書くことも出来る。この場合は、二つの部分に分けて書くことになる。外側の関数 (wrapper) で let 式を使って、グラフの最大の高さのように一度だけ決めればよいような幾つかの変数の値を設定し、内側の関数で、再帰呼び出しを使ってグラフを表示するわけである。

外側の関数 (wrapper) は特に複雑ではない。

```
(defun recursive-graph-body-print (numbers-list)
  "Print a bar graph of the NUMBERS-LIST.
The numbers-list consists of the Y-axis values."
  (let ((height (apply 'max numbers-list))
        (symbol-width (length graph-blank))
        from-position)
    (recursive-graph-body-print-internal
     numbers-list
     height
     symbol-width)))
```

再帰関数の方は、ちょっとばかり難しい。これは 'do-again-test'、グラフ表示コード、再帰呼び出し、そして 'next-step-expression' の四つの部分からなる。'do-again-test' は number-list にまだ要素が残っているかを判定する if 式である。もし残っていれば、一本の棒グラフを書いてからまた自分自身を呼び出す。関数が自分自身をもう一度呼び出すかどうかは、結局は 'next-step-expression' が返す値による。これは、短縮版の number-list に作用するような呼び出しを行う。

```
(defun recursive-graph-body-print-internal
  (numbers-list height symbol-width)
  "Print a bar graph.
Used within recursive-graph-body-print function."

  (when numbers-list
        (setq from-position (point))
        (insert-rectangle
         (column-of-graph height (car numbers-list)))
        (goto-char from-position)
        (forward-char symbol-width)
        (sit-for 0)     ; 各桁ごとのグラフの描写
        (recursive-graph-body-print-internal
         (cdr numbers-list) height symbol-width)))
```

これをインストールして実際にテストしてみることが出来る。次にサンプルを挙げる。

```
(recursive-graph-body-print '(3 2 5 6 7 5 3 4 6 4 3 2 1))
```

これを評価すると、次のようなグラフが描かれる。

graph-body-print も recursive-graph-body-print もグラフの本体部分のみを描く関数である。

15.3 軸を表示する

グラフには、それが何を表わすかを示すために軸を表示する必要がある。一度だけしかグラフを描かないのならば、Emacs の Picture mode を利用して手で軸を描くのも結構である。しかし、この関数は何度も利用するかもしれない。

というわけで、基本的な `print-graph-body` 関数を拡張して、自動的に横軸と縦軸のラベルを表示するようにしてみた。この関数には特に新しい事柄は含まれていないので、これについての説明は次の所ですることにする。Appendix C “ラベルと軸が付いたグラフ”, page 171.

15.4 練習問題

上と同様なグラフを表示する関数で、横棒のグラフを描くバージョンを作りなさい。

16 .emacs ファイル

「Emacs を好きになるために Emacs を好きになる必要はない。」– この、一見パラドクスのように聞こえる格言こそが、GNU Emacs の秘密である。ただ箱を開けて取り出したままの Emacs は一般的な道具でしかない。大抵の人は、それを自分自身が使いやすいようにカスタマイズする。

GNU Emacs は多くの部分が Emacs Lisp で書かれている。これは、Emacs Lisp の式を書くことで、変更したり拡張したり出来るということを意味する。

Emacs のデフォルトの設定を素直にありがたく使う人達もいる。それでも Emacs は C のファイルを編集する際は C mode になるし、Fortran のファイルを編集する際は Fortran mode になるし、特に設定されていないファイルについては Fundamental mode になる。誰が Emacs を使うか分らない状況では、これは合理的な設定である。人々がどのファイルに対して何をしたいかを把握出来る人など何処にもいないだろう。Fundamental mode は把握出来ないようなファイルに対しての正しいデフォルトである。ちょうど C mode が C のコードを編集する際の正しいデフォルトであるのと同じである。

しかし、もし誰が Emacs を使おうとしているかが分る場合—例えば自分自身が使う場合—なら Emacs をカスタマイズするのが合理的である。

例えば私は、適当なモードが見つからないようなファイルに対しては Fundamental mode にしたいことは殆どない。そういう場合には Text mode を使うことが多い。そこで Emacs をカスタマイズする。結果として、私にとって使いやすいものになる。

Emacs をカスタマイズしたり拡張したりするには `~/.emacs` ファイルを書くことになる。これは個人用の初期化ファイルである。[1]

中身は Emacs Lisp で書かれ、Emacs に何をするかを伝える。

A `~/.emacs` file contains Emacs Lisp code. You can write this code yourself; or you can use Emacs's `customize` feature to write the code for you. You can combine your own expressions and auto-written Customize expressions in your `.emacs` file.

(I myself prefer to write my own expressions, except for those, particularly fonts, that I find easier to manipulate using the `customize` command. I combine the two methods.)

この章では、簡単な `.emacs` の説明をする。より多くの情報については、以下を参照して欲しい。Section "The Init File" in *The GNU Emacs Manual*, Section "The Init File" in *The GNU Emacs Lisp Reference Manual*.

16.1 サイトごとの初期化ファイル

Emacs は個人的な初期化ファイルに加えて、サイト用の初期化ファイルが存在すれば、それも自動的にロードする。これらは `.emacs` と同様な形式で書かれるが、全員に対してロードされる。

このサイト用の初期化ファイルの中の `site-load.el` と `site-init.el` の二つは、Emacs が dump される際に、ロードされ、かつ標準的なものとして 'dump' される。(Dump することで、Emacs はより速くロード出来る。しかしながら、一旦ファイルがロードされ dump されると、もう一度自分でロードしたり再 dump したりしない限り、変更は Emacs に反映されない。これについては Section "Building Emacs" in *The GNU Emacs Lisp Reference Manual*, 及び `INSTALL` ファイルを参照のこと。)

これら以外にも三つのサイト用の初期化ファイルがあり、それらは存在すれば Emacs の起動時に自動的にロードされる。この三つのファイルというのは、`.emacs` の前にロードされる `site-start.el`、及び `.emacs` の後にロードされる `default.el` と端末タイプファイルである。

`.emacs` での設定や定義は、`site-start.el` での設定や定義を上書きする。そして、`default.el` や端末タイプファイルが更にその設定や定義を上書きすることになる。(端末タイプファイルによる上書きを防ぐには、`term-file-prefix` を `nil` にする。Section 16.11 "ちょっとした拡張", page 151, を参照。)

配布の中に含まれる `INSTALL` ファイルには、`site-init.el` と `site-load.el` についての説明がある。

---

[1] You may also add `.el` to `~/.emacs` and call it a `~/.emacs.el` file. In the past, you were forbidden to type the extra keystrokes that the name `~/.emacs.el` requires, but now you may. The new format is consistent with the Emacs Lisp file naming conventions; the old format saves typing.

loadup.el、startup.el、そして loaddefs.el ファイルがロードをコントロールしている。これらのファイルは Emacs の配布の中の lisp ディレクトリの中にあるので、読んでみると良い。

loaddefs.el ファイルには、何を個人的な .emacs ファイルの中に書き、何をサイト用の初期化ファイルに書くべきなのかということについて多くの有益な示唆が書かれている。

16.2 defcustom による変数の設定

You can specify variables using `defcustom` so that you and others can then use Emacs's `customize` feature to set their values. (You cannot use `customize` to write function definitions; but you can write `defuns` in your `.emacs` file. Indeed, you can write any Lisp expression in your `.emacs` file.)

The `customize` feature depends on the `defcustom` macro. Although you can use `defvar` or `setq` for variables that users set, the `defcustom` macro is designed for the job.

You can use your knowledge of `defvar` for writing the first three arguments for `defcustom`. The first argument to `defcustom` is the name of the variable. The second argument is the variable's initial value, if any; and this value is set only if the value has not already been set. The third argument is the documentation.

The fourth and subsequent arguments to `defcustom` specify types and options; these are not featured in `defvar`. (These arguments are optional.)

Each of these arguments consists of a keyword followed by a value. Each keyword starts with the colon character '`:`'.

For example, the customizable user option variable `text-mode-hook` looks like this:

```
(defcustom text-mode-hook nil
  "Normal hook run when entering Text mode and many related modes."
  :type 'hook
  :options '(turn-on-auto-fill flyspell-mode)
  :group 'wp)
```

The name of the variable is `text-mode-hook`; it has no default value; and its documentation string tells you what it does.

The `:type` keyword tells Emacs the kind of data to which `text-mode-hook` should be set and how to display the value in a Customization buffer.

The `:options` keyword specifies a suggested list of values for the variable. Usually, `:options` applies to a hook. The list is only a suggestion; it is not exclusive; a person who sets the variable may set it to other values; the list shown following the `:options` keyword is intended to offer convenient choices to a user.

Finally, the `:group` keyword tells the Emacs Customization command in which group the variable is located. This tells where to find it.

The `defcustom` macro recognizes more than a dozen keywords. For more information, see Section "Writing Customization Definitions" in *The GNU Emacs Lisp Reference Manual*.

Consider `text-mode-hook` as an example.

There are two ways to customize this variable. You can use the customization command or write the appropriate expressions yourself.

Using the customization command, you can type:

```
M-x customize
```

and find that the group for editing files of data is called 'data'. Enter that group. Text Mode Hook is the first member. You can click on its various options, such as `turn-on-auto-fill`, to set the values. After you click on the button to

```
Save for Future Sessions
```

Emacs will write an expression into your `.emacs` file. It will look like this:

```
(custom-set-variables
  ;; custom-set-variables was added by Custom.
  ;; If you edit it by hand, you could mess it up, so be careful.
  ;; Your init file should contain only one such instance.
  ;; If there is more than one, they won't work right.
 '(text-mode-hook (quote (turn-on-auto-fill text-mode-hook-identify))))
```

(The `text-mode-hook-identify` function tells `toggle-text-mode-auto-fill` which buffers are in Text mode. It comes on automatically.)

The `custom-set-variables` function works somewhat differently than a `setq`. While I have never learned the differences, I modify the `custom-set-variables` expressions in my `.emacs` file by hand: I make the changes in what appears to me to be a reasonable manner and have not had any problems. Others prefer to use the Customization command and let Emacs do the work for them.

Another `custom-set-...` function is `custom-set-faces`. This function sets the various font faces. Over time, I have set a considerable number of faces. Some of the time, I re-set them using `customize`; other times, I simply edit the `custom-set-faces` expression in my `.emacs` file itself.

The second way to customize your `text-mode-hook` is to set it yourself in your `.emacs` file using code that has nothing to do with the `custom-set-...` functions.

When you do this, and later use `customize`, you will see a message that says

```
CHANGED outside Customize; operating on it here may be unreliable.
```

This message is only a warning. If you click on the button to

```
Save for Future Sessions
```

Emacs will write a `custom-set-...` expression near the end of your `.emacs` file that will be evaluated after your hand-written expression. It will, therefore, overrule your hand-written expression. No harm will be done. When you do this, however, be careful to remember which expression is active; if you forget, you may confuse yourself.

So long as you remember where the values are set, you will have no trouble. In any event, the values are always set in your initialization file, which is usually called `.emacs`.

I myself use `customize` for hardly anything. Mostly, I write expressions myself.

Incidentally, to be more complete concerning defines: `defsubst` defines an inline function. The syntax is just like that of `defun`. `defconst` defines a symbol as a constant. The intent is that neither programs nor users should ever change a value set by `defconst`. (You can change it; the value set is a variable; but please do not.)

16.3 .emacs の書き方

Emacs を起動すると、コマンドラインに '`-q`' を指定して読み込まないようにした場合を除き、`.emacs` ファイルがロードされる。(`emacs -q` コマンドでは、素のままの Emacs が起動される。)

`.emacs` ファイルは Lisp 式を含んでいる。大抵は値を設定するだけだが、時には関数定義が書かれていることもある。

初期化ファイルの簡単な説明については、次を参照。See Section "The Init File `~/.emacs`" in *The GNU Emacs Manual*.

この章でも同じようなことを説明するのだが、その際、私が長く使ってきた私自身の `.emacs` を素材にすることにする。

ファイルの最初の部分はコメントからなる。これは個人的なメモとして利用している。勿論、今はその内容を把握しているが、最初はそうではなかった。

```
;;;; Bob's .emacs file
; Robert J. Chassell
; 26 September 1985
```

日付を見てみよう！ このファイルを使い初めたのは随分と昔だ。それ以来、内容を追加してきたわけである。

```
; Each section in this file is introduced by a
; line beginning with four semicolons; and each
; entry is introduced by a line beginning with
; three semicolons.
```

(日本語訳)

```
; このファイルの各セクションは行頭がセミコロンであるような
; 行から始まる。また、各項目は行頭が三つのセミコロンである
; ような行から始まる。
```

ここでは Emacs Lisp のコメントの通常の便宜上の慣習を説明している。セミコロンで始まる行は全てコメントである。二つ、三つ、ないしは四つのセミコロンがセクションやサブセクションのために使われる。(コメントについての説明は次を参照。Section “Comments” in *The GNU Emacs Lisp Reference Manual*.)

```
;;;; The Help Key
; Control-h is the help key;
; after typing control-h, type a letter to
; indicate the subject about which you want help.
; For an explanation of the help facility,
; type control-h two times in a row.
```

(日本語訳)

```
;;;; ヘルプキーについて
; Control-h がヘルプキーである。
; control-h をタイプした後、見たいヘルプの内容を示す
; 文字をタイプする。ヘルプ自体の簡単な説明を見るには
; control-h を二回タイプすればよい。
```

ここでは特に *C-h* を二回タイプすることでヘルプが表示されることを注意しておく。

```
; To find out about any mode, type control-h m
; while in that mode.  For example, to find out
; about mail mode, enter mail mode and then type
; control-h m.
```

(日本語訳)

```
; あるモードがどんなものか知りたい場合は、そのモードで
; control-h m とタイプする。例えば、mail mode について
; 知りたい場合は mail mode に入ってから control-h m と
; タイプする。
```

私はこれを「モードヘルプ (mode help)」と読んでいるのだが、大変便利である。普通はこれを読むだけで必要なことが殆ど分ってしまう。

勿論、このようなコメントは .emacs ファイルに含める必要はない。私がこれらを含めたのは、よくモードヘルプのことやコメントの書き方を忘れたためだ。で、これを見れば思い起こすことが出来たわけである。

16.4 Text モードと Auto Fill モード

さて、Text mode や Auto Fill mode をオンにする箇所まで来た。

```
;;; Text mode and Auto Fill mode
;; The next two lines put Emacs into Text mode
;; and Auto Fill mode, and are for writers who
;; want to start writing prose rather than code.

(setq-default major-mode 'text-mode)
(add-hook 'text-mode-hook 'turn-on-auto-fill)
```

(日本語訳)

```
;;; Text mode と Auto Fill mode
; 次の二つの行は、Emacs を Text mode や Auto Fill mode
; にするためのものである。コードを書くよりは普通の文章
; を書くことが多い人向けの設定である。
(setq-default major-mode 'text-mode)
(add-hook 'text-mode-hook 'turn-on-auto-fill)
```

これが、この .emacs の中で、どこかの忘れっぽい人間のための備忘録以外のちゃんとした役目を果たす最初の部分である。

この括弧で挟まれた二行の内の最初の一行は、ファイルを読み込む際に、そのファイルに C mode のような特定のモードが設定されていない場合には Text mode に入るようにするものである。

Emacs は、ファイルを読む際にその拡張子を見る。(拡張子というのは ‘.’ の後に続く部分である。もしファイルが ‘.c’ や ‘.h’ という拡張子で終わっていれば、Emacs は C mode になる。また、Emacs はファイルの最初の空行でない行も見る。例えば最初の行に ‘-*- C -*-’ と書かれていれば、Emacs は C mode になる。Emacs は自動判別のための拡張子とモード指定のリストを持っている。更に Emacs は、そのバッファの最後のページ付近に “local variable list” があれば、それも見る。

これについては、*The GNU Emacs Manual* の “How Major Modes are Chosen” や “Local Variables in Files” 等のセクションを参照のこと。

さて、`.emacs` ファイルに戻ろう。

もう一度、さっきの行を書いておく。これは何をするのものだろう？

```
(setq major-mode 'text-mode)
```

この行は、短いながらも完全な Emacs Lisp の式である。

我々は既に `setq` には慣れ親しんでいる。これは、後に続く変数 `default-major-mode` をその後の `text-mode` という値にセットするものだ。`text-mode` の先頭の引用符は、Emacs に `text-mode` がシンボルとして何を表しているかを見るのではなく、文字通りそのまま扱うよう指示するものである。`setq` のより詳しい説明については次を参照。Section 1.9 “変数の値の設定”, page 12. 大切なポイントは、`.emacs` で値をセットする手続きは、Emacs の他の場所での手続きと全く違いはないということである。

次は二行目である。

```
(add-hook 'text-mode-hook 'turn-on-auto-fill)
```

この行では、`add-hook` コマンドで `turn-on-auto-fill` を `text-mode-hook` という変数に加えている。

`turn-on-auto-fill` はプログラムの名前であり、名前から推察される通りのことを行う。Auto Fill mode をオンにするのである。

Emacs は Text mode に入るたびに、Text mode にフック (hook) されたコマンドを実行する。従って、Emacs が Text mode になるごとに Auto Fill mode もオンになるわけである。

まとめると、最初の行は、特に拡張子や最初の空でない行や局所変数からどのモードにしてよいか分らない場合には Text mode になるよう設定するものである。

Text mode では、物書きには便利な構文テーブルが設定されたりする。例えば、アポストロフィ ‘’’ は普通の文字ように単語の一部だと解釈するが、終止符や空白は単語の一部だとは考えない。従って *M-f* では ‘`it's`’ 全体を飛び越えて進む。一方、C mode では、‘`it's`’ のちょうど ‘`t`’ の直後の位置で止まる。

また二行目は、Text mode に入る際に同時に Auto Fill mode もオンにするためのものである。Auto Fill mode では、Emacs は自動的に長い行を次の行に折り返してくれる。その際は、単語を寸断したりせずちゃんと単語と単語の間を折り返すようになっている。

Auto Fill mode がオフになっている場合は、行はあなたがタイプしていくとそのまま右に伸びていく。行が右端まで到達した場合、`truncate-lines` の値をどう設定しているかによって、右の部分が見えなくなったり、あるいは汚い形で表示されたりする。

In addition, in this part of my `.emacs` file, I tell the Emacs fill commands to insert two spaces after a colon:

```
(setq colon-double-space t)
```

16.5 メールのエイリアス

以下が mail aliases 等を使えるようにする `setq` 式、及びそれについての備忘録である。

```
;;; Mail mode
; To enter mail mode, type `C-x m'
; To enter RMAIL (for reading mail),
; type `M-x rmail'
(setq mail-aliases t)
```

(日本語訳)

```
;;; Mail mode
; Mail mode に入るには、`C-x m' とタイプする。
; (mail を読むために) RMAIL にはいるには、
; `M-x rmail' とタイプする。
(setq mail-aliases t)
```

この `setq` コマンドは、変数 `mail-aliases` の値を `t` にセットする。`t` は真を意味するので、この行は「どうぞ、mail aliases を使って下さい」という意味になる。

Mail aliases というのは、長い電子メールアドレスや電子メールアドレスのリストに対する便宜上の短い名前のことである。あなたの ‘aliases’ を保存しておくファイルは `~/.mailrc` である。alias は次のような形式で書けばよい。

```
alias geo george@foobar.wiz.edu
```

George にメッセージを書く場合は、アドレスに 'geo' と書くだけでよい。するとメイラーは自動的に 'geo' をフルアドレスに展開してくれる。

16.6 Indent Tabs モード

Emacs はデフォルトでは、リージョンを整形する際に沢山の空白の連続にはタブを挿入する。(例えばあなたは、沢山の行を一度にインデントするために `indent-region` を使ったりすることだろう。) タブは端末や普通の印刷では綺麗に見える。しかし TEX ではタブは無視されるために、TEX や Texinfo を使う際には出力が汚くなってしまう。

次のようにすれば、Indent Tabs mode をオフに出来る。

```
;;; Prevent Extraneous Tabs
(setq-default indent-tabs-mode nil)
```

この行では、今まで使ってきた `setq` ではなく `setq-default` を使っていることに注意しよう。`setq-default` コマンドは、その変数に対するバッファローカルな値がない場合に限り、その値をセットする。

The GNU Emacs Manual の中の "Tabs vs. Spaces" や "Local Variables in Files" を参照。

16.7 幾つかのキーバインディング

次は、幾つかの個人的なキーバインディングの設定である。

```
;;; Compare windows
(global-set-key "\C-cw" 'compare-windows)
```

`compare-windows` は、現在のウィンドウと次のウィンドウの中のテキストを比較してくれるという、洒落たコマンドである。これは、各々のウィンドウの中のポイントの位置から比較を開始して、両方が同じである所まで進む。私はこのコマンドをいつも利用している。

また、これを見れば、全てのモードに対する大域的なキーバインディングを設定する方法が分る。

コマンドは `global-set-key` だ。この後にキーバインディングが続く。`.emacs` ファイルではキーバインディングは上に示した通りの方法で書かれる。`\C-c` は 'control-c', つまり、「control キーと *c* を同時に押す」ことを表わす。また、`w` は「`w` を押す」ことを表わす。説明の中では、*C-c w* のように書かれる。(`CTL` キーではなく *M-c* のような `META` キーを使ったバインドをしたい場合は、`\M-c` のように書く。詳しくは次を参照。Section "Rebinding Keys in Your Init File" in *The GNU Emacs Manual*.)

このキーで呼び出されるコマンドは `compare-windows` である。この時、`compare-windows` の先頭に引用符を付けることに注意しよう。そうしないと、Emacs は最初にこのシンボルの値を評価しようとしてしまう。

二重引用符と、'C' の前のバックスラッシュ、及び単独の引用符の三つはキーバインディングには必須であるにもかかわらず、私はよく忘れてしまう。幸いなことに、これらは `.emacs` を見れば思い出すことが出来る。

C-c w というキーバインディングそのものについても書いておく。これは、*C-c* という前置キーと一文字—今の場合は *w*—からなっている。このような *C-c* に続けて一文字というキー設定は、特に個人的な用途に残されている。もしあなたが Emacs を拡張したい場合、これらのキーを公の用途に使わないで欲しい。その代わり、*C-c C-w* のようなキーを使うべきである。そうでないと個人用にバインドするキーがなくなってしまう。

以下にコメント付きで他のキーバインディングを書いておく。

```
;;; Keybinding for 'occur'
; I use occur a lot, so let's bind it to a key:
; (私は occur をよく使うので、キーにバインドしておく。)
(global-set-key "\C-co" 'occur)
```

`occur` コマンドは、現在のバッファの中で正規表現にマッチする部分を含む行を全て表示してくれるコマンドである。マッチした行は、`*Occur*` というバッファに表示される。このバッファはその位置に簡単に飛ぶためにも使われる。

次に、キーバインディングの解除の仕方を示す。特定のキーバインディングを使えなくするわけである。

```
;;; Unbind ‘C-x f’
(global-unset-key "\C-xf")
```

このキーバインディングを解除するのには理由がある。私は、自分がよく *C-x C-f* とタイプしようとして、間違って *C-x f* とタイプしてしまうことに気が付いた。ファイルを読み込もうとして、偶然行詰めの幅を設定してしまうわけである。結果的に大抵は変な幅が設定されてしまう。私はこの幅を変更することは殆どないので、単にキーバインディングを解除することにしたのだ。

次は現在のキーをバインドし直すものである。

```
;;; Rebind ‘C-x C-b’ for ‘buffer-menu’
(global-set-key "\C-x\C-b" 'buffer-menu)
```

デフォルトでは、*C-x C-b* は `list-buffer` コマンドを起動するものである。このコマンドは現在のバッファのリストを別のウィンドウに表示する。私は大抵はそのウィンドウで何かをしたいことが多いので `buffer-menu` コマンドの方が好きである。これは単にバッファのリストを表示するだけでなく、ポイントをそのウィンドウに移動してくれる。

16.8 キーマップ

Emacs はどのキーをどのコマンドにバインドするかを *keymaps* を使って記録する。When you use `global-set-key` to set the keybinding for a single command in all parts of Emacs, you are specifying the keybinding in `current-global-map`.

C mode や Text mode のような特定のモードは、それ自身のキーマップを持っている。モード独自のキーマップは、全てのバッファで共有される global map を上書きする。

`global-set-key` 関数は global keymap のキーをバインドしたり再バインドしたりする。例えば次の S 式は、キー *C-c C-l* を関数 `line-to-top-of-window` にバインドする。

```
(global-set-key "\C-x\C-b" 'buffer-menu)
```

モード独自のキーマップは `define-key` 関数を使ってバインドされる。これは、キーやコマンドと同時に特定のキーマップも引数に取る。例えば私の `.emacs` ファイルでは、`texinfo-insert-@group` コマンドを *C-c C-c g* にバインドするために、次のような S 式が書かれている。

```
(define-key texinfo-mode-map "\C-c\C-cg" 'texinfo-insert-@group)
```

`texinfo-insert-@group` 関数そのものは、Texinfo ファイルに ‘`@group`’ をインサートするためのちょっとした Texinfo mode の拡張である。私はこのコマンドをいつも使うので、六つのキーストローク *@ g r o u p* よりも三つのキーストロークである *C-c C-c g* を使う方がずっと便利なのだ。(‘`@group`’ と対応する ‘`@end group`’ は中のテキストを一つのページに収めるためのコマンドである。この本の中の、沢山の複数行に渡る例が ‘`@group ... @end group`’ で囲まれている。)

以下が `texinfo-insert-@group` の関数定義である。

```
(defun texinfo-insert-@group ()
  "Insert the string @group in a Texinfo buffer."
  (interactive)
  (beginning-of-line)
  (insert "@group\n"))
```

(タイプ数を節約するためには、勿論、単語を挿入する関数を定義したりする代わりに Abbrev mode を使うことも出来る。しかし、キーストロークが他の Texinfo mode のキーバインディングと調和が取れているという意味で、私は上の方法の方が好みである。)

`loaddefs.el` を見れば `c-mode.el` や `lisp-mode.el` のような様々なモードライブラリと同時に沢山の *define-key* 式が書かれているのが分る。

キーマップについての詳細は次を参照のこと。Section “Customizing Key Bindings” in *The GNU Emacs Manual*, 及び Section “Keymaps” in *The GNU Emacs Lisp Reference Manual*.

16.9 ファイルのロード

GNU Emacs のコミュニティの沢山の人々が Emacs の拡張を行ってきた。時が経つにつれ、これらの拡張が新しいリリースに含まれることも多くなった。例えばカレンダーや日記のパッケージは今や Emacs version 19 の配布の一部になっている。これらは標準の Emacs version 18 の配布には含まれていなかったものだ。

load コマンドを使うと、ファイルの中身全てを評価することが出来る。従って、そのファイルの関数や変数を全て Emacs にインストールすることが可能だ。例えば、

```
(load "~/emacs/slowsplit")
```

というS式は、ホームディレクトリの emacs というサブディレクトリの下の slowsplit.el ファイルを評価、即ちロードする。(もしバイトコンパイルされた slowsplit.elc があれば、そちらがロードされる。こっちの方が速くロード、実行出来る。) The file contains the function split-window-quietly, which John Robinson wrote in 1989.

The split-window-quietly function splits a window with the minimum of redisplay. I installed it in 1989 because it worked well with the slow 1200 baud terminals I was then using. Nowadays, I only occasionally come across such a slow connection, but I continue to use the function because I like the way it leaves the bottom half of a buffer in the lower of the new windows and the top half in the upper window.

To replace the key binding for the default split-window-vertically, you must also unset that key and bind the keys to split-window-quietly, like this:

```
(global-unset-key "\C-x2")
(global-set-key "\C-x2" 'split-window-quietly)
```

沢山の拡張パッケージをロードしたい場合は、上のようにそのファイルの位置を正確に指定するのではなく、あるディレクトリを Emacs の load-path に指定するとよい。私はそうしている。この場合には、Emacs はファイルをロードする際にデフォルトのディレクトリのリストに加えてそのディレクトリも検索してくれるようになる。(デフォルトのディレクトリは Emacs を作成する際に paths.h で指定される。)

次のコマンドで、~/emacs ディレクトリを現在のロードパスに含めることが出来る。

```
;;; Emacs Load Path
(setq load-path (cons "~/emacs" load-path))
```

ついでに言っておくと、load-library は関数をロードする際のインタラクティブなインターフェースも提供してくれる。この関数の完全なコードは次のようになっている。

```
(defun load-library (library)
  "Load the library named LIBRARY.
This is an interface to the function `load'."
  (interactive
   (list (completing-read "Load library: "
                          (apply-partially 'locate-file-completion-table
                                           load-path
                                           (get-load-suffixes)))))
  (load library))
```

load-library という関数の名前では 'library' という言葉を便宜的に 'file' の同意語として見ている。load-library コマンドのソースは files.el ライブラリにある。

これと若干違う動作をするインタラクティブなコマンドとして、load-file がある。これが load-library とどう違うかについてのより詳しい情報は次を参照。Section "Libraries of Lisp Code for Emacs" in *The GNU Emacs Manual*.

16.10 オートロード

その関数を含むファイルをロードしたり、関数定義を直接評価したりしてインストールする代わりに、その関数が実際に最初に使われるまではインストールはしないが、いつでも呼び出せる状態にはしておくということも出来る。これはオートロード (*autoloading*) と呼ばれる。

オートロードされた関数を実行すると、Emacs は自動的にその定義を含むファイルを評価して、その関数を呼び出す。

オートロード関数を使うと Emacs をより速く起動出来る。というのも起動時にはそのライブラリはロードされないからだ。その代わり、最初にオートロードされた関数を使う際には、それを含むファイルを評価する間ちょっと待たされることになる。

めったに使われない関数はしばしばオートロードされる。loaddefs.el ライブラリには bookmark-set から wordstar-mode まで何百ものオートロードされる関数が含まれている。勿論、個人的にこれらの「めったに使わない」関数を頻繁に使うことになることもあるかもしれない。この場合は、その関数のファイルを load 式を使って .emacs ファイルの中でロードしておくべきだろう。

私の Emacs version 19.23 用の .emacs ファイルでは、元々オートロードされるような関数を含む 17 のライブラリがロードされている。(実際には、これらの関数を「dump」した Emacs を作るべきなのだが、忘れていたのだ。dump についての詳しい情報は、Section “Building Emacs” in *The GNU Emacs Lisp Reference Manual*, や INSTALL を見て欲しい。)

あるいはまた .emacs ファイルにオートロード式を書きたいこともあるだろう。autoload は五つの引数を取る組み込み関数であり、最後の三つの引数は省略可能である。最初の引数はオートロードする関数の名前、二番目の引数はロードされるファイル名、三番目は関数の説明文、四番目はその関数をインタラクティブに呼び出せるかどうか、そして最後の五番目の引数は、オブジェクトのタイプ—autoload は関数だけでなく、キーマップやマクロも扱えるので—である。(デフォルトは関数である。)

以下に典型的な例を挙げる。

```
(autoload 'html-helper-mode
  "html-helper-mode" "Edit HTML documents" t)
```

(html-helper-mode is an older alternative to html-mode, which is a standard part of the distribution.)

この式は、html-helper-mode 関数をオートロードするものである。これは、html-helper-mode.el ファイルの中でロードされる。(あるいは、もし html-helper-mode.elc があれば、そちらから読み込まれる。) このファイルは load-path で指定されたディレクトリに置かれていなければならない。説明文を読むと、これは HyperText Markup Language で書かれた文書を編集するためのモードであることが分る。このモードは *M-x html-helper-mode* とタイプすることでインタラクティブに呼び出すことが出来る。(この場合、オートロード式の説明文には正規の説明文をそのまま複写する必要がある。正規の説明文はまだロードされていないので、まだ読めないわけである。)

詳しくは、次を参照。See Section “Autoload” in *The GNU Emacs Lisp Reference Manual*.

16.11 ちょっとした拡張: line-to-top-of-window

次に、ポイントのある行をウィンドウの最上段に移動する簡単な Emacs の拡張を書いておく。これを使うとテキストを簡単に見ることが出来るので、私は普段よく利用している。

次のコードを別ファイルにして、それを .emacs でロードしても良いし、コードを直接 .emacs の中に含めてしまうことも出来る。

以下が定義である。

```
;;; Line to top of window;
;;; replace three keystroke sequence  C-u 0 C-l
(defun line-to-top-of-window ()
  "Move the line point is on to top of window."
  (interactive)
  (recenter 0))
```

次にキーバインディングである。

Nowadays, function keys as well as mouse button events and non-ASCII characters are written within square brackets, without quotation marks. (In Emacs version 18 and before, you had to write different function key bindings for each different make of terminal.)

I bind line-to-top-of-window to my F6 function key like this:

```
(global-set-key [f6] 'line-to-top-of-window)
```

より詳しいことについては、次を見て欲しい。Section “Rebinding Keys in Your Init File” in *The GNU Emacs Manual*.

If you run two versions of GNU Emacs, such as versions 22 and 23, and use one .emacs file, you can select which code to evaluate with the following conditional:

```
(cond
 ((= 22 emacs-major-version)
  ;; evaluate version 22 code
  ( ... ))
 ((= 23 emacs-major-version)
  ;; evaluate version 23 code
  ( ... )))
```

For example, recent versions blink their cursors by default. I hate such blinking, as well as other features, so I placed the following in my .emacs file[2]:

```
(when (>= emacs-major-version 21)
  (blink-cursor-mode 0)
  ;; Insert newline when you press `C-n' (next-line)
  ;; at the end of the buffer
  (setq next-line-add-newlines t)
  ;; Turn on image viewing
  (auto-image-file-mode t)
  ;; Turn on menu bar (this bar has text)
  ;; (Use numeric argument to turn on)
  (menu-bar-mode 1)
  ;; Turn off tool bar (this bar has icons)
  ;; (Use numeric argument to turn on)
  (tool-bar-mode nil)
  ;; Turn off tooltip mode for tool bar
  ;; (This mode causes icon explanations to pop up)
  ;; (Use numeric argument to turn on)
  (tooltip-mode nil)
  ;; If tooltips turned on, make tips appear promptly
  (setq tooltip-delay 0.1)  ; default is 0.7 second
   )
```

16.12 X11 でのカラー表示

X Window System を使うと、Emacs をカラーで使うことが出来る。

私はデフォルトのカラーが気に入らないので、自分独自の色を指定している。

Here are the expressions in my .emacs file that set values:

```
;; Set cursor color
(set-cursor-color "white")

;; Set mouse color
(set-mouse-color "white")

;; Set foreground and background
(set-foreground-color "white")
(set-background-color "darkblue")

;;; Set highlighting colors for isearch and drag
(set-face-foreground 'highlight "white")
(set-face-background 'highlight "blue")

(set-face-foreground 'region "cyan")
(set-face-background 'region "blue")

(set-face-foreground 'secondary-selection "skyblue")
(set-face-background 'secondary-selection "darkblue")

;; Set calendar highlighting colors
(setq calendar-load-hook
      (lambda ()
        (set-face-foreground 'diary-face   "skyblue")
        (set-face-background 'holiday-face "slate blue")
        (set-face-foreground 'holiday-face "white")))
```

色々な色合いの青が、スクリーンのちらつきから私の眼を守り、そして癒してくれる。

[2] When I start instances of Emacs that do not load my .emacs file or any site file, I also turn off blinking:

```
emacs -q --no-site-file -eval '(blink-cursor-mode nil)'
```

Or nowadays, using an even more sophisticated set of options,

```
emacs -Q -D
```

Alternatively, I could have set my specifications in various X initialization files. For example, I could set the foreground, background, cursor, and pointer (i.e., mouse) colors in my ~/.Xresources file like this:

```
Emacs*foreground:   white
Emacs*background:   darkblue
Emacs*cursorColor:  white
Emacs*pointerColor: white
```

In any event, since it is not part of Emacs, I set the root color of my X window in my ~/.xinitrc file, like this[3]:

```
xsetroot -solid Navy -fg white &
```

16.13 .emacs での雑多な設定

以下は、雑多な設定である。

– Set the shape and color of the mouse cursor:

```
; Cursor shapes are defined in
; `/usr/include/X11/cursorfont.h';
; for example, the `target' cursor is number 128;
; the `top_left_arrow' cursor is number 132.

(let ((mpointer (x-get-resource "*mpointer"
                                "*emacs*mpointer")))
  ;; If you have not set your mouse pointer
  ;;     then set it, otherwise leave as is:
  (if (eq mpointer nil)
      (setq mpointer "132")) ; top_left_arrow
  (setq x-pointer-shape (string-to-int mpointer))
  (set-mouse-color "white"))
```

– Or you can set the values of a variety of features in an alist, like this:

```
(setq-default
 default-frame-alist
 '((cursor-color . "white")
   (mouse-color . "white")
   (foreground-color . "white")
   (background-color . "DodgerBlue4")
   ;; (cursor-type . bar)
   (cursor-type . box)
   (tool-bar-lines . 0)
   (menu-bar-lines . 1)
   (width . 80)
   (height . 58)
   (font .
         "-Misc-Fixed-Medium-R-Normal--20-200-75-75-C-100-ISO8859-1")
   ))
```

– Convert *CTRL-h* into DEL and DEL into *CTRL-h*.
(Some older keyboards needed this, although I have not seen the problem recently.)

```
;; Translate `C-h' to <DEL>.
; (keyboard-translate ?\C-h ?\C-?)

;; Translate <DEL> to `C-h'.
(keyboard-translate ?\C-? ?\C-h)
```

– Turn off a blinking cursor!

```
(if (fboundp 'blink-cursor-mode)
    (blink-cursor-mode -1))
```

or start GNU Emacs with the command emacs -nbc.

[3] I also run more modern window managers, such as Enlightenment, Gnome, or KDE; in those cases, I often specify an image rather than a plain color.

– When using 'grep'
  '-i' Ignore case distinctions
  '-n' Prefix each line of output with line number
  '-H' Print the filename for each match.
  '-e' Protect patterns beginning with a hyphen character, '-'

```
(setq grep-command "grep -i -nH -e ")
```

– Find an existing buffer, even if it has a different name
  This avoids problems with symbolic links.

```
(setq find-file-existing-other-name t)
```

– Set your language environment and default input method

```
(set-language-environment "latin-1")
;; Remember you can enable or disable multilingual text input
;; with the toggle-input-method' (C-\) command
(setq default-input-method "latin-1-prefix")
```

  If you want to write with Chinese 'GB' characters, set this instead:

```
(set-language-environment "Chinese-GB")
(setq default-input-method "chinese-tonepy")
```

Fixing Unpleasant Key Bindings

Some systems bind keys unpleasantly. Sometimes, for example, the CTRL key appears in an awkward spot rather than at the far left of the home row.

Usually, when people fix these sorts of keybindings, they do not change their ~/.emacs file. Instead, they bind the proper keys on their consoles with the loadkeys or install-keymap commands in their boot script and then include xmodmap commands in their .xinitrc or .Xsession file for X Windows.

For a boot script:

```
loadkeys /usr/share/keymaps/i386/qwerty/emacs2.kmap.gz
```

or

```
install-keymap emacs2
```

For a .xinitrc or .Xsession file when the Caps Lock key is at the far left of the home row:

```
# Bind the key labeled 'Caps Lock' to 'Control'
# (Such a broken user interface suggests that keyboard manufacturers
# think that computers are typewriters from 1885.)

xmodmap -e "clear Lock"
xmodmap -e "add Control = Caps_Lock"
```

In a .xinitrc or .Xsession file, to convert an ALT key to a META key:

```
# Some ill designed keyboards have a key labeled ALT and no Meta
xmodmap -e "keysym Alt_L = Meta_L Alt_L"
```

16.14 モード行の修正

最後に、モード行の修正について述べる。これが変更出来るようになっているのは本当に嬉しい。

私は時々ネットワーク上で仕事をするので、私が現在どのマシンの上で仕事をしているか忘れてしまうことがある。Also, I tend to I lose track of where I am, and which line point is on.

So I reset my mode line to look like this:

```
-:-- foo.texi   rattlesnake:/home/bob/  Line 1  (Texinfo Fill) Top
```

I am visiting a file called foo.texi, on my machine rattlesnake in my /home/bob buffer. I am on line 1, in Texinfo mode, and am at the top of the buffer.

My .emacs file has a section that looks like this:

```
;; Set a Mode Line that tells me which machine, which directory,
;; and which line I am on, plus the other customary information.
(setq-default mode-line-format
 (quote
  (#("-" 0 1
     (help-echo
      "mouse-1: select window, mouse-2: delete others ..."))
   mode-line-mule-info
   mode-line-modified
   mode-line-frame-identification
   "    "
   mode-line-buffer-identification
   "    "
   (:eval (substring
           (system-name) 0 (string-match "\\..+" (system-name))))
   ":"
   default-directory
   #(" " 0 1
     (help-echo
      "mouse-1: select window, mouse-2: delete others ..."))
   (line-number-mode " Line %l ")
   global-mode-string
   #("   %[(" 0 6
     (help-echo
      "mouse-1: select window, mouse-2: delete others ..."))
   (:eval (mode-line-mode-name))
   mode-line-process
   minor-mode-alist
   #("%n" 0 2 (help-echo "mouse-2: widen" local-map (keymap ...)))
   ")%] "
   (-3 . "%P")
   ;;   "-%-"
   )))
```

Here, I redefine the default mode line. Most of the parts are from the original; but I make a few changes. I set the *default* mode line format so as to permit various modes, such as Info, to override it.

(旧訳) *default* のモード行のフォーマットは Info のような様々なモードが上書きすることを許すようにした。リストの中の多くの要素は、名前から役目が推測出来るようになっている。例えば `mode-line-modified` はバッファが修正された場合、それを教えてくれるようにする変数だし、`mode-name` はモードの名前を教えてくれる、等々。

Many elements in the list are self-explanatory: `mode-line-modified` is a variable that tells whether the buffer has been modified, `mode-name` tells the name of the mode, and so on. However, the format looks complicated because of two features we have not discussed.

(旧訳) '`"%14b"`' は現在のバッファの名前を (お馴染みの `buffer-name` 関数を使って) 表示する。'14' は表示する文字数の最大の指定である。それよりも文字数が少ない場合は空白文字で埋めてくれる。'`%[`' と '`%]`' は各々の再帰編集レベルに応じて角括弧の組を表示するものである。'`%n`' はナローイングをしている時に 'Narrow' と表示してくれる。'`%P`' はバッファ全体でウィンドウの最下段より上にある部分が何パーセントあるか、もしくは 'Top' か 'Bottom' かを表示してくれる。(小文字の '`p`' にすると、ウィンドウの最上段より上の部分がどのくらいかを表示する。) '`%-`' は行の残りをダッシュ (訳註：'-' のこと) で埋めてくれる。

(旧訳) Emacs version 19.29 以降では、`frame-title-format` を使って Emacs のフレームにタイトルを表示することが出来る。この変数は `mode-line-format` と同じ構造をしている。

The first string in the mode line is a dash or hyphen, '`-`'. In the old days, it would have been specified simply as `"-"`. But nowadays, Emacs can add properties to a string, such as highlighting or, as in this case, a help feature. If you place your mouse cursor over the hyphen, some help information appears (By default, you must wait seven-tenths of a second before the information appears. You can change that timing by changing the value of `tooltip-delay`.)

The new string format has a special syntax:

```
#("-" 0 1 (help-echo "mouse-1: select window, ..."))
```

The `#(` begins a list. The first element of the list is the string itself, just one '`-`'. The second and third elements specify the range over which the fourth element applies. A range starts *after* a character, so a zero means the range starts just before the first character; a 1 means that the range ends just after the first character. The third element is the property for the range. It consists of a property list, a property name, in this case, '`help-echo`', followed by a value, in this case, a string. The second, third, and fourth elements of this new string format can be repeated.

より詳細な情報については、See Section "Text Properties" in *The GNU Emacs Lisp Reference Manual*, および Section "Mode Line Format" in *The GNU Emacs Lisp Reference Manual.* を参照されたい。

`mode-line-buffer-identification` displays the current buffer name. It is a list beginning `(#("%12b" 0 4 ....` The `#(` begins the list.

The '`"%12b"`' displays the current buffer name, using the `buffer-name` function with which we are familiar; the '12' specifies the maximum number of characters that will be displayed. When a name has fewer characters, whitespace is added to fill out to this number. (Buffer names can and often should be longer than 12 characters; this length works well in a typical 80 column wide window.)

`:eval` says to evaluate the following form and use the result as a string to display. In this case, the expression displays the first component of the full system name. The end of the first component is a '`.`' ('period'), so I use the `string-match` function to tell me the length of the first component. The substring from the zeroth character to that length is the name of the machine.

This is the expression:

```
(:eval (substring
        (system-name) 0 (string-match "\\..+" (system-name))))
```

'`%[`' and '`%]`' cause a pair of square brackets to appear for each recursive editing level. '`%n`' says 'Narrow' when narrowing is in effect. '`%P`' tells you the percentage of the buffer that is above the bottom of the window, or 'Top', 'Bottom', or 'All'. (A lower case '`p`' tell you the percentage above the *top* of the window.) '`%-`' inserts enough dashes to fill out the line.

もう一度「Emacs を好きになるために Emacs を好きになる必要はない。」という言葉を思い出して欲しい—あなた自身の Emacs はデフォルトの Emacs とは異なる色や異なるコマンド、そして異なるキーマップを持つことが出来るのだ。

一方で、もし「箱から取り出したままの」カスタマイズされていない素の Emacs を起動したい場合は、次のようにタイプしよう。

```
emacs -q
```

こうすると、Emacs はあなたの初期化ファイル `~/.emacs` をロードしないで起動する。これが素のままのデフォルトの Emacs である。

17 デバッグ

GNU Emacs は二つのデバッガを持っている。debug と edebug である。最初のものは、Emacs の内部に組み込まれていて、常に使える状態にある。二番目のものは拡張機能として与えられているもので、version 19 では標準の配布の中に含まれている。

これらのデバッガは共に、以下の所に非常に詳しく説明されている。Section “Debugging Lisp Program” in *The GNU Emacs Lisp Reference Manual*. この章では、各々の簡単な例を見ていくことにする。

17.1 debug

例えばあなたが、1 から与えられた数までの和を計算するための関数の定義を書いたとしよう。(これは以前に説明した triangle 関数である。次の所で詳しく議論されている。Section 11.1.4 “減少するカウンタを使ったループ”, page 87.)

しかしながら、あなたの関数にはバグがあったとしよう。例えば ‘1-’ と書くべきところを ‘1=’ と書いてしまったとする。この間違いをしたプログラムは以下の通りである。

```
(defun triangle-bugged (number)
  "Return sum of numbers 1 through NUMBER inclusive."
  (let ((total 0))
    (while (> number 0)
      (setq total (+ total number))
      (setq number (1= number)))      ; ここでエラー
    total))
```

もし、この文章を Info で読んでいるなら、この関数をいつものように評価することが出来る。そうすると、エコー領域には triangle-bugged と表示されるはずだ。

さて、triangle-bugged を引数 4 とともに評価してみよう。

```
(triangle-bugged 4)
```

In a recent GNU Emacs, you will create and enter a *Backtrace* buffer that says:

```
---------- Buffer: *Backtrace* ----------
Debugger entered--Lisp error: (void-function 1=)
  (1= number)
  (setq number (1= number))
  (while (> number 0) (setq total (+ total number))
         (setq number (1= number)))
  (let ((total 0)) (while (> number 0) (setq total ...)
    (setq number ...)) total)
  triangle-bugged(4)
  eval((triangle-bugged 4))
  eval-last-sexp-1(nil)
  eval-last-sexp(nil)
  call-interactively(eval-last-sexp)
---------- Buffer: *Backtrace* ----------
```

(I have reformatted this example slightly; the debugger does not fold long lines. As usual, you can quit the debugger by typing *q* in the *Backtrace* buffer.)

実際は、このような単純なバグなら、エラーメッセージだけでどう修正すればよいか分るだろう。The function 1= is ‘void’.

しかし、分らない場合はどうすればよいか？ You can read the complete backtrace.

In this case, you need to run a recent GNU Emacs, which automatically starts the debugger that puts you in the *Backtrace* buffer; or else, you need to start the debugger manually as described below.

Read the *Backtrace* buffer from the bottom up; it tells you what Emacs did that led to the error. Emacs made an interactive call to *C-x C-e* (eval-last-sexp), which led to the evaluation of the triangle-bugged expression. Each line above tells you what the Lisp interpreter evaluated next.

上から三行目は

```
(setq number (1= number))
```

である。Emacs はこの S 式を評価しようとした。そのために、内部の S 式を評価しようとした。それが上から二行目の式である。

```
(1= number)
```

エラーが生じたのは、ここである。一行目にはそう書かれている。

```
Debugger entered--Lisp error: (void-function 1=)
```

そこでエラーを修正し、再び関数定義を評価して、もう一度テストしてみることになる。

17.2 debug-on-entry

A recent GNU Emacs starts the debugger automatically when your function has an error.

Incidentally, you can start the debugger manually for all versions of Emacs; the advantage is that the debugger runs even if you do not have a bug in your code. Sometimes your code will be free of bugs!

ある関数に対して debug を実行するには debug-on-entry を使う。

まず、次のようにタイプしてみよう。

```
M-x debug-on-entry RET triangle-bugged RET
```

そして、次を評価する。

```
(triangle-bugged 5)
```

All versions of Emacs will create a *Backtrace* buffer and tell you that it is beginning to evaluate the triangle-bugged function:

(旧訳) Emacs は *Backtrace* バッファを作成し、triangle-bugged 関数の評価の開始を知らせてくる。

```
---------- Buffer: *Backtrace* ----------
Debugger entered--entering a function:
* triangle-bugged(5)
  eval((triangle-bugged 5))
  eval-last-sexp-1(nil)
  eval-last-sexp(nil)
  call-interactively(eval-last-sexp)
---------- Buffer: *Backtrace* ----------
```

Backtrace バッファの中で、*d* とタイプしてみよう。すると、Emacs は triangle-bugged の最初の S 式を評価する。このときバッファは次のようになる。

```
---------- Buffer: *Backtrace* ----------
Beginning evaluation of function call form:
* (let ((total 0)) (while (> number 0) (setq total ...)
         (setq number ...)) total))
  triangle-bugged(5)
* eval((triangle-bugged 5))
  eval-last-sexp(nil)
* call-interactively(eval-last-sexp)
---------- Buffer: *Backtrace* ----------
```

ここで、ゆっくりと、更に 8 回 *d* をタイプしてみよう。*d* をタイプするごとに、Emacs は関数定義の中の別の S 式を評価していく。

結果として、このバッファは次のようになる。

```
---------- Buffer: *Backtrace* ----------
Debugger entered--beginning evaluation of function call form:
* (setq number (1= number))
* (while (> number 0) (setq total (+ total number))
        (setq number (1= number)))

* (let ((total 0)) (while (> number 0) (setq total ...)
        (setq number ...)) total)
* triangle-bugged(5)
  eval((triangle-bugged 5))

  eval-last-sexp-1(nil)
  eval-last-sexp(nil)
  call-interactively(eval-last-sexp)
---------- Buffer: *Backtrace* ----------
```

最後に、更に 2 回 *d* をタイプすると Emacs はエラーの箇所まで辿り着く。その結果 *Backtrace* の上から二行は次のようになる。

```
---------- Buffer: *Backtrace* ----------
Debugger entered--Lisp error: (void-function 1=)
* (1= number)
...
---------- Buffer: *Backtrace* ----------
```

d をタイプすることで、関数を一つずつ実行していくことが出来る。

Backtrace バッファを抜けるには *q* をタイプする。こうするとトレースは抜けるが、debug-on-entry はキャンセルしない。

debug-on-entry そのものを無効にするには、cancel-debug-on-entry に呼び出して関数名を入力すればよい。次のような感じだ。

```
M-x cancel-debug-on-entry RET triangle-bugged RET
```

(もしこれを Info で読んでいるなら、ここで debug-on-entry を無効にしておこう。)

17.3 debug-on-quit と (debug)

debug-on-error をセットしたり debug-on-entry を呼び出したりする以外にも二つ程、debug を開始する方法がある。

一つは変数 debug-on-quit を t にセットすることで、こうすると *C-g* (keyboard-quit) をタイプすることで常に debug を開始することが出来る。これは無限ループをデバッグする際に有効である。

もう一つは、コードの中のデバッグを開始したい所に (debug) を呼び出す行を挿入する方法である。例えば次のような感じである。

```
(defun triangle-bugged (number)
  "Return sum of numbers 1 through NUMBER inclusive."
  (let ((total 0))
    (while (> number 0)
      (setq total (+ total number))
      (debug)                          ; デバッガ起動
      (setq number (1= number)))       ; ここでエラー
    total))
```

debug 関数は次の中で詳しく説明されている。 Section “The Lisp Debugger” in *The GNU Emacs Lisp Reference Manual*.

17.4 ソースレベルのデバッガ edebug

Edebug は通常デバッグをしているコードのソースを表示してくれる。そして、現在どの行を実行しているかを左側の矢印で示してくれる。

関数を一行ごとに実行することも出来るし、実行を止める *breakpoint* の所まで一気に走らせることも出来る。

Edebug については次の所に説明がある。Section “Edebug” in *The GNU Emacs Lisp Reference Manual*.

次にバグを含んだ `triangle-recursively` の関数定義を挙げておく。これについて復習したい場合には、次を参照。Section 11.3.4 “カウンタの代わりに再帰を使う”, page 93. この例は、後で説明されているように `defun` の左のインデント無しで表示されている。

```
(defun triangle-recursively-bugged (number)
  "Return sum of numbers 1 through NUMBER inclusive.
Uses recursion."
  (if (= number 1)
      1
    (+ number
       (triangle-recursively-bugged
        (1= number)))))               ; ここでエラー
```

普通は、この関数定義をインストールするにはカーソルを関数の最後の閉じ括弧の後で *C-x C-e* (`eval-last-sexp`) とタイプするか、関数定義内にカーソルを置いて *C-M-x* (`eval-defun`) とタイプする。(デフォルトでは `eval-defun` コマンドは Emacs Lisp mode か Lisp Interaction mode でしか動作しない。)

しかしながら、Edebug を使ってこの関数をデバッグする際には、まず最初に別の関数を使って、コードの 膳立て (*instrument*) をする必要がある。Emacs version 19 では関数定義内で次のようにタイプする。

```
M-x edebug-defun RET
```

こうすることで、もし Edebug が Emacs にインストールされていなければ自動的にインストールし、関数を適切に膳立てする。

関数の膳立てをした後、カーソルを次の S 式のすぐ後に持っていって *C-x C-e* (`eval-last-sexp`) とタイプしよう。

```
(triangle-recursively-bugged 3)
```

You will be jumped back to the source for `triangle-recursively-bugged` and the cursor positioned at the beginning of the `if` line of the function. Also, you will see an arrowhead at the left hand side of that line. The arrowhead marks the line where the function is executing. (In the following examples, we show the arrowhead with ‘`=>`’; in a windowing system, you may see the arrowhead as a solid triangle in the window ‘fringe’.)

（旧訳）すると `triangle-recursively-bugged` のソースに戻ってカーソルがこの関数内の `if` の行の最初に移動する。また、‘`=>`’ という矢印がその行の左側に表示される。この矢印は関数が実行されている場所を指している。

```
=>*(if (= number 1)
```

例の中で、星印 ‘*’ はポイントを表している (これは Info では ‘`-!-`’ と表示される。)。

さて、ここで *SPC* を押すと、ポイントは次に実行される S 式に移動する。従って、この行が次のようになる。

```
=>(if *(= number 1)
```

続けて *SPC* を押していくと、ポイントは S 式から S 式へと移動していく。同時に S 式が値を返すごとにその値がエコー領域に表示される。例えばポイントが `number` の所を通過すると、次のように表示されるはずだ。

```
Result: 3 (#o3, #x3, ?\C-c)
```

これは `number` の値が 3、つまり三番目の ASCII コードである ASCII `CTL-C` であることを意味している。(C はアルファベットの三番目の文字である。) (訳註：暇な人は `(char-to-string 3)` 等を評価してみよう。)

こうしてエラーのある行まで移動することが出来る。これを表示する直前には、この行は次のように表示されている。

```
=>        *(1= number)))))              ; ここでエラー
```

ここでもう一度 *SPC* を押すと次のようなエラーメッセージが表示される。

```
Symbol's function definition is void: 1=
```

これがバグである。

Edebug を終了するには ‘`q`’ を押せばよい。

この関数定義の膳立てを解くには、 単に膳立てしないようなコマンドでその関数を評価しなおせばよい。例えば、関数定義の最後の閉じ括弧の後で *C-x C-e* とタイプするだけでよい。

Edebug は単に関数を一つずつ実行していくだけではなく他にも沢山のことをしてくれる。自動的に関数を実行していってエラーが起きた所で止まるように設定することも出来るし、特定の場所で止まるようにすることも可能である。様々なS式の値の変化を表示させることも出来るし、ある関数が何回実行されたかも調べることも出来る。その他にもいろいろなことが出来る。

Edebug については次で説明されている。 Section “Edebug” in *The GNU Emacs Lisp Reference Manual*.

17.5 デバッグについての練習問題

- `count-words-region` 関数をインストールし、それを呼び出した時に内部デバッガが起動するようにしなさい。二つの単語を含むような region の上でこの関数を実行しなさい。*`d`* をかなりの回数押す必要があるだろう。あなたの システムではこのコマンドが終了した時点で何かフックが呼ばれますか？ (フックについての情報は次を参照。 Section “Command Loop Overview” in *The GNU Emacs Lisp Reference Manual*.)
- `count-words-region` を `*scratch*` バッファにコピーして、必要なら `defun` の先頭のスペースを取り除いてから、この関数を Edebug のために膳立てしなさい。そして、この関数を 1 ステップずつ実行しなさい。この場合、必ずしも関数にはバグがある必要はない。バグがあった方がよければバグを設定してもよい。もしこの関数にバグがなければ、問題なく最後まで実行出来るはずだ。
- Edebug を実行している間に、`?` とタイプして全ての Edebug のコマンドのリストを見なさい。(`global-edebug-prefix` は普通 *`C-x X`*、つまり *`CTL-x`* に続く大文字の *`X`* になっている。Edebug のデバッグバッファの外でのコマンドにはこの前置キーを使うようにする。)
- Edebug のデバッギングバッファで *`p`* (`edebug-bounce-point`) コマンドを使ってリージョンの何処で `count-words-region` が動作しているかを見なさい。
- ポイントを関数の下の方に持って行き、*`h`* (`edebug-goto-here`) コマンドをタイプすることでその場所にジャンプしなさい。
- *`t`* (`edebug-trace-mode`) コマンドを使って、Edebug が自分自身で関数の上を走るようにしなさい。同様に大文字の *`T`* を使って `edebug-Trace-fast-mode` に入りなさい。
- ブレークポイントを設定し、その位置まで Edebug を Trace mode で走らせなさい。

18 まとめ

この入門書はこれで終わりである。以上であなたは Emacs Lisp でのプログラミングについて十分学んだので、値をセットしたり、簡単な `.emacs` ファイルを自分自身や友人のために書いたり、Emacs にちょっとしたカスタマイズや拡張を施すことが出来るようになっているはずである。

もっとも、これは一つの段階に過ぎない。望むなら、もっと先に進み、自分自身で学んでいくことが出来る。

You have learned some of the basic nuts and bolts of programming. But only some. There are a great many more brackets and hinges that are easy to use that we have not touched.

ここからは GNU Emacs のソースや *The GNU Emacs Lisp Reference Manual.* などが参考になるだろう。

Emacs Lisp のソースを見ることは一種の探検である。ソースを読んで見慣れない関数や S 式に出逢った場合には、それが何であるかを解読したり調べたりしなければならない。

まずリファレンスマニュアルを開こう。これは徹底して書かれた、完全な、しかも読みやすい Emacs Lisp の説明書である。これは熟練者のためだけではなく、あなたが現在知っているような知識しかない人向けにも書かれている。(*Reference Manual* は標準的な GNU Emacs の配布に含まれている。この入門書と同様、これも Texinfo のソースファイルとして配布されている。従って、オンラインでも読めるし、タイプセットして印刷された本としても読むことも出来る。)

また、他の GNU Emacs 附属のオンラインヘルプも読んでみよう。全ての関数には説明文字列がついているし、`find-tag` というソースを見つけてくれるプログラムもある。

例として私がソースを探検する方法を紹介しよう。もう随分と昔のことになるが、最初に私は、その名前に魅かれて `simple.el` というファイルを見た。往々にしてそんなものだが、`simple.el` の中の幾つかの関数は複雑だった。少なくとも、ぱっと見には複雑そうに見えた。例えば一番最初の `open-line` 関数からして、いかにも複雑という感じだ。

あなたは、この関数を `forward-sentence` 関数の時のように、ゆっくりと眺めたいかもしれない。(See Section 12.3 “forward-sentence”, page 102.) あるいは、この関数は飛ばして他のもっと簡単そうな `split-line` 等の関数を見たいかもしれない。これらの関数を全て読む必要はない。`count-words-in-defun` を使うと、`split-line` 関数は 102 個の単語とシンボルを含んでいることが分る。

`split-line` は、その短さにも関わらず、まだ我々が学んでいない四つの S 式を含んでいる。`skip-chars-forward`、`indent-to`、`current-column`、そして ‘`insert-and-inherit`’ である。

例えば `skip-chars-forward` 関数を考えてみよう。(これについては復習のセクションである次の所でちょこっと触れられている。) Section 3.11 “Review”, page 30

In GNU Emacs, you can find out more about `skip-chars-forward` by typing *`C-h f`* (`describe-function`) and the name of the function. This gives you the function documentation.

You may be able to guess what is done by a well named function such as `indent-to`; or you can look it up, too. Incidentally, the `describe-function` function itself is in `help.el`; it is one of those long, but decipherable functions. You can look up `describe-function` using the *`C-h f`* command!

In this instance, since the code is Lisp, the `*Help*` buffer contains the name of the library containing the function’s source. You can put point over the name of the library and press the RET key, which in this situation is bound to `help-follow`, and be taken directly to the source, in the same way as *`M-.`* (`find-tag`).

The definition for `describe-function` illustrates how to customize the `interactive` expression without using the standard character codes; and it shows how to create a temporary buffer.

(The `indent-to` function is written in C rather than Emacs Lisp; it is a ‘built-in’ function. `help-follow` takes you to its source as does `find-tag`, when properly set up.)

You can look at a function’s source using `find-tag`, which is bound to *`M-.`* Finally, you can find out what the Reference Manual has to say by visiting the manual in Info, and typing *`i`* (`Info-index`) and the name of the function, or by looking up the function in the index to a printed copy of the manual.

（旧訳）Emacs 上で *C-h f* (describe-function) に続けて skip-chars-forward とタイプすれば、この関数についてより詳しいことを知ることが出来る。これは関数の説明文字列を表示してくれる関数である。ソースを見る場合は find-tag を使う。これは *M-.* にバインドされている。(もっとも今の場合にはあまり役に立たない。この関数は Lisp ではなく C で書かれたプリミティブだからだ。) 最後に、この関数についてのリファレンスマニュアルの記述は次のようにして探せる。まず Info を起動する。そして *i* (Info-index) に続けてこの関数の名前をタイプする。もしくは印刷されたマニュアルのインデックスから insert を探してもよい。

Similarly, you can find out what is meant by insert-and-inherit.

（旧訳）同様に、‘?\n’ の意味も調べることが出来る。Info-index で ‘?\n’ を探してみるとよい。やってみると分るがこれは失敗する。しかし諦めてはいけない。インデックスで ‘?’ 無しの ‘\n’ を探してみれば、マニュアルの中で関係するセクションを見つけることが出来る。(‘?\n’ が改行文字を表していることは次の所に出ている。Section “Character Type” in *The GNU Emacs Lisp Reference Manual*.)

（旧訳）skip-chars-forward と indent-to で何が行われているかは推察することが出来るだろう。勿論詳しく見ることだって出来る。(ついでだが、describe-function 関数そのものは help.el にある。これは、長い割には解読しやすい関数の一つである。これの定義を見れば、どうやって標準の文字コードを使わずに interactive 式をカスタマイズ出来るかとか、どうやってテンポラリバッファを生成するかということが分る。

Other interesting source files include paragraphs.el, loaddefs.el, and loadup.el. The paragraphs.el file includes short, easily understood functions as well as longer ones. The loaddefs.el file contains the many standard autoloads and many keymaps. I have never looked at it all; only at parts. loadup.el is the file that loads the standard parts of Emacs; it tells you a great deal about how Emacs is built. (See Section “Building Emacs” in *The GNU Emacs Lisp Reference Manual*, for more about building.)

（旧訳）他の面白いソースとしては、paragraphs.el や loaddefs.el それに loadup.el などがあげられる。paragraphs.el ファイルには、長い関数だけでなく短くて簡単に理解出来る関数も含まれている。loaddefs.el ファイルには沢山の標準オートロード式や沢山のキーマップが含まれている。loadup.el は Emacs の標準部分をロードするファイルである。これを見れば、Emacs がどういうふうに作成されるかがよく分る。(Emacs の作成については次を参照。Section “Building Emacs” in *The GNU Emacs Lisp Reference Manual*.)

前にも言ったように、これまでに Emacs の部品についてはいくらか学習してきた。しかし、これは大切なことなのだが、プログラミングの多くの部分についてはほとんど触れることは出来なかった。私は情報のソートについては既に定義された sort 関数を使うこと以外全く触れることが出来なかった。また、変数やリストを使うこと以外の事柄については、例えば情報を貯めておくにはどうしたらいいかといったことにも触れられなかった。プログラムを書くプログラムについても同様だ。これらの話題については、他の別の種類の本で本書とは異なる学習形態で学ぶべきであろう。

ともかく、以上で GNU Emacs を使って沢山の実際の仕事をする準備が整った。出発点に辿り着いたわけである。これは始まりの終わりなのだ。

Appendix A 関数 the-the

文章を書いていると、時々単語を二重に書いてしまうことがある—例えば “you you” (訳註：原文では実際にこの文の先頭で you が二重になっている。) と書いてしまったりする。個人的には “the” をよく二重に書いてしまう。そこで、二重になっている単語を見つける関数のことを the-the と呼ぶことにする。

まず最初の段階として、取り敢えず次の正規表現を使えば、このような二重の単語を表現出来そうなことが分る。

```
\\(\\w+[ \t\n]+\\)\\1
```

この正規表現 (の前半) は一つ以上の単語構成文字に続いて一つ以上のスペース、タブないしは改行がくるものにマッチする。しかし、これでは二行に渡るものは見つけられない。というのも、最初の単語の終わりに改行が来た場合に、次の行の同じ単語の後にスペースが来ればマッチしないからである。(正規表現についての詳細は、Chapter 12 “正規表現の検索”, page 101, や Section “Syntax of Regular Expressions” in *The GNU Emacs Manual*, それに Section “Regular Expressions” in *The GNU Emacs Lisp Reference Manual*, を参照。)

単に単語の部分だけを検索すればよいのではと考えるかもしれないが、これでは ‘with the’ の中の ‘th’ のようなものまで引っかけてしまう。

上とは別の正規表現検索で、単語構成文字に続いて非単語構成文字が続き、更に最初の単語が来るものにマッチするものがある。次の式で ‘\\w+’ は一つ以上の単語構成文字にマッチし、‘\\W*’ は零個以上の非単語構成文字にマッチする。

```
\\(\\(\\w+\\)\\W*\\)\\1
```

が、これも役に立たない。

次に私が使っているものを挙げる。これは完全ではないが、十分使いものになる。‘\\b’ は単語の始まりもしくは終わりの空文字列にマッチし、‘[^@ \n\t]+’ は一つ以上の @ マーク、空白、改行、もしくはタブのいずれでもない文字の連続にマッチする。

```
\\b\\([^@ \n\t]+\\)[ \n\t]+\\1\\b
```

もっと複雑な式を書くことも出来るが、私自身はこの式で十分であると分ったので、これを使っている。

以下が the-the 関数である。これを使いやすいキーバインディングと共に .emacs ファイルに記述している。

```
(defun the-the ()
  "Search forward for for a duplicated word."
  (interactive)
  (message "Searching for for duplicated words ...")
  (push-mark)
  ;; This regexp is not perfect
  ;; but is fairly good over all:
  (if (re-search-forward
       "\\b\\([^@ \n\t]+\\)[ \n\t]+\\1\\b" nil 'move)
      (message "Found duplicated word.")
    (message "End of buffer")))

;; Bind `the-the' to  C-c \
(global-set-key "\C-c\\" 'the-the)
```

以下がテスト用の文章である。

```
one two two three four five
five six seven
```

関数定義の中の正規表現を上に挙げた他の正規表現で置き換えて、このリストの上で試してみることも出来る。

Appendix B Kill リングの扱い

kill リングは rotate-yank-pointer の働きによってリングの形態に変換されたリストである。yank 及び yank-pop コマンドは rotate-yank-pointer 関数を利用している。

この Appendix では、yank や yank-pop と共に、rotate-yank-pointer 関数も説明することにする。

The kill ring has a default maximum length of sixty items; this number is too large for an explanation. Instead, set it to four. Please evaluate the following:

```
(setq old-kill-ring-max kill-ring-max)
(setq kill-ring-max 4)
```

Then, please copy each line of the following indented example into the kill ring. You may kill each line with *C-k* or mark it and copy it with *M-w*.

(In a read-only buffer, such as the **info** buffer, the kill command, *C-k* (kill-line), will not remove the text, merely copy it to the kill ring. However, your machine may beep at you. Alternatively, for silence, you may copy the region of each line with the *M-w* (kill-ring-save) command. You must mark each line for this command to succeed, but it does not matter at which end you put point or mark.)

Please invoke the calls in order, so that five elements attempt to fill the kill ring:

```
first some text
second piece of text
third line
fourth line of text
fifth bit of text
```

Then find the value of kill-ring by evaluating

```
kill-ring
```

It is:

```
("fifth bit of text" "fourth line of text"
"third line" "second piece of text")
```

The first element, ‘first some text’, was dropped.

To return to the old value for the length of the kill ring, evaluate:

```
(setq kill-ring-max old-kill-ring-max)
```

B.1 関数 current-kill

The current-kill function changes the element in the kill ring to which kill-ring-yank-pointer points. (Also, the kill-new function sets kill-ring-yank-pointer to point to the latest element of the kill ring. The kill-new function is used directly or indirectly by kill-append, copy-region-as-kill, kill-ring-save, kill-line, and kill-region.)

The current-kill function is used by yank and by yank-pop. Here is the code for current-kill:

```
(defun current-kill (n &optional do-not-move)
  "Rotate the yanking point by N places, and then return that kill.
If N is zero, ‘interprogram-paste-function’ is set, and calling it
returns a string, then that string is added to the front of the
kill ring and returned as the latest kill.
If optional arg DO-NOT-MOVE is non-nil, then don’t actually move the
yanking point; just return the Nth kill forward."
  (let ((interprogram-paste (and (= n 0)
                                 interprogram-paste-function
                                 (funcall interprogram-paste-function))))
    (if interprogram-paste
        (progn
          ;; Disable the interprogram cut function when we add the new
          ;; text to the kill ring, so Emacs doesn’t try to own the
          ;; selection, with identical text.
          (let ((interprogram-cut-function nil))
            (kill-new interprogram-paste))
          interprogram-paste)
```

```
(or kill-ring (error "Kill ring is empty"))
(let ((ARGth-kill-element
       (nthcdr (mod (- n (length kill-ring-yank-pointer))
                    (length kill-ring))
               kill-ring)))
  (or do-not-move
      (setq kill-ring-yank-pointer ARGth-kill-element))
  (car ARGth-kill-element)))))
```

Remember also that the `kill-new` function sets `kill-ring-yank-pointer` to the latest element of the kill ring, which means that all the functions that call it set the value indirectly: `kill-append`, `copy-region-as-kill`, `kill-ring-save`, `kill-line`, and `kill-region`.

Here is the line in `kill-new`, which is explained in "The `kill-new` function", page 70.

```
(setq kill-ring-yank-pointer kill-ring)
```

The `current-kill` function looks complex, but as usual, it can be understood by taking it apart piece by piece. First look at it in skeletal form:

```
(defun current-kill (n &optional do-not-move)
  "Rotate the yanking point by N places, and then return that kill."
  (let varlist
    body...)
```

This function takes two arguments, one of which is optional. It has a documentation string. It is *not* interactive.

The body of the function definition is a `let` expression, which itself has a body as well as a *varlist*.

The `let` expression declares a variable that will be only usable within the bounds of this function. This variable is called `interprogram-paste` and is for copying to another program. It is not for copying within this instance of GNU Emacs. Most window systems provide a facility for interprogram pasting. Sadly, that facility usually provides only for the last element. Most windowing systems have not adopted a ring of many possibilities, even though Emacs has provided it for decades.

The `if` expression has two parts, one if there exists `interprogram-paste` and one if not.

Let us consider the 'if not' or else-part of the `current-kill` function. (The then-part uses the `kill-new` function, which we have already described. See "The `kill-new` function", page 70.)

```
(or kill-ring (error "Kill ring is empty"))
(let ((ARGth-kill-element
       (nthcdr (mod (- n (length kill-ring-yank-pointer))
                    (length kill-ring))
               kill-ring)))
  (or do-not-move
      (setq kill-ring-yank-pointer ARGth-kill-element))
  (car ARGth-kill-element))
```

The code first checks whether the kill ring has content; otherwise it signals an error.

Note that the `or` expression is very similar to testing length with an `if`:

```
(if (zerop (length kill-ring))          ; if-part
    (error "Kill ring is empty"))       ; then-part
  ;; No else-part
```

If there is not anything in the kill ring, its length must be zero and an error message sent to the user: '`Kill ring is empty`'. The `current-kill` function uses an `or` expression which is simpler. But an `if` expression reminds us what goes on.

This `if` expression uses the function `zerop` which returns true if the value it is testing is zero. When `zerop` tests true, the then-part of the `if` is evaluated. The then-part is a list starting with the function `error`, which is a function that is similar to the `message` function (see Section 1.8.5 "The `message` Function", page 11) in that it prints a one-line message in the echo area. However, in addition to printing a message, `error` also stops evaluation of the function within which it is embedded. This means that the rest of the function will not be evaluated if the length of the kill ring is zero.

Then the `current-kill` function selects the element to return. The selection depends on the number of places that `current-kill` rotates and on where `kill-ring-yank-pointer` points.

Next, either the optional `do-not-move` argument is true or the current value of `kill-ring-yank-pointer` is set to point to the list. Finally, another expression returns the first element of the list even if the `do-not-move` argument is true.

In my opinion, it is slightly misleading, at least to humans, to use the term 'error' as the name of the `error` function. A better term would be 'cancel'. Strictly speaking, of course, you cannot point to, much less rotate a pointer to a list that has no length, so from the point of view of the computer, the word 'error' is correct. But a human expects to attempt this sort of thing, if only to find out whether the kill ring is full or empty. This is an act of exploration.

From the human point of view, the act of exploration and discovery is not necessarily an error, and therefore should not be labeled as one, even in the bowels of a computer. As it is, the code in Emacs implies that a human who is acting virtuously, by exploring his or her environment, is making an error. This is bad. Even though the computer takes the same steps as it does when there is an 'error', a term such as 'cancel' would have a clearer connotation.

Among other actions, the else-part of the `if` expression sets the value of `kill-ring-yank-pointer` to `ARGth-kill-element` when the kill ring has something in it and the value of `do-not-move` is `nil`.

The code looks like this:

```
(nthcdr (mod (- n (length kill-ring-yank-pointer))
             (length kill-ring))
        kill-ring)))
```

This needs some examination. Unless it is not supposed to move the pointer, the `current-kill` function changes where `kill-ring-yank-pointer` points. That is what the `(setq kill-ring-yank-pointer ARGth-kill-element))` expression does. Also, clearly, `ARGth-kill-element` is being set to be equal to some CDR of the kill ring, using the `nthcdr` function that is described in an earlier section. (See Section 8.3 "copy-region-as-kill", page 67.) How does it do this?

As we have seen before (see Section 7.3 "nthcdr", page 57), the `nthcdr` function works by repeatedly taking the CDR of a list—it takes the CDR of the CDR of the CDR . . .

The two following expressions produce the same result:

```
(setq kill-ring-yank-pointer (cdr kill-ring))

(setq kill-ring-yank-pointer (nthcdr 1 kill-ring))
```

However, the `nthcdr` expression is more complicated. It uses the `mod` function to determine which CDR to select.

(You will remember to look at inner functions first; indeed, we will have to go inside the `mod`.)

The `mod` function returns the value of its first argument modulo the second; that is to say, it returns the remainder after dividing the first argument by the second. The value returned has the same sign as the second argument.

Thus,

```
(mod 12 4)
  ⇒ 0  ;; because there is no remainder
(mod 13 4)
  ⇒ 1
```

In this case, the first argument is often smaller than the second. That is fine.

```
(mod 0 4)
  ⇒ 0
(mod 1 4)
  ⇒ 1
```

We can guess what the `-` function does. It is like `+` but subtracts instead of adds; the `-` function subtracts its second argument from its first. Also, we already know what the `length` function does (see Section 7.2.1 "length", page 57). It returns the length of a list.

And `n` is the name of the required argument to the `current-kill` function.

So when the first argument to `nthcdr` is zero, the `nthcdr` expression returns the whole list, as you can see by evaluating the following:

```
;; kill-ring-yank-pointer and kill-ring have a length of four
;; and (mod (- 0 4) 4) ⇒ 0
(nthcdr (mod (- 0 4) 4)
        '("fourth line of text"
          "third line"
          "second piece of text"
          "first some text"))
```

When the first argument to the `current-kill` function is one, the `nthcdr` expression returns the list without its first element.

```
(nthcdr (mod (- 1 4) 4)
        '("fourth line of text"
          "third line"
          "second piece of text"
          "first some text"))
```

Incidentally, both `kill-ring` and `kill-ring-yank-pointer` are *global variables*. That means that any expression in Emacs Lisp can access them. They are not like the local variables set by `let` or like the symbols in an argument list. Local variables can only be accessed within the `let` that defines them or the function that specifies them in an argument list (and within expressions called by them).

B.2 yank

`current-kill` について学んだ後では、`yank` 関数のコードは全く易しく見えるだろう。

The `yank` function does not use the `kill-ring-yank-pointer` variable directly. It calls `insert-for-yank` which calls `current-kill` which sets the `kill-ring-yank-pointer` variable.

（旧訳）これには一箇所だけトリッキーな部分がある。`rotate-yank-pointer` に渡す引数の計算の所だ。

コードは以下の通りである。

```
(defun yank (&optional arg)
  "Reinsert (\"paste\") the last stretch of killed text.
More precisely, reinsert the stretch of killed text most recently
killed OR yanked.  Put point at end, and set mark at beginning.
With just \\[universal-argument] as argument, same but put point at
beginning (and mark at end).  With argument N, reinsert the Nth most
recently killed stretch of killed text.

When this command inserts killed text into the buffer, it honors
‘yank-excluded-properties’ and ‘yank-handler’ as described in the
doc string for ‘insert-for-yank-1’, which see.

See also the command \\[yank-pop]."
  (interactive "*P")
  (setq yank-window-start (window-start))
  ;; If we don’t get all the way thru, make last-command indicate that
  ;; for the following command.
  (setq this-command t)
  (push-mark (point))
  (insert-for-yank (current-kill (cond
                                  ((listp arg) 0)
                                  ((eq arg '-) -2)
                                  (t (1- arg)))))
  (if (consp arg)
      ;; This is like exchange-point-and-mark,
      ;;     but doesn’t activate the mark.
      ;; It is cleaner to avoid activation, even though the command
      ;; loop would deactivate the mark because we inserted text.
      (goto-char (prog1 (mark t)
                   (set-marker (mark-marker) (point) (current-buffer))))))
```

```
    ;; If we do get all the way thru, make this-command indicate that.
    (if (eq this-command t)
        (setq this-command 'yank))
    nil)
```

The key expression is `insert-for-yank`, which inserts the string returned by `current-kill`, but removes some text properties from it.

However, before getting to that expression, the function sets the value of `yank-window-start` to the position returned by the `(window-start)` expression, the position at which the display currently starts. The `yank` function also sets `this-command` and pushes the mark.

After it yanks the appropriate element, if the optional argument is a CONS rather than a number or nothing, it puts point at beginning of the yanked text and mark at its end.

(The `prog1` function is like `progn` but returns the value of its first argument rather than the value of its last argument. Its first argument is forced to return the buffer's mark as an integer. You can see the documentation for these functions by placing point over them in this buffer and then typing *C-h f* (`describe-function`) followed by a *RET*; the default is the function.)

The last part of the function tells what to do when it succeeds.

B.3 yank-pop

yank と current-kill を理解してしまえば、yank-pop 関数は簡単に理解出来る。スペースの節約のために説明文字列を省くと、コードは次の通りである。

```
(defun yank-pop (&optional arg)
  "..."
  (interactive "*p")
  (if (not (eq last-command 'yank))
      (error "Previous command was not a yank"))
  (setq this-command 'yank)
  (unless arg (setq arg 1))
  (let ((inhibit-read-only t)
        (before (< (point) (mark t))))
    (if before
        (funcall (or yank-undo-function 'delete-region) (point) (mark t))
      (funcall (or yank-undo-function 'delete-region) (mark t) (point)))
    (setq yank-undo-function nil)
    (set-marker (mark-marker) (point) (current-buffer))
    (insert-for-yank (current-kill arg))
    ;; Set the window start back where it was in the yank command,
    ;; if possible.
    (set-window-start (selected-window) yank-window-start t)
    (if before
        ;; This is like exchange-point-and-mark,
        ;;     but doesn't activate the mark.
        ;; It is cleaner to avoid activation, even though the command
        ;; loop would deactivate the mark because we inserted text.
        (goto-char (prog1 (mark t)
                     (set-marker (mark-marker)
                                 (point)
                                 (current-buffer))))))
  nil)
```

The function is interactive with a small 'p' so the prefix argument is processed and passed to the function. The command can only be used after a previous yank; otherwise an error message is sent. This check uses the variable `last-command` which is set by `yank` and is discussed elsewhere. (See Section 8.3 "copy-region-as-kill", page 67.)

The `let` clause sets the variable `before` to true or false depending whether point is before or after mark and then the region between point and mark is deleted. This is the region that was just inserted by the previous yank and it is this text that will be replaced.

`funcall` calls its first argument as a function, passing remaining arguments to it. The first argument is whatever the `or` expression returns. The two remaining arguments are the positions of point and mark set by the preceding `yank` command.

There is more, but that is the hardest part.

B.4 The `ring.el` File

Interestingly, GNU Emacs posses a file called `ring.el` that provides many of the features we just discussed. But functions such as `kill-ring-yank-pointer` do not use this library, possibly because they were written earlier.

Appendix C ラベルと軸が付いたグラフ

座標軸を描くと、グラフを理解する手助けになる。目盛もつけたい。以前の章では (Chapter 15 “グラフを描く準備”, page 136, 参照) グラフの本体を描くコードを書いたのであった。ここではその本体にそって縦軸と横軸を表示し、ラベルをつけるコードを書くことにしよう。

ここでの挿入は、バッファのポイントの下方を右方向に向って埋めていくので、新しいグラフ表示関数では、まず Y 軸ないしは縦軸を表示し、ついで本体を、そして最後に X 軸ないしは横軸を表示するという順番になる。この流れから、関数の中身のレイアウトは次のような感じになる。

1. コードの下準備。
2. Y 軸の表示。
3. グラフ本体の表示。
4. X 軸の表示。

以下に出来上がったグラフの例がどうあるべきかを載せておく。

```
10 -
                *
                *  *
                *  **
                *  ***
 5 -      *   *******
        * *** *******
        ************
      ***************
 1 - ****************
     |   |    |    |
     1   5   10   15
```

このグラフでは、縦軸と横軸には数によるラベルがついている。しかしながら、グラフの種類によっては横軸が時間で、月によるラベルをつけた方が良い場合も多いだろう。次のような感じだ。

```
 5 -      *
        * ** *
        *******
      ********** **
 1 - **************
     |    ^      |
     Jan  June   Jan
```

実際には、ちょっと考えれば横軸や縦軸について様々な形式を思いつく。複雑な関数にすることも可能だ。しかし複雑さは混乱を招く。それよりも、最初は単純な形式を選んで、後でそれを修正ないしは置き換えるのが良いだろう。

以上のことを考慮すると、次のようなアウトラインで print-graph 関数を書いていくのがいいだろう。

```
(defun print-graph (numbers-list)
  "説明文字列..."
  (let ((height  ...
        ...))
    (print-Y-axis height ... )
    (graph-body-print numbers-list)
    (print-X-axis ... )))
```

この print-graph 関数の雛型を元にして、各々の部分を順に書いていくことにしよう。

C.1 print-graph の変数リスト

print-graph 関数を書く際に、まずやらなければならないことは、let 式の変数リストを書くことである。(取り敢えず、インタラクティブ式や説明文字列のことは置いておく。)

変数リストでは幾つかの値を設定しなければならない。縦軸のラベルの一番上は少なくともグラフの高さでなくてはいけないので、まずこれについての情報が必要になる。print-graph-body 関数でもこの情報が必要だったことを思い出そう。同じ計算を別の場所で二度行う理由はどこにもないので、以前定義した print-graph-body を書き換えてこの計算を一度だけにするべきである。

同様に、X 軸のラベルを表示する関数と print-graph-body 関数のどちらにおいても記号の幅の値を知る必要がある。これも以前の章の print-graph-body の定義を書き換えて、最初に計算をすませてしまうべきであろう。

横軸のラベルの長さは少なくともグラフの長さの分はなければならない。しかし、この情報は横軸を表示する関数だけに必要なものである。従って、ここでやらなくとも良い。

以上のことから、print-graph の let 式の変数リストは次のような感じになる。

```
(let ((height (apply 'max numbers-list)) ; 最初のバージョン
      (symbol-width (length graph-blank)))
```

が、後で見るように、これではちょっとまずい。

C.2 関数 print-Y-axis

print-Y-axis 関数の仕事は、縦軸として次のようなラベルを表示することである。

```
10 -



 5 -



 1 -
```

この関数には、グラフの高さを渡してやる必要がある。それを元に適切な目盛と数を構成し、挿入するわけである。

図で見ると、Y 軸のラベルがどうなるべきかは一目瞭然である。しかし、これを言葉で表しその仕事をする関数を書くとなると、少々ややこしくなる。単に 5 行ごとに数や目盛をつければよいというのは間違いである。'1' と '5' の間には三行しかない (2、3、4 行)。しかし、'5' と '10' の間には四行ある (6、7、8、9 行)。まずは一行目にベースとなる行として数 1 と目盛をつけ、その他に最下行から 5 行ごとに目盛と数をつけていく方がよいだろう。

次の問題は、ラベルの高さをどうするかである。例えば、グラフの縦の列の中で最も高いものの高さが 7 だったとしよう。この場合、最も上のラベルは '5 -' にして、グラフを上に突き出させるべきだろうか。それとも 7 以上の最大の 5 の倍数として '10 -' の所までラベルを付けるべきだろうか。

これは後者の方がよいだろう。大抵のグラフは長方形のなかに描かれており、側面は刻み幅の整数倍になっている。今の場合なら、5、10、15 というふうな 5 の倍数である。しかし、縦軸の仕様としてこのようなものを採用するとなると、以前の関数の変数リストでの、単純に高さを計算する S 式ではまずいことに気付く。これは (apply 'max numbers-list) というものだったが、これでは単に正確なグラフの縦の列の高さの最大値が出るだけで、5 の倍数になるような調整はしてはくれない。もっと複雑な関数が必要なわけである。

いつものことだが、複雑な問題も幾つかの小さな問題に分割して考えれば単純な問題になることが多い。

最初に、いつグラフの高さが 5 の整数倍になるかを考えてみよう。これは 5、10、15 等の 5 の倍数の時である。この場合は、この値を即、Y 軸の高さとしてよい。

ある数が 5 の倍数になるかどうかを見るには、この数を 5 で割ってみて、余りがどうなるかを見るのが早い。もし余りが無ければ、その数は 5 の倍数である。7 の場合は余りが 2 になるので、これは 5 の倍数ではない。ちょっと言い方を変えて小学校ふうに説明するなら、5 は 7 の中に一回だけ含まれて、残りは 2 になる。一方、5 は 10 の中には 2 回含まれ、残りはない。従って、10 は 5 の倍数ということになる。

C.2.1 寄り道: 剰余の計算

Lisp では余りを計算する関数は % である。この関数は最初の引数を二番目の引数で割った時の余りを返す。あいにく、% という Emacs Lisp 関数は apropos では見つけられない。つまり、*M-x apropos RET remainder RET* とすると、何も見つけられない。% という関数の存在を知るためにはこの文書などの本を見るか、Emacs Lisp のソースを眺めるしか方法がない。

% 関数を試してみるには、次の二つの S 式を評価してみると良い。

```
(% 7 5)

(% 10 5)
```

最初のS式は2を返すし、二番目のS式は0を返す。

返された値が0かそうでないかのテストには zerop 関数が使える。この関数は引数として数値を取り、それが零なら t を返す。

```
(zerop (% 7 5))
     ⇒ nil

(zerop (% 10 5))
     ⇒ t
```

というわけで、次のようなS式を書けば、高さが5で割切れる場合には t が返される。

```
(zerop (% height 5))
```

(height の値は勿論 (apply 'max numbers-list). で分る。)

一方、もし height が5の倍数でない場合、値が5の倍数になるように再設定する必要がある。これは既にお馴染みの関数を使えば単純な算数にすぎない。まずは height を5で割ってこの中に何回5が含まれるかを調べる。例えば12なら2回含まれる。この値に1加えて5倍すれば、棒グラフの最大の高さよりも大きい最初の5の倍数が得られる。5は12の中には2回含まれるので、これに1を足して5倍すると15が求まる。これが12を越える最初の5の倍数であ。この作業を行うS式は次の通りである。

```
(* (1+ (/ height 5)) 5)
```

例えばもし次のS式を評価したなら、結果は15になる。

```
(* (1+ (/ 12 5)) 5)
```

これまでの議論では「5」を Y 軸についての目盛幅としてきたわけだが、ここは他の値を使っても構わない。一般性のために、5を他の値も設定出来るように変数で置き換えよう。私としては、この変数は Y-axis-label-spacing と名付けるのが良いと思う。この変数と if 式を使うと、次のS式が出来る。

Using this term, and an if expression, we produce the following:

```
(if (zerop (% height Y-axis-label-spacing))
    height
  ;; else
  (* (1+ (/ height Y-axis-label-spacing))
     Y-axis-label-spacing))
```

このS式は、もし height が Y-axis-label-spacing の値の倍数なら height の値そのものを返し、そうでない場合は、それより大きい最小の Y-axis-label-spacing の倍数の値を計算して返す。

それでは、このS式を print-graph 関数の中の let 式の中に (Y-axis-label-spacing の設定の後で) 埋め込んでみよう。

```
(defvar Y-axis-label-spacing 5
  "Number of lines from one Y axis label to next.")

...
(let* ((height (apply 'max numbers-list))
       (height-of-top-line
        (if (zerop (% height Y-axis-label-spacing))
            height
          ;; else
          (* (1+ (/ height Y-axis-label-spacing))
             Y-axis-label-spacing)))
       (symbol-width (length graph-blank))))
...
```

(let* 関数を使っていることに注意。始めに変数 height の初期値が (apply 'max numbers-list) 式で計算され、その値を使って他の変数の値を計算しているためである。let* についての詳細は、次を参照。 “let * 式”, page 105.)

C.2.2 Y 軸の要素の構成

縦軸を表示する際には、‘5 -’ とか ‘10 -’ 等を 5 行ごとに挿入したい。更に、数は下の部分を揃えたい。つまり、小さな桁の数は、前に空白を置くことになる。もし二桁の数が出てきた時には、一桁の数の先頭には一つの空白を置くことになるわけである。

数の長さを求めるには、`length` 関数が使われる。しかし、`length` 関数は文字列に対してしかうまく動作せず、数値は扱えない。そこでまず、数を数値から文字列に変換する必要がある。これは、`number-to-string` 関数を使って行うことが出来る。例えば、次のような感じである。

```
(length (number-to-string 35))
     ⇒ 2

(length (number-to-string 100))
     ⇒ 3
```

(`number-to-string` is also called `int-to-string`; you will see this alternative name in various sources.)

更に、各ラベルにおいて、各々の数には ‘ - ’ のような文字列を付け加えたい。これを、`Y-axis-tic` マーカと呼ぶことにしよう。この変数は、`defvar` を使って定義する。

```
(defvar Y-axis-tic " - "
   "String that follows number in a Y axis label.")
```

Y ラベルの長さは、Y axis tick mark とグラフの一番上の数の桁数の和になる。

```
(length (concat (number-to-string height) Y-axis-tic)))
```

この値は、`print-graph` 関数の中の変数リストの所で、`full-Y-label-width` の値として計算される。(我々が最初にこの関数を考えた時には、この関数は変数リストの中に入れることは考えていなかったことに注意しよう。)

縦軸を完成させるには、目盛記号を数と結合する必要がある。そして、その前には数の桁数によって空白が付いたりする。というわけで、ラベルには三つのパートがあることになる。先頭の空白 (無い場合もある)、数、そして目盛記号である。この関数には、指定された行に対する数の値と一番上の行の幅が渡される。これらの値は `print-graph` によって (一回だけ) 計算される。

```
(defun Y-axis-element (number full-Y-label-width)
  "Construct a NUMBERed label element.
A numbered element looks like this ‘  5 - ’,
and is padded as needed so all line up with
the element for the largest number."
  (let* ((leading-spaces
         (- full-Y-label-width
            (length
             (concat (number-to-string number)
                     Y-axis-tic)))))
    (concat
     (make-string leading-spaces ? )
     (number-to-string number)
     Y-axis-tic)))
```

`Y-axis-element` 関数は、(もしあれば) 先頭の空白と、文字列としての数と、目盛記号を結合するものである。

先頭にいくつ空白をつければよいかは、ラベルの実質部分の長さ—数の長さと目盛記号の長さの和—を望まれるラベルの長さから引くことで求められる。

空白は `make-string` 関数を使って挿入される。この関数は、二つの引数を取る。一つは文字列の長さ、もう一つは特定の形式で書かれた挿入する文字のシンボルである。今の場合、この形式は ‘? ’ のように、疑問符に続く空白という形をしている。このような文字の表し方についての説明は、次を参照。Section “Character Type” in *The GNU Emacs Lisp Reference Manual*.

`number-to-string` 関数は、結合式の中で数を文字列に変換するために使われている。文字列に変換してから先頭の空白と目盛記号と結合するのである。

C.2.3 Y 軸全体の構成

前節までの関数で、縦軸のラベルとして挿入する番号と空白のついた文字列のリストを生成する関数を構成するのに必要なツールが全て揃う。

```
(defun Y-axis-column (height width-of-label)
  "Construct list of Y axis labels and blank strings.
For HEIGHT of line above base and WIDTH-OF-LABEL."
  (let (Y-axis)

    (while (> height 1)
      (if (zerop (% height Y-axis-label-spacing))
          ;; ラベル挿入
          (setq Y-axis
                (cons
                 (Y-axis-element height width-of-label)
                 Y-axis))

        ;; そうでなければ空白挿入
        (setq Y-axis
              (cons
               (make-string width-of-label ? )
               Y-axis)))
      (setq height (1- height)))
    ;; ベースライン挿入
    (setq Y-axis
          (cons (Y-axis-element 1 width-of-label) Y-axis))
    (nreverse Y-axis)))
```

この関数では、まず height の値から出発してその値を一つずつ減らしていき、各々の引き算が終わった所でその値が Y-axis-label-spacing の整数倍になっているかどうかを判定する。そして、もしそうなっていれば、Y-axis-element 関数を使って番号付きのラベルを作成し、そうでなければ make-string 関数を使って空白のラベルを作成する。最下行は、番号 1 と目盛記号からなっている。

C.2.4 print-Y-axis 最終版

(訳註：題名に最終版と書いてあるが、実際には Section C.4 “グラフ全体の表示”, page 179, に出ているものが真の最終版である。原文ではここに最終版が載っていたのだが、そうすると、中で使っている y-axis-column 関数が最終版でないために、その次に述べられているテストでエラーが出てしまう。)

Y-axis-column 関数によって構成されたリストは print-Y-axis 関数に渡される。これが実際にそのリストを挿入する。

```
(defun print-Y-axis (height full-Y-label-width)
  "Insert Y axis using HEIGHT and FULL-Y-LABEL-WIDTH.
Height must be the maximum height of the graph.
Full width is the width of the highest label element."
;; Value of height and full-Y-label-width
;; are passed by ‘print-graph’.
  (let ((start (point)))
    (insert-rectangle
     (Y-axis-column height full-Y-label-width))
    ;; グラフ挿入の準備としてポイントを移動
    (goto-char start)
    ;; full-Y-label-width の値の分だけポイントを前方に移動
    (forward-char full-Y-label-width)))
```

print-Y-axis は Y-axis-column によって作成された Y 軸のラベルを挿入するのに insert-rectangle を使っている。更に、グラフの本体部分を挿入するための適切な位置にポイントを移動している。

以下のようにして print-Y-axis をテストしてみることが出来る。

1. まず次の変数や関数をインストールする。

```
Y-axis-label-spacing
Y-axis-tic
Y-axis-element
Y-axis-column
print-Y-axis
```

2. 次の S 式をコピーする。

```
(print-Y-axis 12 5)
```

3. *scratch* バッファに移り、カーソルを縦軸のラベルを挿入したい位置まで移動する。
4. *M-:* (eval-expression) とタイプする。(訳註：Emacs 19.28 ベースの Mule では *M-ESC*.)
5. print-Y-axis 式をミニバッファに *C-y* (yank) を使ってヤンクする。
6. この S 式を評価するために RET を押す。

Emacs はラベルを縦に表示する。一番上は '10 - ' である。(print-graph 関数は、height-of-top-line の値を渡すのだが、今の場合これは 15 である。) (訳註：ここは原文も混乱しているみたいである。print-Y-axis は print-graph の中で使われている。その中では print-Y-axis に第一引数として渡されるのは height-of-top-line の値なのであるが、print-graph 関数の中の局所変数としての *height* の値が 12 で *Y-axis-label-spacing* の値が 5 だったとすると、この height-of-top-line の値が 15 になるということを言いたかったのであろう。詳しくは最終版 print-graph を参照のこと。)

C.3 関数 print-X-axis

X 軸のラベルは Y 軸のラベルと大変似ている。違うのは、目盛記号が数の上にのっていることである。ラベルは次のようになる。

```
    |   |    |    |
    1   5   10   15
```

最初の目盛はグラフの最初の桁の下の部分にきており、その前には幾つかの空白がある。これらの空白は、Y 軸のラベルが上に来る部分である。二番目、三番目、四番目、あるいはそれに続く目盛は全て等間隔で並んでいる。この間隔は X-axis-label-spacing の値で決まる。

X 軸の二行目は、空白に続く数からなる。この数字の間隔もまた X-axis-label-spacing の値による。

変数 X-axis-label-spacing の値そのものは symbol-width という単位を元に決められるべきである。棒グラフを表示するためのシンボルの幅を変更する時に、ラベルの付け方まで変えなければならないなんて事態は避けたいだろう。

print-X-axis 関数は多かれ少なかれ print-Y-axis 関数と同じ形をしている。違うのは、目盛の行と数字の行の二行あるということである。この二つは各々別の関数で書いて、print-X-axis の中で一緒にすることにする。

以下が、これから行う三段階のステップである。

1. X 軸の目盛記号を表示する関数 print-X-axis-tic-line を書く。
2. X 軸の数字を表示する関数 print-X-axis-numbered-line を書く。
3. この print-X-axis-tic-line と print-X-axis-numbered-line を使って、上の二つの行を両方とも表示する関数 print-X-axis を書く。

C.3.1 X 軸の目盛記号

最初の関数では、X 軸の目盛記号を表示する。まずは、目盛記号とその間の記号を指定しないといけない。

```
(defvar X-axis-label-spacing
  (if (boundp 'graph-blank)
      (* 5 (length graph-blank)) 5)
  "Number of units from one X axis label to next.")
```

(graph-blank の値はまた別の defvar で設定されることに注意しよう。boundp という述語は、既に何かの値がセットされているかどうかを判定するものである。もし何もセットされていなければ、boundp は nil を返す。もし graph-blank が設定されず、また、この条件分岐式がなかったとすると、Symbol's value as variable is void というエラーメッセージを受けとることになる。

Here is the defvar for X-axis-tic-symbol:

```
(defvar X-axis-tic-symbol "|"
  "String to insert to point to a column in X axis.")
```

目標は、次のような行を作成することである。

```
     |   |    |    |
```

最初の目盛は、Y 軸のラベルのためのスペースを空けるためにインデントされた最初の棒グラフの開始の桁と同じだけインデントしないといけない。

目盛の要素は、ある目盛から次の目盛までのスペースと目盛記号からなる。空白の数は目盛記号の幅と X-axis-label-spacing から決まる。

コードは次の通りである。

```
;;; X-axis-tic-element
...
(concat
 (make-string
  ;; 空白の文字列を作成
  (-  (* symbol-width X-axis-label-spacing)
      (length X-axis-tic-symbol))
  ? )
 ;; 空白を目盛記号と結合
 X-axis-tic-symbol)
...
```

次に、最初の目盛をグラフの開始位置までインデントするために、どれだけ空白が要るかを数える必要がある。これには、print-graph 関数によって渡された full-Y-label-width の値を利用する。

X-axis-leading-spaces を作るためのコードは次の通りである。

```
;; X-axis-leading-spaces
...
(make-string full-Y-label-width ? )
...
```

また、横軸の長さも決めなければならない。これは、数のリストの長である。更に、横軸の目盛の数も決める必要がある。

```
;; X-length
...
(length numbers-list)

;; tic-width
...
(* symbol-width X-axis-label-spacing)

;; number-of-X-tics
(if (zerop (% X-length tic-width))
    (/ X-length tic-width)
  (1+ (/ X-length tic-width)))
```

これらを使うと、X 軸の目盛の行を表示する関数は次のようになる。

```
(defun print-X-axis-tic-line
  (number-of-X-tics X-axis-leading-spaces X-axis-tic-element)
  "Print ticks for X axis."
    (insert X-axis-leading-spaces)
    (insert X-axis-tic-symbol)  ; 最初の桁の下に
    ;; 右の位置に二番目の目盛記号を挿入
    (insert (concat
             (make-string
              (-  (* symbol-width X-axis-label-spacing)
                  ;; 二番目の目盛記号まで空白を挿入
                  (* 2 (length X-axis-tic-symbol)))
              ? )
             X-axis-tic-symbol))
    ;; 残りの目盛記号を挿入
    (while (> number-of-X-tics 1)
      (insert X-axis-tic-element)
      (setq number-of-X-tics (1- number-of-X-tics))))
```

数字の行も同様にして簡単に書ける。

まずは前に空白の付いた番号を作成する。

```
(defun X-axis-element (number)
  "Construct a numbered X axis element."
  (let ((leading-spaces
         (-  (* symbol-width X-axis-label-spacing)
             (length (number-to-string number)))))
    (concat (make-string leading-spaces ? )
            (number-to-string number))))
```

次に、数字のついた行を表示するための関数を作る。これはまず最初の桁に数 “1” を付けるところから始まる。

```
(defun print-X-axis-numbered-line
  (number-of-X-tics X-axis-leading-spaces)
  "Print line of X-axis numbers"
  (let ((number X-axis-label-spacing))
    (insert X-axis-leading-spaces)
    (insert "1")
    (insert (concat
             (make-string
              ;; 次の数字の所まで空白を挿入
              (-  (* symbol-width X-axis-label-spacing) 2)
              ? )
             (number-to-string number)))
    ;; 残りの数字を挿入
    (setq number (+ number X-axis-label-spacing))
    (while (> number-of-X-tics 1)
      (insert (X-axis-element number))
      (setq number (+ number X-axis-label-spacing))
      (setq number-of-X-tics (1- number-of-X-tics)))))
```

最後に、print-X-axis-tic-line と print-X-axis-numbered-line を使って、print-X-axis を書く。

この関数では print-X-axis-tic-line と print-X-axis-numbered-line の両方で使われている局所変数の値を決定してから、これらの関数を呼び出さなければならない。また、二つの行の区切りに復帰コードを表示する必要もある。

この関数は、5 つの局所変数を指定する変数リストと二つの行を表示する関数の呼び出しからなる。

```
(defun print-X-axis (numbers-list)
  "Print X axis labels to length of NUMBERS-LIST."
  (let* ((leading-spaces
          (make-string full-Y-label-width ? ))
         ;; symbol-width は graph-body-print で与えられる。
         (tic-width (* symbol-width X-axis-label-spacing))
         (X-length (length numbers-list))
         (X-tic
          (concat
           (make-string
            ;; 空白の文字列を作成
            (-  (* symbol-width X-axis-label-spacing)
                (length X-axis-tic-symbol))
            ? )
           ;; 空白を目盛記号と結合
           X-axis-tic-symbol))
         (tic-number
          (if (zerop (% X-length tic-width))
              (/ X-length tic-width)
            (1+ (/ X-length tic-width)))))
    (print-X-axis-tic-line tic-number leading-spaces X-tic)
    (insert "\n")
    (print-X-axis-numbered-line tic-number leading-spaces)))
```

これで、print-X-axis のテストが出来る。

1. まずは X-axis-tic-symbol, X-axis-label-spacing, print-X-axis-tic-line, などの変数とともに X-axis-element, print-X-axis-numbered-line, print-X-axis という関数をインストールする。
2. 次の S 式をコピーする。

```
(progn
 (let ((full-Y-label-width 5)
       (symbol-width 1))
   (print-X-axis
    '(1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16))))
```

3. *scratch* バッファに移り、カーソルを軸のラベルを表示したい位置にまで動かす。
4. *M-:* (eval-expression) とタイプする。(訳註：Emacs 19.28 ベースの Mule では *M-ESC*.)
5. *C-y* (yank) を使って、テストの S 式をヤンクする。
6. その式を評価するために、RET を押す。

これで Emacs は次のような横軸を表示してくれるはずだ。

```
|   |    |    |    |
1   5   10   15   20
```

C.4 グラフ全体の表示

さて、これでグラフ全体を表示する準備が整った。

きちんとしたラベルが付いたグラフを表示する関数を書く際には、以前作成したアウトラインに従うわけだが (Appendix C “ラベルと軸の付いたグラフ”, page 171, 参照)、幾つか追加することがある。

アウトラインは以下のようなものだった。

```
(defun print-graph (numbers-list)
  "説明文字列..."
  (let ((height  ...
        ...))
    (print-Y-axis height ... )
    (graph-body-print numbers-list)
    (print-X-axis ... )))
```

最終的なバージョンは、上の計画とは二つの点で違っている。まず、変数リストの中で追加して計算するものがある。次に、ラベルの増加のさせ方を指定するオプションがある。後者の特徴はかなり本質的である。そうしないことには、画面や紙面に合わなくなってしまう可能性があるからだ。

この新しい特徴を実現するには、Y-axis-column 関数にいくらか変更を加えて vertical-step という変数を加える必要がある。同時に、print-Y-axis も書き換える。新しい関数は次のようになる。

```
;;; 最終バージョン。
(defun Y-axis-column
  (height width-of-label &optional vertical-step)
  "Construct list of labels for Y axis.
HEIGHT is maximum height of graph.
WIDTH-OF-LABEL is maximum width of label.
VERTICAL-STEP, an option, is a positive integer
that specifies how much a Y axis label increments
for each line.  For example, a step of 5 means
that each line is five units of the graph."
  (let (Y-axis
        (number-per-line (or vertical-step 1)))
    (while (> height 1)
      (if (zerop (% height Y-axis-label-spacing))
          ;; ラベルの挿入
          (setq Y-axis
                (cons
                 (Y-axis-element
                  (* height number-per-line)
                  width-of-label)
                 Y-axis))
```

```
        ;; そうでない場合は, 空白の挿入
        (setq Y-axis
              (cons
               (make-string width-of-label ? )
               Y-axis)))
      (setq height (1- height)))
    ;; ベースラインの挿入
    (setq Y-axis (cons (Y-axis-element
                        (or vertical-step 1)
                        width-of-label)
                       Y-axis))
    (nreverse Y-axis)))
```

グラフの最大の高さの値とシンボルの幅は print-graph の中の let 式によって計算される。従って、graph-body-print はそれらを受け取ることが出来るように変更しなければならない。

```
;;; 最終バージョン。
(defun graph-body-print (numbers-list height symbol-width)
  "Print a bar graph of the NUMBERS-LIST.
The numbers-list consists of the Y-axis values.
HEIGHT is maximum height of graph.
SYMBOL-WIDTH is number of each column."
  (let (from-position)
    (while numbers-list
      (setq from-position (point))
      (insert-rectangle
       (column-of-graph height (car numbers-list)))
      (goto-char from-position)
      (forward-char symbol-width)
      ;; 各桁ごとにグラフを描写
      (sit-for 0)
      (setq numbers-list (cdr numbers-list)))
    ;; X 軸のラベルのためにポイントを移動
    (forward-line height)
    (insert "\n")))
```

最後に、print-graph 関数のコードを載せておこう。

```
;;; 最終バージョン。
(defun print-graph
  (numbers-list &optional vertical-step)
  "Print labelled bar graph of the NUMBERS-LIST.
The numbers-list consists of the Y-axis values.

Optionally, VERTICAL-STEP, a positive integer,
specifies how much a Y axis label increments for
each line.  For example, a step of 5 means that
each row is five units."
  (let* ((symbol-width (length graph-blank))
         ;; height は最大の数でもあり、
         ;; 最大桁の数でもある。
         (height (apply 'max numbers-list))
         (height-of-top-line
          (if (zerop (% height Y-axis-label-spacing))
              height
            ;; else
            (* (1+ (/ height Y-axis-label-spacing))
               Y-axis-label-spacing)))
         (vertical-step (or vertical-step 1))
         (full-Y-label-width
          (length
           (concat
            (number-to-string
             (* height-of-top-line vertical-step))
            Y-axis-tic))))
```

```
(print-Y-axis
 height-of-top-line full-Y-label-width vertical-step)
(graph-body-print
 numbers-list height-of-top-line symbol-width)
(print-X-axis numbers-list)))
```

C.4.1 print-graph のテスト

print-graph 関数を短い数のリストで試してみよう。

1. (他のコードに加えて) Y-axis-column、print-Y-axis、graph-body-print、そして print-graph の各最終バージョンをインストールする。
2. 次のS式をコピーする。

```
(print-graph '(3 2 5 6 7 5 3 4 6 4 3 2 1))
```

3. *scratch* バッファに移り、カーソルをグラフを表示したい位置にまで移動する。
4. *M-:* (eval-expression) とタイプする。(訳註：Emacs 19.28 ベースの Mule では *M-ESC.*)
5. *C-y* (yank) を使って、上のS式をミニバッファにヤンクする。
6. このS式を評価するために *RET* を押す。

Emacs は次のようなグラフを表示する。

```
10 -


          *
         **   *
 5 -    ****  *
        **** ***
      * *********
      ************
 1 - **************

      |   |    |    |
      1   5   10   15
```

一方、次のS式を評価して print-graph に vertical-step として 2 を与えてみたとする。

```
(print-graph '(3 2 5 6 7 5 3 4 6 4 3 2 1) 2)
```

するとグラフは次のようになるはずだ。

```
20 -


          *
         **   *
10 -    ****  *
        **** ***
      * *********
      ************
 2 - **************

      |   |    |    |
      1   5   10   15
```

(疑問：横軸の左の '2' はバグだろうか、仕様だろうか。もしバグであると思うのなら、そして '2' ではなく '1' であるべきだと思うのなら (あるいは '0' であるべきだと思うのなら)、ソースを修正すればよい。)

C.4.2 単語やシンボルの数のグラフ化

さて、やっと目的のグラフの所まで来た。いよいよ単語やシンボルの数が 10 より少ない関数定義がどれだけあるか、10 から 19 までの間だとどれだけか、20 から 29 までだとどうか、といったことを示してくれるグラフを描く関数を書くわけである。

これは幾つかのプロセスに分けて行う。まずは、必要なコードを全てロードしてあることを確認しよう。

異なる値を設定してしまった場合に備えて、top-of-ranges の値を再設定しておく方がよいだろう。それには以下を評価すればよい。

```
(setq top-of-ranges
 '(10  20  30  40  50
   60  70  80  90 100
  110 120 130 140 150
  160 170 180 190 200
  210 220 230 240 250
  260 270 280 290 300)
```

次に、各々の範囲に入っている単語やシンボルの数のリストを作成しよう。

まず次を評価する。

```
(setq list-for-graph
      (defuns-per-range
        (sort
         (recursive-lengths-list-many-files
          (directory-files "/usr/local/emacs/lisp"
                           t ".+el$"))
         '<)
        top-of-ranges))
```

私の機械では、これに一時間程かかった。私の持っている Emacs version 19.23 だと 303 の Lisp ファイルを見ていることになる。この計算の後では、list-for-graph の値は次のようになる。

```
(537 1027 955 785 594 483 349 292 224 199 166 120 116 99
90 80 67 48 52 45 41 33 28 26 25 20 12 28 11 13 220)
```

これは、私の Emacs には 10 以下の単語やシンボルしか持たない関数定義が 537 個あり、10 以上 19 未満の単語やシンボルを持つものが 1,027 個、20 から 29 までだと 955 個、などというふうになっていることを示している。

このリストから明らかに見て取れることは、殆どの関数定義では、中に含まれる単語やシンボルの数は、10 から 30 までだということである。

さて、これをグラフに表示することにしよう。我々は、1,030 行もの高さのグラフは描きたくない。そうではなく、25 行以下程度のグラフを描きたいのである。この高さのグラフであれば、大抵のモニタで表示出来るだろう。また、紙にも印刷しやすい。

これは、今の場合、list-for-graph の各々の値を 50 分の 1 に縮小しないといけないということである。

この操作をやってくれる簡単な関数を、以下に挙げる。この関数には、今までに出てこなかった二つの関数 mapcar と lambda が出てくる。

```
(defun one-fiftieth (full-range)
  "Return list, each number one-fiftieth of previous."
 (mapcar (lambda (arg) (/ arg 50)) full-range))
```

C.4.3 lambda 式

lambda は無名関数、即ち名前のない関数のシンボルである。無名関数を使う場合には毎回本体を全て含めて書かなければならない。

従って、

```
(lambda (arg) (/ arg 50))
```

であれば、常に arg として渡されたものを 50 で割った値を返す、という関数定義になる。

例えば、以前、multiply-by-seven 関数というのを説明したことがある。これは引数を 7 倍するものであった。今回の関数も同様である。異なる点は、引数を 50 で割るということと、名前がないということだけである。multiply-by-seven を無名関数で書き直すと次のようになる。

```
(lambda (number) (* 7 number))
```

(Section 3.1 “特殊形式 defun”, page 19, を参照。)

もし 3 を 7 倍したいなら、次のように書くことが出来る。

```
(multiply-by-seven 3)
 \---------------/ ^
        |        |
     function  argument
```

この S 式は 21 を返す。

同様に、次のようにも書ける。

```
((lambda (number) (* 7 number)) 3)
 \----------------------------/ ^
              |                 |
     anonymous function     argument
```

100 を 50 で割る場合は次のように書く。

```
((lambda (arg) (/ arg 50)) 100)
 \----------------------/  \_/
            |              |
   anonymous function   argument
```

この S 式は 2 を返す。上の 100 は関数に渡されるもので、その数を関数が 50 で割っているのである。

`lambda` についてのより詳しいことは、Section “Lambda Expressions” in *The GNU Emacs Lisp Reference Manual*, を参照せよ。Lisp や lambda 式は、Lambda 計算から派生したものである。

C.4.4 関数 `mapcar`

`mapcar` は最初の引数を二番目の引数の各々の要素にともに、順に呼び出す関数である。二番目の引数はシーケンス (sequence) でないといけない。

The ‘`map`’ part of the name comes from the mathematical phrase, ‘mapping over a domain’, meaning to apply a function to each of the elements in a domain. The mathematical phrase is based on the metaphor of a surveyor walking, one step at a time, over an area he is mapping. And ‘`car`’, of course, comes from the Lisp notion of the first of a list.

例えば、

```
(mapcar '1+ '(2 4 6))
     ⇒ (3 5 7)
```

という感じである。関数 `1+` は引数に 1 を加える関数であるが、これがその後のリストの各要素について実行され、新しいリストを返すわけである。

これを `apply` と比較してみよう。こちらは最初の要素に対し、二番目以降の全ての引数を引数として渡して実行させるものだった。(`apply` の説明については Chapter 15 “グラフを描く準備”, page 136, を参照。)

`one-fiftieth` の定義では、最初の引数は、

```
(lambda (arg) (/ arg 50))
```

という無名関数である。また、二番目の引数は `full-range` であり、これは `list-for-graph` にバインドされる。

S 式全体は次の通りである。

```
(mapcar (lambda (arg) (/ arg 50)) full-range))
```

`mapcar` についての詳細は Section “Mapping Functions” in *The GNU Emacs Lisp Reference Manual*, を参照。

`one-fiftieth` 関数を使うと、各々の要素が、対応する `list-for-graph` の要素の 50 分の 1 であるようなリストを生成することが出来る。

```
(setq fiftieth-list-for-graph
      (one-fiftieth list-for-graph))
```

結果として出来るリストは次のようになる。

```
(10 20 19 15 11 9 6 5 4 3 3 2 2
1 1 1 1 0 1 0 0 0 0 0 0 0 0 0 0 4)
```

これは、今すぐにでも表示出来そうである！ (もっともここで情報が失われていることにも気付く。上の方の範囲の多くの要素は 0 である。これは、その範囲の単語やシンボルを持つ関数定義の数が 50 未満であることを示しているだけであって、そういう関数定義が全くないといっているわけではない。)

C.4.5 まだバグがある...

私は、「今すぐにでも表示出来そう」と書いた。勿論、print-graph にはバグがあるのだ... これには vertical-step オプションがあるが、horizontal-step オプションはない。top-of-range の範囲は、10 ごとに 10 から 300 まである。しかし、print-graph は 1 ずつしか表示出来ない。

これは、ある人々が最も見つけにくいタイプのバグだと考えている、考慮不足によるバグ (訳註：原文では bug of omission. これはどう訳すべきか) の古典的な例である。これは、コードを調べることで見つかるようなバグではない。コード自体に誤りはないからだ。これは単に何かが欠落しているということである。このようなバグに対するもっとも正しい態度は、プログラムをなるべく早く、何回もテストしてみることである。そして、コードを出来るだけ理解しやすく簡単に変更出来るようにアレンジしていくのである。また、一度書いたものは、すぐにではなくともいつかは書き換えられるものだということを頭に入れておくようにしよう。これはなかなか実行が困難な格言ではあるが。

作業対象となるのは print-X-axis-numbered-line 関数である。また、print-X-axis と print-graph もそれに合わせて書き換える必要がある。もっともやることはそれ程多いわけではない。一つだけ面倒な部分がある。数字が目盛記号の下に綺麗に並ばなければならないということである。この点はちょっとばかし頭をひねる必要がある。

以下が修正した print-X-axis-numbered-line である。

```
(defun print-X-axis-numbered-line
  (number-of-X-tics X-axis-leading-spaces
   &optional horizontal-step)
  "Print line of X-axis numbers"
  (let ((number X-axis-label-spacing)
        (horizontal-step (or horizontal-step 1)))
    (insert X-axis-leading-spaces)
    ;; 頭の余計な空白を削除
    (delete-char
     (- (1-
         (length (number-to-string horizontal-step)))))
    (insert (concat
             (make-string
              ;; 空白を挿入
              (-  (* symbol-width
                     X-axis-label-spacing)
                  (1-
                   (length
                    (number-to-string horizontal-step)))
                  2)
              ? )
             (number-to-string
              (* number horizontal-step))))
    ;; 残りの数を挿入
    (setq number (+ number X-axis-label-spacing))
    (while (> number-of-X-tics 1)
      (insert (X-axis-element
               (* number horizontal-step)))
      (setq number (+ number X-axis-label-spacing))
      (setq number-of-X-tics (1- number-of-X-tics)))))
```

もしあなたがこれを Info の中で読んでいるなら、print-X-axis と print-graph の新しいバージョンを見ることが出来る。もしこれを印刷された本で読んでいるとしたら、変更する行が下に書かれているはずだ。(印刷するにはコードがちょっと多過ぎるので。)

```
(defun print-X-axis (numbers-list horizontal-step)
  ...
    (print-X-axis-numbered-line
     tic-number leading-spaces horizontal-step))
(defun print-graph
  (numbers-list
   &optional vertical-step horizontal-step)
  ...
    (print-X-axis numbers-list horizontal-step))
```

C.4.6 表示されたグラフ

全てをインストールし終ったら、print-graph コマンドを次のようにして呼び出そう。

```
(print-graph fiftieth-list-for-graph 50 10)
```

次のようなグラフが表示されるはずだ。



最も大きな関数のグループは、10 から 19 までの単語やシンボルを含んでいるもののグループである。

Appendix D Free Software and Free Manuals

by Richard M. Stallman

The biggest deficiency in free operating systems is not in the software—it is the lack of good free manuals that we can include in these systems. Many of our most important programs do not come with full manuals. Documentation is an essential part of any software package; when an important free software package does not come with a free manual, that is a major gap. We have many such gaps today.

Once upon a time, many years ago, I thought I would learn Perl. I got a copy of a free manual, but I found it hard to read. When I asked Perl users about alternatives, they told me that there were better introductory manuals—but those were not free.

Why was this? The authors of the good manuals had written them for O'Reilly Associates, which published them with restrictive terms—no copying, no modification, source files not available—which exclude them from the free software community.

That wasn't the first time this sort of thing has happened, and (to our community's great loss) it was far from the last. Proprietary manual publishers have enticed a great many authors to restrict their manuals since then. Many times I have heard a GNU user eagerly tell me about a manual that he is writing, with which he expects to help the GNU project—and then had my hopes dashed, as he proceeded to explain that he had signed a contract with a publisher that would restrict it so that we cannot use it.

Given that writing good English is a rare skill among programmers, we can ill afford to lose manuals this way.

Free documentation, like free software, is a matter of freedom, not price. The problem with these manuals was not that O'Reilly Associates charged a price for printed copies—that in itself is fine. The Free Software Foundation sells printed copies (`http://shop.fsf.org`) of free GNU manuals (`http://www.gnu.org/doc/doc.html`), too. But GNU manuals are available in source code form, while these manuals are available only on paper. GNU manuals come with permission to copy and modify; the Perl manuals do not. These restrictions are the problems.

The criterion for a free manual is pretty much the same as for free software: it is a matter of giving all users certain freedoms. Redistribution (including commercial redistribution) must be permitted, so that the manual can accompany every copy of the program, on-line or on paper. Permission for modification is crucial too.

As a general rule, I don't believe that it is essential for people to have permission to modify all sorts of articles and books. The issues for writings are not necessarily the same as those for software. For example, I don't think you or I are obliged to give permission to modify articles like this one, which describe our actions and our views.

But there is a particular reason why the freedom to modify is crucial for documentation for free software. When people exercise their right to modify the software, and add or change its features, if they are conscientious they will change the manual too—so they can provide accurate and usable documentation with the modified program. A manual which forbids programmers to be conscientious and finish the job, or more precisely requires them to write a new manual from scratch if they change the program, does not fill our community's needs.

While a blanket prohibition on modification is unacceptable, some kinds of limits on the method of modification pose no problem. For example, requirements to preserve the original author's copyright notice, the distribution terms, or the list of authors, are ok. It is also no problem to require modified versions to include notice that they were modified, even to have entire sections that may not be deleted or changed, as long as these sections deal with nontechnical topics. (Some GNU manuals have them.)

These kinds of restrictions are not a problem because, as a practical matter, they don't stop the conscientious programmer from adapting the manual to fit the modified program. In other words, they don't block the free software community from making full use of the manual.

However, it must be possible to modify all the technical content of the manual, and then distribute the result in all the usual media, through all the usual channels; otherwise, the restrictions do block the community, the manual is not free, and so we need another manual.

Unfortunately, it is often hard to find someone to write another manual when a proprietary manual exists. The obstacle is that many users think that a proprietary manual is good enough—so they don't see the need to write a free manual. They do not see that the free operating system has a gap that needs filling.

Why do users think that proprietary manuals are good enough? Some have not considered the issue. I hope this article will do something to change that.

Other users consider proprietary manuals acceptable for the same reason so many people consider proprietary software acceptable: they judge in purely practical terms, not using freedom as a criterion. These people are entitled to their opinions, but since those opinions spring from values which do not include freedom, they are no guide for those of us who do value freedom.

Please spread the word about this issue. We continue to lose manuals to proprietary publishing. If we spread the word that proprietary manuals are not sufficient, perhaps the next person who wants to help GNU by writing documentation will realize, before it is too late, that he must above all make it free.

We can also encourage commercial publishers to sell free, copylefted manuals instead of proprietary ones. One way you can help this is to check the distribution terms of a manual before you buy it, and prefer copylefted manuals to non-copylefted ones.

Note: The Free Software Foundation maintains a page on its Web site that lists free books available from other publishers:
`http://www.gnu.org/doc/other-free-books.html`

Appendix E GNU Free Documentation License

Version 1.3, 3 November 2008

0. PREAMBLE

The purpose of this License is to make a manual, textbook, or other functional and useful document *free* in the sense of freedom: to assure everyone the effective freedom to copy and redistribute it, with or without modifying it, either commercially or noncommercially. Secondarily, this License preserves for the author and publisher a way to get credit for their work, while not being considered responsible for modifications made by others.

This License is a kind of "copyleft", which means that derivative works of the document must themselves be free in the same sense. It complements the GNU General Public License, which is a copyleft license designed for free software.

We have designed this License in order to use it for manuals for free software, because free software needs free documentation: a free program should come with manuals providing the same freedoms that the software does. But this License is not limited to software manuals; it can be used for any textual work, regardless of subject matter or whether it is published as a printed book. We recommend this License principally for works whose purpose is instruction or reference.

1. APPLICABILITY AND DEFINITIONS

This License applies to any manual or other work, in any medium, that contains a notice placed by the copyright holder saying it can be distributed under the terms of this License. Such a notice grants a world-wide, royalty-free license, unlimited in duration, to use that work under the conditions stated herein. The "Document", below, refers to any such manual or work. Any member of the public is a licensee, and is addressed as "you". You accept the license if you copy, modify or distribute the work in a way requiring permission under copyright law.

A "Modified Version" of the Document means any work containing the Document or a portion of it, either copied verbatim, or with modifications and/or translated into another language.

A "Secondary Section" is a named appendix or a front-matter section of the Document that deals exclusively with the relationship of the publishers or authors of the Document to the Document's overall subject (or to related matters) and contains nothing that could fall directly within that overall subject. (Thus, if the Document is in part a textbook of mathematics, a Secondary Section may not explain any mathematics.) The relationship could be a matter of historical connection with the subject or with related matters, or of legal, commercial, philosophical, ethical or political position regarding them.

The "Invariant Sections" are certain Secondary Sections whose titles are designated, as being those of Invariant Sections, in the notice that says that the Document is released under this License. If a section does not fit the above definition of Secondary then it is not allowed to be designated as Invariant. The Document may contain zero Invariant Sections. If the Document does not identify any Invariant Sections then there are none.

The "Cover Texts" are certain short passages of text that are listed, as Front-Cover Texts or Back-Cover Texts, in the notice that says that the Document is released under this License. A Front-Cover Text may be at most 5 words, and a Back-Cover Text may be at most 25 words.

A "Transparent" copy of the Document means a machine-readable copy, represented in a format whose specification is available to the general public, that is suitable for revising the document straightforwardly with generic text editors or (for images composed of pixels)

generic paint programs or (for drawings) some widely available drawing editor, and that is suitable for input to text formatters or for automatic translation to a variety of formats suitable for input to text formatters. A copy made in an otherwise Transparent file format whose markup, or absence of markup, has been arranged to thwart or discourage subsequent modification by readers is not Transparent. An image format is not Transparent if used for any substantial amount of text. A copy that is not "Transparent" is called "Opaque".

Examples of suitable formats for Transparent copies include plain ASCII without markup, Texinfo input format, LaTeX input format, SGML or XML using a publicly available DTD, and standard-conforming simple HTML, PostScript or PDF designed for human modification. Examples of transparent image formats include PNG, XCF and JPG. Opaque formats include proprietary formats that can be read and edited only by proprietary word processors, SGML or XML for which the DTD and/or processing tools are not generally available, and the machine-generated HTML, PostScript or PDF produced by some word processors for output purposes only.

The "Title Page" means, for a printed book, the title page itself, plus such following pages as are needed to hold, legibly, the material this License requires to appear in the title page. For works in formats which do not have any title page as such, "Title Page" means the text near the most prominent appearance of the work's title, preceding the beginning of the body of the text.

The "publisher" means any person or entity that distributes copies of the Document to the public.

A section "Entitled XYZ" means a named subunit of the Document whose title either is precisely XYZ or contains XYZ in parentheses following text that translates XYZ in another language. (Here XYZ stands for a specific section name mentioned below, such as "Acknowledgements", "Dedications", "Endorsements", or "History".) To "Preserve the Title" of such a section when you modify the Document means that it remains a section "Entitled XYZ" according to this definition.

The Document may include Warranty Disclaimers next to the notice which states that this License applies to the Document. These Warranty Disclaimers are considered to be included by reference in this License, but only as regards disclaiming warranties: any other implication that these Warranty Disclaimers may have is void and has no effect on the meaning of this License.

2. VERBATIM COPYING

You may copy and distribute the Document in any medium, either commercially or noncommercially, provided that this License, the copyright notices, and the license notice saying this License applies to the Document are reproduced in all copies, and that you add no other conditions whatsoever to those of this License. You may not use technical measures to obstruct or control the reading or further copying of the copies you make or distribute. However, you may accept compensation in exchange for copies. If you distribute a large enough number of copies you must also follow the conditions in section 3.

You may also lend copies, under the same conditions stated above, and you may publicly display copies.

3. COPYING IN QUANTITY

If you publish printed copies (or copies in media that commonly have printed covers) of the Document, numbering more than 100, and the Document's license notice requires Cover Texts, you must enclose the copies in covers that carry, clearly and legibly, all these Cover Texts: Front-Cover Texts on the front cover, and Back-Cover Texts on the back cover. Both covers must also clearly and legibly identify you as the publisher of these copies. The front cover must present the full title with all words of the title equally prominent and visible. You may add other material on the covers in addition. Copying with changes limited to the covers, as long as they preserve the title of the Document and satisfy these conditions, can be treated as verbatim copying in other respects.

If the required texts for either cover are too voluminous to fit legibly, you should put the first ones listed (as many as fit reasonably) on the actual cover, and continue the rest onto adjacent pages.

If you publish or distribute Opaque copies of the Document numbering more than 100, you must either include a machine-readable Transparent copy along with each Opaque copy, or state in or with each Opaque copy a computer-network location from which the general network-using public has access to download using public-standard network protocols a complete Transparent copy of the Document, free of added material. If you use the latter option, you must take reasonably prudent steps, when you begin distribution of Opaque copies in quantity, to ensure that this Transparent copy will remain thus accessible at the stated location until at least one year after the last time you distribute an Opaque copy (directly or through your agents or retailers) of that edition to the public.

It is requested, but not required, that you contact the authors of the Document well before redistributing any large number of copies, to give them a chance to provide you with an updated version of the Document.

4. MODIFICATIONS

You may copy and distribute a Modified Version of the Document under the conditions of sections 2 and 3 above, provided that you release the Modified Version under precisely this License, with the Modified Version filling the role of the Document, thus licensing distribution and modification of the Modified Version to whoever possesses a copy of it. In addition, you must do these things in the Modified Version:

A. Use in the Title Page (and on the covers, if any) a title distinct from that of the Document, and from those of previous versions (which should, if there were any, be listed in the History section of the Document). You may use the same title as a previous version if the original publisher of that version gives permission.

B. List on the Title Page, as authors, one or more persons or entities responsible for authorship of the modifications in the Modified Version, together with at least five of the principal authors of the Document (all of its principal authors, if it has fewer than five), unless they release you from this requirement.

C. State on the Title page the name of the publisher of the Modified Version, as the publisher.

D. Preserve all the copyright notices of the Document.

E. Add an appropriate copyright notice for your modifications adjacent to the other copyright notices.

F. Include, immediately after the copyright notices, a license notice giving the public permission to use the Modified Version under the terms of this License, in the form shown in the Addendum below.

G. Preserve in that license notice the full lists of Invariant Sections and required Cover Texts given in the Document's license notice.

H. Include an unaltered copy of this License.

I. Preserve the section Entitled "History", Preserve its Title, and add to it an item stating at least the title, year, new authors, and publisher of the Modified Version as given on the Title Page. If there is no section Entitled "History" in the Document, create one stating the title, year, authors, and publisher of the Document as given on its Title Page, then add an item describing the Modified Version as stated in the previous sentence.

J. Preserve the network location, if any, given in the Document for public access to a Transparent copy of the Document, and likewise the network locations given in the Document for previous versions it was based on. These may be placed in the "History" section. You may omit a network location for a work that was published at least four years before the Document itself, or if the original publisher of the version it refers to gives permission.

K. For any section Entitled "Acknowledgements" or "Dedications", Preserve the Title of the section, and preserve in the section all the substance and tone of each of the contributor acknowledgements and/or dedications given therein.

L. Preserve all the Invariant Sections of the Document, unaltered in their text and in their titles. Section numbers or the equivalent are not considered part of the section titles.

M. Delete any section Entitled "Endorsements". Such a section may not be included in the Modified Version.

N. Do not retitle any existing section to be Entitled "Endorsements" or to conflict in title with any Invariant Section.

O. Preserve any Warranty Disclaimers.

If the Modified Version includes new front-matter sections or appendices that qualify as Secondary Sections and contain no material copied from the Document, you may at your option designate some or all of these sections as invariant. To do this, add their titles to the list of Invariant Sections in the Modified Version's license notice. These titles must be distinct from any other section titles.

You may add a section Entitled "Endorsements", provided it contains nothing but endorsements of your Modified Version by various parties—for example, statements of peer review or that the text has been approved by an organization as the authoritative definition of a standard.

You may add a passage of up to five words as a Front-Cover Text, and a passage of up to 25 words as a Back-Cover Text, to the end of the list of Cover Texts in the Modified Version. Only one passage of Front-Cover Text and one of Back-Cover Text may be added by (or through arrangements made by) any one entity. If the Document already includes a cover text for the same cover, previously added by you or by arrangement made by the same entity you are acting on behalf of, you may not add another; but you may replace the old one, on explicit permission from the previous publisher that added the old one.

The author(s) and publisher(s) of the Document do not by this License give permission to use their names for publicity for or to assert or imply endorsement of any Modified Version.

5. COMBINING DOCUMENTS

You may combine the Document with other documents released under this License, under the terms defined in section 4 above for modified versions, provided that you include in the combination all of the Invariant Sections of all of the original documents, unmodified, and list them all as Invariant Sections of your combined work in its license notice, and that you preserve all their Warranty Disclaimers.

The combined work need only contain one copy of this License, and multiple identical Invariant Sections may be replaced with a single copy. If there are multiple Invariant Sections with the same name but different contents, make the title of each such section unique by adding at the end of it, in parentheses, the name of the original author or publisher of that section if known, or else a unique number. Make the same adjustment to the section titles in the list of Invariant Sections in the license notice of the combined work.

In the combination, you must combine any sections Entitled "History" in the various original documents, forming one section Entitled "History"; likewise combine any sections Entitled "Acknowledgements", and any sections Entitled "Dedications". You must delete all sections Entitled "Endorsements."

6. COLLECTIONS OF DOCUMENTS

You may make a collection consisting of the Document and other documents released under this License, and replace the individual copies of this License in the various documents with a single copy that is included in the collection, provided that you follow the rules of this License for verbatim copying of each of the documents in all other respects.

You may extract a single document from such a collection, and distribute it individually under this License, provided you insert a copy of this License into the extracted document, and follow this License in all other respects regarding verbatim copying of that document.

7. AGGREGATION WITH INDEPENDENT WORKS

   A compilation of the Document or its derivatives with other separate and independent documents or works, in or on a volume of a storage or distribution medium, is called an "aggregate" if the copyright resulting from the compilation is not used to limit the legal rights of the compilation's users beyond what the individual works permit. When the Document is included in an aggregate, this License does not apply to the other works in the aggregate which are not themselves derivative works of the Document.

   If the Cover Text requirement of section 3 is applicable to these copies of the Document, then if the Document is less than one half of the entire aggregate, the Document's Cover Texts may be placed on covers that bracket the Document within the aggregate, or the electronic equivalent of covers if the Document is in electronic form. Otherwise they must appear on printed covers that bracket the whole aggregate.

8. TRANSLATION

   Translation is considered a kind of modification, so you may distribute translations of the Document under the terms of section 4. Replacing Invariant Sections with translations requires special permission from their copyright holders, but you may include translations of some or all Invariant Sections in addition to the original versions of these Invariant Sections. You may include a translation of this License, and all the license notices in the Document, and any Warranty Disclaimers, provided that you also include the original English version of this License and the original versions of those notices and disclaimers. In case of a disagreement between the translation and the original version of this License or a notice or disclaimer, the original version will prevail.

   If a section in the Document is Entitled "Acknowledgements", "Dedications", or "History", the requirement (section 4) to Preserve its Title (section 1) will typically require changing the actual title.

9. TERMINATION

   You may not copy, modify, sublicense, or distribute the Document except as expressly provided under this License. Any attempt otherwise to copy, modify, sublicense, or distribute it is void, and will automatically terminate your rights under this License.

   However, if you cease all violation of this License, then your license from a particular copyright holder is reinstated (a) provisionally, unless and until the copyright holder explicitly and finally terminates your license, and (b) permanently, if the copyright holder fails to notify you of the violation by some reasonable means prior to 60 days after the cessation.

   Moreover, your license from a particular copyright holder is reinstated permanently if the copyright holder notifies you of the violation by some reasonable means, this is the first time you have received notice of violation of this License (for any work) from that copyright holder, and you cure the violation prior to 30 days after your receipt of the notice.

   Termination of your rights under this section does not terminate the licenses of parties who have received copies or rights from you under this License. If your rights have been terminated and not permanently reinstated, receipt of a copy of some or all of the same material does not give you any rights to use it.

10. FUTURE REVISIONS OF THIS LICENSE

    The Free Software Foundation may publish new, revised versions of the GNU Free Documentation License from time to time. Such new versions will be similar in spirit to the present version, but may differ in detail to address new problems or concerns. See `http://www.gnu.org/copyleft/`.

    Each version of the License is given a distinguishing version number. If the Document specifies that a particular numbered version of this License "or any later version" applies to it, you have the option of following the terms and conditions either of that specified version or of any later version that has been published (not as a draft) by the Free Software Foundation. If the Document does not specify a version number of this License, you may choose any version ever published (not as a draft) by the Free Software Foundation. If the Document specifies that a proxy can decide which future versions of this License can be

used, that proxy's public statement of acceptance of a version permanently authorizes you to choose that version for the Document.

11. RELICENSING

"Massive Multiauthor Collaboration Site" (or "MMC Site") means any World Wide Web server that publishes copyrightable works and also provides prominent facilities for anybody to edit those works. A public wiki that anybody can edit is an example of such a server. A "Massive Multiauthor Collaboration" (or "MMC") contained in the site means any set of copyrightable works thus published on the MMC site.

"CC-BY-SA" means the Creative Commons Attribution-Share Alike 3.0 license published by Creative Commons Corporation, a not-for-profit corporation with a principal place of business in San Francisco, California, as well as future copyleft versions of that license published by that same organization.

"Incorporate" means to publish or republish a Document, in whole or in part, as part of another Document.

An MMC is "eligible for relicensing" if it is licensed under this License, and if all works that were first published under this License somewhere other than this MMC, and subsequently incorporated in whole or in part into the MMC, (1) had no cover texts or invariant sections, and (2) were thus incorporated prior to November 1, 2008.

The operator of an MMC Site may republish an MMC contained in the site under CC-BY-SA on the same site at any time before August 1, 2009, provided the MMC is eligible for relicensing.

ADDENDUM: How to use this License for your documents

To use this License in a document you have written, include a copy of the License in the document and put the following copyright and license notices just after the title page:

```
Copyright (C)  year  your name.
Permission is granted to copy, distribute and/or modify this document
under the terms of the GNU Free Documentation License, Version 1.3
or any later version published by the Free Software Foundation;
with no Invariant Sections, no Front-Cover Texts, and no Back-Cover
Texts.  A copy of the license is included in the section entitled ``GNU
Free Documentation License''.
```

If you have Invariant Sections, Front-Cover Texts and Back-Cover Texts, replace the "with. . . Texts." line with this:

```
with the Invariant Sections being list their titles, with
the Front-Cover Texts being list, and with the Back-Cover Texts
being list.
```

If you have Invariant Sections without Cover Texts, or some other combination of the three, merge those two alternatives to suit the situation.

If your document contains nontrivial examples of program code, we recommend releasing these examples in parallel under your choice of free software license, such as the GNU General Public License, to permit their use in free software.

Index

(Index is nonexistent)

About the Author

(訳註：以下は原文をそのまま載せておくことにする。)

Robert J. Chassell has worked with GNU Emacs since 1985. He writes and edits, teaches Emacs and Emacs Lisp, and speaks throughout the world on software freedom. Chassell was a founding Director and Treasurer of the Free Software Foundation, Inc. He is co-author of the *Texinfo* manual, and has edited more than a dozen other books. He graduated from Cambridge University, in England. He has an abiding interest in social and economic history and flies his own airplane.